
Preface

Systems of polynomial equations arise throughout mathematics, science, and
engineering. Algebraic geometry provides powerful theoretical techniques for
studying the qualitative and quantitative features of their solution sets. Re-
cently developed algorithms have made theoretical aspects of the subject
accessible to a broad range of mathematicians and scientists. The algorith-
mic approach to the subject has two principal aims: developing new tools for
research within mathematics, and providing new tools for modeling and solv-
ing problems that arise in the sciences and engineering. A healthy synergy
emerges, as new theorems yield new algorithms and emerging applications
lead to new theoretical questions.

This book presents algorithmic tools for algebraic geometry and experi-
mental applications of them. It also introduces a software system in which
the tools have been implemented and with which the experiments can be
carried out. Macaulay 2 is a computer algebra system devoted to supporting
research in algebraic geometry, commutative algebra, and their applications.
The reader of this book will encounter Macaulay 2 in the context of concrete
applications and practical computations in algebraic geometry.

The expositions of the algorithmic tools presented here are designed to
serve as a useful guide for those wishing to bring such tools to bear on their
own problems. A wide range of mathematical scientists should find these
expositions valuable. This includes both the users of other programs similar
to Macaulay 2 (for example, Singular and CoCoA) and those who are not
interested in explicit machine computations at all.

The chapters are ordered roughly by increasing mathematical difficulty.
The first part of the book is meant to be accessible to graduate students and
computer algebra users from across the mathematical sciences and is pri-
marily concerned with introducing Macaulay 2. The second part emphasizes
the mathematics: each chapter exposes some domain of mathematics at an
accessible level, presents the relevant algorithms, sometimes with proofs, and
illustrates the use of the program. In both parts, each chapter comes with
its own abstract and its own bibliography; the index at the back of the book
covers all of them.

One of the first computer algebra packages aimed at algebraic geometry
was Macaulay, the predecessor of Macaulay 2, written during the years 1983-
1993 by Dave Bayer and Mike Stillman. Worst-case estimates suggested that
trying to compute Gröbner bases might be a hopeless approach to solving
problems. But from the first prototype, Macaulay was successful surprisingly
often, perhaps because of the geometrical origin of the problems attacked.
Macaulay improved steadily during its first decade. It helped transform the
theoretical notion of a projective resolution into an exciting new practical

research tool, and became widely used for research and teaching in com-
mutative algebra and algebraic geometry. It was possible to write routines
in the top-level language, and many important algorithms were added by
David Eisenbud and other users, enhancing the system and broadening its
usefulness.

There were certain practical drawbacks for the researcher who wanted
to use Macaulay effectively. A minor annoyance was that only finite prime
fields were available as coefficient rings. The major problem was that the
language made available to users was primitive and barely supported high-
level development of new algorithms; it had few basic data types and didn’t
support the addition of new ones.

Macaulay 2 is based on experience gained from writing and using its pre-
decessor Macaulay , but is otherwise a fresh start. It was written by Dan
Grayson and Mike Stillman with the generous financial support of the U.S.
National Science Foundation, with the work starting in 19931. It also incor-
porates some code from other authors: the package SINGULAR-FACTORY2

provides for factorization of polynomials; SINGULAR-LIBFAC3 uses FAC-
TORY to enable the computation of characteristic sets and thus the decom-
position of subvarieties into their irreducible components; and GNU MP4 by
Torbjörn Granlund and others provides for multiple precision arithmetic.

Macaulay 2 aims to support efficient computation associated with a wide
variety of high level mathematical objects, including Galois fields, number
fields, polynomial rings, exterior algebras, Weyl algebras, quotient rings, ide-
als, modules, homomorphisms of rings and modules, graded modules, maps
between graded modules, chain complexes, maps between chain complexes,
free resolutions, algebraic varieties, and coherent sheaves. To make the system
easily accessible, standard mathematical notation is followed closely.

As with Macaulay , it was hoped that users would join in the further
development of new algorithms for Macaulay 2, so the developers tried to
make the language available to the users as powerful as possible, yet easy to
use. Indeed, much of the high-level part of the system is written in the same
language available to the user. This ensures that the user will find it just as

1 NSF grants DMS 92-10805, 92-10807, 96-23232, 96-22608, 99-70085, and 99-
70348.

2 SINGULAR-FACTORY, a subroutine library for factorization, by G.-M. Greuel,
R. Stobbe, G. Pfister, H. Schoenemann, and J. Schmidt; available at
ftp://helios.mathematik.uni-kl.de/pub/Math/Singular/Factory/.

3 SINGULAR-LIBFAC, a subroutine library for characteristic sets and irreducible
decomposition, by M. Messollen; available at ftp://helios.mathematik.uni-
kl.de/pub/Math/Singular/Libfac/.

4 GMP, a library for arbitrary precision arithmetic, by Torbjörn Granlund, John
Amanatides, Paul Zimmermann, Ken Weber, Bennet Yee, Andreas Schwab,
Robert Harley, Linus Nordberg, Kent Boortz, Kevin Ryde, and Guillaume Han-
rot; available at ftp://ftp.gnu.org/gnu/gmp/.

easy as the developers did to implement a new type of mathematical object
or to modify the high-level aspects of the current algorithms.

The language available to the user is interpreted. The interpreter itself
is written in a convenient language designed to be mostly type-safe and to
handle memory allocation and initialization automatically. For maximum effi-
ciency, the core mathematical algorithms are written in C++ and compiled,
not interpreted. This includes the arithmetic operations of rings, modules,
and matrices, the Gröbner basis algorithm (in several enhanced versions, tai-
lored for various situations), several algorithms for computing free resolutions
of modules, the algorithm for computing the Hilbert series of a graded ring or
module, the algorithms for computing determinants and Pfaffians, the basis
reduction algorithm, factoring, etc.

In one way Macaulay 2 is like a standard computer algebra system, such
as Mathematica or Maple: the user enters mathematical expressions at the
keyboard, and the program computes the value of the expression and displays
the answer.

Here is the first input prompt offered to the user.

i1 :

In response to the prompt, the user may enter, for example, a simple arith-
metic expression.

i1 : 3/5 + 7/11

68
o1 = --

55

o1 : QQ

The answer itself is displayed to the right of the output label

o1 =

and its type (or class) is displayed to the right of the following label.

o1 :

The symbol QQ appearing in this example denotes the class of all rational
numbers, and is meant to be reminiscent of the notation Q.

Macaulay 2 often finds itself being run in a window with horizontal scroll
bars, so by default it does not wrap output lines, but instead lets them grow
without bound. This book was produced by an automated mechanism that
submits code provided by the authors to Macaulay 2 and incorporates the
result into the text. Output lines that exceed the width of the pages of this
book are indicated with ellipses, as in the following example.

i2 : 100!

o2 = 93326215443944152681699238856266700490715968264381621468592963895 · · ·

Next we describe an important difference between general computer al-
gebra systems (such as Maple and Mathematica) and Macaulay 2. Before
entering an expression involving variables (such as x+y) into Macaulay 2 the
user must first create a ring containing those variables. Rings are important
objects of study in algebraic geometry; quotient rings of polynomial rings,
for example, encapsulate the essential information about a system of polyno-
mial equations, including, for example, the field from which the coefficients
are drawn. Often one has several rings under consideration at once, along
with ring homomorphisms between them, so it is important to treat them
as first-class objects in the computer, capable of being named and manipu-
lated the same way numbers and characters can be manipulated in simpler
programming languages.

Let’s give a hint of the breadth of types of mathematical objects available
in Macaulay 2 with some examples. In Macaulay 2 one defines a quotient ring
of a polynomial ring R over the rational numbers by entering a command such
as the one below.

i3 : R = QQ[x,y,z]/(x^3-y^3-z^3)

o3 = R

o3 : QuotientRing

Having done that, we can compute in the ring.
i4 : (x+y+z)^3

2 2 3 2 2 2 2 3
o4 = 3x y + 3x*y + 2y + 3x z + 6x*y*z + 3y z + 3x*z + 3y*z + 2z

o4 : R

We can make matrices over the ring.
i5 : b = vars R

o5 = | x y z |

1 3
o5 : Matrix R <--- R

i6 : c = matrix {{x^2,y^2,z^2}}

o6 = | x2 y2 z2 |

1 3
o6 : Matrix R <--- R

We can make modules over the ring.
i7 : M = coker b

o7 = cokernel | x y z |

1
o7 : R-module, quotient of R

i8 : N = ker c

o8 = image {2} | x 0 -y2 -z2 |
{2} | -y -z2 x2 0 |
{2} | -z y2 0 x2 |

3
o8 : R-module, submodule of R

We can make projective resolutions of modules.
i9 : res M

1 3 4 4 4
o9 = R <-- R <-- R <-- R <-- R

0 1 2 3 4

o9 : ChainComplex

We can make projective varieties.
i10 : X = Proj R

o10 = X

o10 : ProjectiveVariety

We can make coherent sheaves and compute their cohomology.
i11 : HH^1 cotangentSheaf X

1
o11 = QQ

o11 : QQ-module, free

At this writing, Macaulay 2 is available for GNU/Linux and other flavors
of Unix, and also for Microsoft Windows and the Macintosh operating system.
Although it can be used as a free-standing program, it is most convenient to
use it in an editor’s buffer; Emacs (on Unix or Windows systems) or MPW on
Macintosh systems are currently the editors of choice. To obtain Macaulay 2,
download it from the website5 and unpack the file. Among the resulting files
will be a file called Macaulay2/README.txt, which you should read. It will
tell you how to run the setup script, and how to install a few lines of code
in your emacs init file to enable you to run M2 in an emacs buffer and to edit
Macaulay 2 code. A system administrator of a Unix system may optionally
arrange for those lines of code to be available to every emacs user.

The editors thank the authors of the chapters for their valuable contri-
butions and hard work, and the National Science Foundation for funding the
development of Macaulay 2 and for partial funding of the authors who have
contributed to this volume.

May, 2001 David Eisenbud
Daniel R. Grayson

Michael E. Stillman
Bernd Sturmfels

5 Macaulay 2, a software system for research in algebraic geometry, by Daniel
R. Grayson and Michael E. Stillman; available online in source code form and
compiled for various architectures at http://www.math.uiuc.edu/Macaulay2/.

Part I

Introducing Macaulay 2





Ideals, Varieties and Macaulay 2

Bernd Sturmfels?

This chapter introduces Macaulay 2 commands for some elementary compu-
tations in algebraic geometry. Familiarity with Gröbner bases is assumed.

Many students and researchers alike have their first encounter with Gröb-
ner bases through the delightful text books [1] and [2] by David Cox, John
Little and Donal O’Shea. This chapter illustrates the use of Macaulay 2 for
some computations discussed in these books. It can be used as a supplement
for an advanced undergraduate course or first-year graduate course in com-
putational algebraic geometry. The mathematically advanced reader will find
this chapter a useful summary of some basic Macaulay 2 commands.

1 A Curve in Affine Three-Space

Our first example concerns geometric objects in (complex) affine 3-space. We
start by setting up the ring of polynomial functions with rational coefficients.

i1 : R = QQ[x,y,z]

o1 = R

o1 : PolynomialRing

Various monomial orderings are available in Macaulay 2; since we did not
specify one explicitly, the monomials in the ring R will be sorted in graded
reverse lexicographic order [1, §I.2, Definition 6]. We define an ideal generated
by two polynomials in this ring and assign it to the variable named curve.

i2 : curve = ideal( x^4-y^5, x^3-y^7 )

5 4 7 3
o2 = ideal (- y + x , - y + x )

o2 : Ideal of R

We compute the reduced Gröbner basis of our ideal:
i3 : gb curve

o3 = | y5-x4 x4y2-x3 x8-x3y3 |

o3 : GroebnerBasis

By inspecting leading terms (and using [1, §9.3, Theorem 8]), we see that our
ideal curve does indeed define a one-dimensional affine variety. This can be
tested directly with the following commands in Macaulay 2:

i4 : dim curve

o4 = 1

? Partially supported by the National Science Foundation (DMS-9970254).

i5 : codim curve

o5 = 2

The degree of a curve in complex affine 3-space is the number of intersection
points with a general plane. It coincides with the degree [2, §6.4] of the
projective closure [1, §8.4] of our curve, which we compute as follows:

i6 : degree curve

o6 = 28

The Gröbner basis in o3 contains two polynomials which are not irreducible:
they contain a factor of x3. This shows that our curve is not irreducible over
Q. We first extract the components which are transverse to the plane x = 0:

i7 : curve1 = saturate(curve,ideal(x))

2 5 4 5 3
o7 = ideal (x*y - 1, y - x , x - y )

o7 : Ideal of R

And next we extract the component which lies in the plane x = 0:
i8 : curve2 = saturate(curve,curve1)

3 5
o8 = ideal (x , y )

o8 : Ideal of R

The second component is a multiple line. Hence our input ideal was not
radical. To test equality of ideals we use the command == .

i9 : curve == radical curve

o9 = false

We now replace our curve by its first component:
i10 : curve = curve1

2 5 4 5 3
o10 = ideal (x*y - 1, y - x , x - y )

o10 : Ideal of R

i11 : degree curve

o11 = 13

The ideal of this curve is radical:
i12 : curve == radical curve

o12 = true

Notice that the variable z does not appear among the generators of the ideal.
Our curve consists of 13 straight lines (over C) parallel to the z-axis.

2 Intersecting Our Curve With a Surface

In this section we explore basic operations on ideals, starting with those
described in [1, §4.3]. Consider the following surface in affine 3-space:

i13 : surface = ideal( x^5 + y^5 + z^5 - 1)

5 5 5
o13 = ideal(x + y + z - 1)

o13 : Ideal of R

The union of the curve and the surface is represented by the intersection of
their ideals:

i14 : theirunion = intersect(curve,surface)

6 2 7 2 5 5 5 5 2 5 5 1 · · ·
o14 = ideal (x y + x*y + x*y z - x - y - z - x*y + 1, x y + y · · ·
o14 : Ideal of R

In this example this coincides with the product of the two ideals:
i15 : curve*surface == theirunion

o15 = true

The intersection of the curve and the surface is represented by the sum of
their ideals. We get a finite set of points:

i16 : ourpoints = curve + surface

2 5 4 5 3 5 5 5
o16 = ideal (x*y - 1, y - x , x - y , x + y + z - 1)

o16 : Ideal of R

i17 : dim ourpoints

o17 = 0

The number of points is sixty five:
i18 : degree ourpoints

o18 = 65

Each of the points is multiplicity-free:
i19 : degree radical ourpoints

o19 = 65

The number of points coincides with the number of monomials not in the
initial ideal [2, §2.2]. These are called the standard monomials.

i20 : staircase = ideal leadTerm ourpoints

2 5 5 5
o20 = ideal (x*y , z , y , x )

o20 : Ideal of R

The basis command can be used to list all the standard monomials
i21 : T = R/staircase;

i22 : basis T

o22 = | 1 x x2 x3 x4 x4y x4yz x4yz2 x4yz3 x4yz4 x4z x4z2 x4z3 x4z4 x3y · · ·
1 65

o22 : Matrix T <--- T

The assignment of the quotient ring to the global variable T had a side
effect: the variables x, y, and z now have values in that ring. To bring the
variables of R to the fore again, we must say:

i23 : use R;

Every polynomial function on our 65 points can be written uniquely as a
linear combination of these standard monomials. This representation can be
computed using the normal form command %.

i24 : anyOldPolynomial = y^5*x^5-x^9-y^8+y^3*x^5

5 5 9 5 3 8
o24 = x y - x + x y - y

o24 : R

i25 : anyOldPolynomial % ourpoints

4 3
o25 = x y - x y

o25 : R

Clearly, the normal form is zero if and only the polynomial is in the ideal.
i26 : anotherPolynomial = y^5*x^5-x^9-y^8+y^3*x^4

5 5 9 8 4 3
o26 = x y - x - y + x y

o26 : R

i27 : anotherPolynomial % ourpoints

o27 = 0

o27 : R

3 Changing the Ambient Polynomial Ring

During a Macaulay 2 session it sometimes becomes necessary to change the
ambient ring in which the computations takes place. Our original ring, defined
in i1, is the polynomial ring in three variables over the field Q of rational
numbers with the graded reverse lexicographic order. In this section two
modifications are made: first we replace the field of coefficients by a finite
field, and later we replace the monomial order by an elimination order.

An important operation in algebraic geometry is the decomposition of
algebraic varieties into irreducible components [1, §4.6]. Algebraic algorithms
for this purpose are based on the primary decomposition of ideals [1, §4.7]. A
future version of Macaulay 2 will have an implementation of primary decom-
position over any polynomial ring. The current version of Macaulay 2 has a
command decompose for finding all the minimal primes of an ideal, but, as
it stands, this works only over a finite field.

Let us change our coefficient field to the field with 101 elements:
i28 : R’ = ZZ/101[x,y,z];

We next move our ideal from the previous section into the new ring (fortu-
nately, none of the coefficients of its generators have 101 in the denominator):

i29 : ourpoints’ = substitute(ourpoints,R’)

2 5 4 5 3 5 5 5
o29 = ideal (x*y - 1, y - x , x - y , x + y + z - 1)

o29 : Ideal of R’

i30 : decompose ourpoints’

· · ·
o30 = {ideal (z + 36, y - 1, x - 1), ideal (z + 1, y - 1, x - 1), idea · · ·
o30 : List

Oops, that didn’t fit on the display, so let’s print them out one per line.
i31 : oo / print @@ print;
ideal (z + 36, y - 1, x - 1)

ideal (z + 1, y - 1, x - 1)

ideal (z - 6, y - 1, x - 1)

ideal (z - 14, y - 1, x - 1)

ideal (z - 17, y - 1, x - 1)

3 2 2 3 2 2 · · ·
ideal (x - 46x + 28x*y - 27y + 46x + y + 27, - 16x + x y + x - 15 · · ·

2 2 · · ·
ideal (- 32x - 16x*y + x*z - 16x - 27y - 30z - 14, - 34x - 14x*y + y · · ·

2 2 2 · · ·
ideal (44x + 22x*y + x*z + 22x - 26y - 30z - 6, 18x + 12x*y + y + 1 · · ·

2 2 2 · · ·
ideal (- 41x + 30x*y + x*z + 30x + 38y - 30z + 1, - 26x - 10x*y + y · · ·

2 2 2 · · ·
ideal (39x - 31x*y + x*z - 31x - 46y - 30z + 36, - 32x - 13x*y + y · · ·

2 2 2 · · ·
ideal (- 10x - 5x*y + x*z - 5x - 40y - 30z - 17, - 37x + 35x*y + y · · ·

If we just want to see the degrees of the irreducible components, then we say:
i32 : ooo / degree

o32 = {1, 1, 1, 1, 1, 30, 6, 6, 6, 6, 6}

o32 : List

Note that the expressions oo and ooo refer to the previous and prior-to-
previous output lines respectively.

Suppose we wish to compute the x-coordinates of our sixty five points.
Then we must use an elimination order, for instance, the one described in
[1, §3.2, Exercise 6.a]. We define a new polynomial ring with the elimination
order for {y, z} > {x} as follows:

i33 : S = QQ[z,y,x, MonomialOrder => Eliminate 2]

o33 = S

o33 : PolynomialRing

We move our ideal into the new ring,
i34 : ourpoints’’ = substitute(ourpoints,S)

2 5 4 3 5 5 5 5
o34 = ideal (y x - 1, y - x , - y + x , z + y + x - 1)

o34 : Ideal of S

and we compute the reduced Gröbner basis in this new order:
i35 : G = gens gb ourpoints’’

o35 = | x13-1 y-x6 z5+x5+x4-1 |

1 3
o35 : Matrix S <--- S

To compute the elimination ideal we use the following command:
i36 : ideal selectInSubring(1,G)

13
o36 = ideal(x - 1)

o36 : Ideal of S

4 Monomials Under the Staircase

Invariants of an algebraic variety, such as its dimension and degree, are com-
puted from an initial monomial ideal. This computation amounts to the com-
binatorial task of analyzing the collection of standard monomials, that is, the
monomials under the staircase [1, Chapter 9]. In this section we demonstrate
some basic operations on monomial ideals in Macaulay 2.

Let us create a non-trivial staircase in three dimensions by taking the
third power of the initial monomial from line i20.

i37 : M = staircase^3

3 6 2 4 5 2 9 7 4 2 10 7 5 6 2 5 12 · · ·
o37 = ideal (x y , x y z , x y , x y , x*y z , x*y z , x y z , x*y , · · ·
o37 : Ideal of R

The number of current generators of this ideal equals
i38 : numgens M

o38 = 20

To see all generators we can transpose the matrix of minimal generators:
i39 : transpose gens M

o39 = {-9} | x3y6 |
{-11} | x2y4z5 |
{-11} | x2y9 |
{-11} | x7y4 |
{-13} | xy2z10 |
{-13} | xy7z5 |
{-13} | x6y2z5 |
{-13} | xy12 |
{-13} | x6y7 |
{-13} | x11y2 |
{-15} | z15 |
{-15} | y5z10 |
{-15} | x5z10 |
{-15} | y10z5 |
{-15} | x5y5z5 |
{-15} | x10z5 |
{-15} | y15 |
{-15} | x5y10 |
{-15} | x10y5 |
{-15} | x15 |

20 1
o39 : Matrix R <--- R

Note that this generating set is not minimal; see o48 below. The number of
standard monomials equals

i40 : degree M

o40 = 690

To list all the standard monomials we first create the residue ring
i41 : S = R/M

o41 = S

o41 : QuotientRing

and then we ask for a vector space basis of the residue ring:
i42 : basis S

o42 = | 1 x x2 x3 x4 x5 x6 x7 x8 x9 x10 x11 x12 x13 x14 x14y x14yz x14 · · ·
1 690

o42 : Matrix S <--- S

Let us count how many standard monomials there are of a given degree. The
following table represents the Hilbert function of the residue ring.

i43 : tally apply(flatten entries basis(S),degree)

o43 = Tally{{0} => 1 }
{1} => 3
{10} => 63
{11} => 69
{12} => 73
{13} => 71
{14} => 66
{15} => 53
{16} => 38
{17} => 23
{18} => 12



10 B. Sturmfels

{19} => 3
{2} => 6
{3} => 10
{4} => 15
{5} => 21
{6} => 28
{7} => 36
{8} => 45
{9} => 54

o43 : Tally

Thus the largest degree of a standard monomial is nineteen, and there are
three standard monomials of that degree:

i44 : basis(19,S)

o44 = | x14yz4 x9yz9 x4yz14 |

1 3
o44 : Matrix S <--- S

The most recently defined ring involving x, y, and z was S, so all computations
involving those variables are done in the residue ring S. For instance, we can
also obtain the standard monomials of degree nineteen as follows:

i45 : (x+y+z)^19

14 4 9 9 4 14
o45 = 58140x y*z + 923780x y*z + 58140x y*z

o45 : S

An operation on ideals which will occur frequently throughout this book is
the computation of minimal free resolutions. This is done as follows:

i46 : C = res M

1 16 27 12
o46 = R <-- R <-- R <-- R <-- 0

0 1 2 3 4

o46 : ChainComplex

This shows that our ideal M has sixteen minimal generators. They are the
entries in the leftmost matrix of the chain complex C:

i47 : C.dd_1

o47 = | x3y6 x7y4 x2y9 x2y4z5 x11y2 xy12 x6y2z5 xy7z5 xy2z10 x15 y15 x · · ·
1 16

o47 : Matrix R <--- R

This means that four of the twenty generators in o39 were redundant. We
construct the set consisting of the four redundant generators as follows:

i48 : set flatten entries gens M - set flatten entries C.dd_1

6 7 10 5 5 10 5 5 5
o48 = Set {x y , x y , x y , x y z }

o48 : Set

Here flatten entries turns the matrix M into a single list. The command
set turns that list into a set, to which we can apply the difference operation
for sets.

Let us now take a look at the first syzygies (or minimal S-pairs [1, §2.9])
among the sixteen minimal generators. They correspond to the columns of
the second matrix in our resolution C:

i49 : C.dd_2

o49 = {9} | -y3 -x4 0 -z5 0 0 0 0 0 0 0 0 0 0 0 · · ·
{11} | 0 y2 0 0 0 -x4 0 0 -z5 0 0 0 0 0 0 · · ·
{11} | x 0 -y3 0 0 0 0 0 0 -z5 0 0 0 0 0 · · ·
{11} | 0 0 0 xy2 -y3 0 -x4 0 x5 y5 0 -z5 0 0 0 · · ·
{13} | 0 0 0 0 0 y2 0 0 0 0 0 0 0 -x4 0 · · ·
{13} | 0 0 x 0 0 0 0 -y3 0 0 0 0 0 0 0 · · ·
{13} | 0 0 0 0 0 0 y2 0 0 0 0 0 0 0 - · · ·
{13} | 0 0 0 0 x 0 0 0 0 0 -y3 0 0 0 0 · · ·
{13} | 0 0 0 0 0 0 0 0 0 0 0 xy2 -y3 0 0 · · ·
{15} | 0 0 0 0 0 0 0 0 0 0 0 0 0 y2 0 · · ·
{15} | 0 0 0 0 0 0 0 x 0 0 0 0 0 0 0 · · ·
{15} | 0 0 0 0 0 0 0 0 0 0 0 0 0 0 y · · ·
{15} | 0 0 0 0 0 0 0 0 0 0 x 0 0 0 0 · · ·
{15} | 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 · · ·
{15} | 0 0 0 0 0 0 0 0 0 0 0 0 x 0 0 · · ·
{15} | 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 · · ·

16 27
o49 : Matrix R <--- R

The first column represents the S-pair between the first generator x3y6 and
the third generator x2y9. It is natural to form the S-pair graph with 16
vertices and 27 edges represented by this matrix. According to the general
theory described in [3], this is a planar graph with 12 regions. The regions
correspond to the 12 second syzygies, that is, to the columns of the matrix

i50 : C.dd_3

o50 = {12} | z5 0 0 0 0 0 0 0 0 0 0 0 |
{13} | 0 z5 0 0 0 0 0 0 0 0 0 0 |
{14} | 0 0 z5 0 0 0 0 0 0 0 0 0 |
{14} | -y3 -x4 0 0 0 0 0 0 0 0 0 0 |
{14} | 0 0 -y5 z5 0 0 0 0 0 0 0 0 |
{15} | 0 0 0 0 z5 0 0 0 0 0 0 0 |
{15} | 0 0 0 0 -x5 z5 0 0 0 0 0 0 |
{16} | 0 0 0 0 0 0 z5 0 0 0 0 0 |
{16} | 0 y2 0 0 -x4 0 0 0 0 0 0 0 |
{16} | x 0 -y3 0 0 0 0 0 0 0 0 0 |
{16} | 0 0 0 0 0 0 -y5 z5 0 0 0 0 |
{16} | 0 0 0 -y3 0 -x4 0 0 0 0 0 0 |
{16} | 0 0 0 0 0 0 0 -y5 z5 0 0 0 |
{17} | 0 0 0 0 0 0 0 0 0 z5 0 0 |
{17} | 0 0 0 0 0 0 0 0 0 -x5 z5 0 |
{17} | 0 0 0 0 0 0 0 0 0 0 -x5 z5 |
{18} | 0 0 0 0 y2 0 0 0 0 -x4 0 0 |
{18} | 0 0 x 0 0 0 -y3 0 0 0 0 0 |
{18} | 0 0 0 0 0 y2 0 0 0 0 -x4 0 |
{18} | 0 0 0 x 0 0 0 -y3 0 0 0 0 |
{18} | 0 0 0 0 0 0 0 0 -y3 0 0 -x4 |
{20} | 0 0 0 0 0 0 0 0 0 y2 0 0 |

{20} | 0 0 0 0 0 0 x 0 0 0 0 0 |
{20} | 0 0 0 0 0 0 0 0 0 0 y2 0 |
{20} | 0 0 0 0 0 0 0 x 0 0 0 0 |
{20} | 0 0 0 0 0 0 0 0 0 0 0 y2 |
{20} | 0 0 0 0 0 0 0 0 x 0 0 0 |

27 12
o50 : Matrix R <--- R

But we are getting ahead of ourselves. Homological algebra and resolutions
will be covered in the next chapter, and monomial ideals will appear in the
chapter of Hoşten and Smith. Let us return to Cox, Little and O’Shea [2].

5 Pennies, Nickels, Dimes and Quarters

We now come to an application of Gröbner bases which appears in [2, Sec-
tion 8.1]: Integer Programming. This is the problem of minimizing a linear
objective function over the set of non-negative integer solutions of a system
of linear equations. We demonstrate some techniques for doing this in Mac-
aulay 2. Along the way, we learn about multigraded polynomial rings and
how to compute Gröbner bases with respect to monomial orders defined by
weights. Our running example is the linear system defined by the matrix:

i51 : A = {{1, 1, 1, 1},
{1, 5,10,25}}

o51 = {{1, 1, 1, 1}, {1, 5, 10, 25}}

o51 : List

For the algebraic study of integer programming problems, a good starting
point is to work in a multigraded polynomial ring, here in four variables:

i52 : R = QQ[p,n,d,q, Degrees => transpose A]

o52 = R

o52 : PolynomialRing

The degree of each variable is the corresponding column vector of the matrix
Each variable represents one of the four coins in the U.S. currency system:

i53 : degree d

o53 = {1, 10}

o53 : List

i54 : degree q

o54 = {1, 25}

o54 : List

Each monomial represents a collection of coins. For instance, suppose you
own four pennies, eight nickels, ten dimes, and three quarters:

i55 : degree(p^4*n^8*d^10*q^3)

o55 = {25, 219}

o55 : List

Then you have a total of 25 coins worth two dollars and nineteen cents. There
are nine other possible ways of having 25 coins of the same value:

i56 : h = basis({25,219}, R)

o56 = | p14n2d2q7 p9n8d2q6 p9n5d6q5 p9n2d10q4 p4n14d2q5 p4n11d6q4 p4n8 · · ·
1 9

o56 : Matrix R <--- R

For just counting the number of columns of this matrix we can use the com-
mand

i57 : rank source h

o57 = 9

How many ways can you make change for ten dollars using 100 coins?
i58 : rank source basis({100,1000}, R)

o58 = 182

A typical integer programming problem is this: among all 182 ways of ex-
pressing ten dollars using 100 coins, which one uses the fewest dimes? We set
up the Conti-Traverso algorithm [2, §8.1] for answering this question. We use
the following ring with the lexicographic order and with the variable order:
dimes (d) before pennies (p) before nickels (n) before quarters (q).

i59 : S = QQ[x, y, d, p, n, q,
MonomialOrder => Lex, MonomialSize => 16]

o59 = S

o59 : PolynomialRing

The option MonomialSize advises Macaulay 2 to use more space to store the
exponents of monomials, thereby avoiding a potential overflow.

We define an ideal with one generator for each column of the matrix A.
i60 : I = ideal( p - x*y, n - x*y^5, d - x*y^10, q - x*y^25)

5 10 25
o60 = ideal (- x*y + p, - x*y + n, - x*y + d, - x*y + q)

o60 : Ideal of S

The integer program is solved by normal form reduction with respect to the
following Gröbner basis consisting of binomials.

i61 : transpose gens gb I

o61 = {-6} | p5q-n6 |
{-4} | d4-n3q |
{-3} | yn2-dp |
{-6} | yp4q-dn4 |
{-4} | yd3-pnq |
{-6} | y2p3q-d2n2 |
{-5} | y2d2n-p2q |
{-7} | y2d2p3-n5 |
{-6} | y3p2q-d3 |
{-6} | y3dp2-n3 |
{-5} | y4p-n |
{-6} | y5n-d |

{-8} | y6d2-pq |
{-16} | y15d-q |
{-7} | xq-y5d2 |
{-5} | xn-y3p2 |
{-2} | xd-n2 |
{-2} | xy-p |

18 1
o61 : Matrix S <--- S

We fix the quotient ring, so the reduction to normal form will happen auto-
matically.

i62 : S’ = S/I

o62 = S’

o62 : QuotientRing

You need at least two dimes to express one dollar with ten coins.
i63 : x^10 * y^100

2 6 2
o63 = d n q

o63 : S’

But you can express ten dollars with a hundred coins none of which is a dime.
i64 : x^100 * y^1000

75 25
o64 = n q

o64 : S’

The integer program is infeasible if and only if the normal form still contains
the variable x or the variable y. For instance, you cannot express ten dollars
with less than forty coins:

i65 : x^39 * y^1000

25 39
o65 = y q

o65 : S’

We now introduce a new term order on the polynomial ring, defined by as-
signing a weight to each variable. Specifically, we assign weights for each of
the coins. For instance, let pennies have weight 5, nickels weight 7, dimes
weight 13 and quarters weight 17.

i66 : weight = (5,7,13,17)

o66 = (5, 7, 13, 17)

o66 : Sequence

We set up a new ring with the resulting weight term order, and work modulo
the same ideal as before in this new ring.

i67 : T = QQ[x, y, p, n, d, q,
Weights => {{1,1,0,0,0,0},{0,0,weight}},
MonomialSize => 16]/

(p - x*y, n - x*y^5, d - x*y^10, q - x*y^25);

One dollar with ten coins:
i68 : x^10 * y^100

5 2 3
o68 = p d q

o68 : T

Ten dollars with one hundred coins:
i69 : x^100 * y^1000

60 3 37
o69 = p n q

o69 : T

Here is an optimal solution which involves all four types of coins:
i70 : x^234 * y^5677

2 4 3 225
o70 = p n d q

o70 : T

Projective Geometry and Homological Algebra

David Eisenbud?

We provide an introduction to many of the homological commands in Mac-
aulay 2 (modules, free resolutions, Ext and Tor. . . ) by means of examples
showing how to use homological tools to study projective varieties.

In this chapter we will illustrate how one can manipulate projective vari-
eties and sheaves, using the rich collection of tools Macaulay 2 provides. One
of our goals is to show how homological methods can be effective in solving
concrete geometric problems.

The first four sections can be read by anyone who knows about projective
varieties at the level of a first graduate course and knows the definitions of
Ext and Tor. The last section assumes that the reader is familiar with the
theory of curves and surfaces roughly at the level of the books of Hartshorne
[7] and Harris [6].

We will work with projective schemes over a field kk. Macaulay 2 can
work over any finite field of characteristic at most 32749, and also a variety
of fields in characteristic 0 (except for the primary decomposition commands,
which at this writing are still restricted to positive characteristics). Our main
interest is in geometry over an algebraically closed field of characteristic 0.
Nevertheless, it is most convenient to work over a large prime field. It is
known that the intermediate results in Gröbner basis computations (as in
the Euclidean Algorithm computations they generalize) often involve coeffi-
cients far larger than those in the input data, so that work in characteristic
zero essentially requires infinite precision arithmetic, a significant additional
overhead. If we work over a finite field where the scalars can be represented
in one machine word, we avoid this coefficient explosion. Experience with the
sort of computations we will be doing shows that working over Z/p, where
p is a moderately large prime, gives results identical to the results we would
get in characteristic 0. Of course one still has to be careful about the fact
that our fields are not algebraically closed, especially when using primary de-
composition. The largest prime p we can work with being 32749, we choose
the field Z/32749. The name of the Macaulay 2 constant representing the
integers is ZZ, and by analogy we will call our field kk:

i1 : kk = ZZ/32749

o1 = kk

o1 : QuotientRing

In Macaulay 2 we will represent projective space Pn by its homogeneous
coordinate ring ringPn = kk[x0, . . . , xn]. A projective scheme X in Pn may
? Supported by the NSF.

be most conveniently represented, depending on the situation, by its homo-
geneous ideal idealX or its homogeneous coordinate ring, represented either
as a ring ringPn/idealX or as a module OX over ringPn. Coherent sheaves
on the projective space, or on its subvarieties, will be represented by finitely
generated graded modules over ringPn, using the Serre correspondence. For
example, the structure sheaf OX of the subvariety X would be represented
by the module ringPn^1/idealX; here ringPn^1 denotes the free module of
rank one over the ring ringPn.

1 The Twisted Cubic

As a first illustration, we give three constructions of the twisted cubic curve
in P3. We represent P3 by

i2 : ringP3 = kk[x_0..x_3]

o2 = ringP3

o2 : PolynomialRing

The twisted cubic is the image of the map P1 → P
3 sending a point with

homogeneous coordinates (s, t) to the point with homogeneous coordinates
(s3, s2t, st2, t3). We can compute its relations directly with

i3 : ringP1 = kk[s,t]

o3 = ringP1

o3 : PolynomialRing

i4 : cubicMap = map(ringP1,ringP3,{s^3, s^2*t, s*t^2, t^3})

3 2 2 3
o4 = map(ringP1,ringP3,{s , s t, s*t , t })

o4 : RingMap ringP1 <--- ringP3

i5 : idealCubic = kernel cubicMap

2 2
o5 = ideal (x - x x , x x - x x , x - x x )

2 1 3 1 2 0 3 1 0 2

o5 : Ideal of ringP3

We could also use Macaulay 2’s built-in facility, and say
i6 : idealCubic2 = monomialCurveIdeal(ringP3,{1,2,3})

2 2
o6 = ideal (x x - x x , x - x x , x - x x )

1 2 0 3 2 1 3 1 0 2

o6 : Ideal of ringP3

which uses precisely the same method.
Of course we might remember that the ideal of the twisted cubic is gen-

erated by the 2× 2 minors of the matrix(
x0 x1 x2

x1 x2 x3

)
,

which we can realize with the commands
i7 : M = matrix{{x_0,x_1,x_2},{x_1,x_2,x_3}}

o7 = | x_0 x_1 x_2 |
| x_1 x_2 x_3 |

2 3
o7 : Matrix ringP3 <--- ringP3

i8 : idealCubic3 = minors(2, M)

2 2
o8 = ideal (- x + x x , - x x + x x , - x + x x )

1 0 2 1 2 0 3 2 1 3

o8 : Ideal of ringP3

We can get some useful information about the ideal idealCubic with
i9 : codim idealCubic

o9 = 2

i10 : degree idealCubic

o10 = 3

This shows that we do indeed have a cubic curve. Note that the command
i11 : dim idealCubic

o11 = 2

gives 2, not 1; it represents the dimension of the ideal in ringP3, the dimen-
sion of the affine cone over the curve.

We can easily assure ourselves that these ideals are the same. For example,
to see whether the ideal idealCubic is contained in the ideal of minors of M,
we can reduce the former modulo the latter, and see whether we get zero. The
reduction operator % takes two maps with the same target as its arguments,
so we must replace each ideal by a matrix whose entries generate it. This is
done by the function gens as in

i12 : gens idealCubic

o12 = | x_2^2-x_1x_3 x_1x_2-x_0x_3 x_1^2-x_0x_2 |

1 3
o12 : Matrix ringP3 <--- ringP3

Thus for one of the inclusions we check

i13 : 0 == (gens idealCubic)%(gens idealCubic3)

o13 = true

Both inclusions can be checked automatically in this way with
i14 : idealCubic == idealCubic3

o14 = true

2 The Cotangent Bundle of P3

Many invariants of varieties are defined in terms of their tangent and cotan-
gent bundles. We identify a bundle with its sheaf of sections, which is locally
free. Any coherent locally free sheaf arises this way. (One can also regard a
bundle as a variety in its own right, but this view is used in algebraic geom-
etry more rarely.) In this section and the next we construct the cotangent
bundle ΩP3 of P3 and its restriction to the twisted cubic above.

Consulting Hartshorne [7, Theorem II.8.13], we find that the cotangent
bundle to Pn can be described by the cotangent sequence:

0 - ΩPn - OPn(−1)n+1 f- OPn - 0

where f is defined by the matrix of variables (x0, . . . , xn). We can translate
this description directly into the language of Macaulay 2, here in the case
n = 3:

i15 : f = vars ringP3

o15 = | x_0 x_1 x_2 x_3 |

1 4
o15 : Matrix ringP3 <--- ringP3

i16 : OmegaP3 = kernel f

o16 = image {1} | 0 0 0 -x_1 -x_2 -x_3 |
{1} | 0 -x_2 -x_3 x_0 0 0 |
{1} | -x_3 x_1 0 0 x_0 0 |
{1} | x_2 0 x_1 0 0 x_0 |

4
o16 : ringP3-module, submodule of ringP3

Note that the module which we specified as a kernel is now given as the image
of a matrix. We can recover this matrix with

i17 : g=generators OmegaP3

o17 = {1} | 0 0 0 -x_1 -x_2 -x_3 |
{1} | 0 -x_2 -x_3 x_0 0 0 |
{1} | -x_3 x_1 0 0 x_0 0 |
{1} | x_2 0 x_1 0 0 x_0 |

4 6
o17 : Matrix ringP3 <--- ringP3

and we could correspondingly write
i18 : OmegaP3=image g

o18 = image {1} | 0 0 0 -x_1 -x_2 -x_3 |
{1} | 0 -x_2 -x_3 x_0 0 0 |
{1} | -x_3 x_1 0 0 x_0 0 |
{1} | x_2 0 x_1 0 0 x_0 |

4
o18 : ringP3-module, submodule of ringP3

An even more elementary way to give a module is by generators and relations,
and we can see this “free presentation” too with

i19 : presentation OmegaP3

o19 = {2} | x_1 0 0 x_0 |
{2} | x_3 x_0 0 0 |
{2} | -x_2 0 x_0 0 |
{2} | 0 x_2 x_3 0 |
{2} | 0 -x_1 0 x_3 |
{2} | 0 0 -x_1 -x_2 |

6 4
o19 : Matrix ringP3 <--- ringP3

The astute reader will have noticed that we have just been computing the
first few terms in the free resolution of the cokernel of the map of free modules
corresponding to f. We could see the whole resolution at once with

i20 : G = res coker f

1 4 6 4 1
o20 = ringP3 <-- ringP3 <-- ringP3 <-- ringP3 <-- ringP3 <-- 0

0 1 2 3 4 5

o20 : ChainComplex

and then see all the matrices in the resolution with
i21 : G.dd

1 4
o21 = 0 : ringP3 <----------------------- ringP3 : 1

| x_0 x_1 x_2 x_3 |

4 6
1 : ringP3 <----------------------------------------- ringP3 : 2

{1} | -x_1 -x_2 0 -x_3 0 0 |
{1} | x_0 0 -x_2 0 -x_3 0 |
{1} | 0 x_0 x_1 0 0 -x_3 |
{1} | 0 0 0 x_0 x_1 x_2 |

6 4
2 : ringP3 <------------------------------- ringP3 : 3

{2} | x_2 x_3 0 0 |
{2} | -x_1 0 x_3 0 |
{2} | x_0 0 0 x_3 |
{2} | 0 -x_1 -x_2 0 |
{2} | 0 x_0 0 -x_2 |
{2} | 0 0 x_0 x_1 |

4 1
3 : ringP3 <---------------- ringP3 : 4

{3} | -x_3 |
{3} | x_2 |
{3} | -x_1 |
{3} | x_0 |

1
4 : ringP3 <----- 0 : 5

0

o21 : ChainComplexMap

or just one of them, say the second, with
i22 : G.dd_2

o22 = {1} | -x_1 -x_2 0 -x_3 0 0 |
{1} | x_0 0 -x_2 0 -x_3 0 |
{1} | 0 x_0 x_1 0 0 -x_3 |
{1} | 0 0 0 x_0 x_1 x_2 |

4 6
o22 : Matrix ringP3 <--- ringP3

Note that this matrix does not look exactly the same as the matrix produced
by computing the kernel of f. This is because when Macaulay 2 is asked to
compute a whole resolution, it does not do the “obvious” thing and compute
kernels over and over; it defaults to a more efficient algorithm, first proposed
by Frank Schreyer [10, Appendix].

Any graded map of free modules, such as a map in a graded free reso-
lution of a graded module, comes with some numerical data: the degrees of
the generators of the source and target free modules. We can extract this
information one module at a time with the command degrees, as in

i23 : degrees source G.dd_2

o23 = {{2}, {2}, {2}, {2}, {2}, {2}}

o23 : List

i24 : degrees target G.dd_2

o24 = {{1}, {1}, {1}, {1}}

o24 : List

Macaulay 2 has a more convenient mechanism for examining this numer-
ical data, which we take time out to explain. First, for the resolution just
computed, we can call

i25 : betti G

o25 = total: 1 4 6 4 1
0: 1 4 6 4 1

The diagram shows the degrees of the generators of each free module in the
resolution in coded form. To understand the code, it may be helpful to look
at a less symmetric example, say the free resolution of ringP3^1/I where I
is the ideal generated by the minors of the following 2× 4 matrix.

i26 : m = matrix{{x_0^3, x_1^2, x_2,x_3},{x_1^3,x_2^2,x_3,0}}

o26 = | x_0^3 x_1^2 x_2 x_3 |
| x_1^3 x_2^2 x_3 0 |

2 4
o26 : Matrix ringP3 <--- ringP3

We do this with
i27 : I = minors(2,m)

5 3 2 3 3 3 2 3 2 2
o27 = ideal (- x + x x , - x x + x x , - x + x x , -x x , -x x , -x )

1 0 2 1 2 0 3 2 1 3 1 3 2 3 3

o27 : Ideal of ringP3

i28 : F = res(ringP3^1/I)

1 6 8 3
o28 = ringP3 <-- ringP3 <-- ringP3 <-- ringP3 <-- 0

0 1 2 3 4

o28 : ChainComplex

i29 : betti F

o29 = total: 1 6 8 3
0: 1 . . .
1: . 1 . .
2: . 2 2 .
3: . 2 2 .
4: . 1 4 3

The resulting Betti diagram should be interpreted as follows. First, the
maps go from right to left, so the beginning of the resolution is on the left.
The given Betti diagram thus corresponds to an exact sequence of graded
free modules

F0
� F1

� F2
� F3

� 0.

The top row of the diagram, 1,6,8,3, shows the ranks of the free modules Fi in
the resolution. For example the 1 on the left means that F0 has rank 1 (and,
indeed, the module ringP3^1/I we are resolving is cyclic). The 6 shows that
the rank of F1 is 6, or equivalently that the ideal I is minimally generated by
6 elements—in this case the 6 =

(
4
2

)
minors of size 2 of the 2× 4 matrix m.

The first column of the diagram shows degrees. The successive columns
indicate how many generators of each degree occur in the successive Fi. The
free module F0 has a single generator in degree 0, and this is the signifi-
cance of the second column. Note that F1 could not have any generators of
degree less than or equal to zero, because the resolution is minimal! Thus
for compactness, the diagram is skewed: in each successive column the places
correspond to larger degrees. More precisely, a number a occurring opposite
the degree indication “i:” in the column corresponding to Fj signifies that
Fj has a generators in degree i + j. Thus for example the 1 in the third
column opposite the one on the left corresponds to a generator of degree 2

in the free module F1; and altogether F1 has one generator of degree 2, two
generators of degree 3, two of degree 4 and one of degree 5.

Returning to the diagram
i30 : betti G

o30 = total: 1 4 6 4 1
0: 1 4 6 4 1

we see that the successive free modules of G are each generated in degree 1
higher than the previous one; that is, the matrices in G.dd all have linear
entries, as we have already seen.

3 The Cotangent Bundle of a Projective Variety

It is easy to construct the cotangent bundle ΩX of a projective variety X
starting from the cotangent bundle of the ambient projective space. We use
the conormal sequence (Hartshorne [7, Proposition II.8.12] or Eisenbud [4,
Proposition 16.3]). Writing I for the ideal of a variety X in Pn there is an
exact sequence of sheaves

I
δ- ΩPn ⊗OX - ΩX - 0

where the map δ takes a function f to the element df ⊗ 1. If I is generated
by forms f1, . . . , fm then δ is represented by the Jacobian matrix (dfi/dxj).

First of all, we must compute a module corresponding to ΩPn ⊗OX , the
restriction of the sheaf ΩPn to X. The simplest approach would be to take
the tensor product of graded modules representing ΩPn and OX . The result
would represent the right sheaf, but would not be the module of twisted
global sections of ΩPn ⊗ OX (the unique module of depth two representing
the sheaf). This would make further computations less efficient.

Thus we take a different approach: since the cotangent sequence given in
the previous section is a sequence of locally free sheaves, it is locally split,
and thus remains exact when tensored by OX . Consequently ΩPn ⊗ OX is
also represented by the kernel of the map f ⊗OX , where f is the map used
in the definition of the cotangent bundle of Pn. In Macaulay 2, working on
P

3, with X the twisted cubic, we can translate this into
i31 : OmegaP3res = kernel (f ** (ringP3^1/idealCubic))

o31 = subquotient ({1} | -x_3 0 0 -x_2 -x_3 0 -x_1 -x_2 -x_3 · · ·
{1} | x_2 -x_3 0 x_1 0 -x_3 x_0 0 0 · · ·
{1} | 0 x_2 -x_3 0 x_1 0 0 x_0 0 · · ·
{1} | 0 0 x_2 0 0 x_1 0 0 x_0 · · ·

4
o31 : ringP3-module, subquotient of ringP3

(The operator ** is Macaulay 2’s symbol for tensor product.) Since the map is
a map between free modules over ringP3/idealCubic, the kernel has depth
(at least) two.

Next, we form the Jacobian matrix of the generators of idealCubic, which
represents a map from this ideal to the free module ringP3^4.

i32 : delta1 = jacobian idealCubic

o32 = {1} | 0 -x_3 -x_2 |
{1} | -x_3 x_2 2x_1 |
{1} | 2x_2 x_1 -x_0 |
{1} | -x_1 -x_0 0 |

4 3
o32 : Matrix ringP3 <--- ringP3

We need to make this into a map to OmegaP3res, which as defined is a
subquotient of ringP3^4. To this end we must first express the image of
delta1 in terms of the generators of OmegaP3res. The division command //
does this with

i33 : delta2 = delta1 // (gens OmegaP3res)

o33 = {2} | 0 1 0 |
{2} | 2 0 0 |
{2} | 0 0 0 |
{2} | 0 0 2 |
{2} | 0 1 0 |
{2} | -1 0 0 |
{2} | 0 0 0 |
{2} | 0 0 -1 |
{2} | 0 -1 0 |

9 3
o33 : Matrix ringP3 <--- ringP3

Once this is done we can use this matrix to form the necessary map δ : I →
ΩP3 ⊗OX :

i34 : delta = map(OmegaP3res, module idealCubic, delta2)

o34 = {2} | 0 1 0 |
{2} | 2 0 0 |
{2} | 0 0 0 |
{2} | 0 0 2 |
{2} | 0 1 0 |
{2} | -1 0 0 |
{2} | 0 0 0 |
{2} | 0 0 -1 |
{2} | 0 -1 0 |

o34 : Matrix

A minimal free presentation of ΩX — or rather of one module over ringP3
that represents it — can be obtained with

i35 : OmegaCubic = prune coker delta

o35 = cokernel {2} | -10917x_3 0 -10917x_3 x_2 0 0 · · ·
{2} | 0 0 x_2 0 16374x_3 0 · · ·
{2} | 0 -x_3 0 16373x_3 0 x_2 · · ·
{2} | x_3 x_2 0 0 0 0 · · ·
{2} | 0 0 0 0 0 -2x_3 · · ·
{2} | 0 0 0 0 x_2 0 · · ·

6
o35 : ringP3-module, quotient of ringP3

We have used the function prune to compute minimal presentation matrices;
these often make subsequent computations faster, and also allow us to inspect
the final answer more easily.

The module OmegaCubic represents the sheaf ΩX , where X is the cubic,
but it is not the simplest possibility. A better representative is the graded
module ⊕d∈ZH0(ΩX(d)). We can at least find a minimal presentation of the
submodule ⊕d≥0H0(ΩX(d)) with

i36 : prune HH^0((sheaf OmegaCubic)(>=0))

o36 = cokernel {1} | 16374x_3 16374x_2 16374x_1 |
{1} | x_2 x_1 x_0 |

2
o36 : ringP3-module, quotient of ringP3

The large coefficients appearing in the matrix arise in finite characteristic
as the result of chance division by small integers. We see from the degrees
labeling the rows of the matrix in the output of this command that the
generators of the submodule are in degree 1, so in particular H0(ΩX) = 0. It
follows that that H0(ΩX(d)) = 0 for all d ≤ 0, so the submodule we computed
was actually the whole module that we wanted! (If this had not been the
case we could have tried HH^0((sheaf OmegaCubic)(>=d)) to compute the
cohomology of all the twists greater than a given negative integer d, or simply
used the submodule we had already computed, since it also represents the
sheaf ΩX .)

The sequence of commands we have used to construct the cotangent sheaf
can be obtained also with the following built-in commands.

i37 : Cubic = Proj(ringP3/idealCubic)

o37 = Cubic

o37 : ProjectiveVariety

i38 : cotangentSheaf Cubic

o38 = cokernel {1} | x_2 x_1 x_0 |
{1} | -x_3 -x_2 -x_1 |

2
o38 : coherent sheaf on Cubic, quotient of OO (-1)

Cubic

Since X is a smooth curve, its cotangent bundle is equal to its canonical
bundle, and also to its dualizing sheaf (see Hartshorne [7, sections II.8 and
III.7] for definitions). We will see another (generally more efficient) method
of computing this dualizing sheaf by using Ext and duality theory.

4 Intersections by Serre’s Method

To introduce homological algebra in a simple geometric context, consider
the problem of computing the intersection multiplicities of two varieties X
and Y in Pn, assuming for simplicity that dimX + dimY = n and that the

two meet in a zero-dimensional scheme. Beginning in the 19th century, many
people struggled to make a definition of local intersection multiplicity that
would make Bézout’s Theorem true: the product of the degrees of X and Y
should be the number of points of intersection, each counted with its local
intersection multiplicity (multiplied by the degree of the point, if the point
is not rational over the ground field). In the simplest case, where the two
varieties are Cohen-Macaulay, the right answer is that a point p should count
with multiplicity equal to the length of the local ring OX,p ⊗OPn,p OY,p, and
at first it was naively assumed that this would be the right answer in general.

Here is a famous example in P4 showing that the naive value can be
wrong: in it, the scheme X is a 2-plane and the scheme Y = L1 ∪ L2 is the
union of two 2-planes. The planes L1 and L2 meet at just one point p, and
we assume that X passes through p as well, and is general enough so that it
meets Y only in p. Since degree(X) = 1,degree(Y ) = 2, Bézout’s Theorem
requires that the multiplicity of the intersection at p should be 2. However,
we have:

i39 : ringP4 = kk[x_0..x_4]

o39 = ringP4

o39 : PolynomialRing

i40 : idealX = ideal(x_1+x_3, x_2+x_4)

o40 = ideal (x + x , x + x )
1 3 2 4

o40 : Ideal of ringP4

i41 : idealL1 = ideal(x_1,x_2)

o41 = ideal (x , x )
1 2

o41 : Ideal of ringP4

i42 : idealL2 = ideal(x_3,x_4)

o42 = ideal (x , x )
3 4

o42 : Ideal of ringP4

i43 : idealY = intersect(idealL1,idealL2)

o43 = ideal (x x , x x , x x , x x )
2 4 1 4 2 3 1 3

o43 : Ideal of ringP4

i44 : degree(idealX+idealY)

o44 = 3

That is, the length of OX,p ⊗OPn,p OY,p is 3 rather than 2. (We can do this
computation without first passing to local rings because there is only one
point of intersection, and because all the constructions we are using commute
with localization.)

It was the happy discovery of Jean-Pierre Serre [11, V.B.3] that the naive
measure of intersection multiplicity can be fixed in a simple way that works
for all intersections in smooth varieties. One simply replaces the length of the
tensor product

OX,p ⊗OPn,p OY,p = TorOPn,p0 (OX,p,OY,p)

with the alternating sum of the Tor functors∑
i

(−1)i length TorOPn,pi (OX,p,OY,p).

In Macaulay 2 we can proceed as follows:
i45 : degree Tor_0(ringP4^1/idealX, ringP4^1/idealY)

o45 = 3

i46 : degree Tor_1(ringP4^1/idealX, ringP4^1/idealY)

o46 = 1

i47 : degree Tor_2(ringP4^1/idealX, ringP4^1/idealY)

o47 = 0

The other Tor’s are 0 because the projective dimension of ringP4^1/idealX
is only two, as we see from

i48 : res (ringP4^1/idealX)

1 2 1
o48 = ringP4 <-- ringP4 <-- ringP4 <-- 0

0 1 2 3

o48 : ChainComplex

Thus, indeed, the alternating sum is 2, and Bézout’s Theorem is upheld.

5 A Mystery Variety in P3

In the file mystery.m2 is a function called mystery that will compute the
ideal of a subvariety X of P3. We’ll reveal what it does at the end of the
chapter. Let’s run it.

i49 : ringP3 = kk[x_0..x_3];

i50 : load "mystery.m2"

i51 : idealX = mystery ringP3

4 2 2 2 2 2 2 2 · · ·
o51 = ideal (x - 2x x x - x x x + x x , x x - 10915x x x - 10917x · · ·

1 0 1 3 1 2 3 0 3 0 1 0 1 2 · · ·
o51 : Ideal of ringP3

We can’t see all the generators of the ideal; the same file contains a function
prettyPrint which will display the generators visibly.

i52 : prettyPrint gens idealX
x_1^4-2*x_0*x_1^2*x_3-x_1^2*x_2*x_3+x_0^2*x_3^2,
x_0^2*x_1^2-10915*x_0*x_1^2*x_2-10917*x_0^3*x_3+10916*x_0^2*x_2*x_3-

10916*x_0*x_2^2*x_3-10916*x_1*x_3^3,
x_0*x_1^2*x_2^2+11909*x_0^4*x_3+5954*x_0^3*x_2*x_3+2977*x_0^2*x_2^2*x_3+

11910*x_0*x_2^3*x_3-2978*x_1^3*x_3^2+14887*x_0*x_1*x_3^3+
11910*x_1*x_2*x_3^3,

x_0*x_1^3*x_2-13099*x_1^3*x_2^2-6550*x_0^3*x_1*x_3-
13100*x_0^2*x_1*x_2*x_3-6550*x_0*x_1*x_2^2*x_3+13099*x_1*x_2^3*x_3+
13100*x_1^2*x_3^3+13099*x_0*x_3^4,

x_0^5+5*x_0^2*x_2^3+5*x_0*x_2^4-3*x_0*x_1^3*x_3-4*x_1^3*x_2*x_3+
4*x_0^2*x_1*x_3^2+10*x_0*x_1*x_2*x_3^2+5*x_1*x_2^2*x_3^2,

x_1^2*x_2^4-8932*x_0^4*x_2*x_3+11909*x_0^3*x_2^2*x_3+5954*x_0^2*x_2^3*x_3-
8934*x_0*x_2^4*x_3-x_2^5*x_3+2*x_0*x_1^3*x_3^2-5952*x_1^3*x_2*x_3^2-
x_0^2*x_1*x_3^3-2979*x_0*x_1*x_2*x_3^3-8934*x_1*x_2^2*x_3^3+x_3^6

Imagine that you found yourself looking at the scheme X in P3 defined by
the 6 equations above.

i53 : X = variety idealX

o53 = X

o53 : ProjectiveVariety

How would you analyze the scheme X? We will illustrate one approach.
In outline, we will first look at the topological invariants: the number

and dimensions of the irreducible components, and how they meet if there
is more than one; the topological type of each component; and the degree of
each component in P3. We will then see what we can say about the analytic
invariants of X using adjunction theory (we give some references at the end).

Since we are interested in the projective scheme defined by idealX we
could work with any ideal having the same saturation. It is usually the case
that working with the saturation itself greatly eases subsequent computation
so, as a matter of good practice, we begin by checking whether the ideal is
saturated. If not, we should replace it with its saturation.

i54 : idealX == saturate idealX

o54 = true

Thus we see that idealX is already saturated. Perhaps the most basic invari-
ant of X is its dimension:

i55 : dim X

o55 = 1

This shows that X consists of a curve, and possibly some zero-dimensional
components. The command

i56 : idealXtop = top idealX

4 2 2 2 2 2 2 2 · · ·
o56 = ideal (x - 2x x x - x x x + x x , x x - 10915x x x - 10917x · · ·

1 0 1 3 1 2 3 0 3 0 1 0 1 2 · · ·
o56 : Ideal of ringP3

returns the ideal of the largest dimensional components of X. If there were
0-dimensional components (or if idealX were not saturated) then idealXtop
would be larger than idealX. To test this we reduce idealXtop modulo
idealX and see whether we get 0:

i57 : (gens idealXtop)%(gens idealX) == 0

o57 = true

Thus X is a purely one-dimensional scheme.
Is X singular?
i58 : codim singularLocus idealX

o58 = 4

A variety of codimension 4 in P3 must be empty, so X is a nonsingular curve.
A nonsingular curve in P3 could still be reducible, but since the inter-

section of two components would be a singular point, the curve would then
be disconnected. A straightforward way to decide is to use the command
decompose, which returns a list of irreducible components defined over kk.
The length of this list,

i59 : # decompose idealX

o59 = 1

is thus the number of irreducible components that are defined over kk, and
we see there is only one. (Warning: at this writing (December 2000), the
command “decompose” works only in positive characteristic).

Often what we really want to know is whether X is absolutely irreducible
(that is, irreducible over the algebraic closure of kk). The property of being
smooth transfers to the algebraic closure, so again the question is the number
of connected components we would get over the algebraic closure. For any
reduced scheme X over a perfect field (such as our finite field kk) this number
is h0OX := dimkk H0OX . We compute it with

i60 : HH^0 OO_X

1
o60 = kk

o60 : kk-module, free

i61 : rank oo

o61 = 1

This command works much faster than the decompose command. (You can
compute the time by adding the command time to the beginning of the line
where the command to be timed starts.) Since we already know that idealX
is saturated, this also shows that idealX is prime.

We next ask for the genus of the curve X. Here the genus may be defined
as the dimension of the space H1OX . We can get this space with

i62 : HH^1 OO_X

6
o62 = kk

o62 : kk-module, free

The genus of the curve is the dimension of this space, which we can see to be
6. Next, the cohomology class of X in P3 is determined by the degree of X:

i63 : degree idealX

o63 = 10

In sum: X is a smooth, absolutely irreducible curve of genus 6 and degree 10.
We next ask for analytic information about the curve and the embedding.

A reasonable place to start is with the relation between the line bundle defin-
ing the embedding and the canonical sheaf ωX . Notice first that the degree
of the hyperplane divisor (the degree of the curve) is 10 = 2g-2, the same as
the canonical bundle. By Riemann-Roch the embedding line bundle either is
the canonical bundle or has first cohomology 0, which we can check with

i64 : P3 = Proj ringP3

o64 = P3

o64 : ProjectiveVariety

i65 : HH^1((OO_P3(1)/idealX)(>=0))

o65 = cokernel | x_3 x_2 x_1 x_0 |

1
o65 : ringP3-module, quotient of ringP3

Let’s examine the degree of the generator of that module.
i66 : degrees oo

o66 = {{0}}

o66 : List

From that and the presentation matrix above we see that this cohomology
module is the residue class field ringP3/(x0, x1, x2, x3), concentrated in degree
0. Thus the embedding line bundle OX(1) is isomorphic to ωX . On the other
hand the dimension of the space of sections of this line bundle has already
been computed; it is g = 6. The curve is embedded in P

3, so only 4 of
these sections were used—the embedding is a projection of the same curve,
embedded in P6 by the canonical map.

We next ask more about the curve itself. After the genus, the gonality and
the Clifford index are among the most interesting invariants. Recall that the
gonality of X is the smallest degree of a mapping from X to P1. To define
the Clifford index of X we first define the Clifford index of a line bundle
L on X to be degree(L) − 2(h0(L) − 1). For example, the Clifford indices
of the structure sheaf OX and the canonical sheaf ωX are both equal to 0.
The Clifford index of the curve X is defined to be the minimum value of
the Clifford index of a line bundle L on X for which both h0(L) ≥ 2 and
h1(L) ≥ 2. The Clifford index of a curve of genus g lies between 0 (for a
hyperelliptic curve) and b(g− 1)/2c (for a general curve). The Clifford index
of any curve is bounded above by the gonality minus 2.

For a curve of genus 6 such as X, the gonality is either 2 (the hyperelliptic
case), 3 (the trigonal case) or 4 (the value for general curves). The Clifford
index, on the other hand is either 0 (the hyperelliptic case) or 1 (the case of
a trigonal curve OR a smooth plane quintic curve—which is necessarily of
gonality 4) or 2 (the case of a general curve). Thus for most curves (and this
is true in any genus) the Clifford index is equal to the gonality minus 2.

We can make a start on distinguishing these cases already: since our
curve is embedded in P3 by a subseries of the canonical series, X cannot be
hyperelliptic (for hyperelliptic curves, the canonical series maps the curve
two-to-one onto a rational curve.)

To make further progress we use an idea of Mark Green (see Green and
Lazarsfeld [5]). Green conjectured a formula for the Clifford index that de-
pends only on numerical data about the free resolution of the curve in its
complete canonical embedding (where the hyperplanes cut out all the canon-
ical divisors). The conjecture is known for genus 6 and in many other cases;
see for example Schreyer [9].

We therefore begin by computing the canonical embedding of X. We
could proceed to find the canonical bundle as in the computation for P3

above, or indeed as OX(1), but instead we describe the general method that
is most efficient: duality, as described (for example) in the book of Altman
and Kleiman [1]. The module ⊕d∈ZH0(ωX(d)) can be computed as

i67 : omegaX = Ext^(codim idealX)(ringP3^1/idealX, ringP3^{-4})

o67 = cokernel {0} | 9359x_3 -4677x_3 -10105x_1 · · ·
{0} | 12014x_1 2552x_1 2626x_0 · · ·
{-1} | x_0x_3-2553x_2x_3 x_1^2-1702x_2x_3 x_0x_1-8086x_ · · ·

3
o67 : ringP3-module, quotient of ringP3

To find the equations of the canonical embedding of X, we first compute
a basis of H0(ωX), which is the degree 0 part of the module omegaX. The
desired equations are computed as the algebraic relations among the images
of this basis under any monomorphism ωX → OX .

As the ring ringP3/idealX is a domain, and ωX is the module corre-
sponding to a line bundle, any nonzero map from ωX to ringP3/idealX will
be an embedding. We can compute the module of such maps with

i68 : dualModule = Hom(omegaX, ringP3^1/idealX)

o68 = subquotient ({0} | x_0^3x_2^2+10915x_0^2x_2^3+807x_0x_2^4+4043x_ · · ·
{0} | 10105x_0x_1x_2^3+6063x_1x_2^4+11820x_0x_1^2x_ · · ·
{1} | 10105x_0^2x_2^2-11322x_0x_2^3+11322x_2^4+8396 · · ·

3
o68 : ringP3-module, subquotient of ringP3

and examine it with
i69 : betti prune dualModule

o69 = relations : total: 10 26
3: 3 2



Projective Geometry and Homological Algebra 33

4: 6 14
5: 1 9
6: . 1

For want of a better idea we take the first generator, dualModule_{0}, which
we can turn into an actual homomorphism with

i70 : f = homomorphism dualModule_{0}

o70 = | x_0^3x_2^2+10915x_0^2x_2^3+807x_0x_2^4+4043x_2^5+7655x_0x_1x_2 · · ·
o70 : Matrix

The image of a basis of ωX is given by the columns of the matrix
i71 : canGens = f*basis(0,omegaX)

o71 = | x_0^3x_2^2+10915x_0^2x_2^3+807x_0x_2^4+4043x_2^5+7655x_0x_1x_2 · · ·
o71 : Matrix

regarded as elements of
i72 : ringX = ringP3/idealX

o72 = ringX

o72 : QuotientRing

Because of the particular homomorphism we chose, they have degree 5.
We can now compute the defining ideal for X in its canonical embedding

as the relations on these elements. We first define a ring with 6 variables
corresponding to the columns of canGens

i73 : ringP5 = kk[x_0..x_5]

o73 = ringP5

o73 : PolynomialRing

and then compute the canonical ideal as the kernel of the corresponding map
from this ring to ringX with

i74 : idealXcan = trim kernel map(ringX, ringP5,
substitute(matrix canGens,ringX),
DegreeMap => i -> 5*i)

2 · · ·
o74 = ideal (x + 5040x x - 8565x x - 11589x x , x x - 6048x x - 1 · · ·

3 0 5 2 5 4 5 1 3 0 5 · · ·
o74 : Ideal of ringP5

Here the command trim is used to extract a minimal set of generators of
the desired ideal, and the command matrix replaces the map of (nonfree)
modules canGens by the matrix that gives its action on the generators. The
DegreeMap option specifies a function which transforms degrees (represented
as lists of integers) as the ring homomorphism does; using it here makes the
ring map homogeneous.

To get information about the Clifford index, we examine the free resolu-
tion with

i75 : betti res idealXcan

o75 = total: 1 9 16 9 1
0: 1 . . . .
1: . 6 8 3 .
2: . 3 8 6 .
3: . . . . 1

Quite generally, for a non-hyperelliptic curve of genus g ≥ 3 the ideal of the
canonical embedding requires g −

(
2
2

)
quadratic generators, in our case 6. It is

known that the curve is trigonal (Clifford index 1) if and only if the ideal also
requires cubic generators, that is, the first term in the free resolution requires
generators of degree 3 = 1 + 2; and Green’s conjecture says in general that
the curve has Clifford index c if the c − 1 term in the resolution does not
require generators of degree (c − 1) + 2 = c + 1 but the c term does require
generators of degree c+2. Thus from the Betti diagram above, and the truth
of Green’s conjecture in low genus, we see that our curve has Clifford index
1 and is thus either trigonal or a plane quintic.

If X is trigonal, that is, X has a map of degree 3 to P1, then the fibers
of this map form a linear series whose elements are divisors of degree three.
The geometric form of the Riemann-Roch theorem says that if

p1, . . . , pd ∈ X ⊂ Pg

are points on a canonically embedded curve X, then the dimension of the
linear system in which the divisor p1 + · · · + pd moves is the amount by
which the points fail to be linearly independent: d − 1 minus the dimension
of the projective plane spanned by the points. In particular, the 3 points in
the fiber of a three-to-one map to P1 are linearly dependent, that is, they
span a projective line. This “explains” why the ideal of a trigonal curve
requires cubic generators: the quadrics all contain three points of these lines
and thus contain the whole lines! It is known (see St-Donat [8]) that, in
the trigonal case, the 6 quadrics in the ideal of the canonical curve generate
the defining ideal of the variety which is the union of these lines, and that
variety is a rational normal scroll. In case X is a plane quintic, the adjunction
formula (Hartshorne [7, II.8.20.3]) shows that the canonical embedding of
X is obtained from the plane embedding by composing with the Veronese
embedding of the plane in P5 as the Veronese surface; and the 6 quadrics in
the ideal of the canonical curve generate the defining ideal of the Veronese
surface.

Thus if we let S denote the variety defined by the quadrics in the ideal
of X, we can decide whether X is a trigonal curve or a plane quintic by
deciding whether S is a rational normal scroll or a Veronese surface. To
compute the ideal of S we first ascertain which of the generators of the ideal
of the canonical curve have degree 2 with

i76 : deg2places = positions(degrees idealXcan, i->i=={2})

o76 = {0, 1, 2, 3, 4, 5}

o76 : List

and then compute
i77 : idealS= ideal (gens idealXcan)_deg2places

2 · · ·
o77 = ideal (x + 5040x x - 8565x x - 11589x x , x x - 6048x x - 1 · · ·

3 0 5 2 5 4 5 1 3 0 5 · · ·
o77 : Ideal of ringP5

One of the scrolls that could appear is singular, the cone over the rational
quartic in P4. We check for singularity first:

i78 : codim singularLocus idealS

o78 = 6

Since the codimension is 6, the surface S is nonsingular, and thus must be
one of the nonsingular scrolls or the Veronese surface (which is by definition
the image of P2, embedded in P5 by the linear series of conics.)

The ideals defining any rational normal scroll of codimension 3, and the
ideal of a Veronese surface all have free resolutions with the same Betti dia-
grams, so we need a subtler method to determine the identity of S. The most
powerful tool for such purposes is adjunction theory; we will use a simple
version.

The idea is to compare the embedding bundle (the “hyperplane bundle”)
with the canonical bundle. On the Veronese surface, the canonical bundle
is the bundle associated to −3 lines in P2, while the hyperplane bundle is
associated to 2 lines in P2. Thus the inverse of the square of the canonical
bundle is the cube of the hyperplane bundle, OS(3). For a scroll on the other
hand, these two bundles are different.

As before we follow the homological method for computing the canonical
bundle:

i79 : omegaS = Ext^(codim idealS)(ringP5^1/idealS, ringP5^{-6})

o79 = cokernel {2} | 4032x_5 0 14811x_5 -4032x_3 6549x_3 · · ·
{2} | x_3 x_2 x_1 -x_4 x_0-14291x_ · · ·
{2} | -6852x_5 6549x_3 362x_5 x_1-6248x_3 0 · · ·

3
o79 : ringP5-module, quotient of ringP5

i80 : OS = ringP5^1/idealS

o80 = cokernel | x_3^2+5040x_0x_5-8565x_2x_5-11589x_4x_5 x_1x_3-6048x_ · · ·
1

o80 : ringP5-module, quotient of ringP5

We want the square of the canonical bundle, which we can compute as
the tensor square

i81 : omegaS**omegaS

o81 = cokernel {4} | 4032x_5 0 14811x_5 -4032x_3 6549x_3 · · ·
{4} | x_3 x_2 x_1 -x_4 x_0-14291x_ · · ·
{4} | -6852x_5 6549x_3 362x_5 x_1-6248x_3 0 · · ·
{4} | 0 0 0 0 0 · · ·

{4} | 0 0 0 0 0 · · ·
{4} | 0 0 0 0 0 · · ·
{4} | 0 0 0 0 0 · · ·
{4} | 0 0 0 0 0 · · ·
{4} | 0 0 0 0 0 · · ·

9
o81 : ringP5-module, quotient of ringP5

But while this module represents the correct sheaf, it is hard to interpret,
since it may not be (is not, in this case) the module of all twisted global
sections of the square of the line bundle. Since the free resolution of OS (visible
inside the Betti diagram of the resolution of idealXcan) has length 3, the
module OS has depth 2. Thus we can find the module of all twisted global
sections of omega2S by taking the double dual

i82 : omega2S = Hom(Hom(omegaS**omegaS, OS),OS)

o82 = cokernel {3} | x_3^2+5040x_0x_5-8565x_2x_5-11589x_4x_5 x_1x_3-60 · · ·
1

o82 : ringP5-module, quotient of ringP5

We see from the output that this module is generated by 1 element of
degree 3. It follows that ω2

S
∼= OS(−3). This in turn shows that S is the

Veronese surface.
We now know that the canonical embedding of the curve X is the Veronese

map applied to a planar embedding of X of degree 5, and we can ask to see
the plane embedding. Since the anticanonical bundle ω−1

S on S corresponds
to 3 lines in the plane and the hyperplane bundle to 2 lines, we can recover
the line bundle corresponding to 1 line, giving the isomorphism of X to the
plane, as the quotient

i83 : L = Hom(omegaS, OS**(ringP5^{-1}))

o83 = subquotient ({-1} | 14401x_2+16185x_4 x_0-14291x_4 -5359x_1+1 · · ·
{-1} | -1488x_1-10598x_3 -6549x_3 -11789x_5 · · ·
{-1} | x_0+7742x_2-15779x_4 x_2 x_1+6551x_ · · ·

3
o83 : ringP5-module, subquotient of ringP5

and the line bundle on Xcan that gives the embedding in P2 will be the
restriction of L to Xcan. To realize the map from X to P2, we proceed as
before:

i84 : dualModule = Hom(L, OS)

o84 = subquotient (| x_0+7742x_2-15779x_4 14401x_2+16185x_4 x_1-301x_3 · · ·
| x_2 x_0-14291x_4 4032x_3 · · ·
| x_1+6551x_3 -5359x_1+14409x_3 -9874x_5 · · ·

3
o84 : ringP5-module, subquotient of ringP5

i85 : betti generators dualModule

o85 = total: 3 3
0: 3 3

Again, we may choose any homomorphism from L to OS, for example
i86 : g = homomorphism dualModule_{0}

o86 = | x_0+7742x_2-15779x_4 x_2 x_1+6551x_3 |

o86 : Matrix

i87 : toP2 = g*basis(0,L)

o87 = | x_0+7742x_2-15779x_4 x_2 x_1+6551x_3 |

o87 : Matrix

i88 : ringXcan = ringP5/idealXcan

o88 = ringXcan

o88 : QuotientRing

i89 : ringP2 = kk[x_0..x_2]

o89 = ringP2

o89 : PolynomialRing

i90 : idealXplane = trim kernel map(ringXcan, ringP2,
substitute(matrix toP2,ringXcan))

5 4 3 2 2 3 4 5 · · ·
o90 = ideal(x + 13394x x - 13014x x + 9232x x + 12418x x - 2746x · · ·

0 0 1 0 1 0 1 0 1 1 · · ·
o90 : Ideal of ringP2

We have effectively computed the square root of the line bundle embed-
ding X in P3 with which we started, and exchanged a messy set of defining
equations of an unknown scheme for a single equation defining a smooth
plane curve whose properties are easy to deduce. The same curve may also
be defined by a much simpler plane equation (see Appendix A below). I do
not know any general method for choosing a coordinate transformation to
simplify a given equation! Can the reader find one that will work at least in
this case?

There is not yet a textbook-level exposition of the sort of methods we have
used (although an introduction will be contained in a forthcoming elementary
book of Decker and Schreyer). The reader who would like to go further into
such ideas can find a high-level survey of how adjunction theory is used in
the paper of Decker and Schreyer [3]. For a group of powerful methods with
a different flavor, see Aure, Decker, Hulek, Popescu, and Ranestad [2].

Appendix A. How the “Mystery Variety” was Made

For those who would like to try out the computations above over a different
field (perhaps the field of rational numbers QQ), and for the curious, we include
the code used to produce the equations of the variety X above.

Start with the Fermat quintic in the plane

i91 : ringP2 = kk[x_0..x_2]

o91 = ringP2

o91 : PolynomialRing

i92 : idealC2 = ideal(x_0^5+x_1^5+x_2^5)

5 5 5
o92 = ideal(x + x + x )

0 1 2

o92 : Ideal of ringP2

Embed it by the Veronese map in P5:
i93 : ringC2 = ringP2/idealC2

o93 = ringC2

o93 : QuotientRing

i94 : ringP5 = kk[x_0..x_5]

o94 = ringP5

o94 : PolynomialRing

i95 : idealC5 = trim kernel map(ringC2, ringP5,
gens (ideal vars ringC2)^2)

2 2 · · ·
o95 = ideal (x - x x , x x - x x , x x - x x , x - x x , x x - x · · ·

4 3 5 2 4 1 5 2 3 1 4 2 0 5 1 2 0 · · ·
o95 : Ideal of ringP5

Finally, choose a projection into P3, from a line not meeting C5, which is an
isomorphism onto its image. (This requires the image to be a smooth curve
of degree 10).

i96 : ringC5 = ringP5/idealC5

o96 = ringC5

o96 : QuotientRing

i97 : use ringC5

o97 = ringC5

o97 : QuotientRing

i98 : idealC = trim kernel map(ringC5, ringP3,
matrix{{x_0+x_1,x_2,x_3,x_5}})

4 2 2 2 2 2 2 2 · · ·
o98 = ideal (x - 2x x x - x x x + x x , x x - 10915x x x - 10917x · · ·

1 0 1 3 1 2 3 0 3 0 1 0 1 2 · · ·
o98 : Ideal of ringP3

Let’s check that this is the same ideal as that of the mystery variety.
i99 : idealC == idealX

o99 = true

Here is the code of the function mystery, which does the steps above.
i100 : code mystery

o100 = -- mystery.m2:1-13
mystery = ringP3 -> (

kk := coefficientRing ringP3;
x := local x;
ringP2 := kk[x_0..x_2];
idealC2 := ideal(x_0^5+x_1^5+x_2^5);
ringC2 := ringP2/idealC2;
ringP5 := kk[x_0..x_5];
idealC5 := trim kernel map(ringC2, ringP5,

gens (ideal vars ringC2)^2);
ringC5 := ringP5/idealC5;
use ringC5;
trim kernel map(ringC5, ringP3,

matrix{{x_0+x_1,x_2,x_3,x_5}}))

And here is the code of the function prettyPrint.
i101 : code prettyPrint

o101 = -- mystery.m2:15-51
prettyPrint = f -> (

-- accept a matrix f and print its entries prettily,
-- separated by commas
wid := 74;
-- page width
post := (c,s) -> (

-- This function concatenates string c to end of each
-- string in list s except the last one
concatenate \ pack_2 between_c s);

strings := post_"," (toString \ flatten entries f);
-- list of strings, one for each polynomial, with commas
istate := ("",0);
-- initial state = (out : output string, col : column number)
strings = apply(

strings,
poly -> first fold(

-- break each poly into lines
(state,term) -> (

(out,col) -> (
if col + #term > wid -- too wide?
then (

out = out | "\n ";
col = 3;
-- insert line break
);

(out | term, col + #term) -- new state
)

) state,
istate,
fold( -- separate poly into terms

{"+","-"},
{poly},
(delimiter,poly) -> flatten(

post_delimiter \ separate_delimiter \ poly
))));

print stack strings; -- stack them vertically, then print
)

Data Types, Functions, and Programming

Daniel R. Grayson? and Michael E. Stillman??

In this chapter we present an introduction to the structure of Macaulay 2
commands and the writing of functions in the Macaulay 2 language. For
further details see the Macaulay 2 manual distributed with the program [1].

1 Basic Data Types

The basic data types of Macaulay 2 include numbers of various types (in-
tegers, rational numbers, floating point numbers, complex numbers), lists
(basic lists, and three types of visible lists, depending on the delimiter used),
hash tables, strings of characters (both 1-dimensional and 2-dimensional),
Boolean values (true and false), symbols, and functions. Higher level types
useful in mathematics are derived from these basic types using facilities pro-
vided in the Macaulay 2 language. Except for the simplest types (integers
and Boolean values), Macaulay 2 normally displays the type of the output
value on a second labeled output line.

Symbols have a name which consists of letters, digits, or apostrophes, the
first of which is a letter. Values can be assigned to symbols and recalled later.

i1 : w

o1 = w

o1 : Symbol

i2 : w = 2^100

o2 = 1267650600228229401496703205376

i3 : w

o3 = 1267650600228229401496703205376

Multiple values can be assigned in parallel.
i4 : (w,w’) = (33,44)

o4 = (33, 44)

o4 : Sequence

i5 : w

o5 = 33

i6 : w’

o6 = 44

? Supported by NSF grant DMS 99-70085.
?? Supported by NSF grant 99-70348.



42 D. R. Grayson and M. E. Stillman

Comments are initiated by -- and extend to the end of the line.
i7 : (w,w’) = (33, -- this is a comment

44)

o7 = (33, 44)

o7 : Sequence

Strings of characters are delimited by quotation marks.
i8 : w = "abcdefghij"

o8 = abcdefghij

They may be joined horizontally to make longer strings, or vertically to make
a two-dimensional version called a net.

i9 : w | w

o9 = abcdefghijabcdefghij

i10 : w || w

o10 = abcdefghij
abcdefghij

Nets are used in the preparation of two dimensional output for polynomials.
Floating point numbers are distinguished from integers by the presence

of a decimal point, and rational numbers are entered as fractions.
i11 : 2^100

o11 = 1267650600228229401496703205376

i12 : 2.^100

o12 = 1.26765 10^30

o12 : RR

i13 : (36 + 1/8)^6

582622237229761
o13 = ---------------

262144

o13 : QQ

Parentheses, braces, and brackets are used as delimiters for the three types
of visible lists: lists, sequences, and arrays.

i14 : x1 = {1,a}

o14 = {1, a}

o14 : List

i15 : x2 = (2,b)

o15 = (2, b)

o15 : Sequence

i16 : x3 = [3,c,d,e]

o16 = [3, c, d, e]

o16 : Array

Even though they use braces, lists should not be confused with sets, which
will be treated later. A double period can be used to construct a sequence of
consecutive elements in various contexts.

i17 : 1 .. 6

o17 = (1, 2, 3, 4, 5, 6)

o17 : Sequence

i18 : a .. f

o18 = (a, b, c, d, e, f)

o18 : Sequence

Lists can be nested.
i19 : xx = {x1,x2,x3}

o19 = {{1, a}, (2, b), [3, c, d, e]}

o19 : List

The number of entries in a list is provided by #.
i20 : #xx

o20 = 3

The entries in a list are numbered starting with 0, and can be recovered with
# used as a binary operator.

i21 : xx#0

o21 = {1, a}

o21 : List

i22 : xx#0#1

o22 = a

o22 : Symbol

We can join visible lists and append or prepend an element to a visible list.
The output will be the same type of visible list that was provided in the
input: a list, a sequence, or an array; if the arguments are various types of
lists, the output will be same type as the first argument.

i23 : join(x1,x2,x3)

o23 = {1, a, 2, b, 3, c, d, e}

o23 : List

i24 : append(x3,f)

o24 = [3, c, d, e, f]

o24 : Array

i25 : prepend(f,x3)

o25 = [f, 3, c, d, e]

o25 : Array

Use sum or product to produce the sum or product of all the elements in a
list.

i26 : sum {1,2,3,4}

o26 = 10

i27 : product {1,2,3,4}

o27 = 24

2 Control Structures

Commands for later execution are encapsulated in functions. A function is
created using the operator -> to separate the parameter or sequence of pa-
rameters from the code to be executed later. Let’s try an elementary example
of a function with two arguments.

i28 : f = (x,y) -> 1000 * x + y

o28 = f

o28 : Function

The parameters x and y are symbols that will acquire a value later when the
function is executed. They are local in the sense that they are completely
different from any symbols with the same name that occur elsewhere. Ad-
ditional local variables for use within the body of a function can be created
by assigning a value to them with := (first time only). We illustrate this by
rewriting the function above.

i29 : f = (x,y) -> (z := 1000 * x; z + y)

o29 = f

o29 : Function

Let’s apply the function to some arguments.
i30 : f(3,7)

o30 = 3007

The sequence of arguments can be assembled first, and then passed to the
function.

i31 : s = (3,7)

o31 = (3, 7)

o31 : Sequence

i32 : f s

o32 = 3007

As above, functions receiving one argument may be called without parenthe-
ses.

i33 : sin 2.1

o33 = 0.863209

o33 : RR

A compact notation for functions makes it convenient to apply them without
naming them first. For example, we may use apply to apply a function to
every element of a list and to collect the results into a list.

i34 : apply(1 .. 10, i -> i^3)

o34 = (1, 8, 27, 64, 125, 216, 343, 512, 729, 1000)

o34 : Sequence

The function scan will do the same thing, but discard the results.
i35 : scan(1 .. 5, print)
1
2
3
4
5

Use if ... then ... else ... to perform alternative actions based on the
truth of a condition.

i36 : apply(1 .. 10, i -> if even i then 1000*i else i)

o36 = (1, 2000, 3, 4000, 5, 6000, 7, 8000, 9, 10000)

o36 : Sequence

A function can be terminated prematurely with return.
i37 : apply(1 .. 10, i -> (if even i then return 1000*i; -i))

o37 = (-1, 2000, -3, 4000, -5, 6000, -7, 8000, -9, 10000)

o37 : Sequence

Loops in a program can be implemented with while ... do ....
i38 : i = 1; while i < 50 do (print i; i = 2*i)
1
2
4
8
16
32

Another way to implement loops is with for and do or list, with optional
clauses introduced by the keywords from, to, and when.

i40 : for i from 1 to 10 list i^3

o40 = {1, 8, 27, 64, 125, 216, 343, 512, 729, 1000}

o40 : List

i41 : for i from 1 to 4 do print i
1
2
3
4

A loop can be terminated prematurely with break, which accepts an optional
value to return as the value of the loop expression.

i42 : for i from 2 to 100 do if not isPrime i then break i

o42 = 4

If no value needs to be returned, the condition for continuing can be provided
with the keyword when; iteration continues only as long as the predicate
following the keyword returns true.

i43 : for i from 2 to 100 when isPrime i do print i
2
3

3 Input and Output

The function print can be used to display something on the screen.
i44 : print 2^100
1267650600228229401496703205376

For example, it could be used to display the elements of a list on separate
lines.

i45 : (1 .. 5) / print;
1
2
3
4
5

The operator << can be used to display something on the screen, without the
newline character.

i46 : << 2^100
1267650600228229401496703205376
o46 = stdio

o46 : File

-- the standard input output file

Notice the value returned is a file. A file in Macaulay 2 is a data type that
represents a channel through which data can be passed, as input, as output,
or in both directions. The file stdio encountered above corresponds to your
shell window or terminal, and is used for two-way communication between
the program and the user. A file may correspond to what one usually calls
a file, i.e., a sequence of data bytes associated with a given name and stored
on your disk drive. A file may also correspond to a socket, a channel for
communication with other programs over the network.

Files can be used with the binary form of the operator << to display
something else on the same line.

i47 : << "the value is : " << 2^100
the value is : 1267650600228229401496703205376
o47 = stdio

o47 : File

-- the standard input output file

Using endl to represent the new line character or character sequence, we can
produce multiple lines of output.

i48 : << "A = " << 2^100 << endl << "B = " << 2^200 << endl;
A = 1267650600228229401496703205376
B = 1606938044258990275541962092341162602522202993782792835301376

We can send the same output to a disk file named foo, but we must remember
to close it with close.

i49 : "foo" << "A = " << 2^100 << endl << close

o49 = foo

o49 : File

The contents of the file can be recovered as a string with get.
i50 : get "foo"

o50 = A = 1267650600228229401496703205376

If the file contains valid Macaulay 2 commands, as it does in this case, we
can execute those commands with load.

i51 : load "foo"

We can verify that the command took effect by evaluating A.
i52 : A

o52 = 1267650600228229401496703205376

Alternatively, if we want to see those commands and the output they produce,
we may use input.

i53 : input "foo"

i54 : A = 1267650600228229401496703205376

o54 = 1267650600228229401496703205376

i55 :

Let’s set up a ring for computation in Macaulay 2.
i56 : R = QQ[x,y,z]

o56 = R

o56 : PolynomialRing

i57 : f = (x+y)^3

3 2 2 3
o57 = x + 3x y + 3x*y + y

o57 : R

Printing, and printing to files, works for polynomials, too.
i58 : "foo" << f << close;

The two-dimensional output is readable by humans, but is not easy to convert
back into a polynomial.

i59 : get "foo"

o59 = 3 2 2 3
x + 3x y + 3x*y + y

Use toString to create a 1-dimensional form of the polynomial that can be
stored in a file in a format readable by Macaulay 2 and by other symbolic
algebra programs, such as Mathematica or Maple.

i60 : toString f

o60 = x^3+3*x^2*y+3*x*y^2+y^3

Send it to the file.
i61 : "foo" << toString f << close;

Get it back.
i62 : get "foo"

o62 = x^3+3*x^2*y+3*x*y^2+y^3

Convert the string back to a polynomial with value, using oo to recover the
value of the expression on the previous line.

i63 : value oo

3 2 2 3
o63 = x + 3x y + 3x*y + y

o63 : R

The same thing works for matrices, and a little more detail is provided by
toExternalString, if needed.

i64 : vars R

o64 = | x y z |

1 3
o64 : Matrix R <--- R

i65 : toString vars R

o65 = matrix {{x, y, z}}

i66 : toExternalString vars R

o66 = map(R^{{0}}, R^{{-1}, {-1}, {-1}}, {{x, y, z}})

4 Hash Tables

Recall how one sets up a quotient ring for computation in Macaulay 2.

i67 : R = QQ[x,y,z]/(x^3-y)

o67 = R

o67 : QuotientRing

i68 : (x+y)^4

2 2 3 4 2
o68 = 6x y + 4x*y + y + x*y + 4y

o68 : R

How does Macaulay 2 represent a ring like R in the computer? To answer
that, first think about what sort of information needs to be retained about
R. We may need to remember the coefficient ring of R, the names of the
variables in R, the monoid of monomials in the variables, the degrees of the
variables, the characteristic of the ring, whether the ring is commutative,
the ideal modulo which we are working, and so on. We also may need to
remember various bits of code: the code for performing the basic arithmetic
operations, such as addition and multiplication, on elements of R; the code for
preparing a readable representation of an element of R, either 2-dimensional
(with superscripts above the line and subscripts below), or 1-dimensional.
Finally, we may want to remember certain things that take a lot of time to
compute, such as the Gröbner basis of the ideal.

A hash table is, by definition, a way of representing (in the computer)
a function whose domain is a finite set. In Macaulay 2, hash tables are ex-
tremely flexible: the elements of the domain (or keys) and the elements of the
range (or values) of the function may be any of the other objects represented
in the computer. It’s easy to come up with uses for functions whose domain
is finite: for example, a monomial can be represented by the function that as-
sociates to a variable its nonzero exponent; a polynomial can be represented
by a function that associates to a monomial its nonzero coefficient; a set can
be represented by any function with that set as its domain; a (sparse) ma-
trix can be represented as a function from pairs of natural numbers to the
corresponding nonzero entry.

Let’s create a hash table and name it.
i69 : f = new HashTable from { a=>444, Daniel=>555, {c,d}=>{1,2,3,4}}

o69 = HashTable{{c, d} => {1, 2, 3, 4}}
a => 444
Daniel => 555

o69 : HashTable

The operator => is used to represent a key-value pair. We can use the operator
# to recover the value from the key.

i70 : f#Daniel

o70 = 555

i71 : f#{c,d}

o71 = {1, 2, 3, 4}

o71 : List

If the key is a symbol, we can use the operator . instead; this is convenient
if the symbol has a value that we want to ignore.

i72 : Daniel = a

o72 = a

o72 : Symbol

i73 : f.Daniel

o73 = 555

We can use #? to test whether a given key occurs in the hash table.
i74 : f#?a

o74 = true

i75 : f#?c

o75 = false

Finite sets are implemented in Macaulay 2 as hash tables: the elements of the
set are stored as the keys in the hash table, with the accompanying values all
being 1. (Multisets are implemented by using values larger than 1, and are
called tallies.)

i76 : x = set{1,a,{4,5},a}

o76 = Set {{4, 5}, 1, a}

o76 : Set

i77 : x#?a

o77 = true

i78 : peek x

o78 = Set{{4, 5} => 1}
1 => 1
a => 1

i79 : y = tally{1,a,{4,5},a}

o79 = Tally{{4, 5} => 1}
1 => 1
a => 2

o79 : Tally

i80 : y#a

o80 = 2

We might use tally to tally how often a function attains its various possible
values. For example, how often does an integer have 3 prime factors? Or 4?
Use factor to factor an integer.

i81 : factor 60

2
o81 = 2 3*5

o81 : Product

Then use # to get the number of factors.
i82 : # factor 60

o82 = 3

Use apply to list some values of the function.
i83 : apply(2 .. 1000, i -> # factor i)

o83 = (1, 1, 1, 1, 2, 1, 1, 1, 2, 1, 2, 1, 2, 2, 1, 1, 2, 1, 2, 2, 2, · · ·

o83 : Sequence

Finally, use tally to summarize the results.
i84 : tally oo

o84 = Tally{1 => 193}
2 => 508
3 => 275
4 => 23

o84 : Tally

Hash tables turn out to be convenient entities for storing odd bits and
pieces of information about something in a way that’s easy to think about
and use. In Macaulay 2, rings are represented as hash tables, as are ideals,
matrices, modules, chain complexes, and so on. For example, although it isn’t
a documented feature, the key ideal is used to preserve the ideal that was
used above to define the quotient ring R, as part of the information stored in
R.

i85 : R.ideal

3
o85 = ideal(x - y)

o85 : Ideal of QQ [x, y, z]

The preferred and documented way for a user to recover this information is
with the function ideal.

i86 : ideal R

3
o86 = ideal(x - y)

o86 : Ideal of QQ [x, y, z]

Users who want to introduce a new high-level mathematical concept to Mac-
aulay 2 may learn about hash tables by referring to the Macaulay 2 manual
[1].

5 Methods

You may use the code command to locate the source code for a given function,
at least if it is one of those functions written in the Macaulay 2 language.
For example, here is the code for demark, which may be used to put commas
between strings in a list.

i87 : code demark

o87 = -- ../../../m2/fold.m2:23
demark = (s,v) -> concatenate between(s,v)

The code for tensoring a ring map with a module can be displayed in this
way.

i88 : code(symbol **, RingMap, Module)

o88 = -- ../../../m2/ringmap.m2:294-298
RingMap ** Module := Module => (f,M) -> (

R := source f;
S := target f;
if R =!= ring M then error "expected module over source ring";
cokernel f(presentation M));

The code implementing the ideal function when applied to a quotient ring
can be displayed as follows.

i89 : code(ideal, QuotientRing)

o89 = -- ../../../m2/quotring.m2:7
ideal QuotientRing := R -> R.ideal

Notice that it uses the key ideal to extract the information from the ring’s
hash table, as you might have guessed from the previous discussion. The bit
of code displayed above may be called a method as a way of indicating that
several methods for dealing with various types of arguments are attached
to the function named ideal. New such method functions may be created
with the function method. Let’s illustrate that with an example: we’ll write
a function called denom which should produce the denominator of a rational
number. When applied to an integer, it should return 1. First we create the
method function.

i90 : denom = method();

Then we tell it what to do with an argument from the class QQ of rational
numbers.

i91 : denom QQ := x -> denominator x;

And also what to do with an argument from the class ZZ of integers.
i92 : denom ZZ := x -> 1;

Let’s test it.
i93 : denom(5/3)

o93 = 3

i94 : denom 5

o94 = 1

6 Pointers to the Source Code

A substantial part of Macaulay 2 is written in the same language provided
to the users. A good way to learn more about the Macaulay 2 language
is to peruse the source code that comes with the system in the directory
Macaulay2/m2. Use the code function, as described in the previous section,
for locating the bit of code you wish to view.

The source code for the interpreter of the Macaulay 2 language is in
the directory Macaulay2/d. It is written in another language designed to be
mostly type-safe, which is translated into C by the translator whose own C
source code is in the directory Macaulay2/c. Here is a sample line of code from
the file Macaulay2/d/tokens.d, which shows how the translator provides for
allocation and initialization of dynamic data structures.

globalFrame := Frame(dummyFrame,globalScope.seqno,Sequence(nullE));

And here is the C code produced by the translator.
tokens_Frame tokens_globalFrame;
tokens_Frame tmp__23;
Sequence tmp__24;
tmp__24 = (Sequence) GC_MALLOC(sizeof(struct S259_)+(1-1)*sizeof(Expr));
if (0 == tmp__24) outofmem();
tmp__24->len_ = 1;
tmp__24->array_[0] = tokens_nullE;
tmp__23 = (tokens_Frame) GC_MALLOC(sizeof(struct S260_));
if (0 == tmp__23) outofmem();
tmp__23->next = tokens_dummyFrame;
tmp__23->scopenum = tokens_globalScope->seqno;
tmp__23->values = tmp__24;
tokens_globalFrame = tmp__23;

The core algebraic algorithms constitute the engine of Macaulay 2 and
are written in C++, with the source files in the directory Macaulay2/e. In
the current version of the program, the interface between the interpreter and
the core algorithms consists of a single two-directional stream of bytes. The
manual that comes with the system [1] describes the engine communication
protocol used in that interface.

Teaching the Geometry of Schemes

Gregory G. Smith and Bernd Sturmfels

This chapter presents a collection of graduate level problems in algebraic
geometry illustrating the power of Macaulay 2 as an educational tool.

When teaching an advanced subject, like the language of schemes, we
think it is important to provide plenty of concrete instances of the theory.
Computer algebra systems, such as Macaulay 2, provide students with an
invaluable tool for studying complicated examples. Furthermore, we believe
that the explicit nature of a computational approach leads to a better under-
standing of the objects being examined. This chapter presents some problems
which we feel illustrate this point of view.

Our examples are selected from the homework of an algebraic geometry
class given at the University of California at Berkeley in the fall of 1999. This
graduate course was taught by the second author with assistance from the
first author. Our choice of problems, as the title suggests, follows the material
in David Eisenbud and Joe Harris’ textbook The Geometry of Schemes [5].

1 Distinguished Open Sets

We begin with a simple example involving the Zariski topology of an affine
scheme. This example also indicates some of the subtleties involved in working
with arithmetic schemes.

Problem. Let S = Z[x, y, z] and X = Spec(S). If f = x and Xf is the
corresponding basic open subset in X, then establish the following:

(1) If e1 = x + y + z, e2 = xy + xz + yz and e3 = xyz are the elementary
symmetric functions then the set {Xei}1≤i≤3 is an open cover of Xf .

(2) If p1 = x+ y + z, p2 = x2 + y2 + z2 and p3 = x3 + y3 + z3 are the power
sum symmetric functions then {Xpi}1≤i≤3 is not an open cover of Xf .

Solution. (1) To prove that {Xei}1≤i≤3 is an open cover of Xf , it suffices
to show that e1, e2 and e3 generate the unit ideal in Sf ; see Lemma I-16
in Eisenbud and Harris [5]. This is equivalent to showing that xm belongs
to the S-ideal 〈e1, e2, e3〉 for some m ∈ N. In other words, the saturation(
〈e1, e2, e3〉 : x∞

)
is the unit ideal if and only if {Xei}1≤i≤3 is an open cover

of Xf . We verify this in Macaulay 2 as follows:
i1 : S = ZZ[x, y, z];

i2 : elementaryBasis = ideal(x+y+z, x*y+x*z+y*z, x*y*z);

o2 : Ideal of S

i3 : saturate(elementaryBasis, x)

o3 = ideal 1

o3 : Ideal of S

(2) Similarly, to show that {Xpi}1≤i≤3 is not an open cover of Xf , we prove
that

(
〈p1, p2, p3〉 : x∞

)
is not the unit ideal. Calculating this saturation, we

find
i4 : powerSumBasis = ideal(x+y+z, x^2+y^2+z^2, x^3+y^3+z^3);

o4 : Ideal of S

i5 : saturate(powerSumBasis, x)

2 2
o5 = ideal (6, x + y + z, 2y + 2y*z + 2z , 3y*z)

o5 : Ideal of S

i6 : clearAll

which is not the unit ideal. ut

The fact that 6 is a generator of the ideal
(
〈p1, p2, p3〉 : x∞

)
indicates

that {Xpi}1≤i≤3 does not contain the points in X lying over the points 〈2〉
and 〈3〉 in Spec(Z). If we work over a base ring in which 6 is a unit, then
{Xpi}1≤i≤3 would, in fact, be an open cover of Xf .

2 Irreducibility

The study of complex semisimple Lie algebras gives rise to an important fam-
ily of algebraic varieties called nilpotent orbits. The next problem examines
the irreducibility of a particular nilpotent orbit.

Problem. Let X be the set of nilpotent complex 3 × 3 matrices. Show that
X is an irreducible algebraic variety.

Solution. A 3 × 3 matrix M is nilpotent if and only if its minimal polyno-
mial p(T) equals Tk, for some k ∈ N. Since each irreducible factor of the
characteristic polynomial of M is also a factor of p(T), it follows that the
characteristic polynomial of M is T3. We conclude that the coefficients of
the characteristic polynomial of a generic 3 × 3 matrix define the algebraic
variety X.

To prove that X is irreducible over C, we construct a rational parame-
terization. First, observe that GL3(C) acts on X by conjugation. Jordan’s
canonical form theorem implies that there are exactly three orbits; one for
each of the following matrices:

N(1,1,1) =
[

0 0 0
0 0 0
0 0 0

]
, N(2,1) =

[
0 1 0
0 0 0
0 0 0

]
and N(3) =

[
0 1 0
0 0 1
0 0 0

]
.

Each orbit is defined by a rational parameterization, so it suffices to show
that the closure of the orbit containing N(3) is the entire variety X. We
demonstrate this as follows:

i7 : S = QQ[t, y_0 .. y_8, a..i, MonomialOrder => Eliminate 10];

i8 : N3 = (matrix {{0,1,0},{0,0,1},{0,0,0}}) ** S

o8 = | 0 1 0 |
| 0 0 1 |
| 0 0 0 |

3 3
o8 : Matrix S <--- S

i9 : G = genericMatrix(S, y_0, 3, 3)

o9 = | y_0 y_3 y_6 |
| y_1 y_4 y_7 |
| y_2 y_5 y_8 |

3 3
o9 : Matrix S <--- S

To determine the entries in G · N(3) · G−1, we use the classical adjoint to
construct the matrix det(G) ·G−1.

i10 : classicalAdjoint = (G) -> (
n := degree target G;
m := degree source G;
matrix table(n, n, (i, j) -> (-1)^(i+j) * det(

submatrix(G, {0..j-1, j+1..n-1},
{0..i-1, i+1..m-1}))));

i11 : num = G * N3 * classicalAdjoint(G);

3 3
o11 : Matrix S <--- S

i12 : D = det(G);

i13 : M = genericMatrix(S, a, 3, 3);

3 3
o13 : Matrix S <--- S

The entries in G · N(3) · G−1 give a rational parameterization of the orbit
generated by N(3). Using elimination theory — see section 3.3 in Cox, Little
and O‘Shea [2] — we give an “implicit representation” of this variety.

i14 : elimIdeal = minors(1, (D*id_(S^3))*M - num) + ideal(1-D*t);

o14 : Ideal of S

i15 : closureOfOrbit = ideal selectInSubring(1, gens gb elimIdeal);

o15 : Ideal of S

Finally, we verify that this orbit closure equals X scheme-theoretically.
Recall that X is defined by the coefficients of the characteristic polynomial
of a generic 3× 3 matrix M.

i16 : X = ideal substitute(
contract(matrix{{t^2,t,1}}, det(t-M)),
{t => 0_S})

o16 = ideal (- a - e - i, - b*d + a*e - c*g - f*h + a*i + e*i, c*e*g - · · ·
o16 : Ideal of S

i17 : closureOfOrbit == X

o17 = true

i18 : clearAll

This completes our solution. ut

More generally, Kostant shows that the set of all nilpotent elements in
a complex semisimple Lie algebra form an irreducible variety. We refer the
reader to Chriss and Ginzburg [1] for a proof of this result (Corollary 3.2.8)
and a discussion of its applications in representation theory.

3 Singular Points

In our third question, we study the singular locus of a family of elliptic curves.

Problem. Consider a general form of degree 3 in Q[x, y, z]:

F = ax3 + bx2y + cx2z + dxy2 + exyz + fxz2 + gy3 + hy2z + iyz2 + jz3 .

Give necessary and sufficient conditions in terms of a, . . . , j for the cubic
curve Proj

(
Q[x, y, z]/〈F 〉

)
to have a singular point.

Solution. The singular locus of F is defined by a polynomial of degree 12 in
the 10 variables a, . . . , j. We calculate this polynomial in two different ways.

Our first method is an elementary but time consuming elimination. Car-
rying it out in Macaulay 2, we have

i19 : S = QQ[x, y, z, a..j, MonomialOrder => Eliminate 2];

i20 : F = a*x^3+b*x^2*y+c*x^2*z+d*x*y^2+e*x*y*z+f*x*z^2+g*y^3+h*y^2*z+
i*y*z^2+j*z^3;

i21 : partials = submatrix(jacobian matrix{{F}}, {0..2}, {0})

o21 = {1} | 3x2a+2xyb+y2d+2xzc+yze+z2f |
{1} | x2b+2xyd+3y2g+xze+2yzh+z2i |
{1} | x2c+xye+y2h+2xzf+2yzi+3z2j |

3 1
o21 : Matrix S <--- S

i22 : singularities = ideal(partials) + ideal(F);

o22 : Ideal of S

i23 : elimDiscr = time ideal selectInSubring(1,gens gb singularities);
-- used 64.27 seconds

o23 : Ideal of S

i24 : elimDiscr = substitute(elimDiscr, {z => 1});

o24 : Ideal of S

On the other hand, there is also an elegant and more useful determinantal
formula for this discriminant; it is a specialization of the formula (2.8) in
section 3.2 of Cox, Little and O‘Shea [3]. To apply this determinantal formula,
we first create the coefficient matrix A of the partial derivatives of F .

i25 : A = contract(matrix{{x^2,x*y,y^2,x*z,y*z,z^2}},
diff(transpose matrix{{x,y,z}},F))

o25 = {1} | 3a 2b d 2c e f |
{1} | b 2d 3g e 2h i |
{1} | c e h 2f 2i 3j |

3 6
o25 : Matrix S <--- S

We also construct the coefficient matrix B of the partial derivatives of the
Hessian of F .

i26 : hess = det submatrix(jacobian ideal partials, {0..2}, {0..2});

i27 : B = contract(matrix{{x^2,x*y,y^2,x*z,y*z,z^2}},
diff(transpose matrix{{x,y,z}},hess))

o27 = {1} | -24c2d+24bce-18ae2-24b2f+72adf 4be2-16bdf-48 · · ·
{1} | 2be2-8bdf-24c2g+72afg+16bch-24aeh-8b2i+24adi 4de2-16d2f-48 · · ·
{1} | 2ce2-8cdf-8c2h+24afh+16bci-24aei-24b2j+72adj 2e3-8def-24cf · · ·

3 6
o27 : Matrix S <--- S

To obtain the discriminant, we combine these two matrices and take the
determinant.

i28 : detDiscr = ideal det (A || B);

o28 : Ideal of S

Finally, we check that our two discriminants are equal
i29 : detDiscr == elimDiscr

o29 = true

and examine the generator.
i30 : detDiscr_0

2 4 3 2 5 3 2 6 3 2 2 2 2 · · ·
o30 = 13824c d*e f g - 13824b*c*e f g + 13824a*e f g - 110592c d e · · ·
o30 : S

i31 : numgens detDiscr

o31 = 1

i32 : # terms detDiscr_0

o32 = 2040

i33 : clearAll

Hence, the singular locus is given by a single polynomial of degree 12 with
2040 terms. ut

For a further discussion of singularities and discriminants see Section V.3
in Eisenbud and Harris [5]. For information on resultants and discriminants
see Chapter 2 in Cox, Little and O‘Shea [3].

4 Fields of Definition

Schemes over non-algebraically closed fields arise in number theory. Our
fourth problem looks at one technique for working with number fields in
Macaulay 2.

Problem (Exercise II-6 in [5]). An inclusion of fields K ↪→ L induces a
map AnL → A

n
K . Find the images in A2

Q
of the following points of A2

Q
under

this map.

(1) 〈x−
√

2, y −
√

2〉;
(2) 〈x−

√
2, y −

√
3〉;

(3) 〈x− ζ, y − ζ−1〉 where ζ is a 5-th root of unity ;
(4) 〈

√
2x−

√
3y〉;

(5) 〈
√

2x−
√

3y − 1〉.

Solution. The images can be determined by using the following three step
algorithm: (1) replace the coefficients not contained inK with indeterminates,
(2) add the minimal polynomials of these coefficients to the given ideal in A2

L,
and (3) eliminate the new indeterminates. Here are the five examples:

i34 : S = QQ[a,b,x,y, MonomialOrder => Eliminate 2];

i35 : I1 = ideal(x-a, y-a, a^2-2);

o35 : Ideal of S

i36 : ideal selectInSubring(1, gens gb I1)

2
o36 = ideal (x - y, y - 2)

o36 : Ideal of S

i37 : I2 = ideal(x-a, y-b, a^2-2, b^2-3);

o37 : Ideal of S

i38 : ideal selectInSubring(1, gens gb I2)

2 2
o38 = ideal (y - 3, x - 2)

o38 : Ideal of S

i39 : I3 = ideal(x-a, y-a^4, a^4+a^3+a^2+a+1);

o39 : Ideal of S

i40 : ideal selectInSubring(1, gens gb I3)

2 2 3 2
o40 = ideal (x*y - 1, x + y + x + y + 1, y + y + x + y + 1)

o40 : Ideal of S

i41 : I4 = ideal(a*x+b*y, a^2-2, b^2-3);

o41 : Ideal of S

i42 : ideal selectInSubring(1, gens gb I4)

2 3 2
o42 = ideal(x - -*y )

2

o42 : Ideal of S

i43 : I5 = ideal(a*x+b*y-1, a^2-2, b^2-3);

o43 : Ideal of S

i44 : ideal selectInSubring(1, gens gb I5)

4 2 2 9 4 2 3 2 1
o44 = ideal(x - 3x y + -*y - x - -*y + -)

4 2 4

o44 : Ideal of S

i45 : clearAll

ut

It is worth noting that the points in An
Q

correspond to orbits of the action
of Gal(Q/Q) on the points of An

Q
. For more examples and information, see

section II.2 in Eisenbud and Harris [5].

5 Multiplicity

The multiplicity of a zero-dimensional scheme X at a point p ∈ X is defined
to be the length of the local ringOX,p. Unfortunately, we cannot work directly
in the local ring in Macaulay 2. What we can do, however, is to compute the
multiplicity by computing the degree of the component of X supported at p;
see page 66 in Eisenbud and Harris [5].

Problem. What is the multiplicity of the origin as a zero of the polynomial
equations x5 + y3 + z3 = x3 + y5 + z3 = x3 + y3 + z5 = 0?

Solution. If I is the ideal generated by x5+y3+z3, x3+y5+z3 and x3+y3+z5

in Q[x, y, z], then the multiplicity of the origin is

dimQ
Q[x, y, z]〈x,y,z〉
IQ[x, y, z]〈x,y,z〉

.

It follows that the multiplicity is the vector space dimension of the ring
Q[x, y, z]/ϕ−1(IQ[x, y, z]〈x,y,z〉) where ϕ : Q[x, y, z] → Q[x, y, z]〈x,y,z〉 is the
natural map. Moreover, we can express this using ideal quotients:

ϕ−1(IQ[x, y, z]〈x,y,z〉) =
(
I : (I : 〈x, y, z〉∞)

)
.

Carrying out this calculation in Macaulay 2, we obtain:
i46 : S = QQ[x, y, z];

i47 : I = ideal(x^5+y^3+z^3, x^3+y^5+z^3, x^3+y^3+z^5);

o47 : Ideal of S

i48 : multiplicity = degree(I : saturate(I))

o48 = 27

i49 : clearAll

Thus, we conclude that the multiplicity is 27. ut

There are algorithms (not yet implemented in Macaulay 2) for working
directly in the local ring Q[x, y, z]〈x,y,z〉. We refer the interested reader to
Chapter 4 in Cox, Little and O‘Shea [3].

6 Flat Families

Non-reduced schemes arise naturally as flat limits of a family of reduced
schemes. Our next problem illustrates how a family of skew lines in P3 gives
rise to a double line with an embedded point.

Problem (Exercise III-68 in [5]). Let L and M be the lines in P3
k[t] given

by x = y = 0 and x− tz = y + t2w = 0 respectively. Show that the flat limit
as t→ 0 of the union L ∪M is the double line x2 = y = 0 with an embedded
point of degree 1 located at the point (0 : 0 : 0 : 1).

Solution. We first find the flat limit by saturating the intersection ideal and
setting t = 0.

i50 : PP3 = QQ[t, x, y, z, w];

i51 : L = ideal(x, y);

o51 : Ideal of PP3

i52 : M = ideal(x-t*z, y+t^2*w);

o52 : Ideal of PP3

i53 : X = intersect(L, M);

o53 : Ideal of PP3

i54 : Xzero = trim substitute(saturate(X, t), {t => 0})

2 2
o54 = ideal (y*z, y , x*y, x )

o54 : Ideal of PP3

Secondly, we verify that this is the union of a double line and an embedded
point of degree 1.

i55 : Xzero == intersect(ideal(x^2, y), ideal(x, y^2, z))

o55 = true

i56 : degree(ideal(x^2, y ) / ideal(x, y^2, z))

o56 = 1

i57 : clearAll

ut

Section III.3.4 in Eisenbud and Harris [5] contains several other interesting
limits of various flat families.

7 Bézout’s Theorem

Bézout’s Theorem — Theorem III-78 in Eisenbud and Harris [5] — may fail
without the Cohen-Macaulay hypothesis. Our seventh problem is to demon-
strate this.

Problem (Exercise III-81 in [5]). Find irreducible closed subvarieties X
and Y in P4 such that

codim(X ∩ Y ) = codim(X) + codim(Y )
deg(X ∩ Y ) > deg(X) · deg(Y ) .

Solution. We show that the assertion holds when X is the cone over the
nonsingular rational quartic curve in P3 and Y is a two-plane passing through
the vertex of the cone. First, recall that the rational quartic curve is given by
the 2× 2 minors of the matrix

[
a b2 bd c
b ac c2 d

]
; see Exercise 18.8 in Eisenbud [4].

Thus, we have
i58 : S = QQ[a, b, c, d, e];

i59 : IX = trim minors(2, matrix{{a, b^2, b*d, c},{b, a*c, c^2, d}})

3 2 2 2 3 2
o59 = ideal (b*c - a*d, c - b*d , a*c - b d, b - a c)

o59 : Ideal of S

i60 : IY = ideal(a, d);

o60 : Ideal of S

i61 : codim IX + codim IY == codim (IX + IY)

o61 = true

i62 : (degree IX) * (degree IY)

o62 = 4

i63 : degree (IX + IY)

o63 = 5

which establishes the assertion. ut

To understand how this example works, it is enlightening to express Y as
the intersection of two hyperplanes; one given by a = 0 and the other given
by d = 0. Intersecting X with the first hyperplane yields

i64 : J = ideal mingens (IX + ideal(a))

3 2 2 3
o64 = ideal (a, b*c, c - b*d , b d, b )

o64 : Ideal of S

However, this first intersection has an embedded point;
i65 : J == intersect(ideal(a, b*c, b^2, c^3-b*d^2),

ideal(a, d, b*c, c^3, b^3)) -- embedded point

o65 = true

i66 : clearAll

The second hyperplane passes through this embedded point which explains
the extra intersection.

8 Constructing Blow-ups

The blow-up of a scheme X along a subscheme Y can be constructed from
the Rees algebra associated to the ideal sheaf of Y in X; see Theorem IV-22
in Eisenbud and Harris [5]. Gröbner basis techniques allow one to express the
Rees algebra in terms of generators and relations. We illustrate this method
in the next solution.

Problem (Exercises IV-43 & IV-44 in [5]). Find the blow-up X of the
affine plane A2 = Spec

(
Q[x, y]

)
along the subscheme defined by 〈x3, xy, y2〉.

Show that X is nonsingular and its fiber over the origin is the union of two
copies of P1 meeting at a point.

Solution. We first provide a general function which returns the ideal of rela-
tions for the Rees algebra.

i67 : blowUpIdeal = (I) -> (
r := numgens I;
S := ring I;
n := numgens S;
K := coefficientRing S;
tR := K[t, gens S, vars(0..r-1),

MonomialOrder => Eliminate 1];
f := map(tR, S, submatrix(vars tR, {1..n}));
F := f(gens I);
J := ideal apply(1..r, j -> (gens tR)_(n+j)-t*F_(0,(j-1)));
L := ideal selectInSubring(1, gens gb J);
R := K[gens S, vars(0..r-1)];
g := map(R, tR, 0 | vars R);
trim g(L));

Now, applying the function to our specific case yields:
i68 : S = QQ[x, y];

i69 : I = ideal(x^3, x*y, y^2);

o69 : Ideal of S

i70 : J = blowUpIdeal(I)

2 2 3 2
o70 = ideal (y*b - x*c, x*b - a*c, x b - y*a, x c - y a)

o70 : Ideal of QQ [x, y, a, b, c]

Therefore, the blow-up of the affine plane along the given subscheme is

X = Proj
(

(Q[x, y])[a, b, c]
〈yb− xc, xb2 − ac, x2b− ya, x3c− y2a〉

)
.

Using Macaulay 2, we can also verify that the scheme X is nonsingular;
i71 : J + ideal jacobian J == ideal gens ring J

o71 = true

i72 : clearAll

Since we have

(Q[x, y])[a, b, c]
〈yb− xc, xb2 − ac, x2b− ya, x3c− y2a〉

⊗ Q[x, y]
〈x, y〉

∼=
Q[a, b, c]
〈ac〉

,

the fiber over the origin 〈x, y〉 in A2 is clearly a union of two copies of P1

meeting at one point. In particular, the exceptional fiber is not a projective
space. ut

Many other interesting blow-ups can be found in section II.2 in Eisenbud
and Harris [5].

9 A Classic Blow-up

We consider the blow-up of the projective plane P2 at a point.

Problem. Show that the following varieties are isomorphic.

(a) the image of the rational map from P
2 to P4 given by

(r : s : t) 7→ (r2 : s2 : rs : rt : st) ;

(b) the blow-up of the plane P2 at the point (0 : 0 : 1);
(c) the determinantal variety defined by the 2×2 minors of the matrix

[
a c d
b d e

]
where P4 = Proj

(
k[a, b, c, d, e]

)
.

This surface is called the cubic scroll in P4.

Solution. We find the ideal in part (a) by elimination theory.
i73 : PP4 = QQ[a..e];

i74 : S = QQ[r..t, A..E, MonomialOrder => Eliminate 3];

i75 : I = ideal(A - r^2, B - s^2, C - r*s, D - r*t, E - s*t);

o75 : Ideal of S

i76 : phi = map(PP4, S, matrix{{0_PP4, 0_PP4, 0_PP4}} | vars PP4)

o76 = map(PP4,S,{0, 0, 0, a, b, c, d, e})

o76 : RingMap PP4 <--- S

i77 : surfaceA = phi ideal selectInSubring(1, gens gb I)

2
o77 = ideal (c*d - a*e, b*d - c*e, a*b - c )

o77 : Ideal of PP4

Next, we determine the surface in part (b). We construct the ideal defining
the blow-up of P2

i78 : R = QQ[t, x, y, z, u, v, MonomialOrder => Eliminate 1];

i79 : blowUpIdeal = ideal selectInSubring(1, gens gb ideal(u-t*x,
v-t*y))

o79 = ideal(y*u - x*v)

o79 : Ideal of R

and embed it in P2 × P1.
i80 : PP2xPP1 = QQ[x, y, z, u, v];

i81 : embed = map(PP2xPP1, R, 0 | vars PP2xPP1);

o81 : RingMap PP2xPP1 <--- R

i82 : blowUp = PP2xPP1 / embed(blowUpIdeal);

We then map this surface into P5 using the Segre embedding.
i83 : PP5 = QQ[A .. F];

i84 : segre = map(blowUp, PP5, matrix{{x*u,y*u,z*u,x*v,y*v,z*v}});

o84 : RingMap blowUp <--- PP5

i85 : ker segre

2
o85 = ideal (B - D, C*E - D*F, D - A*E, C*D - A*F)

o85 : Ideal of PP5

Note that the image under the Segre map lies on a hyperplane in P5. To get
the desired surface in P4, we project

i86 : projection = map(PP4, PP5, matrix{{a, c, d, c, b, e}})

o86 = map(PP4,PP5,{a, c, d, c, b, e})

o86 : RingMap PP4 <--- PP5

i87 : surfaceB = trim projection ker segre

2
o87 = ideal (c*d - a*e, b*d - c*e, a*b - c )

o87 : Ideal of PP4

Finally, we compute the surface in part (c).
i88 : determinantal = minors(2, matrix{{a, c, d}, {b, d, e}})

2
o88 = ideal (- b*c + a*d, - b*d + a*e, - d + c*e)

o88 : Ideal of PP4

i89 : sigma = map( PP4, PP4, matrix{{d, e, a, c, b}});

o89 : RingMap PP4 <--- PP4

i90 : surfaceC = sigma determinantal

2
o90 = ideal (c*d - a*e, b*d - c*e, a*b - c )

o90 : Ideal of PP4

By incorporating a permutation of the variables into definition of surfaceC,
we obtain the desired isomorphisms

i91 : surfaceA == surfaceB

o91 = true

i92 : surfaceB == surfaceC

o92 = true

i93 : clearAll

which completes the solution. ut

For more information of the geometry of rational normal scrolls, see Lec-
ture 8 in Harris [6].

10 Fano Schemes

Our final example concerns the family of Fano schemes associated to a flat
family of quadrics. Recall that the k-th Fano scheme Fk(X) of a scheme X ⊆
P
n is the subscheme of the Grassmannian parametrizing k-planes contained

in X.

Problem (Exercise IV-69 in [5]). Consider the one-parameter family of
quadrics tending to a double plane with equation

Q = V (tx2 + ty2 + tz2 + w2) ⊆ P3
Q[t] = Proj

(
Q[t][x, y, z, w]

)
.

What is the flat limit of the Fano schemes F1(Qt)?

Solution. We first compute the ideal defining F1(Qt), the scheme parametriz-
ing lines in Q.

i94 : PP3 = QQ[t, x, y, z, w];

i95 : Q = ideal( t*x^2+t*y^2+t*z^2+w^2 );

o95 : Ideal of PP3

To parametrize a line in our projective space, we introduce indeterminates
u, v and A, . . . ,H.

i96 : R = QQ[t, u, v, A .. H];

We then make a map phi from PP3 to R sending the variables to the coordi-
nates of the general point on a line.

i97 : phi = map(R, PP3, matrix{{t}} |
u*matrix{{A, B, C, D}} + v*matrix{{E, F, G, H}});

o97 : RingMap R <--- PP3

i98 : imageFamily = phi Q;

o98 : Ideal of R

For a line to belong to Q, the imageFamily must vanish identically. In other
words, F1(Q) is defined by the coefficients of the generators of imageFamily.

i99 : coeffOfFamily = contract(matrix{{u^2,u*v,v^2}}, gens imageFamily)

o99 = | tA2+tB2+tC2+D2 2tAE+2tBF+2tCG+2DH tE2+tF2+tG2+H2 |

1 3
o99 : Matrix R <--- R

Since we don’t need the variables u and v, we get rid of them.
i100 : S = QQ[t, A..H];

i101 : coeffOfFamily = substitute(coeffOfFamily, S);

1 3
o101 : Matrix S <--- S

i102 : Sbar = S / (ideal coeffOfFamily);

Next, we move to the Grassmannian G(1, 3) ⊂ P5. Recall the homogeneous
coordinates on P5 correspond to the 2×2 minors of a 2×4 matrix. We obtain
these minors using the exteriorPower function in Macaulay 2.

i103 : psi = matrix{{t}} | exteriorPower(2,
matrix{{A, B, C, D}, {E, F, G, H}})

o103 = | t -BE+AF -CE+AG -CF+BG -DE+AH -DF+BH -DG+CH |

1 7
o103 : Matrix Sbar <--- Sbar

i104 : PP5 = QQ[t, a..f];

i105 : fanoOfFamily = trim ker map(Sbar, PP5, psi);

o105 : Ideal of PP5

Now, to answer the question, we determine the limit as t tends to 0.
i106 : zeroFibre = trim substitute(saturate(fanoOfFamily, t), {t=>0})

2 2 2 · · ·
o106 = ideal (e*f, d*f, e , f , d*e, a*e + b*f, d , c*d - b*e + a*f, b · · ·
o106 : Ideal of PP5

Let’s transpose the matrix of generators so all of its elements are visible on
the printed page.

i107 : transpose gens zeroFibre

o107 = {-2} | ef |
{-2} | df |
{-2} | e2 |
{-2} | f2 |
{-2} | de |
{-2} | ae+bf |
{-2} | d2 |
{-2} | cd-be+af |
{-2} | bd+ce |
{-2} | ad-cf |
{-2} | a2+b2+c2 |

11 1
o107 : Matrix PP5 <--- PP5

We see that F1(Q0) is supported on the plane conic 〈d, e, f, a2 + b2 + c2〉.
However, F1(Q0) is not reduced; it has multiplicity two. On the other hand,
the generic fiber is

i108 : oneFibre = trim substitute(saturate(fanoOfFamily, t), {t => 1})

2 2 2 · · ·
o108 = ideal (a*e + b*f, d + e + f , c*d - b*e + a*f, b*d + c*e, a*d · · ·
o108 : Ideal of PP5

i109 : oneFibre == intersect(ideal(c-d, b+e, a-f, d^2+e^2+f^2),
ideal(c+d, b-e, a+f, d^2+e^2+f^2))

o109 = true



70 G. G. Smith and B. Sturmfels

Hence, for t 6= 0, F1(Qt) is the union of two conics lying in complementary
planes and F1(Q0) is the double conic obtained when the two conics move
together. ut

Monomial Ideals

Serkan Hoşten and Gregory G. Smith

Monomial ideals form an important link between commutative algebra and
combinatorics. In this chapter, we demonstrate how to implement algorithms
in Macaulay 2 for studying and using monomial ideals. We illustrate these
methods with examples from combinatorics, integer programming, and alge-
braic geometry.

An ideal I in S = Q[x1, . . . , xn] is called a monomial ideal if it satisfies
any of the following equivalent conditions:

(a) I is generated by monomials,
(b) if f =

∑
α∈Nn kαx

α belongs to I then xα ∈ I whenever kα 6= 0,
(c) I is torus-fixed; in other words, if (c1, . . . , cn) ∈ (Q∗)n, then I is fixed

under the action xi 7→ cixi for all i.

It follows that a monomial ideal is uniquely determined by the monomials it
contains. Most operations are far simpler for a monomial ideal than for an
ideal generated by arbitrary polynomials. In particular, many invariants can
be effectively determined for monomial ideals. As a result, one can solve a
broad collection of problems by reducing to or encoding data in a monomial
ideal. The aim of this chapter is to develop the computational aspects of
monomial ideals in Macaulay 2 and demonstrate a range of applications.

This chapter is divided into five sections. Each section begins with a dis-
cussion of a computational procedure involving monomial ideals. Algorithms
are presented as Macaulay 2 functions. We illustrate these methods by solv-
ing problems from various areas of mathematics. In particular, we include
the Macaulay 2 code for generating interesting families of monomial ideals.
The first section introduces the basic functions on monomial ideals in Mac-
aulay 2. To demonstrate these functions, we use the Stanley-Reisner ideal
associated to a simplicial complex to compute its f -vector. Next, we present
two algorithms for finding a primary decomposition of a monomial ideal. In
a related example, we use graph ideals to study the complexity of determin-
ing the codimension of a monomial ideal. The third section focuses on the
standard pairs of a monomial ideal; two methods are given for finding the set
of standard pairs. As an application, we use standard pairs to solve integer
linear programming problems. The fourth section examines Borel-fixed ideals
and generic initial ideals. Combining these constructions with distractions,
we demonstrate that the Hilbert scheme Hilb 4t+1(P4) is connected. Finally,
we look at the chains of associated primes in various families of monomial
ideals.

1 The Basics of Monomial Ideals

Creating monomial ideals in Macaulay 2 is analogous to creating general
ideals. The monomial ideal generated by a sequence or list of monomials can
be constructed with the function monomialIdeal.

i1 : S = QQ[a, b, c, d];

i2 : I = monomialIdeal(a^2, a*b, b^3, a*c)

2 3
o2 = monomialIdeal (a , a*b, b , a*c)

o2 : MonomialIdeal of S

i3 : J = monomialIdeal{a^2, a*b, b^2}

2 2
o3 = monomialIdeal (a , a*b, b )

o3 : MonomialIdeal of S

The type MonomialIdeal is the class of all monomial ideals. If an entry in
the sequence or list is not a single monomial, then monomialIdeal takes only
the leading monomial; recall that every polynomial ring in Macaulay 2 is
equipped with a monomial ordering.

i4 : monomialIdeal(a^2+a*b, a*b+3, b^2+d)

2 2
o4 = monomialIdeal (a , a*b, b )

o4 : MonomialIdeal of S

There are also several methods of associating a monomial ideal to an
arbitrary ideal in a polynomial ring. The most important of these is the
initial ideal — the monomial ideal generated by the leading monomials of
all elements in the given ideal. When applied to an Ideal, the function
monomialIdeal returns the initial ideal.

i5 : K = ideal(a^2, b^2, a*b+b*c)

2 2
o5 = ideal (a , b , a*b + b*c)

o5 : Ideal of S

i6 : monomialIdeal K

2 2 2
o6 = monomialIdeal (a , a*b, b , b*c )

o6 : MonomialIdeal of S

This is equivalent to taking the leading monomials of a Gröbner basis for K.
In our example, the given generators for K are not a Gröbner basis.

i7 : monomialIdeal gens K

2 2
o7 = monomialIdeal (a , a*b, b )

o7 : MonomialIdeal of S

One can also test if a general ideal is generated by monomials with the
function isMonomialIdeal.

i8 : isMonomialIdeal K

o8 = false

i9 : isMonomialIdeal ideal(a^5, b^2*c, d^11)

o9 = true

The usual algebraic operations on monomial ideals are the same as on general
ideals. For example, we have

i10 : I+J

2 2
o10 = monomialIdeal (a , a*b, b , a*c)

o10 : MonomialIdeal of S

Example: Stanley-Reisner Ideals and f-vectors

Radical monomial ideals — ideals generated by squarefree monomials — have
a beautiful combinatorial interpretation in terms of simplicial complexes.
More explicitly, a simplicial complex ∆ on the vertex set {x1, . . . , xn} cor-
responds to the ideal I∆ in S = Q[x1, . . . , xn] generated by all monomials
xi1 · · ·xip such that {xi1 , . . . , xip} 6∈ ∆. The ideal I∆ is called the Stanley-
Reisner ideal of ∆.

To illustrate the connections between Stanley-Reisner ideals and simpli-
cial complexes, we consider the f -vector. Perhaps the most important invari-
ant of a simplicial complex, the f -vector of a d-dimensional simplicial complex
∆ is (f0, f1, . . . , fd) ∈ Nd+1, where fi denotes the number of i-dimensional
faces in ∆. From the monomial ideal point of view, the f -vector is encoded
in the Hilbert series of the quotient ring S/I∆ as follows:

Theorem 1.1. If ∆ is a simplicial complex with f-vector (f0, . . . , fd), then
the Hilbert series of S/I∆ is

HS/I∆(t) =
d∑

i=−1

fit
i+1

(1− t)i+1
,

where f−1 = 1.

Proof. Following Stanley [24], we work with the fine grading and then spe-
cialize. The fine grading of S is the Zn-grading defined by deg xi = ei ∈ Zn,
where ei is the i-th standard basis vector. The support of a monomial xα is
defined to be the set supp(xα) = {xi : αi > 0}. Observe that xα 6= 0 in S/I∆
if and only if supp(xα) ∈ ∆. Moreover, the nonzero monomials xα form a

Q-basis of S/I∆. By counting such monomials according to their support, we
obtain the following expression for the Hilbert series with the fine grading:

HS/I∆(t) =
∑
F∈∆

∑
α∈Nn

supp(xα)=F

tα =
∑
F∈∆

∏
xi∈F

ti
1− ti

.

Finally, by replacing each ti with t, we complete the proof. ut

Since HS/I∆(t) is typically expressed in the form h0+h1t+···+hdtd
(1−t)d+1 , we can

obtain the f -vector by using the identity
∑
i hit

i =
∑d
j=0 fj−1t

j(1− t)d−j . In
particular, we can compute f -vectors from Stanley-Reisner ideals as follows:

i11 : fvector = I -> (
R := (ring I)/I;
d := dim R;
N := poincare R;
t := first gens ring N;
while 0 == substitute(N, t => 1) do N = N // (1-t);
h := apply(reverse toList(0..d), i -> N_(t^i));
f := j -> sum(0..j+1, i -> binomial(d-i, j+1-i)*h#(d-i));
apply(toList(0..d-1), j -> f(j)));

For example, we can demonstrate that the f -vector of the octahedron is
(6, 12, 8).

i12 : S = QQ[x_1 .. x_6];

i13 : octahedron = monomialIdeal(x_1*x_2, x_3*x_4, x_5*x_6)

o13 = monomialIdeal (x x , x x , x x )
1 2 3 4 5 6

o13 : MonomialIdeal of S

i14 : fvector octahedron

o14 = {6, 12, 8}

o14 : List

More generally, we can recursively construct simplicial 2-spheres with
f0 ≥ 4, starting with the tetrahedron, by pulling a point in the relative
interior of a facet. This procedure leads to the following family:

i15 : simplicial2sphere = v -> (
S := QQ[x_1..x_v];
if v === 4 then monomialIdeal product gens S
else (

L := {};
scan(1..v-4, i -> L = L | apply(v-i-3,

j -> x_i*x_(i+j+4)));
scan(2..v-3, i -> L = L | {x_i*x_(i+1)*x_(i+2)});
monomialIdeal L));

i16 : apply({4,5,6,7,8}, j -> fvector simplicial2sphere(j))

o16 = {{4, 6, 4}, {5, 9, 6}, {6, 12, 8}, {7, 15, 10}, {8, 18, 12}}

o16 : List

Fig. 1. The octahedron

In fact, it follows from Euler’s formula that the f -vector of any simplicial
2-sphere has the form (v, 3v− 6, 2v− 4) for v ≥ 4. The problem of character-
izing the f -vectors for triangulations of d-spheres is open for d ≥ 3. One of
the most important results in this direction is the upper bound theorem for
simplicial spheres (Corollary 5.4.7 Bruns and Herzog [6]) which states that
the cyclic polytope has the maximal number of i-faces for all i. We point out
that Stanley’s proof of this theorem depends heavily on these methods from
commutative algebra.

On the other hand, the f -vectors for several major classes of simpli-
cial complexes have been characterized. The Kruskal-Katona theorem (The-
orem 8.32 in Ziegler [29]) gives necessary and sufficient conditions for a
sequence of nonnegative integers to be an f -vector of a simplicial com-
plex. Stanley [24] describes the f -vectors of pure shellable complexes and
Cohen-Macaulay complexes. Given the Betti numbers of a simplicial com-
plex, Björner and Kalai [5] specify the f -vectors. Finally, the g-theorem (The-
orem 8.35 in Ziegler [29]) characterizes the f -vectors for boundary complexes
of a simplicial convex polytope.

For a further study of Stanley-Reisner ideals see Bruns and Herzog [6]
and Stanley [24]. For more information of f -vectors, see Ziegler [29] and
Björner [4].

2 Primary Decomposition

A primary decomposition of an ideal I is an expression of I as a finite in-
tersection of primary ideals; an ideal J is called primary if r1r2 ∈ J implies
either r1 ∈ J or r`2 ∈ J for some ` > 0. Providing an algorithm for computing
the primary decomposition of an arbitrary ideal in a polynomial ring is quite
difficult. However, for monomial ideals, there are two algorithms which are
relatively simple to describe.

We first present a recursive method for generating an irreducible primary
decomposition. It is based on the following two observations.

Lemma 2.1. Let I be a monomial ideal in S = Q[x1, . . . , xn].

(1) If I is generated by pure powers of a subset of the variables, then it is a
primary ideal.

(2) If r is minimal generator of I such that r = r1r2 where r1 and r2 are
relatively prime, then I =

(
I + 〈r1〉

)
∩
(
I + 〈r2〉

)
.

Proof. (1) This follows immediately from the definition of primary. (2) Since
I is a monomial ideal, it is enough to show that I and

(
I + 〈r1〉

)
∩
(
I + 〈r1〉

)
contain the same monomials. A monomial r′ belongs to

(
I+ 〈rj〉

)
if and only

if r′ ∈ I or rj divides r′. Because r1 and r2 are relative prime, we have

r′ ∈
(
I + 〈r1〉

)
∩
(
I + 〈r1〉

)
⇔ r′ ∈ I or r1r2 divides r′ ⇔ r′ ∈ I . ut

The following is an implementation of the resulting algorithm:
i17 : supp = r -> select(gens ring r, e -> r % e == 0);

i18 : monomialDecompose = method();

i19 : monomialDecompose List := L -> (
P := select(L, I -> all(first entries gens I,

r -> #supp(r) < 2) === false);
if #P > 0 then (

I := first P;
m := first select(first entries gens I,

r -> #supp(r) > 1);
E := first exponents m;
i := position(E, e -> e =!= 0);
r1 := product apply(E_{0..i}, (gens ring I)_{0..i},

(j, r) -> r^j);
r2 := m // r1;
monomialDecompose(delete(I, L) | {I+monomialIdeal(r1),

I+monomialIdeal(r2)}))
else L);

i20 : monomialDecompose MonomialIdeal := I -> monomialDecompose {I};

Here is a small example illustrating this method.
i21 : S = QQ[a,b,c,d];

i22 : I = monomialIdeal(a^3*b, a^3*c, a*b^3, b^3*c, a*c^3, b*c^3)

3 3 3 3 3 3
o22 = monomialIdeal (a b, a*b , a c, b c, a*c , b*c )

o22 : MonomialIdeal of S

i23 : P = monomialDecompose I;

i24 : scan(P, J -> << endl << J << endl);

monomialIdeal (b, c)

monomialIdeal (a, c)

3 3 3
monomialIdeal (a , b , c )

monomialIdeal (a, b)

3 3
monomialIdeal (a , b, c )

monomialIdeal (a, b)

3 3
monomialIdeal (a, b , c )

i25 : I == intersect(P)

o25 = true

As we see from this example, this procedure doesn’t necessarily yield an
irredundant decomposition.

The second algorithm for finding a primary decomposition of a monomial
ideal I is based on the Alexander dual of I. The Alexander dual was first in-
troduced for squarefree monomial ideals. In this case, it is the monomial ideal
of the dual of the simplicial complex ∆ corresponding to I. By definition the
dual complex of ∆ is ∆∨ = {F : F c 6∈ ∆}, where F c = {x1, . . . , xn} \F . The
following general definition appears in Miller [16], [17]. If I ⊆ Q[x1, . . . , xn]
is a monomial ideal and xλ is the least common multiple of the minimal
generators of I, then the Alexander dual of I is

I∨ =

〈∏
βi>0

xλi+1−βi
i : 〈x

βi
i : βi ≥ 1〉 is an irredundant

irreducible component of I

〉
.

In particular, the minimal generators of I∨ correspond to the irredundant
irreducible components of I. The next proposition provides a useful way of
computing I∨ given a set of generators for I.

Proposition 2.2. If I is a monomial ideal and xλ is the least common mul-
tiple of the minimal generators of I, then the generators for I∨ are those gen-
erators of the ideal

(
〈xλ1+1

1 , . . . , xλn+1
n 〉 : I

)
that are not divisible by xλi+1

i

for 1 ≤ i ≤ n.

Proof. See Theorem 2.1 in Miller [16]. ut

Miller’s definition of Alexander dual is even more general than the one
above. The resulting algorithm for computing this general Alexander dual
and primary decomposition are implemented in Macaulay 2 as follows. For
the Alexander dual we use, the list a that appears as an input argument for
dual should be list of exponents of the least common multiple of the minimal
generators of I.

i26 : code(dual, MonomialIdeal, List)

o26 = -- ../../../m2/monideal.m2:260-278
dual(MonomialIdeal, List) := (I,a) -> ( -- Alexander dual

R := ring I;
X := gens R;
aI := lcmOfGens I;
if aI =!= a then (

if #aI =!= #a
then error (

"expected list of length ",
toString (#aI));

scan(a, aI,
(b,c) -> (

if b<c then
error "exponent vector not large enough"
));

);
S := R/(I + monomialIdeal apply(#X, i -> X#i^(a#i+1)));
monomialIdeal contract(

lift(syz transpose vars S, R),
product(#X, i -> X#i^(a#i))))

i27 : code(primaryDecomposition, MonomialIdeal)

o27 = -- ../../../m2/monideal.m2:286-295
primaryDecomposition MonomialIdeal := (I) -> (

R := ring I;
aI := lcmOfGens I;
M := first entries gens dual I;
L := unique apply(#M, i -> first exponents M_i);
apply(L, i -> monomialIdeal apply(#i, j -> (

if i#j === 0 then 0_R
else R_j^(aI#j+1-i#j)
)))

)

This direct algorithm is more efficient than our recursive algorithm. In par-
ticular, it gives an irredundant decomposition. For example, when we use it
to determine a primary decomposition for the ideal I above, we obtain

i28 : L = primaryDecomposition I;

i29 : scan(L, J -> << endl << J << endl);

3 3 3
monomialIdeal (a , b , c )

monomialIdeal (b, c)

monomialIdeal (a, b)

monomialIdeal (a, c)

i30 : I == intersect L

o30 = true

For a family of larger examples, we consider the tree ideals:〈(∏
i∈Fxi

)n−|F |+1 : ∅ 6= F ⊆ {x1, . . . , xn}
〉
.

These ideals are so named because their standard monomials (the monomials
not in the ideal) correspond to trees on n+ 1 labeled vertices. We determine
the number of irredundant irreducible components as follows:

i31 : treeIdeal = n -> (
S = QQ[vars(0..n-1)];
L := delete({}, subsets gens S);
monomialIdeal apply(L, F -> (product F)^(n - #F +1)));

i32 : apply(2..6, i -> #primaryDecomposition treeIdeal i)

o32 = (2, 6, 24, 120, 720)

o32 : Sequence

Example: Graph Ideals and Complexity Theory

Monomial ideals also arise in graph theory. Given a graph G with vertices
{x1, . . . , xn}, we associate the ideal IG in Q[x1, . . . , xn] generated by the
quadratic monomials xixj such that xi is adjacent to xj . The primary de-
composition of IG is related to the graph G as follows. Recall that a subset
F ⊆ {x1, . . . , xn} is called a vertex cover of G if each edge in G is incident to
at least one vertex in F .

Lemma 2.3. If G is a graph and C is the set of minimal vertex covers of G
then the irreducible irredundant primary decomposition of IG is

⋂
F∈C PF c ,

where PF c is the prime ideal 〈xi : xi 6∈ F c〉 = 〈xi : xi ∈ F 〉.

Proof. Since each generator of IG corresponds to an edge in G, it follows
from the monomialDecompose algorithm that IG has an irreducible primary
decomposition of the form: IG =

⋂
PF c , where F is a vertex cover. To obtain

an irredundant decomposition, one clearly needs only the minimal vertex
covers. ut

As an application of graph ideals, we examine the complexity of deter-
mining the codimension of a monomial ideal. In fact, following Bayer and
Stillman [3], we prove

Proposition 2.4. The following decision problem is NP-complete:

Given a monomial ideal I ⊆ Q[x1, . . . , xn] and
m ∈ N, is codim I ≤ m?

(Codim)

By definition, a decision problem is NP-complete if all other problems
in the class NP can be reduced to it. To prove that a particular problem is
NP-complete, it suffices to show: (1) the problem belongs to the class NP; (2)
some known NP-complete problem reduces to the given decision problem (see
Lemma 2.3 in Garey and Johnson [8]). One of the “standard NP-complete”
problems (see section 3.1 in Garey and Johnson [8]) is the following:

Given a graph G and m ∈ N, is there a vertex
cover F such that |F | ≤ m? (Vertex Cover)

Proof of Proposition. (1) Observe that a monomial ideal I has codimension
at most m if and only if I ⊆ PF c for some F with |F | ≤ m. Now, if I
has codimension at most m, then given an appropriate choice of F , one can
verify in polynomial time that I ⊆ PF c and |F | ≤ m. Therefore, the Codim

problem belongs to the class NP.
(2) Lemma 2.3 implies that IG has codimension m if and only if G has a

vertex cover of size at most m. In particular, the Vertex Cover problem
reduces to the Codim problem. ut

Thus, assuming P 6= NP, there is no polynomial time algorithm for find-
ing the codimension of a monomial ideal. Nevertheless, we can effectively
compute the codimension for many interesting examples.

To illustrate this point, we consider the following family of examples. Let
S = Q[X] denote the polynomial ring generated by the entries of a generic
m×n matrix X = [xi,j ]. Let Ik be the ideal generated by the k×k minors of
X. Since the Hilbert function of S/Ik equals the Hilbert function of S/ in(Ik)
(see Theorem 15.26 in Eisenbud [7]), we can determine the codimension Ik by
working with the monomial ideal in(Ik). Because Sturmfels [25] shows that
the set of k × k-minors of X is the reduced Gröbner basis of Ik with respect
to the lexicographic term order induced from the variable order

x1,n > x1,n−1 > · · · > x1,1 > x2,n > · · · > x2,1 > · · · > xm,n > · · · > xm,1 ,

we can easily calculate in(Ik). In particular, in Macaulay 2 we have
i33 : minorsIdeal = (m,n,k) -> (

S := QQ[x_1..x_(m*n), MonomialOrder => Lex];
I := minors(k, matrix table(m, n, (i,j) -> x_(i*n+n-j)));
forceGB gens I;
I);

i34 : apply(2..8, i -> time codim monomialIdeal minorsIdeal(i,2*i,2))
-- used 0.02 seconds
-- used 0.05 seconds
-- used 0.1 seconds
-- used 0.36 seconds
-- used 1.41 seconds
-- used 5.94 seconds
-- used 25.51 seconds

o34 = (3, 10, 21, 36, 55, 78, 105)

o34 : Sequence

The properties of Ik are further developed in chapter 11 of Sturmfels [26] and
chapter 7 of Bruns and Herzog [6]

For more on the relationships between a graph and its associated ideal,
see Villarreal [28], Simis, Vasconcelos and Villarreal [23], and Ohsugi and
Hibi [19].

i35 : erase symbol x;

3 Standard Pairs

In this section, we examine a combinatorial object associated to a monomial
ideal. In particular, we present two algorithms for computing the standard
pairs of a monomial ideal from its minimal generators. Before giving the
definition of a standard pair, we consider an example.

Fig. 2. Staircase diagram for I = 〈xy3z, xy2z2, y3z2, y2z3〉

Example 3.1. Let I = 〈xy3z, xy2z2, y3z2, y2z3〉 inQ[x, y, z]. We identify the
monomials in Q[x, y, z] with the lattice points in N3; see Figure 2. The stan-
dard monomials of I, those monomials which are not in I, can be enumerated
as follows: (i) monomials corresponding to lattice points in the xy-plane, (ii)
monomials corresponding to lattice points in the xz-plane, (iii) monomials
corresponding to lattice points in the plane parallel to the xz-plane containing
(0, 1, 0), (iv) monomials corresponding to lattice points on the line parallel to
the y-axis containing (0, 0, 1), (v) monomials corresponding to lattices point
on the line parallel to the x-axis containing (0, 2, 1), and (vi) the monomial
y2z2.

Following Sturmfels, Trung and Vogel [27], we make the following defini-
tions. Given a monomial xα and a subset F ⊆ {x1, . . . , xn}, we index the set
of monomials of the form xα · xβ where supp(xβ) ⊆ F by the pair (xα, F ). A
standard pair of a monomial ideal I is a pair (xα, F ) satisfying the following
three conditions:

(1) supp(xα) ∩ F = ∅,

(2) all of the monomials represented by this pair are standard, and
(3) (xα, F ) 6⊆ (xβ , G) for any other pair (xβ , G) satisfying the first two con-

ditions.

Hence, the six standard pairs

(1, {x, y}), (1, {x, z}), (y, {x, z}), (z, {y}), (y2z, {x}), (y2z2, ∅)

in Example 3.1 correspond to (i)–(vi).
Observe that the set of standard pairs of I gives an irreducible decompo-

sition of I =
⋂
〈xαi+1
i : xi /∈ F 〉, where the intersection is over all standard

pairs (xα, F ). Moreover, the prime ideal PF := 〈xi : xi /∈ F 〉 is an associated
prime of I if and only if there exists a standard pair of the form (•, F ); see
Sturmfels, Trung and Vogel [27] for details.

Our first algorithm for computing the set of standard pairs is taken from
Hoşten and Thomas [15]. The ideas behind it are as follows: given a witness
w1 = xα for the associated prime PF := 〈xi : xi /∈ F 〉, that is (I : xα) = PF ,
set w2 =

∏
xi∈supp(w1)∩F c x

αi
i . It follows that (w2, F ) is a standard pair of I.

Now, consider the standard pairs of the slightly larger ideal I+ 〈w1〉. Clearly
(w2, F ) is not a standard pair of this ideal because w1 “destroys” it. This
larger ideal might have standard pairs which cover standard monomials in
(w2, F ) that are not in the pair (w1, F ). However, all other standard pairs
are the same as the original ideal I. Thus, the problem of finding all standard
pairs of I reduces to determining if a standard pair of I + 〈w1〉 is a standard
pair for I. To decide if a pair (xβ , G) of I + 〈w1〉 is a standard pair of I, we
first check that PF is an associated prime of I. If this is true, we determine
if (w2, F ) is covered by (xβ , G).

The Macaulay 2 version of this algorithm takes the following form:
i36 : stdPairs = I -> (

S := ring I;
X := gens S;
std := {};
J := I;
while J != S do (

w1 := 1_S;
F := X;
K := J;
while K != 0 do (

g1 := (ideal mingens ideal K)_0;
x := first supp g1;
w1 = w1 * g1 // x;
F = delete(x, F);
K = K : monomialIdeal(g1 // x);
L := select(first entries gens K,

r -> not member(x, supp r));
if #L > 0 then K = monomialIdeal L
else K = monomialIdeal 0_S;);

w2 := w1;
scan(X, r -> if not member(r, supp w1) or member(r, F)

then w2 = substitute(w2, {r => 1}));
P := monomialIdeal select(X, r -> not member(r, F));

if (I:(I:P) == P) and (all(std, p ->
(w2 % (first p) != 0) or not
isSubset(supp(w2 // first p) | F, last p)))

then std = std | {{w2, F}};
J = J + monomialIdeal(w1););

std);

We can compute the standard pairs of Example 3.1 using this Macaulay 2
function:

i37 : S = QQ[x,y,z];

i38 : I = monomialIdeal(x*y^3*z, x*y^2*z^2, y^3*z^2, y^2*z^3);

o38 : MonomialIdeal of S

i39 : scan(time stdPairs I, P -> << endl << P << endl);
-- used 0.66 seconds

{y, {x, z}}

{1, {x, z}}

2 2
{y z , {}}

{z, {y}}

2
{y z, {x}}

{1, {x, y}}

Our second algorithm is taken from section 3.2 of Saito, Sturmfels and
Takayama [22]. The proposition below provides the main ingredient for this
algorithm. If I is a monomial ideal and F ⊆ {x1, . . . , xn}, we write IF for the
monomial ideal in Q[xi : xi /∈ F ] obtained by replacing each xi ∈ F with 1
in every minimal generator of I.

Proposition 3.2. For (xα, F ) to be a standard pair of I, it is necessary and
sufficient that (xα, ∅) be a standard pair of IF .

Proof. Lemma 3.1 in Sturmfels, Trung and Vogel [27].

The definition of a standard pair implies that (xα, ∅) is a standard pair
of IF if and only if xα is one of the finitely many monomials contained in
(IF : P∞F ) but not contained in IF , where PF = 〈xi : xi /∈ F 〉. Since ideal
quotients and saturations are implemented in Macaulay 2, this reduces the
problem to finding a set D which contains F for every associated prime PF
of I. One approach is to simply compute the associated primes of I from a
primary decomposition.

The method standardPairs uses this algorithm to determine the set of
standard pairs for a monomial ideal.

i40 : code(standardPairs, MonomialIdeal, List)

o40 = -- ../../../m2/monideal.m2:318-341
standardPairs(MonomialIdeal, List) := (I,D) -> (

R := ring I;
X := gens R;
S := {};
k := coefficientRing R;
scan(D, L -> (

Y := X;
m := vars R;
Lset := set L;
Y = select(Y, r -> not Lset#?r);
m = substitute(m, apply(L, r -> r => 1));
-- using monoid to create ring to avoid
-- changing global ring.
A := k (monoid [Y]);
phi := map(A, R, substitute(m, A));
J := ideal mingens ideal phi gens I;
Jsat := saturate(J, ideal vars A);
if Jsat != J then (

B := flatten entries super basis (
trim (Jsat / J));

psi := map(R, A, matrix{Y});
S = join(S, apply(B, b -> {psi(b), L}));
)));

S)

i41 : time standardPairs I;
-- used 0.83 seconds

As an example, we will compute the standard pairs of the permutahedron
ideal. Let S = Q[x1, . . . , xn] and let Sn be the symmetric group of order n.
We write ρ for the vector (1, 2, . . . , n) and σ(ρ) for the vector obtained by
applying σ ∈ Sn to the coordinates of ρ. The n-th permutahedron ideal is
〈xσ(ρ) : σ ∈ Sn〉. We compute the number of standard pairs for 2 ≤ n ≤ 5.

i42 : permutohedronIdeal = n -> (
S := QQ[X_1..X_n];
monomialIdeal terms det matrix table(n ,gens S,

(i,r) -> r^(i+1)));

i43 : L = apply({2,3,4,5}, j -> standardPairs(permutohedronIdeal(j)));

i44 : apply(L, i -> #i)

o44 = {3, 10, 53, 446}

o44 : List

i45 : erase symbol x; erase symbol z;

Example: Integer Programming Problems

As an application of standard pairs, we show how to solve integer linear
programming problems. Let A be a d× n matrix of nonnegative integers, let
ω ∈ Rn and fix β ∈ Zd. We focus on the following optimization problem

IPA,ω(β) : minimize ω · α subject to Aα = β, α ∈ Nn.

We view this integer linear program as a family depending on the vector β.
The algorithm we present for solving IPA,ω(β) depends on the proposition
below.

The toric ideal IA ⊆ S = Q[x1, . . . , xn] associated to A is the binomial
ideal generated by xγ − xδ where γ, δ ∈ Nn and Aγ = Aδ. We write inω(IA)
for the initial ideal of IA with respect to the following order:

xγ ≺ω xδ ⇐⇒

{
ω · γ < ω · δ or
ω · γ = ω · δ and xα ≺rlex x

γ .

For more information on toric ideals and their initial ideals see Sturmfels [26].

Proposition 3.3. (1) A monomial xα is a standard monomial of inω(IA) if
and only if α is the optimal solution to the integer program IPA,ω(Aα).

(2) If (•, F ) is a standard pair of inω(IA), then the columns of A correspond-
ing to F are linearly independent.

Proof. See Proposition 2.1 in Hoşten and Thomas [14] for the proof of the
first statement. The second statement follows from Corollary 2.9 of the same
article. ut

The first statement implies that the standard pairs of inω(IA) cover all
optimal solutions to all integer programs in IPA,ω. If α is the optimal solution
to IPA,ω(β) covered by the standard pair (xγ , F ), then the second statement
guarantees there exists a unique δ ∈ Nn such that Aδ = β − Aγ. Therefore,
α = δ + γ. We point out that the complexity of this algorithm is dominated
by determining the set of standard pairs of inω(IA) which depends only on A
and ω. As a result, this method is particularly well suited to solving IPA,ω(β)
as β varies.

To implement this algorithm in Macaulay 2, we need a function which
returns the toric ideal IA. Following Algorithm 12.3 in Sturmfels [26], we
have

i47 : toBinomial = (b, S) -> (
pos := 1_S;
neg := 1_S;
scan(#b, i -> if b_i > 0 then pos = pos*S_i^(b_i)

else if b_i < 0 then neg = neg*S_i^(-b_i));
pos - neg);

i48 : toricIdeal = (A, omega) -> (
n := rank source A;
S = QQ[x_1..x_n, Weights => omega, MonomialSize => 16];
B := transpose matrix syz A;
J := ideal apply(entries B, b -> toBinomial(b, S));
scan(gens S, r -> J = saturate(J, r));
J);

Thus, we can solve IPA,ω(β) using the following function.

i49 : IP = (A, omega, beta) -> (
std := standardPairs monomialIdeal toricIdeal(A, omega);
n := rank source A;
alpha := {};
Q := first select(1, std, P -> (

F := apply(last P, r -> index r);
gamma := transpose matrix exponents first P;
K := transpose syz (submatrix(A,F) | (A*gamma-beta));
X := select(entries K, k -> abs last(k) === 1);
scan(X, k -> if all(k, j -> j>=0) or all(k, j -> j<=0)

then alpha = apply(n, j -> if member(j, F)
then last(k)*k_(position(F, i -> i === j))
else 0));

#alpha > 0));
if #Q > 0 then (matrix {alpha})+(matrix exponents first Q)
else 0);

We illustrate this with some examples.
i50 : A = matrix{{1,1,1,1,1},{1,2,4,5,6}}

o50 = | 1 1 1 1 1 |
| 1 2 4 5 6 |

2 5
o50 : Matrix ZZ <--- ZZ

i51 : w1 = {1,1,1,1,1};

i52 : w2 = {2,3,5,7,11};

i53 : b1 = transpose matrix{{3,9}}

o53 = | 3 |
| 9 |

2 1
o53 : Matrix ZZ <--- ZZ

i54 : b2 = transpose matrix{{5,16}}

o54 = | 5 |
| 16 |

2 1
o54 : Matrix ZZ <--- ZZ

i55 : IP(A, w1, b1)

o55 = | 1 1 0 0 1 |

1 5
o55 : Matrix ZZ <--- ZZ

i56 : IP(A, w2, b1)

o56 = | 1 0 2 0 0 |

1 5
o56 : Matrix ZZ <--- ZZ

i57 : IP(A, w1, b2)

o57 = | 2 1 0 0 2 |

1 5
o57 : Matrix ZZ <--- ZZ

i58 : IP(A, w2, b2)

o58 = | 2 0 1 2 0 |

1 5
o58 : Matrix ZZ <--- ZZ

4 Generic Initial Ideals

Gröbner basis calculations and initial ideals depend heavily on the given co-
ordinate system. By making a generic change of coordinates before taking the
initial ideal, we may eliminate this dependence. This procedure also endows
the resulting monomial ideal with a rich combinatorial structure.

To describe this structure, we introduce the following definitions and nota-
tion. Let S = Q[x0, . . . , xn]. If g = [gi,j ] ∈ GLn+1(Q) and f ∈ S, then g ·f de-
notes the standard action of the general linear group on S: xi 7→

∑n
j=0 gi,jxj .

For an ideal I ⊆ S, we define g ·I = {g ·f |f ∈ I}. Let B denote the Borel sub-
group of GLn+1(Q) consisting of upper triangular matrices. A monomial ideal
I is called Borel-fixed if it satisfies any of the following equivalent conditions:

(a) g · I = I for every g ∈ B;
(b) if r is a generator of I divisible by xj then rxi

xj
∈ I for all i < j;

(c) in(g · I) = I for every g is some open neighborhood of the identity in B.

For a proof that these conditions are equivalent, see Propositon 1.25 in
Green [9].

In Macaulay 2, the function isBorel tests whether a monomial ideal is
Borel-fixed.

i59 : S = QQ[a,b,c,d];

i60 : isBorel monomialIdeal(a^2, a*b, b^2)

o60 = true

i61 : isBorel monomialIdeal(a^2, b^2)

o61 = false

The function borel generates the smallest Borel-fixed ideal containing the
given monomial ideal.

i62 : borel monomialIdeal(b*c)

2 2
o62 = monomialIdeal (a , a*b, b , a*c, b*c)

o62 : MonomialIdeal of S

i63 : borel monomialIdeal(a,c^3)

3 2 2 3
o63 = monomialIdeal (a, b , b c, b*c , c )

o63 : MonomialIdeal of S

The next theorem provides the main source of Borel-fixed ideals.

Theorem 4.1 (Galligo). Fix a term order on S = Q[x0, . . . , xn] such that
x0 > . . . > xn. If I is a homogeneous ideal in S, then there is a Zariski open
subset U ⊆ GLn+1(Q) such that

(1) there is a monomial ideal J ⊆ S such that J = in(g · I) for all g ∈ U ;
(2) the ideal J is Borel-fixed.

The ideal J is called the generic initial ideal of I.

Proof. See Theorem 1.27 in Green [9]. ut

The following method allows one to compute generic initial ideals.
i64 : gin = method();

i65 : gin Ideal := I -> (
S := ring I;
StoS := map(S, S, random(S^{0}, S^{numgens S:-1}));
monomialIdeal StoS I);

i66 : gin MonomialIdeal := I -> gin ideal I;

This routine assumes that the random function generates a matrix in the
Zariski open subset U . Since we are working over a field of characteristic zero
this occurs with probability one. For example, we can determine the generic
initial ideal of two generic homogeneous polynomials of degree p and q in
Q[a, b, c, d].

i67 : genericForms = (p,q) -> ideal(random(p,S), random(q,S));

i68 : gin genericForms(2,2)

2 3
o68 = monomialIdeal (a , a*b, b )

o68 : MonomialIdeal of S

i69 : gin genericForms(2,3)

2 2 4
o69 = monomialIdeal (a , a*b , b )

o69 : MonomialIdeal of S

Although the generic initial ideal is Borel-fixed, some non-generic initial ideals
may also be Borel-Fixed.

i70 : J = ideal(a^2, a*b+b^2, a*c)

2 2
o70 = ideal (a , a*b + b , a*c)

o70 : Ideal of S

i71 : ginJ = gin J

2 2 2
o71 = monomialIdeal (a , a*b, b , a*c )

o71 : MonomialIdeal of S

i72 : inJ = monomialIdeal J

2 3 2
o72 = monomialIdeal (a , a*b, b , a*c, b c)

o72 : MonomialIdeal of S

i73 : isBorel inJ and isBorel ginJ

o73 = true

Finally, we show that the generic initial ideal does depend on the term or-
der by computing lexicographic generic initial ideal for two generic forms of
degree p and q in Q[a, b, c, d]

i74 : S = QQ[a,b,c,d, MonomialOrder => Lex];

i75 : gin genericForms(2,2)

2 4 2
o75 = monomialIdeal (a , a*b, b , a*c )

o75 : MonomialIdeal of S

i76 : gin genericForms(2,3)

2 2 6 2 6 2 4
o76 = monomialIdeal (a , a*b , b , a*b*c , a*c , a*b*c*d , a*b*d )

o76 : MonomialIdeal of S

A more comprehensive treatment of generic initial ideals can be found in
Green [9]. The properties of Borel-fixed ideals in characteristic p > 0 are
discussed in Eisenbud [7].

Example: Connectedness of the Hilbert Scheme

Generic initial ideals are a powerful tool for studying the structure of the
Hilbert scheme. Intuitively, the Hilbert scheme Hilb p(t)(Pn) parameterizes
subschemes X ⊆ P

n with Hilbert polynomial p(t). For an introduction to
Hilbert schemes see Harris and Morrison [11]. The construction of the Hilbert
scheme Hilb p(t)(Pn) can be found in Grothendieck’s original article [10] or
Altman and Kleiman [1]. While much is known about specific Hilbert schemes,
the general structure remain largely a mystery. In particular, the component
structure — the number of irreducible components, their dimensions, how
they intersect and what subschemes they parameterize — is not well under-
stood.

Reeves [21] uses generic initial ideals to establish the most important theo-
rem to date on the component structure. The incidence graph of Hilb p(t)(Pn)
is defined as follows: to each irreducible component we assign a vertex and we
connect two vertices if the corresponding components intersect. Reeves [21]
proves that the distance (the number of edges in the shortest path) between
any two vertices in the incidence graph of Hilb p(t)(Pn) is at most 2 deg p(t)+2.
Her proof can be divided into three major steps.

Step I: connect an arbitrary ideal to a Borel-fixed ideal. Passing to an ini-
tial ideal corresponds to taking the limit in a flat family, in other words a path
on the Hilbert scheme; see Theorem 15.17 in Eisenbud [7]. Thus, Theorem 4.1
shows that generic initial ideals connect arbitrary ideals to Borel-fixed ideals.

Step II: connect Borel-fixed ideals by projection. For a homogeneous ideal
I ⊆ S = Q[x0, . . . , xn], let π(I) denote the ideal obtained by setting xn = 1
and xn−1 = 1 in I. With this notation, we have

Theorem 4.2. If J is a Borel-fixed ideal, then the set of Borel-fixed ideals
I, with Hilbert polynomial p(t) and π(I) = J , consists of ideals defining
subschemes of Pn which all lie on a single component of Hilb p(t)(Pn).

Proof. See Theorem 6 in Reeves [21]. ut

This gives an easy method for partitioning Borel-fixed ideals into classes,
each of which must lie in a single component.

Step III: connect Borel-fixed ideals by distraction. Given a Borel-fixed
ideal, we produce a new ideal via a two-step process called distraction. First,
one polarizes the Borel-fixed ideal. The polarization of a monomial ideal
I ⊂ S is defined as:〈

n∏
i=0

αi∏
j=1

zi,j : where xα0
0 · · ·xαnn is a minimal generator of I

〉
.

One then pulls the result back to an ideal in the original variables by taking
a linear section of the polarization. Theorem 4.10 in Hartshorne [12] shows
that the distraction is connected to the original Borel-fixed ideal. Now, taking
the lexicographic generic initial ideal of the distraction yields a second Borel-
fixed ideal. Reeves [21] proves that repeating this process, at most deg p(t)+1
times, one arrives at a distinguished component of Hilb p(t)(Pn) called the lex-
icographic component. For more information on the lexicographic component
see Reeves and Stillman [20].

We can implement these operations in Macaulay 2 as follows:
i77 : projection = I -> (

S := ring I;
n := numgens S;
X := gens S;
monomialIdeal mingens substitute(ideal I,

{X#(n-2) => 1, X#(n-1) => 1}));

i78 : polarization = I -> (
n := numgens ring I;
u := apply(numgens I, i -> first exponents I_i);
I.lcm = max \ transpose u;
Z := flatten apply(n, i -> apply(I.lcm#i, j -> z_{i,j}));
R := QQ(monoid[Z]);
Z = gens R;
p := apply(n, i -> sum((I.lcm)_{0..i-1}));
monomialIdeal apply(u, e -> product apply(n, i ->

product(toList(0..e#i-1), j -> Z#(p#i+j)))));

i79 : distraction = I -> (
S := ring I;
n := numgens S;
X := gens S;
J := polarization I;
W := flatten apply(n, i -> flatten apply(I.lcm#i,

j -> X#i));
section := map(S, ring J, apply(W, r -> r -

random(500)*X#(n-2) - random(500)*X#(n-1)));
section ideal J);

For example, we have
i80 : S = QQ[x_0 .. x_4, MonomialOrder => GLex];

i81 : I = monomialIdeal(x_0^2, x_0*x_1^2*x_3, x_1^3*x_4)

2 2 3
o81 = monomialIdeal (x , x x x , x x )

0 0 1 3 1 4

o81 : MonomialIdeal of S

i82 : projection I

2 2 3
o82 = monomialIdeal (x , x x , x )

0 0 1 1

o82 : MonomialIdeal of S

i83 : polarization I

o83 = monomialIdeal (z z , z z z z , z · · ·
{0, 0} {0, 1} {0, 0} {1, 0} {1, 1} {3, 0} {1 · · ·

o83 : MonomialIdeal of QQ [z , z , z , z , z · · ·
{0, 0} {0, 1} {1, 0} {1, 1} {1, 2} · · ·

i84 : distraction I

2 2 2 · · ·
o84 = ideal (x - 398x x - 584x x + 36001x + 92816x x + 47239x , - · · ·

0 0 3 0 4 3 3 4 4 · · ·
o84 : Ideal of S

To illustrate Reeves’ method, we show that the incidence graph of the
Hilbert scheme Hilb4t+1(P4) has diameter at most 2. Note that the rational
quartic curve in P4 has Hilbert polynomial 4t+ 1.

i85 : m = matrix table({0,1,2}, {0,1,2}, (i,j) -> (gens S)#(i+j))

o85 = | x_0 x_1 x_2 |
| x_1 x_2 x_3 |
| x_2 x_3 x_4 |

3 3
o85 : Matrix S <--- S

i86 : rationalQuartic = minors(2, m);

o86 : Ideal of S

i87 : H = hilbertPolynomial(S/rationalQuartic);

i88 : hilbertPolynomial(S/rationalQuartic, Projective => false)

o88 = 4$i + 1

o88 : QQ [$i]

There are 12 Borel-fixed ideals with Hilbert polynomial 4t+1; see Example 1
in Reeves [21].

i89 : L = {monomialIdeal(x_0^2, x_0*x_1, x_0*x_2, x_1^2, x_1*x_2, x_2^ · · ·
i90 : scan(#L, i -> << endl << i+1 << " : " << L#i << endl);

2 2 2
1 : monomialIdeal (x , x x , x , x x , x x , x )

0 0 1 1 0 2 1 2 2

2 2 3
2 : monomialIdeal (x , x x , x , x x , x x , x , x x )

0 0 1 1 0 2 1 2 2 0 3

2 2 3
3 : monomialIdeal (x , x , x x , x , x x x )

0 1 1 2 2 1 2 3

2 4 3
4 : monomialIdeal (x , x , x x , x , x x )

0 1 1 2 2 2 3

5 4 3
5 : monomialIdeal (x , x , x , x x )

0 1 2 2 3

2 5 4 2
6 : monomialIdeal (x , x , x x , x , x x , x x )

0 1 1 2 2 1 3 2 3

2 2 5 4
7 : monomialIdeal (x , x x , x , x x , x x , x , x x , x x , x x )

0 0 1 1 0 2 1 2 2 0 3 1 3 2 3

2 5 4 2
8 : monomialIdeal (x , x , x x , x , x x , x x )

0 1 1 2 2 2 3 1 3

2 2 4 2
9 : monomialIdeal (x , x x , x , x x , x x , x , x x , x x )

0 0 1 1 0 2 1 2 2 0 3 1 3

2 2 4 2
10 : monomialIdeal (x , x , x x , x , x x x , x x )

0 1 1 2 2 1 2 3 1 3

2 4 3
11 : monomialIdeal (x , x , x x , x , x x )

0 1 1 2 2 1 3

6 5 4 2
12 : monomialIdeal (x , x , x , x x , x x )

0 1 2 2 3 2 3

i91 : all(L, I -> isBorel I and hilbertPolynomial(S/I) == H)

o91 = true

The projection operation partitions the list L into 3 classes:
i92 : class1 = projection L#0

2 2 2
o92 = monomialIdeal (x , x x , x , x x , x x , x )

0 0 1 1 0 2 1 2 2

o92 : MonomialIdeal of S

i93 : class2 = projection L#1

2 3
o93 = monomialIdeal (x , x , x x , x )

0 1 1 2 2

o93 : MonomialIdeal of S

i94 : class3 = projection L#4

4
o94 = monomialIdeal (x , x , x )

0 1 2

o94 : MonomialIdeal of S

i95 : all(1..3, i -> projection L#i == class2)

o95 = true

i96 : all(4..11, i -> projection L#i == class3)

o96 = true

Finally, we use the distraction to connect the classes.
i97 : all(L, I -> I == monomialIdeal distraction I)

o97 = true

i98 : all(0..3, i -> projection gin distraction L#i == class3)

o98 = true

Therefore, the components corresponding to class1 and class2 intersect
the one corresponding to class3. Note that class3 corresponds to the lexi-
cographic component.

5 The Chain Property

Hoşten and Thomas [14] recently established that the initial ideals of a toric
ideal have an interesting combinatorial structure called the chain property.
This structure is on the poset of associated primes where the partial order is
given by inclusion. Since a monomial ideal I ⊂ S = Q[x1, . . . , xn] is prime if
and only if it is generated by a subset of the variables {x1, . . . , xn}, the poset
of associated primes of I is contained in the power set of the variables. We
say that a monomial ideal I has the chain property if the following condition
holds:

For any embedded prime PF = 〈xi : xi 6∈ F 〉 of I, there exists an
associated prime PG ⊂ PF such that |G| = |F | − 1.

In other words, there is a saturated chain from every embedded prime to
some minimal prime. Experimental evidence suggests that, in fact, most ini-
tial ideals of prime ideals satisfy this saturated chain condition. Because of
ubiquity and simplicity of this condition, we are interested in understanding
which classes of initial ideals (or more generally monomial ideals) have the
chain property.

More recently, Miller, Sturmfels and Yanagawa [18] provided a large class
of monomial ideals with the chain property. A monomial ideal I is called
generic when the following condition holds: if two distinct minimal generators
r1 and r2 of I have the same positive degree in some variable xi, there is a
third generator r3 which strictly divides the least common multiple of r1 and
r2. In particular, if no two distinct minimal generators have the same positive
degree in any variable, then the monomial ideal is generic. Theorem 2.2 in
Miller, Sturmfels and Yanagawa [18] shows that generic monomial ideals have
the chain property.

Examples and Counterexamples

In this final section, we illustrate how to use Macaulay 2 for further exper-
imentation and investigation of the chain property. The following function
determines whether a monomial ideal has the chain property:

i99 : hasChainProperty = I -> (
L := ass I;
radI := radical I;
all(L, P -> radI : (radI : P) == P or (

gensP := first entries gens P;
all(gensP, r -> (

Q := monomialIdeal delete(r, gensP);
I : (I : Q) == Q)))));

Using hasChainProperty, we examine the initial ideals of four interesting
classes of ideals related to toric ideals.

An Initial Ideal of a Toric Ideal. As mentioned above, Hoşten and
Thomas proved that any initial ideal of a toric ideal satisfies the saturated
chain condition. The following example demonstrates this phenomenon. Con-
sider the matrix A:

i100 : A = matrix{{1,1,1,1,1,1,1}, {2,0,0,0,1,0,0}, {0,2,0,0,0,1,0}, { · · ·
o100 = | 1 1 1 1 1 1 1 |

| 2 0 0 0 1 0 0 |
| 0 2 0 0 0 1 0 |
| 2 2 0 2 1 1 1 |

4 7
o100 : Matrix ZZ <--- ZZ

i101 : IA = toricIdeal(A, {1,1,1,1,1,1,1})

2 2 2
o101 = ideal (x x - x , x x - x , x x - x )

3 4 7 2 3 6 1 3 5

o101 : Ideal of S

i102 : inIA = monomialIdeal IA

2 2 2
o102 = monomialIdeal (x x , x x , x x , x x , x x , x x )

1 3 2 3 3 4 2 5 4 5 4 6

o102 : MonomialIdeal of S

i103 : hasChainProperty inIA

o103 = true

An Initial Ideal of a Prime Ideal. Since toric ideals are prime, one
naturally asks if the initial ideal of any prime ideal has the chain property.
By modifying the previous example, we can show that this is not the case.
In particular, making the linear change of coordinates by x4 7→ x3 − x4, we
obtain a new prime ideal J .

i104 : StoS = map(S, S, {x_1, x_2, x_3, x_3 - x_4, x_5, x_6, x_7});

o104 : RingMap S <--- S

i105 : J = StoS IA

2 2 2 2
o105 = ideal (x - x x - x , x x - x , x x - x )

3 3 4 7 2 3 6 1 3 5

o105 : Ideal of S

Taking the initial ideal with respect to the reverse lexicographic term order
(the default order), we have

i106 : inJ = monomialIdeal J

2 2 2 2 2 2 · · ·
o106 = monomialIdeal (x x , x x , x , x x , x x , x x x , x x , x x x · · ·

1 3 2 3 3 2 5 3 5 1 4 5 3 6 1 4 6 · · ·
o106 : MonomialIdeal of S

i107 : hasChainProperty inJ

o107 = false

An A-graded Monomial Ideal. Let A be a d × n matrix of nonnegative
integers and let ai denote the i-th column of A. Consider the polynomial
ring S = Q[x1, . . . , xn] with the Zd-grading defined by deg xi = ai. An ideal
I ⊂ Q[x1, . . . , xn] is called A-graded provided it is homogeneous with respect
to the A-grading and

dimQ

(
S

I

)
b

=

{
1 if b ∈ NA
0 otherwise

for all b ∈ Nd. Remark 10.1 in Sturmfels [26] shows that the initial ideal of
the toric ideal IA is A-graded. Altmann [2] shows that when A has rank 2
every A-graded monomial ideal has the chain property. However, Altmann [2]
also provides a counterexample when A has rank 3. We can verify his example
in Macaulay 2 as follows:

i108 : A = matrix{{2,0,0,1,0,0,2,1,1,3,2,2,2,3,3,3},
{0,2,0,0,1,0,1,2,1,2,3,2,3,2,3,3},
{0,0,2,0,0,1,1,1,2,2,2,3,3,3,2,3}};

3 16
o108 : Matrix ZZ <--- ZZ

In Macaulay 2, the first entry in degree vector of each variable must be
positive. Hence, we append to A the sum of its rows to get a matrix whose
columns will serve as the degrees of the variables.

i109 : D = A^{0}+A^{1}+A^{2} || A

o109 = | 2 2 2 1 1 1 4 4 4 7 7 7 8 8 8 9 |
| 2 0 0 1 0 0 2 1 1 3 2 2 2 3 3 3 |
| 0 2 0 0 1 0 1 2 1 2 3 2 3 2 3 3 |
| 0 0 2 0 0 1 1 1 2 2 2 3 3 3 2 3 |

4 16
o109 : Matrix ZZ <--- ZZ

i110 : D = entries transpose D;

i111 : S = QQ[vars(0..15), Degrees => D, MonomialSize => 16];

i112 : I = monomialIdeal(d*j, d*k, d*l, d*m, d*n, d*o, d*p, e*j, e*k,
e*l, e*m, e*n, e*o, e*p, f*j, f*k, f*l, f*m, f*n, f*o, f*p,
g*j, g*k, g*l, g*m, g*n, g*o, g*p, h*j, h*k, h*l, h*m, h*n,
h*o, h*p, i*j, i*k, i*l, i*m, i*n, i*o, i*p, g^2, g*h, g*i,
h^2, h*i, i^2, j^2, j*k, j*l, j*m, j*n, j*o, j*p, k^2, k*l,
k*m, k*n, k*o, k*p, l^2, l*m, l*n, l*o, l*p, m^2, m*n, m*o,
m*p, n^2, n*o, n*p, o^2, o*p, p^2, d^2, e^2, f^2, d*h, e*i,
f*g, f*d*i, d*e*g, e*f*h, c*d*g, a*e*h, b*f*i, c*e*g,
a*f*h, b*d*i, c*d*e, a*e*f, b*f*d, c*b*d, a*c*e, b*a*f,
c*b*g, a*c*h, b*a*i);

o112 : MonomialIdeal of S

To help convince you that I is an A-graded ideal, we compute the dimQ
(
S
I

)
ai

for 1 ≤ i ≤ 16.
i113 : apply(D, d -> rank source basis(d, (S^1)/ ideal I))

o113 = {1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1}

o113 : List

Finally, we check the chain property.
i114 : hasChainProperty I

o114 = false

The Vertex Ideal. Lastly, we consider a different family of monomials ideals
arising from toric ideals. The vertex ideal VA is defined as intersection all the

monomial initial ideals of the toric ideal IA. Although there are (in general)
infinitely many distinct term orders on a polynomial ring, an ideal has only
finitely many initial ideals; see Theorem 1.2 in Sturmfels [26]. In particular,
the above intersection is finite. Vertex ideals were introduced and studied by
Hoşten and Maclagan [13]. However, the question “Does the vertex ideal VA
have the chain property?” remains open.

From Enumerative Geometry to Solving
Systems of Polynomial Equations

Frank Sottile?

Solving a system of polynomial equations is a ubiquitous problem in the
applications of mathematics. Until recently, it has been hopeless to find ex-
plicit solutions to such systems, and mathematics has instead developed deep
and powerful theories about the solutions to polynomial equations. Enumer-
ative Geometry is concerned with counting the number of solutions when the
polynomials come from a geometric situation and Intersection Theory gives
methods to accomplish the enumeration.

We use Macaulay 2 to investigate some problems from enumerative geom-
etry, illustrating some applications of symbolic computation to this important
problem of solving systems of polynomial equations. Besides enumerating so-
lutions to the resulting polynomial systems, which include overdetermined,
deficient, and improper systems, we address the important question of real
solutions to these geometric problems.

1 Introduction

A basic question to ask about a system of polynomial equations is its number
of solutions. For this, the fundamental result is the following Bézout Theorem.

Theorem 1.1. The number of isolated solutions to a system of polynomial
equations

f1(x1, . . . , xn) = f2(x1, . . . , xn) = · · · = fn(x1, . . . , xn) = 0

is bounded by d1d2 · · · dn, where di := deg fi. If the polynomials are generic,
then this bound is attained for solutions in an algebraically closed field.

Here, isolated is taken with respect to the algebraic closure. This Bézout
Theorem is a consequence of the refined Bézout Theorem of Fulton and
MacPherson [12, §1.23].

A system of polynomial equations with fewer than this degree bound or
Bézout number of solutions is called deficient, and there are well-defined
classes of deficient systems that satisfy other bounds. For example, fewer
monomials lead to fewer solutions, for which polyhedral bounds [4] on the
number of solutions are often tighter (and no weaker than) the Bézout num-
ber, which applies when all monomials are present. When the polynomials
? Supported in part by NSF grant DMS-0070494.

come from geometry, determining the number of solutions is the central prob-
lem in enumerative geometry.

Symbolic computation can help compute the solutions to a system of equa-
tions that has only isolated solutions. In this case, the polynomials generate
a zero-dimensional ideal I. The degree of I is dimk k[X]/I, the dimension of
the k-vector space k[X]/I, which is also the number of standard monomi-
als in any term order. This degree gives an upper bound on the number of
solutions, which is attained when I is radical.

Example 1.2. We illustrate this discussion with an example. Let f1, f2, f3,
and f4 be random quadratic polynomials in the ring F101[y11, y12, y21, y22].

i1 : R = ZZ/101[y11, y12, y21, y22];

i2 : PolynomialSystem = apply(1..4, i ->
random(0, R) + random(1, R) + random(2, R));

The ideal they generate has dimension 0 and degree 16 = 24, which is the
Bézout number.

i3 : I = ideal PolynomialSystem;

o3 : Ideal of R

i4 : dim I, degree I

o4 = (0, 16)

o4 : Sequence

If we restrict the monomials which appear in the fi to be among

1, y11, y12, y21, y22, y11y22, and y12y21,

then the ideal they generate again has dimension 0, but its degree is now 4.
i5 : J = ideal (random(R^4, R^7) * transpose(

matrix{{1, y11, y12, y21, y22, y11*y22, y12*y21}}));

o5 : Ideal of R

i6 : dim J, degree J

o6 = (0, 4)

o6 : Sequence

If we further require that the coefficients of the quadratic terms sum to zero,
then the ideal they generate now has degree 2.

i7 : K = ideal (random(R^4, R^6) * transpose(
matrix{{1, y11, y12, y21, y22, y11*y22 - y12*y21}}));

o7 : Ideal of R

i8 : dim K, degree K

o8 = (0, 2)

o8 : Sequence

In Example 4.2, we shall see how this last specialization is geometrically
meaningful.

For us, enumerative geometry is concerned with enumerating geometric
figures of some kind having specified positions with respect to general fixed
figures. That is, counting the solutions to a geometrically meaningful system
of polynomial equations. We use Macaulay 2 to investigate some enumerative
geometric problems from this point of view. The problem of enumeration will
be solved by computing the degree of the (0-dimensional) ideal generated by
the polynomials.

2 Solving Systems of Polynomials

We briefly discuss some aspects of solving systems of polynomial equations.
For a more complete survey, see the relevant chapters in [6,7].

Given an ideal I in a polynomial ring k[X], set V(I) := Spec k[X]/I.
When I is generated by the polynomials f1, . . . , fN , V(I) gives the set of
solutions in affine space to the system

f1(X) = · · · = fN (X) = 0 (1)

a geometric structure. These solutions are the roots of the ideal I. The degree
of a zero-dimensional ideal I provides an algebraic count of its roots. The de-
gree of its radical counts roots in the algebraic closure, ignoring multiplicities.

2.1 Excess Intersection

Sometimes, only a proper (open) subset of affine space is geometrically mean-
ingful, and we want to count only the meaningful roots of I. Often the roots
V(I) has positive dimensional components that lie in the complement of the
meaningful subset. One way to treat this situation of excess or improper in-
tersection is to saturate I by a polynomial f vanishing on the extraneous
roots. This has the effect of working in k[X][f−1], the coordinate ring of the
complement of V(f) [9, Exer. 2.3].

Example 2.1. We illustrate this with an example. Consider the following
ideal in F7[x, y].

i9 : R = ZZ/7[y, x, MonomialOrder=>Lex];

i10 : I = ideal (y^3*x^2 + 2*y^2*x + 3*x*y, 3*y^2 + x*y - 3*y);

o10 : Ideal of R

Since the generators have greatest common factor y, I defines finitely many
points together with the line y = 0. Saturate I by the variable y to obtain
the ideal J of isolated roots.

i11 : J = saturate(I, ideal(y))

4 3 2
o11 = ideal (x + x + 3x + 3x, y - 2x - 1)

o11 : Ideal of R

The first polynomial factors completely in F7[x],
i12 : factor(J_0)

o12 = (x)(x - 2)(x + 2)(x + 1)

o12 : Product

and so the isolated roots of I are (2, 5), (−1,−1), (0, 1), and (−2,−3).

Here, the extraneous roots came from a common factor in both equations.
A less trivial example of this phenomenon will be seen in Section 5.2.

2.2 Elimination, Rationality, and Solving

Elimination theory can be used to study the roots of a zero-dimensional ideal
I ⊂ k[X]. A polynomial h ∈ k[X] defines a map k[y]→ k[X] (by y 7→ h) and
a corresponding projection h : Spec k[X] � A

1. The generator g(y) ∈ k[y]
of the kernel of the map k[y] → k[X]/I is called an eliminant and it has
the property that V(g) = h(V(I)). When h is a coordinate function xi, we
may consider the eliminant to be in the polynomial ring k[xi], and we have
〈g(xi)〉 = I ∩ k[xi]. The most important result concerning eliminants is the
Shape Lemma [2].

Shape Lemma. Suppose h is a linear polynomial and g is the corresponding
eliminant of a zero-dimensional ideal I ⊂ k[X] with deg(I) = deg(g). Then
the roots of I are defined in the splitting field of g and I is radical if and only
if g is square-free.

Suppose further that h = x1 so that g = g(x1). Then, in the lexicographic
term order with x1 < x2 < · · · < xn, I has a Gröbner basis of the form:

g(x1), x2 − g2(x1), . . . , xn − gn(x1) , (2)

where deg(g) > deg(gi) for i = 2, . . . , n.

When k is infinite and I is radical, an eliminant g given by a generic linear
polynomial h will satisfy deg(g) = deg(I). Enumerative geometry counts
solutions when the fixed figures are generic. We are similarly concerned with
the generic situation of deg(g) = deg(I). In this case, eliminants provide a
useful computational device to study further questions about the roots of I.
For instance, the Shape Lemma holds for the saturated ideal of Example 2.1.
Its eliminant, which is the polynomial J_0, factors completely over the ground
field F7, so all four solutions are defined in F7. In Section 4.3, we will use
eliminants in another way, to show that an ideal is radical.

Given a polynomial h in a zero-dimensional ring k[X]/I, the procedure
eliminant(h, k[y]) finds a linear relation modulo I among the powers
1, h, h2, . . . , hd of h with d minimal and returns this as a polynomial in k[y].
This procedure is included in the Macaulay 2 package realroots.m2.

i13 : load "realroots.m2"

i14 : code eliminant

o14 = -- realroots.m2:65-80
eliminant = (h, C) -> (

Z := C_0;
A := ring h;
assert( dim A == 0 );
F := coefficientRing A;
assert( isField F );
assert( F == coefficientRing C );
B := basis A;
d := numgens source B;
M := fold((M, i) -> M ||

substitute(contract(B, h^(i+1)), F),
substitute(contract(B, 1_A), F),
flatten subsets(d, d));

N := ((ker transpose M)).generators;
P := matrix {toList apply(0..d, i -> Z^i)} * N;

(flatten entries(P))_0)

Here, M is a matrix whose rows are the normal forms of the powers 1, h,
h2, . . ., hd of h, for d the degree of the ideal. The columns of the kernel
N of transpose M are a basis of the linear relations among these powers.
The matrix P converts these relations into polynomials. Since N is in column
echelon form, the initial entry of P is the relation of minimal degree. (This
method is often faster than näıvely computing the kernel of the map k[Z]→ A
given by Z 7→ h, which is implemented by eliminantNaive(h, Z).)

Suppose we have an eliminant g(x1) of a zero-dimensional ideal I ⊂ k[X]
with deg(g) = deg(I), and we have computed the lexicographic Gröbner
basis (2). Then the roots of I are

{(ξ1, g2(ξ1), . . . , gn(ξ1)) | g(ξ1) = 0} . (3)

Suppose now that k = Q and we seek floating point approximations for
the (complex) roots of I. Following this method, we first compute floating
point solutions to g(ξ) = 0, which give all the x1-coordinates of the roots of
I, and then use (3) to find the other coordinates. The difficulty here is that
enough precision may be lost in evaluating gi(ξ1) so that the result is a poor
approximation for the other components ξi.

2.3 Solving with Linear Algebra

We describe another method based upon numerical linear algebra. When
I ⊂ k[X] is zero-dimensional, A = k[X]/I is a finite-dimensional k-vector
space, and any Gröbner basis for I gives an efficient algorithm to compute



106 F. Sottile

ring operations using linear algebra. In particular, multiplication by h ∈ A is
a linear transformation mh : A→ A and the command regularRep(h) from
realroots.m2 gives the matrix of mh in terms of the standard basis of A.

i15 : code regularRep

o15 = -- realroots.m2:96-100
regularRep = f -> (

assert( dim ring f == 0 );
b := basis ring f;
k := coefficientRing ring f;
substitute(contract(transpose b, f*b), k))

Since the action of A on itself is faithful, the minimal polynomial of
mh is the eliminant corresponding to h. The procedure charPoly(h, Z) in
realroots.m2 computes the characteristic polynomial det(Z · Id−mh) of h.

i16 : code charPoly

o16 = -- realroots.m2:106-113
charPoly = (h, Z) -> (

A := ring h;
F := coefficientRing A;
S := F[Z];
Z = value Z;
mh := regularRep(h) ** S;
Idz := S_0 * id_(S^(numgens source mh));
det(Idz - mh))

When this is the minimal polynomial (the situation of the Shape Lemma),
this procedure often computes the eliminant faster than does eliminant,
and for systems of moderate degree, much faster than näıvely computing the
kernel of the map k[Z]→ A given by Z 7→ h.

The eigenvalues and eigenvectors of mh give another algorithm for finding
the roots of I. The engine for this is the following result.

Stickelberger’s Theorem. Let h ∈ A and mh be as above. Then there is
a one-to-one correspondence between eigenvectors vξ of mh and roots ξ of I,
the eigenvalue of mh on vξ is the value h(ξ) of h at ξ, and the multiplicity
of this eigenvalue (on the eigenvector vξ) is the multiplicity of the root ξ.

Since the linear transformations mh for h ∈ A commute, the eigenvec-
tors vξ are common to all mh. Thus we may compute the roots of a zero-
dimensional ideal I ⊂ k[X] by first computing floating-point approximations
to the eigenvectors vξ of mx1 . Then the root ξ = (ξ1, . . . , ξn) of I corre-
sponding to the eigenvector vξ has ith coordinate satisfying

mxi · vξ = ξi · vξ . (4)

An advantage of this method is that we may use structured numerical lin-
ear algebra after the matrices mxi are precomputed using exact arithmetic.
(These matrices are typically sparse and have additional structures which
may be exploited.) Also, the coordinates ξi are linear functions of the float-
ing point entries of vξ, which affords greater precision than the non-linear

evaluations gi(ξ1) in the method based upon elimination. While in principle
only one of the deg(I) components of the vectors in (4) need be computed,
averaging the results from all components can improve precision.

2.4 Real Roots

Determining the real roots of a polynomial system is a challenging problem
with real world applications. When the polynomials come from geometry, this
is the main problem of real enumerative geometry. Suppose k ⊂ R and I ⊂
k[X] is zero-dimensional. If g is an eliminant of k[X]/I with deg(g) = deg(I),
then the real roots of g are in 1-1 correspondence with the real roots of I.
Since there are effective methods for counting the real roots of a univariate
polynomial, eliminants give a näıve, but useful method for determining the
number of real roots to a polynomial system. (For some applications of this
technique in mathematics, see [20,23,25].)

The classical symbolic method of Sturm, based upon Sturm sequences,
counts the number of real roots of a univariate polynomial in an interval.
When applied to an eliminant satisfying the Shape Lemma, this method
counts the number of real roots of the ideal. This is implemented in Macau-
lay 2 via the command SturmSequence(f) of realroots.m2

i17 : code SturmSequence

o17 = -- realroots.m2:117-131
SturmSequence = f -> (

assert( isPolynomialRing ring f );
assert( numgens ring f === 1 );
R := ring f;
assert( char R == 0 );
x := R_0;
n := first degree f;
c := new MutableList from toList (0 .. n);
if n >= 0 then (

c#0 = f;
if n >= 1 then (

c#1 = diff(x,f);
scan(2 .. n, i -> c#i = - c#(i-2) % c#(i-1));
));

toList c)

The last few lines of SturmSequence construct the Sturm sequence of the
univariate argument f : This is (f0, f1, f2, . . .) where f0 = f , f1 = f ′, and for
i > 1, fi is the normal form reduction of −fi−2 modulo fi−1. Given any real
number x, the variation of f at x is the number of changes in sign of the
sequence (f0(x), f1(x), f2(x), . . .) obtained by evaluating the Sturm sequence
of f at x. Then the number of real roots of f over an interval [x, y] is the
difference of the variation of f at x and at y.

The Macaulay 2 commands numRealSturm and numPosRoots (and also
numNegRoots) use this method to respectively compute the total number of
real roots and the number of positive roots of a univariate polynomial.

i18 : code numRealSturm

o18 = -- realroots.m2:160-163
numRealSturm = f -> (

c := SturmSequence f;
variations (signAtMinusInfinity \ c)

- variations (signAtInfinity \ c))

i19 : code numPosRoots

o19 = -- realroots.m2:168-171
numPosRoots = f -> (

c := SturmSequence f;
variations (signAtZero \ c)

- variations (signAtInfinity \ c))

These use the commands signAt∗(f), which give the sign of f at ∗. (Here, ∗
is one of Infinity, zero, or MinusInfinity.) Also variations(c) computes
the number of sign changes in the sequence c.

i20 : code variations

o20 = -- realroots.m2:183-191
variations = c -> (

n := 0;
last := 0;
scan(c, x -> if x != 0 then (

if last < 0 and x > 0 or last > 0
and x < 0 then n = n+1;

last = x;
));

n)

A more sophisticated method to compute the number of real roots which
can also give information about their location uses the rank and signature
of the symmetric trace form. Suppose I ⊂ k[X] is a zero-dimensional ideal
and set A := k[X]/I. For h ∈ k[X], set Sh(f, g) := trace(mhfg). It is an
easy exercise that Sh is a symmetric bilinear form on A. The procedure
traceForm(h) in realroots.m2 computes this trace form Sh.

i21 : code traceForm

o21 = -- realroots.m2:196-203
traceForm = h -> (

assert( dim ring h == 0 );
b := basis ring h;
k := coefficientRing ring h;
mm := substitute(contract(transpose b, h * b ** b), k);
tr := matrix {apply(first entries b, x ->

trace regularRep x)};
adjoint(tr * mm, source tr, source tr))

The value of this construction is the following theorem.

Theorem 2.2 ([3,19]). Suppose k ⊂ R and I is a zero-dimensional ideal
in k[x1, . . . , xn] and consider V(I) ⊂ Cn. Then, for h ∈ k[x1, . . . , xn], the
signature σ(Sh) and rank ρ(Sh) of the bilinear form Sh satisfy

σ(Sh) = #{a ∈ V(I) ∩ Rn : h(a) > 0} −#{a ∈ V(I) ∩ Rn : h(a) < 0}
ρ(Sh) = #{a ∈ V(I) : h(a) 6= 0} .

That is, the rank of Sh counts roots in Cn−V(h), and its signature counts
the real roots weighted by the sign of h (which is −1, 0, or 1) at each root.
The command traceFormSignature(h) in realroots.m2 returns the rank
and signature of the trace form Sh.

i22 : code traceFormSignature

o22 = -- realroots.m2:208-218
traceFormSignature = h -> (

A := ring h;
assert( dim A == 0 );
assert( char A == 0 );
S := QQ[Z];
TrF := traceForm(h) ** S;
IdZ := Z * id_(S^(numgens source TrF));
f := det(TrF - IdZ);
<< "The trace form S_h with h = " << h <<

" has rank " << rank(TrF) << " and signature " <<
numPosRoots(f) - numNegRoots(f) << endl; )

The Macaulay 2 command numRealTrace(A) simply returns the number of
real roots of I, given A = k[X]/I.

i23 : code numRealTrace

o23 = -- realroots.m2:223-230
numRealTrace = A -> (

assert( dim A == 0 );
assert( char A == 0 );
S := QQ[Z];
TrF := traceForm(1_A) ** S;
IdZ := Z * id_(S^(numgens source TrF));
f := det(TrF - IdZ);
numPosRoots(f)-numNegRoots(f))

Example 2.3. We illustrate these methods on the following polynomial sys-
tem.

i24 : R = QQ[x, y];

i25 : I = ideal (1 - x^2*y + 2*x*y^2, y - 2*x - x*y + x^2);

o25 : Ideal of R

The ideal I has dimension zero and degree 5.
i26 : dim I, degree I

o26 = (0, 5)

o26 : Sequence

We compare the two methods to compute the eliminant of x in the ring R/I .
i27 : A = R/I;

i28 : time g = eliminant(x, QQ[Z])
-- used 0.09 seconds

5 4 3 2
o28 = Z - 5Z + 6Z + Z - 2Z + 1

o28 : QQ [Z]

i29 : time g = charPoly(x, Z)
-- used 0.02 seconds

5 4 3 2
o29 = Z - 5Z + 6Z + Z - 2Z + 1

o29 : QQ [Z]

The eliminant has 3 real roots, which we test in two different ways.
i30 : numRealSturm(g), numRealTrace(A)

o30 = (3, 3)

o30 : Sequence

We use Theorem 2.2 to isolate these roots in the x, y-plane.
i31 : traceFormSignature(x*y);
The trace form S_h with h = x*y has rank 5 and signature 3

Thus all 3 real roots lie in the first and third quadrants (where xy > 0). We
isolate these further.

i32 : traceFormSignature(x - 2);
The trace form S_h with h = x - 2 has rank 5 and signature 1

This shows that two roots lie in the first quadrant with x > 2 and one lies
in the third. Finally, one of the roots lies in the triangle y > 0, x > 2, and
x+ y < 3.

i33 : traceFormSignature(x + y - 3);
The trace form S_h with h = x + y - 3 has rank 5 and signature -1

Figure 1 shows these three roots (dots), as well as the lines x+ y = 3 and
x = 2.

Fig. 1. Location of roots

2.5 Homotopy Methods

We describe symbolic-numeric homotopy continuation methods for finding
approximate complex solutions to a system of equations. These exploit the
traditional principles of conservation of number and specialization from enu-
merative geometry.

Suppose we seek the isolated solutions of a system F (X) = 0 where
F = (f1, . . . , fn) are polynomials in the variables X = (x1, . . . , xN ). First, a
homotopy H(X, t) is found with the following properties:

1. H(X, 1) = F (X).
2. The isolated solutions of the start system H(X, 0) = 0 are known.
3. The system H(X, t) = 0 defines finitely many (complex) curves, and

each isolated solution of the original system F (X) = 0 is connected to an
isolated solution σi(0) of H(X, 0) = 0 along one of these curves.

Next, choose a generic smooth path γ(t) from 0 to 1 in the complex plane.
Lifting γ to the curves H(X, t) = 0 gives smooth paths σi(t) connecting each
solution σi(0) of the start system to a solution of the original system. The
path γ must avoid the finitely many points in C over which the curves are
singular or meet other components of the solution set H(X, t) = 0.

Numerical path continuation is used to trace each path σi(t) from t = 0
to t = 1. When there are fewer solutions to F (X) = 0 than to H(X, 0) = 0,
some paths will diverge or become singular as t → 1, and it is expensive to
trace such a path. The homotopy is optimal when this does not occur.

When N = n and the fi are generic, set G(X) := (g1, . . . , gn) with gi =
(xi − 1)(xi − 2) · · · (xi − di) where di := deg(fi). Then the Bézout homotopy

H(X, t) := tF (X) + (1− t)G(X)

is optimal. This homotopy furnishes an effective demonstration of the bound
in Bézout’s Theorem for the number of solutions to F (X) = 0.

When the polynomial system is deficient, the Bézout homotopy is not op-
timal. When n > N (often the case in geometric examples), the Bézout ho-
motopy does not apply. In either case, a different strategy is needed. Present
optimal homotopies for such systems all exploit some structure of the systems
they are designed to solve. The current state-of-the-art is described in [29].

Example 2.4. The Gröbner homotopy [14] is an optimal homotopy that
exploits a square-free initial ideal. Suppose our system has the form

F := g1(X), . . . , gm(X), Λ1(X), . . . , Λd(X)

where g1(X), . . . , gm(X) form a Gröbner basis for an ideal I with respect
to a given term order ≺, Λ1, . . . , Λd are linear forms with d = dim(V(I)),
and we assume that the initial ideal in≺I is square-free. This last, restrictive,
hypothesis occurs for certain determinantal varieties.

As in [9, Chapter 15], there exist polynomials gi(X, t) interpolating be-
tween gi(X) and their initial terms in≺gi(X)

gi(X; 1) = gi(X) and gi(X; 0) = in≺gi(X)

so that 〈g1(X, t), . . . , gm(X, t)〉 is a flat family with generic fibre isomorphic
to I and special fibre in≺I. The Gröbner homotopy is

H(X, t) := g1(X, t), . . . , gm(X, t), Λ1(X), . . . , Λd(X).

Since in≺I is square-free, V(in≺I) is a union of deg(I)-many coordinate d-
planes. We solve the start system by linear algebra. This conceptually simple
homotopy is in general not efficient as it is typically overdetermined.

3 Some Enumerative Geometry

We use the tools we have developed to explore the enumerative geometric
problems of cylinders meeting 5 general points and lines tangent to 4 spheres.

3.1 Cylinders Meeting 5 Points

A cylinder is the locus of points equidistant from a fixed line in R3. The
Grassmannian of lines in 3-space is 4-dimensional, which implies that the
space of cylinders is 5-dimensional, and so we expect that 5 points in R3 will
determine finitely many cylinders. That is, there should be finitely many lines
equidistant from 5 general points. The question is: How many cylinders/lines,
and how many of them can be real?

Bottema and Veldkamp [5] show there are 6 complex cylinders and Licht-
blau [17] observes that if the 5 points are the vertices of a bipyramid con-
sisting of 2 regular tetrahedra sharing a common face, then all 6 will be real.
We check this reality on a configuration with less symmetry (so the Shape
Lemma holds).

If the axial line has direction V and contains the point P (and hence has
parameterization P + tV), and if r is the squared radius, then the cylinder
is the set of points X satisfying

0 = r −
∥∥∥∥X−P− V · (X−P)

‖V‖2
V
∥∥∥∥2

.

Expanding and clearing the denominator of ‖V‖2 yields

0 = r‖V‖2 + [V · (X−P)]2 − ‖X−P‖2 ‖V‖2 . (5)

We consider cylinders containing the following 5 points, which form an asym-
metric bipyramid.

i34 : Points = {{2, 2, 0 }, {1, -2, 0}, {-3, 0, 0},
{0, 0, 5/2}, {0, 0, -3}};

Suppose that P = (0, y11, y12) and V = (1, y21, y22).
i35 : R = QQ[r, y11, y12, y21, y22];

i36 : P = matrix{{0, y11, y12}};

1 3
o36 : Matrix R <--- R

i37 : V = matrix{{1, y21, y22}};

1 3
o37 : Matrix R <--- R

We construct the ideal given by evaluating the polynomial (5) at each of the
five points.

i38 : Points = matrix Points ** R;

5 3
o38 : Matrix R <--- R

i39 : I = ideal apply(0..4, i -> (
X := Points^{i};
r * (V * transpose V) +
((X - P) * transpose V)^2) -
((X - P) * transpose(X - P)) * (V * transpose V)

);

o39 : Ideal of R

This ideal has dimension 0 and degree 6.
i40 : dim I, degree I

o40 = (0, 6)

o40 : Sequence

There are 6 real roots, and they correspond to real cylinders (with r > 0).
i41 : A = R/I; numPosRoots(charPoly(r, Z))

o42 = 6

3.2 Lines Tangent to 4 Spheres

We now ask for the lines having a fixed distance from 4 general points. Equiv-
alently, these are the lines mutually tangent to 4 spheres of equal radius.
Since the Grassmannian of lines is four-dimensional, we expect there to be
only finitely many such lines. Macdonald, Pach, and Theobald [18] show that
there are indeed 12 lines, and that all 12 may be real. This problem makes
geometric sense over any field k not of characteristic 2, and the derivation of
the number 12 is also valid for algebraically closed fields not of characteristic
2.

A sphere in k3 is given by V(q(1,x)), where q is some quadratic form on
k4. Here x ∈ k3 and we note that not all quadratic forms give spheres. If our
field does not have characteristic 2, then there is a symmetric 4 × 4 matrix
M such that q(u) = uMut.

A line ` having direction V and containing the point P is tangent to the
sphere defined by q when the univariate polynomial in s

q((1,P) + s(0,V)) = q(1,P) + 2s(1,P)M(0,V)t + s2q(0,V) ,

has a double root. Thus its discriminant vanishes, giving the equation(
(1,P)M(0,V)t

)2 − (1,P)M(1,P)t · (0,V)M(0,V)t = 0 . (6)

The matrix M of the quadratic form q of the sphere with center (a, b, c)
and squared radius r is constructed by Sphere(a,b,c,r).

i43 : Sphere = (a, b, c, r) -> (
matrix{{a^2 + b^2 + c^2 - r ,-a ,-b ,-c },

{ -a , 1 , 0 , 0 },
{ -b , 0 , 1 , 0 },
{ -c , 0 , 0 , 1 }}

);

If a line ` contains the point P = (0, y11, y12) and ` has direction V =
(1, y21, y22), then tangentTo(M) is the equation for ` to be tangent to the
quadric uMuT = 0 determined by the matrix M .

i44 : R = QQ[y11, y12, y21, y22];

i45 : tangentTo = (M) -> (
P := matrix{{1, 0, y11, y12}};
V := matrix{{0, 1, y21, y22}};
(P * M * transpose V)^2 -

(P * M * transpose P) * (V * M * transpose V)
);

The ideal of lines having distance
√

5 from the four points (0, 0, 0), (4, 1, 1),
(1, 4, 1), and (1, 1, 4) has dimension zero and degree 12.

i46 : I = ideal (tangentTo(Sphere(0,0,0,5)),
tangentTo(Sphere(4,1,1,5)),
tangentTo(Sphere(1,4,1,5)),
tangentTo(Sphere(1,1,4,5)));

o46 : Ideal of R

i47 : dim I, degree I

o47 = (0, 12)

o47 : Sequence

Thus there are 12 lines whose distance from those 4 points is
√

5. We check
that all 12 are real.

i48 : A = R/I;

i49 : numRealSturm(eliminant(y11 - y12 + y21 + y22, QQ[Z]))

o49 = 12

Since no eliminant given by a coordinate function satisfies the hypotheses
of the Shape Lemma, we took the eliminant with respect to the linear form
y11 − y12 + y21 + y22.

This example is an instance of Lemma 3 of [18]. These four points define a
regular tetrahedron with volume V = 9 where each face has area A =

√
35/2

and each edge has length e =
√

18. That result guarantees that all 12 lines
will be real when e/2 < r < A2/3V , which is the case above.

4 Schubert Calculus

The classical Schubert calculus of enumerative geometry concerns linear sub-
spaces having specified positions with respect to other, fixed subspaces. For
instance, how many lines in P3 meet four given lines? (See Example 4.2.)

More generally, let 1 < r < n and suppose that we are given general
linear subspaces L1, . . . , Lm of kn with dimLi = n − r + 1 − li. When
l1 + · · ·+ lm = r(n− r), there will be a finite number d(r, n; l1, . . . , lm) of r-
planes in kn which meet each Li non-trivially. This number may be computed
using classical algorithms of Schubert and Pieri (see [16]).

The condition on r-planes to meet a fixed (n−r+1−l)-plane non-trivially
is called a (special) Schubert condition, and we call the data (r, n; l1, . . . , lm)
(special) Schubert data. The (special) Schubert calculus concerns this class of
enumerative problems. We give two polynomial formulations of this special
Schubert calculus, consider their solutions over R, and end with a question
for fields of arbitrary characteristic.

4.1 Equations for the Grassmannian

The ambient space for the Schubert calculus is the Grassmannian of r-planes
in kn, denoted Gr,n. For H ∈ Gr,n, the rth exterior product of the embedding
H → kn gives a line

k ' ∧rH −→ ∧rkn ' k(nr) .

This induces the Plücker embedding Gr,n ↪→ P
(nr)−1. If H is the row space

of an r by n matrix, also written H, then the Plücker embedding sends H
to its vector of

(
n
r

)
maximal minors. Thus the r-subsets of {0, . . . , n−1},

Yr,n := subsets(n, r), index Plücker coordinates of Gr,n. The Plücker ideal
of Gr,n is therefore the ideal of algebraic relations among the maximal minors
of a generic r by n matrix.

We create the coordinate ring k[pα | α ∈ Y2,5] of P9 and the Plücker ideal
of G2,5. The Grassmannian Gr,n of r-dimensional subspaces of kn is also the
Grassmannian of r−1-dimensional affine subspaces of Pn−1. Macaulay 2 uses
this alternative indexing scheme.

i50 : R = ZZ/101[apply(subsets(5,2), i -> p_i )];

i51 : I = Grassmannian(1, 4, R)

o51 = ideal (p p - p p + p p , p · · ·
{2, 3} {1, 4} {1, 3} {2, 4} {1, 2} {3, 4} {2, 3} · · ·

o51 : Ideal of R

This projective variety has dimension 6 and degree 5
i52 : dim(Proj(R/I)), degree(I)

o52 = (6, 5)

o52 : Sequence

This ideal has an important combinatorial structure [28, Example 11.9].
We write each α ∈ Yr,n as an increasing sequence α : α1 < · · · < αr. Given
α, β ∈ Yr,n, consider the two-rowed array with α written above β. We say
α ≤ β if each column weakly increases. If we sort the columns of an array

with rows α and β, then the first row is the meet α∧β (greatest lower bound)
and the second row the join α ∨ β (least upper bound) of α and β. These
definitions endow Yr,n with the structure of a distributive lattice. Figure 2
shows Y2,5.

Fig. 2. Y2,5

We give k[pα] the degree reverse lexicographic order, where we first order
the variables pα by lexicographic order on their indices α.

Theorem 4.1. The reduced Gröbner basis of the Plücker ideal with respect to
this degree reverse lexicographic term order consists of quadratic polynomials

g(α, β) = pα · pβ − pα∨β · pα∧β + lower terms in ≺ ,

for each incomparable pair α, β in Yr,n, and all lower terms λpγ ·pδ in g(α, β)
satisfy γ ≤ α ∧ β and α ∨ β ≤ δ.

The form of this Gröbner basis implies that the standard monomials are
the sortable monomials, those pαpβ · · · pγ with α ≤ β ≤ · · · ≤ γ. Thus the
Hilbert function of Gr,n may be expressed in terms of the combinatorics of
Yr,n. For instance, the dimension of Gr,n is the rank of Yr,n, and its degree
is the number of maximal chains. From Figure 2, these are 6 and 5 for Y2,5,
confirming our previous calculations.

Since the generators g(α, β) are linearly independent, this Gröbner basis
is also a minimal generating set for the ideal. The displayed generator in o51,

p{2,3}p{1,4} − p{1,3}p{2,4} − p{1,2}p{3,4} ,

is g(23, 14), and corresponds to the underlined incomparable pair in Figure 2.
Since there are 5 such incomparable pairs, the Gröbner basis has 5 generators.
As G2,5 has codimension 3, it is not a complete intersection. This shows how

the general enumerative problem from the Schubert calculus gives rise to an
overdetermined system of equations in this global formulation.

The Grassmannian has a useful system of local coordinates given by
Matr,n−r as follows

Y ∈ Matr,n−r 7−→ rowspace [Ir : Y ] ∈ Gr,n . (7)

Let L be a (n−r+1−l)-plane in kn which is the row space of a n−r+1−l
by n matrix, also written L. Then L meets X ∈ Gr,n non-trivially if

maximal minors of
[
L
X

]
= 0 .

Laplace expansion of each minor along the rows of X gives a linear equation
in the Plücker coordinates. In the local coordinates (substituting [Ir : Y ] for
X), we obtain multilinear equations of degree min{r, n− r}. These equations
generate a prime ideal of codimension l.

Suppose each li = 1 in our enumerative problem. Then in the Plücker
coordinates, we have the Plücker ideal of Gr,n together with r(n− r) linear
equations, one for each (n−r)-plane Li. By Theorem 4.1, the Plücker ideal
has a square-free initial ideal, and so the Gröbner homotopy of Example 2.4
may be used to solve this enumerative problem.

Example 4.2. G2,4 ⊂ P5 has equation

p{1,2}p{0,3} − p{1,3}p{0,2} + p{2,3}p{0,1} = 0 . (8)

The condition for H ∈ G2,4 to meet a 2-plane L is the vanishing of

p{1,2}L34 − p{1,3}L24 + p{2,3}L14 + p{1,4}L23 − p{2,4}L13 + p{3,4}L12 , (9)

where Lij is the (i, j)th maximal minor of L.
If l1 = · · · = l4 = 1, we have 5 equations in P5, one quadratic and 4

linear, and so by Bézout’s Theorem there are two 2-planes in k4 that meet 4
general 2-planes non-trivially. This means that there are 2 lines in P3 meeting
4 general lines. In local coordinates, (9) becomes

L34 − L14y11 + L13y12 − L24y21 + L23y22 + L12(y11y22 − y12y21) .

This polynomial has the form of the last specialization in Example 1.2.

4.2 Reality in the Schubert Calculus

Like the other enumerative problems we have discussed, enumerative prob-
lems in the special Schubert calculus are fully real in that all solutions can
be real [22]. That is, given any Schubert data (r, n; l1, . . . , lm), there exist

subspaces L1, . . . , Lm ⊂ Rn such that each of the d(r, n; l1, . . . , lm) r-planes
that meet each Li are themselves real.

This result gives some idea of which choices of the Li give all r-planes
real. Let γ be a fixed rational normal curve in Rn. Then the Li are lin-
ear subspaces osculating γ. More concretely, suppose that γ is the stan-
dard rational normal curve, γ(s) = (1, s, s2, . . . , sn−1). Then the i-plane
Li(s) := 〈γ(s), γ′(s), . . . , γ(i−1)(s)〉 osculating γ at γ(s) is the row space of
the matrix given by oscPlane(i, n, s).

i53 : oscPlane = (i, n, s) -> (
gamma := matrix {toList apply(1..n, i -> s^(i-1))};
L := gamma;
j := 0;
while j < i-1 do (gamma = diff(s, gamma);

L = L || gamma;
j = j+1);

L);

i54 : QQ[s]; oscPlane(3, 6, s)

o55 = | 1 s s2 s3 s4 s5 |
| 0 1 2s 3s2 4s3 5s4 |
| 0 0 2 6s 12s2 20s3 |

3 6
o55 : Matrix QQ [s] <--- QQ [s]

(In o55, the exponents of s are displayed in line: s2 is written s2. Macaulay 2
uses this notational convention to display matrices efficiently.)

Theorem 4.3 ([22]). For any Schubert data (r, n; l1, . . . , lm), there exist
real numbers s1, s2, . . . , sm such that there are d(r, n; l1, . . . , lm) r-planes that
meet each osculating plane Li(si), and all are real.

The inspiration for looking at subspaces osculating the rational normal
curve to study real enumerative geometry for the Schubert calculus is the
following very interesting conjecture of Boris Shapiro and Michael Shapiro,
or more accurately, extensive computer experimentation based upon their
conjecture [20,23,25,30].

Shapiros’s Conjecture. For any Schubert data (r, n; l1, . . . , lm) and for
all real numbers s1, s2, . . . , sm there are d(r, n; l1, . . . , lm) r-planes that meet
each osculating plane Li(si), and all are real.

In addition to Theorem 4.3, (which replaces the quantifier for all by there
exist), the strongest evidence for this Conjecture is the following result of
Eremenko and Gabrielov [10].

Theorem 4.4. Shapiros’s Conjecture is true when either r or n− r is 2.

We test an example of this conjecture for the Schubert data (3, 6; 13, 23),
(where ab is a repeated b times). The algorithms of the Schubert calculus
predict that d(3, 6; 13, 23) = 6. The function spSchub(r, L, P) computes
the ideal of r-planes meeting the row space of L in the Plücker coordinates
Pα.

i56 : spSchub = (r, L, P) -> (
I := ideal apply(subsets(numgens source L,

r + numgens target L), S ->
fold((sum, U) -> sum +
fold((term,i) -> term*(-1)^i, P_(S_U) * det(
submatrix(L, sort toList(set(S) - set(S_U)))), U),

0, subsets(#S, r))));

We are working in the Grassmannian of 3-planes in C6.
i57 : R = QQ[apply(subsets(6,3), i -> p_i )];

The ideal I consists of the special Schubert conditions for the 3-planes to
meet the 3-planes osculating the rational normal curve at the points 1, 2,
and 3, and to also meet the 2-planes osculating at 4, 5, and 6, together
with the Plücker ideal Grassmannian(2, 5, R). Since this is a 1-dimensional
homogeneous ideal, we add the linear form p_{0,1,5} - 1 to make the ideal
zero-dimensional. As before, Grassmannian(2, 5, R) creates the Plücker
ideal of G3,6.

i58 : I = fold((J, i) -> J +
spSchub(3, substitute(oscPlane(3, 6, s), {s=> 1+i}), p) +
spSchub(3, substitute(oscPlane(2, 6, s), {s=> 4+i}), p),
Grassmannian(2, 5, R), {0,1,2}) +

ideal (p_{0,1,5} - 1);

o58 : Ideal of R

This has dimension 0 and degree 6, in agreement with the Schubert calculus.
i59 : dim I, degree I

o59 = (0, 6)

o59 : Sequence

As expected, all roots are real.
i60 : A = R/I; numRealSturm(eliminant(p_{2,3,4}, QQ[Z]))

o61 = 6

There have been many checked instances of this conjecture [23,25,30], and it
has some geometrically interesting generalizations [26].

The question remains for which numbers 0 ≤ d ≤ d(r, n; l1, . . . , lm) do
there exist real planes Li with d(r, n; l1, . . . , lm) r-planes meeting each Li,
and exactly d of them are real. Besides Theorem 4.3 and the obvious parity
condition, nothing is known in general. In every known case, every possibility
occurs—which is not the case in all enumerative problems, even those that
are fully real1. Settling this (for d = 0) has implications for linear systems
theory [20].2

1 For example, of the 12 rational plane cubics containing 8 real points in P2, either
8, 10 or 12 can be real, and there are 8 points with all 12 real [8, Proposition
4.7.3].

2 After this was written, Eremenko and Gabrielov [11] showed that d can be zero
for the enumerative problems given by data (2, 2n, 14n−4) and (2n−2, 2n, 14n−4).

4.3 Transversality in the Schubert Calculus

A basic principle of the classical Schubert calculus is that the intersection
number d(r, n; l1, . . . , lm) has enumerative significance—that is, for general
linear subspaces Li, all solutions appear with multiplicity 1. This basic princi-
ple is not known to hold in general. For fields of characteristic zero, Kleiman’s
Transversality Theorem [15] establishes this principle. When r or n−r is 2,
then Theorem E of [21] establishes this principle in arbitrary characteristic.
We conjecture that this principle holds in general; that is, for arbitrary in-
finite fields and any Schubert data, if the planes Li are in general position,
then the resulting zero-dimensional ideal is radical.

We test this conjecture on the enumerative problem of Section 4.2, which
is not covered by Theorem E of [21]. The function testTransverse(F) tests
transversality for this enumerative problem, for a given field F . It does this
by first computing the ideal of the enumerative problem using random planes
Li.

i62 : randL = (R, n, r, l) ->
matrix table(n-r+1-l, n, (i, j) -> random(0, R));

and the Plücker ideal of the Grassmannian G3,6 Grassmannian(2, 5, R).)
Then it adds a random (inhomogeneous) linear relation 1 + random(1, R)
to make the ideal zero-dimensional for generic Li. When this ideal is zero
dimensional and has degree 6 (the expected degree), it computes the char-
acteristic polynomial g of a generic linear form. If g has no multiple roots, 1
== gcd(g, diff(Z, g)), then the Shape Lemma guarantees that the ideal
was radical. testTransverse exits either when it computes a radical ideal,
or after limit iterations (which is set to 5 for these examples), and prints
the return status.

i63 : testTransverse = F -> (
R := F[apply(subsets(6, 3), i -> q_i )];
continue := true;
j := 0;
limit := 5;
while continue and (j < limit) do (

j = j + 1;
I := fold((J, i) -> J +

spSchub(3, randL(R, 6, 3, 1), q) +
spSchub(3, randL(R, 6, 3, 2), q),
Grassmannian(2, 5, R) +
ideal (1 + random(1, R)),
{0, 1, 2});

if (dim I == 0) and (degree I == 6) then (
lin := promote(random(1, R), (R/I));
g := charPoly(lin, Z);
continue = not(1 == gcd(g, diff(Z, g)));
));

if continue then << "Failed for the prime " << char F <<
" with " << j << " iterations" << endl;

if not continue then << "Succeeded for the prime " <<
char F << " in " << j << " iteration(s)" << endl;

);

Since 5 iterations do not show transversality for F2,
i64 : testTransverse(ZZ/2);
Failed for the prime 2 with 5 iterations

we can test transversality in characteristic 2 using the field with four elements,
F4 = GF 4.

i65 : testTransverse(GF 4);
Succeeded for the prime 2 in 3 iteration(s)

We do find transversality for F7.
i66 : testTransverse(ZZ/7);
Succeeded for the prime 7 in 2 iteration(s)

We have tested transversality for all primes less than 100 in every enumer-
ative problem involving Schubert conditions on 3-planes in k6. These include
the problem above as well as the problem of 42 3-planes meeting 9 general
3-planes.3

5 The 12 Lines: Reprise

The enumerative problems of Section 3 were formulated in local coordi-
nates (7) for the Grassmannian of lines in P3 (Grassmannian of 2-dimensional
subspaces in k4). When we formulate the problem of Section 3.2 in the global
Plücker coordinates of Section 4.1, we find some interesting phenomena. We
also consider some related enumerative problems.

5.1 Global Formulation

A quadratic form q on a vector space V over a field k not of characteristic 2
is given by q(u) = (ϕ(u),u), where ϕ : V → V ∗ is a symmetric linear map,
that is (ϕ(u),v) = (ϕ(v),u). Here, V ∗ is the linear dual of V and ( · , · ) is
the pairing V ⊗V ∗ → k. The map ϕ induces a quadratic form ∧rq on the rth
exterior power ∧rV of V through the symmetric map ∧rϕ : ∧r V → ∧rV ∗ =
(∧rV )∗. The action of ∧rV ∗ on ∧rV is given by

(x1 ∧ x2 ∧ · · · ∧ xr, y1 ∧ y2 ∧ · · · ∧ yr) = det |(xi,yj)| , (10)

where xi ∈ V ∗ and yj ∈ V .
When we fix isomorphisms V ' kn ' V ∗, the map ϕ is given by a

symmetric n × n matrix M as in Section 3.2. Suppose r = 2. Then for
u,v ∈ kn,

∧2q(u ∧ v) = det
[

uMut uMvt

vMut vMvt

]
,

which is Equation (6) of Section 3.2.
3 After this was written, we discovered an elementary proof of transversality for

the enumerative problems given by data (r, n; 1r(n−r)), where the conditions are
all codimension 1 [24].

Proposition 5.1. A line ` is tangent to a quadric V(q) in Pn−1 if and only
if its Plücker coordinate ∧2` ∈ P(n2)−1 lies on the quadric V(∧2q).

Thus the Plücker coordinates for the set of lines tangent to 4 general
quadrics in P3 satisfy 5 quadratic equations: The single Plücker relation (8)
together with one quadratic equation for each quadric. Thus we expect the
Bézout number of 25 = 32 such lines. We check this.

The procedure randomSymmetricMatrix(R, n) generates a random sym-
metric n× n matrix with entries in the base ring of R.

i67 : randomSymmetricMatrix = (R, n) -> (
entries := new MutableHashTable;
scan(0..n-1, i -> scan(i..n-1, j ->

entries#(i, j) = random(0, R)));
matrix table(n, n, (i, j) -> if i > j then

entries#(j, i) else entries#(i, j))
);

The procedure tangentEquation(r, R, M) gives the equation in Plücker
coordinates for a point in P(nr)−1 to be isotropic with respect to the bilinear
form ∧rM (R is assumed to be the coordinate ring of P(nr)−1). This is the
equation for an r-plane to be tangent to the quadric associated to M .

i68 : tangentEquation = (r, R, M) -> (
g := matrix {gens(R)};
(entries(g * exteriorPower(r, M) * transpose g))_0_0
);

We construct the ideal of lines tangent to 4 general quadrics in P3.
i69 : R = QQ[apply(subsets(4, 2), i -> p_i )];

i70 : I = Grassmannian(1, 3, R) + ideal apply(0..3, i ->
tangentEquation(2, R, randomSymmetricMatrix(R, 4)));

o70 : Ideal of R

As expected, this ideal has dimension 0 and degree 32.
i71 : dim Proj(R/I), degree I

o71 = (0, 32)

o71 : Sequence

5.2 Lines Tangent to 4 Spheres

That calculation raises the following question: In Section 3.2, why did we
obtain only 12 lines tangent to 4 spheres? To investigate this, we generate
the global ideal of lines tangent to the spheres of Section 3.2.

i72 : I = Grassmannian(1, 3, R) +
ideal (tangentEquation(2, R, Sphere(0,0,0,5)),

tangentEquation(2, R, Sphere(4,1,1,5)),
tangentEquation(2, R, Sphere(1,4,1,5)),
tangentEquation(2, R, Sphere(1,1,4,5)));

o72 : Ideal of R

We compute the dimension and degree of V(I).
i73 : dim Proj(R/I), degree I

o73 = (1, 4)

o73 : Sequence

The ideal is not zero dimensional; there is an extraneous one-dimensional
component of zeroes with degree 4. Since we found 12 lines in Section 3.2
using the local coordinates (7), the extraneous component must lie in the
complement of that coordinate patch, which is defined by the vanishing of
the first Plücker coordinate, p{0,1}. We saturate I by p{0,1} to obtain the
desired lines.

i74 : Lines = saturate(I, ideal (p_{0,1}));

o74 : Ideal of R

This ideal does have dimension 0 and degree 12, so we have recovered the
zeroes of Section 3.2.

i75 : dim Proj(R/Lines), degree(Lines)

o75 = (0, 12)

o75 : Sequence

We investigate the rest of the zeroes, which we obtain by taking the ideal
quotient of I and the ideal of lines. As computed above, this has dimension
1 and degree 4.

i76 : Junk = I : Lines;

o76 : Ideal of R

i77 : dim Proj(R/Junk), degree Junk

o77 = (1, 4)

o77 : Sequence

We find the support of this extraneous component by taking its radical.
i78 : radical(Junk)

2 2 2
o78 = ideal (p , p , p , p + p + p )

{0, 3} {0, 2} {0, 1} {1, 2} {1, 3} {2, 3}

o78 : Ideal of R

From this, we see that the extraneous component is supported on an imagi-
nary conic in the P2 of lines at infinity.

To understand the geometry behind this computation, observe that the
sphere with radius r and center (a, b, c) has homogeneous equation

(x− wa)2 + (y − wb)2 + (z − wc)2 = r2w2 .

At infinity, w = 0, this has equation

x2 + y2 + z2 = 0 .

The extraneous component is supported on the set of tangent lines to this
imaginary conic. Aluffi and Fulton [1] studied this problem, using geometry
to identify the extraneous ideal and the excess intersection formula [13] to
obtain the answer of 12. Their techniques show that there will be 12 isolated
lines tangent to 4 quadrics which have a smooth conic in common.

When the quadrics are spheres, the conic is the imaginary conic at infinity.
Fulton asked the following question: Can all 12 lines be real if the (real) four
quadrics share a real conic? We answer his question in the affirmative in the
next section.

5.3 Lines Tangent to Real Quadrics Sharing a Real Conic

We consider four quadrics in P3
R

sharing a non-singular conic, which we will
take to be at infinity so that we may use local coordinates for G2,4 in our
computations. The variety V(q) ⊂ P3

R
of a nondegenerate quadratic form q

is determined up to isomorphism by the absolute value of the signature σ of
the associated bilinear form. Thus there are three possibilities, 0, 2, or 4, for
|σ|.

When |σ| = 4, the real quadric V(q) is empty. The associated symmetric
matrix M is conjugate to the identity matrix, so ∧2M is also conjugate to
the identity matrix. Hence V(∧2q) contains no real points. Thus we need not
consider quadrics with |σ| = 4.

When |σ| = 2, we have V(q) ' S2, the 2-sphere. If the conic at infinity
is imaginary, then V(q) ⊂ R3 is an ellipsoid. If the conic at infinity is real,
then V(q) ⊂ R3 is a hyperboloid of two sheets. When σ = 0, we have V(q) '
S1 × S1, a torus. In this case, V(q) ⊂ R3 is a hyperboloid of one sheet and
the conic at infinity is real.

Thus either we have 4 ellipsoids sharing an imaginary conic at infinity,
which we studied in Section 3.2; or else we have four hyperboloids sharing a
real conic at infinity, and there are five possible combinations of hyperboloids
of one or two sheets sharing a real conic at infinity. This gives six topologically
distinct possibilities in all.

Theorem 5.2. For each of the six topologically distinct possibilities of four
real quadrics sharing a smooth conic at infinity, there exist four quadrics
having the property that each of the 12 lines in C3 simultaneously tangent to
the four quadrics is real.

Proof. By the computation in Section 3.2, we need only check the five possi-
bilities for hyperboloids. We fix the conic at infinity to be x2 + y2 − z2 = 0.
Then the general hyperboloid of two sheets containing this conic has equation
in R3

(x− a)2 + (y − b)2 − (z − c)2 + r = 0 , (11)

(with r > 0). The command Two(a,b,c,r) generates the associated sym-
metric matrix.

i79 : Two = (a, b, c, r) -> (
matrix{{a^2 + b^2 - c^2 + r ,-a ,-b , c },

{ -a , 1 , 0 , 0 },
{ -b , 0 , 1 , 0 },
{ c , 0 , 0 ,-1 }}

);

The general hyperboloid of one sheet containing the conic x2 + y2 − z2 = 0
at infinity has equation in R3

(x− a)2 + (y − b)2 − (z − c)2 − r = 0 , (12)

(with r > 0). The command One(a,b,c,r) generates the associated sym-
metric matrix.

i80 : One = (a, b, c, r) -> (
matrix{{a^2 + b^2 - c^2 - r ,-a ,-b , c },

{ -a , 1 , 0 , 0 },
{ -b , 0 , 1 , 0 },
{ c , 0 , 0 ,-1 }}

);

We consider i quadrics of two sheets (11) and 4 − i quadrics of one
sheet (12). For each of these cases, the table below displays four 4-tuples
of data (a, b, c, r) which give 12 real lines. (The data for the hyperboloids of
one sheet are listed first.)

i Data
0 (5, 3, 3, 16), (5,−4, 2, 1), (−3,−1, 1, 1), (2,−7, 0, 1)
1 (3,−2,−3, 6), (−3,−7,−6, 7), (−6, 3,−5, 2), (1, 6,−2, 5)
2 (6, 4, 6, 4), (−1, 3, 3, 6), (−7,−2, 3, 3), (−6, 7,−2, 5)
3 (−1,−4,−1, 1), (−3, 3,−1, 1), (−7, 6, 2, 9), (5, 6,−1, 12)
4 (5, 2,−1, 25), (6,−6, 2, 25), (−7, 1, 6, 1), (3, 1, 0, 1)

We test each of these, using the formulation in local coordinates of Sec-
tion 3.2.

i81 : R = QQ[y11, y12, y21, y22];

i82 : I = ideal (tangentTo(One( 5, 3, 3,16)),
tangentTo(One( 5,-4, 2, 1)),
tangentTo(One(-3,-1, 1, 1)),
tangentTo(One( 2,-7, 0, 1)));

o82 : Ideal of R

i83 : numRealSturm(charPoly(promote(y22, R/I), Z))

o83 = 12

i84 : I = ideal (tangentTo(One( 3,-2,-3, 6)),
tangentTo(One(-3,-7,-6, 7)),
tangentTo(One(-6, 3,-5, 2)),
tangentTo(Two( 1, 6,-2, 5)));

o84 : Ideal of R

i85 : numRealSturm(charPoly(promote(y22, R/I), Z))

o85 = 12

i86 : I = ideal (tangentTo(One( 6, 4, 6, 4)),
tangentTo(One(-1, 3, 3, 6)),
tangentTo(Two(-7,-2, 3, 3)),
tangentTo(Two(-6, 7,-2, 5)));

o86 : Ideal of R

i87 : numRealSturm(charPoly(promote(y22, R/I), Z))

o87 = 12

i88 : I = ideal (tangentTo(One(-1,-4,-1, 1)),
tangentTo(Two(-3, 3,-1, 1)),
tangentTo(Two(-7, 6, 2, 9)),
tangentTo(Two( 5, 6,-1,12)));

o88 : Ideal of R

i89 : numRealSturm(charPoly(promote(y22, R/I), Z))

o89 = 12

i90 : I = ideal (tangentTo(Two( 5, 2,-1,25)),
tangentTo(Two( 6,-6, 2,25)),
tangentTo(Two(-7, 1, 6, 1)),
tangentTo(Two( 3, 1, 0, 1)));

o90 : Ideal of R

i91 : numRealSturm(charPoly(promote(y22, R/I), Z))

o91 = 12

ut

In each of these enumerative problems there are 12 complex solutions. For
each, we have done other computations showing that every possible number
of real solutions (0, 2, 4, 6, 8, 10, or 12) can occur.

5.4 Generalization to Higher Dimensions

We consider lines tangent to quadrics in higher dimensions. First, we rein-
terpret the action of ∧rV ∗ on ∧rV described in (10) as follows. The vec-
tors x1, . . . ,xr and y1, . . . ,yr define maps α : kr → V ∗ and β : kr → V .
The matrix [(xi, yj)] is the matrix of the bilinear form on kr given by
〈u, v〉 := (α(u), β(v)). Thus (10) vanishes when the bilinear form 〈 · , · 〉
on kr is degenerate.

Now suppose that we have a quadratic form q on V given by a symmetric
map ϕ : V → V ∗. This induces a quadratic form and hence a quadric on
any r-plane H in V (with H 6⊂ V(q)). This induced quadric is singular when
H is tangent to V(q). Since a quadratic form is degenerate only when the
associated projective quadric is singular, we see that H is tangent to the
quadric V(q) if and only if (∧rϕ(∧rH), ∧rH) = 0. (This includes the case
H ⊂ V(q).) We summarize this argument.

Theorem 5.3. Let ϕ : V → V ∗ be a linear map with resulting bilinear form
(ϕ(u), v). Then the locus of r-planes in V for which the restriction of this
form is degenerate is the set of r-planes H whose Plücker coordinates are
isotropic, (∧rϕ(∧rH), ∧rH) = 0, with respect to the induced form on ∧rV .

When ϕ is symmetric, this is the locus of r-planes tangent to the associated
quadric in P(V ).

We explore the problem of lines tangent to quadrics in Pn. From the
calculations of Section 5.1, we do not expect this to be interesting if the
quadrics are general. (This is borne out for P4: we find 320 lines in P4 tangent
to 6 general quadrics. This is the Bézout number, as deg G2,5 = 5 and the
condition to be tangent to a quadric has degree 2.) This problem is interesting
if the quadrics in Pn share a quadric in a Pn−1. We propose studying such
enumerative problems, both determining the number of solutions for general
such quadrics, and investigating whether or not it is possible to have all
solutions be real.

We use Macaulay 2 to compute the expected number of solutions to this
problem when r = 2 and n = 4. We first define some functions for this
computation, which will involve counting the degree of the ideal of lines
in P4 tangent to 6 general spheres. Here, X gives local coordinates for the
Grassmannian, M is a symmetric matrix, tanQuad gives the equation in X
for the lines tangent to the quadric given by M .

i92 : tanQuad = (M, X) -> (
u := X^{0};
v := X^{1};
(u * M * transpose v)^2 -
(u * M * transpose u) * (v * M * transpose v)
);

nSphere gives the matrix M for a sphere with center V and squared radius
r, and V and r give random data for a sphere.

i93 : nSphere = (V, r) ->
(matrix {{r + V * transpose V}} || transpose V ) |
( V || id_((ring r)^n)
);

i94 : V = () -> matrix table(1, n, (i,j) -> random(0, R));

i95 : r = () -> random(0, R);

We construct the ambient ring, local coordinates, and the ideal of the enu-
merative problem of lines in P4 tangent to 6 random spheres.

i96 : n = 4;

i97 : R = ZZ/1009[flatten(table(2, n-1, (i,j) -> z_(i,j)))];

i98 : X = 1 | matrix table(2, n-1, (i,j) -> z_(i,j))

o98 = | 1 0 z_(0,0) z_(0,1) z_(0,2) |
| 0 1 z_(1,0) z_(1,1) z_(1,2) |

2 5
o98 : Matrix R <--- R

i99 : I = ideal (apply(1..(2*n-2),
i -> tanQuad(nSphere(V(), r()), X)));

o99 : Ideal of R

We find there are 24 lines in P4 tangent to 6 general spheres.
i100 : dim I, degree I

o100 = (0, 24)

o100 : Sequence

The expected numbers of solutions we have obtained in this way are displayed
in the table below. The numbers in boldface are those which are proven.4

n 2 3 4 5 6
# expected 4 12 24 48 96

Acknowledgments. We thank Dan Grayson and Bernd Sturmfels: some of the
procedures in this chapter were written by Dan Grayson and the calculation in
Section 5.2 is due to Bernd Sturmfels.

Resolutions and Cohomology
over Complete Intersections

Luchezar L. Avramov and Daniel R. Grayson?

This chapter contains a new proof and new applications of a theorem of
Shamash and Eisenbud, providing a construction of projective resolutions of
modules over a complete intersection. The duals of these infinite projective
resolutions are finitely generated differential graded modules over a graded
polynomial ring, so they can be represented in the computer, and can be
used to compute Ext modules simultaneously in all homological degrees. It
is shown how to write Macaulay 2 code to implement the construction, and
how to use the computer to determine invariants of modules over complete
intersections that are difficult to obtain otherwise.

Introduction

Let A = K[x1, . . . , xe] be a polynomial ring with variables of positive degree
over a field K, and B = A/J a quotient ring modulo a homogeneous ideal.

In this paper we consider the case when B is a graded complete inter-
section, that is, when the defining ideal J is generated by a homogeneous
A-regular sequence. We set up, describe, and illustrate a routine Ext, now
implemented in Macaulay 2. For any two finitely generated graded B-modules
M and N it yields a presentation of Ext•B(M,N) as a bigraded module over
an appropriately bigraded polynomial ring S = A[X1, . . . , Xc].

A novel feature of our routine is that it computes the modules ExtnB(M,N)
simultaneously in all cohomological degrees n ≥ 0. This is made possible by
the use of cohomology operations, a technique usually confined to theoretical
considerations. Another aspect worth noticing is that, although the result is
over a ring B with nontrivial relations, all the computations are made over
the polynomial ring S; this may account for the effectiveness of the algorithm.

To explain the role of the complete intersection hypothesis, we cast it into
the broader context of homological algebra over graded rings.

Numerous results indicate that the high syzygy modules of M exhibit
‘similar’ properties. For an outrageous example, assume that M has finite
projective dimension. Its distant syzygies are then all equal to 0, and so—for
trivial reasons—display an extremely uniform behavior. However, even this
case has a highly nontrivial aspect: due to the Auslander-Buchsbaum Equality
asymptotic information is available after at most (e+ 1) steps. This accounts
for the effectiveness of computer constructions of finite free resolutions.

? Authors supported by the NSF, grants DMS 99-70375 and DMS 99-70085.

Problems that computers are not well equipped to handle arise unavoid-
ably when studying asymptotic behavior of infinite resolutions. We describe
some, using graded Betti numbers βBns(M) = dimK ExtnB(M,k)−s, where
k = B/(x1, . . . , xe)B, and regularity regB(M) = supn,s{s− n |βBns(M) 6= 0}.

• Irrationality . There are rings B for which no recurrent relation with con-
stant coefficients exists among the numbers βBn (k) =

∑
s β

B
ns(k), see [1].

• Irregularity . For each r ≥ 2 there exists a ring B(r) with β
B(r)
ns (k) = 0

for s 6= n and 0 ≤ n ≤ r, but with β
B(r)
r,r+1(k) 6= 0, see [14].

• Span. If B is generated over K by elements of degree one and regB(k) 6= 0,
then regB(k) =∞, see [7].

• Size. There are inequalities βBn (k) ≥ βn for all n ≥ 0 and for some
constant β > 1, unless B is a complete intersection, see [3].

These obstructions vanish miraculously when B is a graded complete in-
tersection: For each M and all n� 0 the number βn+1(M) is a linear combi-
nation with constant coefficients of βBn−2c(M), . . . , βBn (M). If B is generated
in degree one, then regB(k) = 0 if and only if the ideal J is generated by
quadratic forms. There are inequalities βBn (M) ≤ β(M)nc−1 for all n ≥ 1
and for some constant β(M) > 0.

The algebra behind the miracle is a theorem of Gulliksen [12], who proves
that Ext•B(M,N) is a finitely generated bigraded module over a polynomial
ring of cohomology operators S = A[X1, . . . , Xc], where each variable Xi

has cohomological degree 2. As a consequence of this result, problems in
Homological Algebra can be answered in terms of Commutative Algebra.

Gulliksen’s definition of the operators Xi as iterated connecting homo-
morphisms is badly suited for use by a computer. Other definitions have
been given subsequently by several authors, see Remark 4.6. We take the
approach of Eisenbud [11], who derives the operators from a specific B-free
resolution of M , obtained by extending a construction of Shamash [15].

The resolution of Shamash and Eisenbud, and Gulliksen’s Finiteness The-
orem, are presented with detailed proofs in Section 4. They are obtained
through a new construction—that of an intermediate resolution of M over
the polynomial ring—that encodes C and all the null-homotopies of C corre-
sponding to multiplication with elements of J ; this material is contained in
Section 3. It needs standard multilinear algebra, developed ad hoc in Section
2. Rules for juggling several gradings are discussed in an Appendix.

In Section 5 we present and illustrate the code for the routine Ext, which
runs remarkably close to the proofs in Sections 3 and 4. Section 6 contains
numerous computations of popular numerical invariants of a graded mod-
ule, like its complexity, Poincaré series, and Bass series. They are extracted
from knowledge of the bigraded modules Ext•B(M,k) and Ext•B(k,M), whose
computation is also illustrated by examples, and is further used to obtain ex-
plicit equations for the cohomology variety V∗B(M) defined in [2]. For most
invariants we include some short code that automates their computation. In
Section 7 we extend these procedures to invariants of pairs of modules.

1 Matrix Factorizations

We start the discussion of homological algebra over a complete intersection
with a very special case, that can be packaged attractively in matrix terms.

Let f be a non-zero-divisor in a commutative ring A.
Following Eisenbud [11, Sect. 5] we say that a pair (U, V ) of matrices with

entries in A, of sizes k × ` and `× k, is a matrix factorization of −f if

U · V = −f · Ik and V · U = −f · I`

where Im denotes the m × m unit matrix. Localizing at f , one sees that
−f−1 · U and V are inverse matrices over Af ; as a consequence ` = k, and
each equality above implies the other, for instance:

V · U =
(
− f−1 · U

)−1 · U = −f · U−1 · U = −f · Ik

Here is a familiar example of matrix factorization, with f = xy − wz:(
w x
y z

)
·
(
z −y
−x w

)
= −(xy − wz) ·

(
1 0
0 1

)
=
(
z −y
−x w

)
·
(
w x
y z

)
Let now C1 and C0 be free A-modules of rank r, and let

d1 : C1 → C0 and s0 : C0 → C1

be A-linear homomorphisms defined by the matrices U and V , respectively,
after bases have been tacitly chosen.

The second condition on the matrices U and V implies that d1 is injective,
while the first condition on these matrices shows that fC0 is contained in
Im(d1). Setting L = Coker(d1), one sees that the chosen matrix factorization
defines a commutative diagram with exact rows

0 // C1
d1 //

−f ·1C1

��

C0
//

−f ·1C0

��

s0

{{xxxxxxxxxxxxx
L //

0L= −f ·1L

��

0

0 // C1
d1 // C0

// L // 0

which expresses the following facts: C = 0 → C1
d1−→ C0 → 0 is a free

resolution of the A-module L, this module is annihilated by f , and s0 is a
homotopy between the maps −f · 1C and 0C , both of which lift −f · 1L.

Conversely, if an A-module L annihilated by f has a free resolution (C, d1)
of length 1, then rankA C1 = rankA C0, and any choice of homotopy s0 be-
tween −f · 1C and 0C provides a matrix factorization of −f .

When we already have an A-module L with a presentation matrix U that
defines an injective A-linear map, we can use Macaulay 2 to create a matrix
factorization (U, V ) of −f .

Example 1.1. We revisit the familiar example from a higher perspective.

i1 : A = QQ[w,x,y,z]

o1 = A

o1 : PolynomialRing

i2 : U = matrix {{w,x},{y,z}}

o2 = | w x |
| y z |

2 2
o2 : Matrix A <--- A

i3 : C = chainComplex U

2 2
o3 = A <-- A

0 1

o3 : ChainComplex

i4 : L = HH_0 C

o4 = cokernel | w x |
| y z |

2
o4 : A-module, quotient of A

i5 : f = -det U

o5 = x*y - w*z

o5 : A

Let’s verify that f annihilates L.
i6 : f * L == 0

o6 = true

We use the nullhomotopy function.
i7 : s = nullhomotopy (-f * id_C)

2 2
o7 = 1 : A <----------------- A : 0

{1} | z -x |
{1} | -y w |

o7 : ChainComplexMap

Let’s verify that s is a null-homotopy for −f , using C.dd to obtain the dif-
ferential of the chain complex C as a map of graded modules.

i8 : s * C.dd + C.dd * s == -f

o8 = true

We extract the matrix V from the null-homotopy to get our factorization.
i9 : V = s_0

o9 = {1} | z -x |
{1} | -y w |

2 2
o9 : Matrix A <--- A

For every f and every r ≥ 1 there exists a trivial matrix factorization of
−f , namely, (f ·Ik,−Ik); it can be obtained from the A-module L = Ak/fAk.
In general, it may not be clear how to find an A-module L with the properties
necessary to obtain an ‘interesting’ matrix factorization of −f .

However, in some cases the supply is plentiful.

Remark 1.2. Let A be a graded polynomial ring in e variables of positive
degree over a field K, let f be a homogeneous polynomial in A, and set
B = A/(f). Every B-module M of infinite projective dimension generates a
matrix factorization (U, V ) of −f , as follows.

Let (F, dF ) be a minimal graded free resolution of M over B, and set
L = Coker

(
dF : Fe+1 → Fe

)
. Since M has infinite projective dimension,

we have L 6= 0. By the Depth Lemma, depthB L = depthB. On the other
hand, depthB L = depthA L and depthB = depthA−1. By Hilbert’s Syzygy
Theorem, the minimal graded free resolution (C, dC) of L over A is finite. By
the Auslander-Buchsbaum Equality, Cn = 0 for n > depthA−depthA L = 1.

The minimality of F ensures that all nonzero entries of the presentation
matrix U of L are forms of positive degree. On the other hand, by [11, Sect. 0]
the module L has no direct summand isomorphic to B: it follows that all
nonzero entries of the homotopy matrix V are forms of positive degree (this is
the reason for choosing L as above—stopping one step earlier in the resolution
F could have produced a module L with a non-zero free direct summand).

Our reader would have noticed that Macaulay 2 can read all the data and
perform all the operations needed to construct a module L by the procedure
described in the preceding remark. Here is how it does it.

Example 1.3. We produce a matrix factorization of −f , where

f = x3 + 3y3 − 2yz2 + 5z3 ∈ Q[x, y, z] = A

generated by the module M = B/m2, where B = A/(f) and m = (x, y, z)B.
i10 : A = QQ[x,y,z];

i11 : f = x^3 + 3*y^3 - 2*y*z^2 + 5*z^3;

i12 : B = A/f;

i13 : m = ideal(x,y,z)

o13 = ideal (x, y, z)

o13 : Ideal of B

Let’s take the B-module M and compute its minimal B-free resolution.
i14 : M = B^1/m^2;

i15 : F = resolution(M, LengthLimit=>8)

1 6 9 9 9 9 9 9 9
o15 = B <-- B <-- B <-- B <-- B <-- B <-- B <-- B <-- B

0 1 2 3 4 5 6 7 8

o15 : ChainComplex

We introduce a function restrict1 N which accepts a B-module N and
restricts scalars to produce an A-module.

i16 : restrict1 = N -> coker(lift(presentation N,A) | f);

Now make L as described in Remark 1.2
i17 : L = restrict1 cokernel F.dd_4;

We proceed as in Example 1.1 to get a matrix factorization.
i18 : C = res L;

i19 : U = C.dd_1;

9 9
o19 : Matrix A <--- A

i20 : print U
{4} | 0 xy x2 y2 0 0 0 yz-5/2z2 0 |
{4} | 0 x2 -3y2 xy yz-5/2z2 0 yz-5/2z2 0 0 |
{4} | x2 0 -2yz+5z2 0 y2-5/2yz yz-5/2z2 -5/2yz 0 0 |
{5} | 0 0 0 1/3z 0 0 0 1/2y x |
{5} | 0 0 -z 0 1/2y 0 1/2y -1/2x 0 |
{5} | 0 -z 0 0 -1/2x 0 -1/2x 0 3y |
{5} | 0 0 0 -1/3x 0 1/2y -1/3z 0 0 |
{5} | -z y x 0 0 -1/2x 0 0 0 |
{5} | y 0 0 0 0 0 1/3x 0 -2y+5z |

i21 : s = nullhomotopy (-f * id_C);

i22 : V = s_0;

9 9
o22 : Matrix A <--- A

i23 : print V
{6} | 0 0 -x 0 0 -2y2+5yz 0 -2yz+5z2 -3y2 |
{6} | 0 -x 0 0 0 -2yz+5z2 -3xy -3y2 -3yz |
{6} | -x y 0 0 -2yz+5z2 0 0 0 0 |
{6} | -3y 0 0 6yz-15z2 0 0 3x2 3xy 3xz |
{6} | 0 2z -3y -15xz -15yz 2x2 6yz-15z2 0 3x2 |
{6} | -2x 0 2z 0 -4yz+10z2 0 -6y2 2x2 0 |
{6} | 0 0 3y -6xy+15xz -6y2+15yz 0 -6yz+15z2 0 -3x2 |
{6} | 2z 0 0 -6y2 2x2 2xy 0 0 0 |
{6} | 0 0 0 -x2 -xy -y2 -xz -yz -z2 |

i24 : U*V+f==0

o24 = true

i25 : V*U+f==0

o25 = true

The procedure described above can be automated for more pleasant usage.

Code 1.4. The function matrixFactorization M produces a matrix factor-
ization (U, V ) of −f generated by a module M over B = A/(f).

i26 : matrixFactorization = M -> (
B := ring M;
f := (ideal B)_0;
e := numgens B;
F := resolution(M, LengthLimit => e+1);
L := restrict1 cokernel F.dd_(e+1);
C := res L;
U := C.dd_1;
s := nullhomotopy (-f * id_C);
V := s_0;
assert( U*V + f == 0 );
assert( V*U + f == 0 );
return (U,V));

We use the assert command to signal an error in case the matrices found
don’t satisfy our requirement for a matrix factorization.

Let’s illustrate the new code with a slightly bigger module M than before.

Example 1.5. With the same A, f , B, and m as in Example 1.3, we produce
a matrix factorization generated by the B-module M = B/m3.

i27 : time (U,V) = matrixFactorization(B^1/m^3);
-- used 0.21 seconds

The parallel assignment statement above provides both variables U and V
with matrix values. We examine their shapes without viewing the matrices
themselves by appending a semicolon to the appropriate command.

i28 : U;

15 15
o28 : Matrix A <--- A

i29 : V;

15 15
o29 : Matrix A <--- A

Matrix factorizations were introduced to construct resolutions over the
the residue ring B = A/(f), using the following observation.

Remark 1.6. If (U, V ) is a factorization of −f by k × k matrices and the
maps d1 : C1 → C0 and s0 : C0 → C1 are homomorphisms of free A-modules
defined by U and V , respectively, then the sequence

· · · → C1 ⊗A B
d1⊗1B−−−−→ C0 ⊗A B

s0⊗1B−−−−→ C1 ⊗A B
d1⊗1B−−−−→ C0 ⊗A B → 0

of B-linear maps is a free resolution of the B-module L = Coker(d1).
Indeed, freeness is clear, and we have a complex because d1s0 = −f · 1C0

and s0d1 = −f ·1C1 . If x ∈ C1 satisfies
(
d1⊗1B

)
(x⊗1) = 0, then d1(x) = fy

for some y ∈ C0, hence d1x = d1s0(y). As d1 is injective, we get x = s0(y),
so Ker

(
d1 ⊗ 1B

)
⊆ Im

(
s0 ⊗ 1B

)
; the reverse inclusion follows by symmetry.

Pooling Remarks 1.2 and 1.6 we recover Eisenbud’s result [11, Sect. 6].

Theorem 1.7. Let A be a graded polynomial ring in e variables of positive
degree over a field K, and f a homogeneous polynomial in A. The minimal
graded free resolution of every finitely generated graded module over B =
A/(f) becomes periodic of period 2 after at most e steps. The periodic part
of the resolution is given by a matrix factorization of −f generated by M .

We illustrate the theorem on an already computed example.

Example 1.8. Let A, f , B, M , and F be as in Example 1.3.
To verify the periodicity of F we subtract pairs of differentials and com-

pare the result with 0: direct comparison of the differentials would not work,
because the free modules involved have different degrees.

i30 : F.dd_3 - F.dd_5 == 0

o30 = false

i31 : F.dd_4 - F.dd_6 == 0

o31 = false

i32 : F.dd_5 - F.dd_7 == 0

o32 = true

The first two answers above come as a surprise—and suggest a property of
F that is weaker than the one we already know to be true!

There is an easy explanation: we checked the syzygy modules for equality ,
rather than for isomorphism. We do not know why Macaulay 2 didn’t produce
an equality at the earliest possible stage, nor why it eventually produced one.
The program has other strategies for computing resolutions, so let’s try one.

i33 : M = B^1/m^2;

i34 : G = resolution(M, LengthLimit => 8, Strategy => 0)

1 6 9 9 9 9 9 9 9
o34 = B <-- B <-- B <-- B <-- B <-- B <-- B <-- B <-- B

0 1 2 3 4 5 6 7 8

o34 : ChainComplex

i35 : G.dd_3 - G.dd_5 == 0

o35 = true

i36 : G.dd_4 - G.dd_6 == 0

o36 = true

i37 : G.dd_5 - G.dd_7 == 0

o37 = true

The strategy paid off, revealing periodicity at the earliest possible stage.
However, the algorithm used may be a lot slower that the default algorithm.

2 Graded Algebras

We describe some standard universal algebras over a commutative ring A.
Let Q denote a free A-module of rank c, and set Q∗ = HomA(Q,A). We

assign degree 2 to the elements of Q, and degree −2 to those of Q∗. We let Q∧

denote a copy of Q whose elements are assigned degree 1; if x is an element
of Q, then x∧ denotes the corresponding element of Q∧.

We use α = (α1, . . . , αc) ∈ Zc as a multi-index, set |α| =
∑
i αi, and order

Z
c by the rule: α ≥ β means αi ≥ βi for each i. We let o denote the trivial

element of Zc, and εi the i’th element of its standard basis.

Construction 2.1. For each integer m ≥ 0 we form free A-modules

Sm(Q∗) with basis
{
Xα : |α| = m

}
Dm(Q) with basis

{
Y (α) : |α| = m

}∧m(Q∧) with basis
{
Y ∧α : |α| = m and α ≤ (ε1 + · · ·+ εc)

}
For m < 0 we declare the modules Sm(Q∗), Dm(Q), and

∧m(Q∧) to be equal
to 0, and define the symbols Xα, Y (α), and Y ∧α accordingly; in addition, we
set
∧m(Q∧) = 0 and Y ∧α = 0 if |α| 6≤ (ε1 + · · ·+ εc), and we set

Xi = Xεi Yi = Y (εi) Y ∧i = Y ∧εi for i = 1, . . . , c

Taking Sm(Q∗), Dm(Q), and
∧
m(Q∧) as homogeneous components of degree

−2m, 2m, and m, respectively, we introduce graded algebras

S = S(Q∗) D = D(Q) E =
∧

(Q∧)

by defining products of basis elements by the formulas

Xα ·Xβ = Xα+β

Y (α) · Y (β) =
c∏
i=1

(αi + βi)!
αi!βi!

Y (α+β)

Y ∧α · Y ∧β = inv(α, β)Y ∧α+β

where inv(α, β) denotes the number of pairs (i, j) with αi = βj = 1 and i > j.
Thus, S is the symmetric algebra of Q∗, with Xo = 1, while D is the divided
powers algebra of Q, with Y (o) = 1, and E is the exterior algebra of Q∧, with
Y ∧ o = 1. We identify S and the polynomial ring A[X1, . . . , Xc].

A homogeneous derivation of a graded A-algebra W is a homogeneous
A-linear map d : W →W such that the Leibniz rule

d(xy) = d(x)y + (−1)deg x·deg dxd(y)

holds for all homogeneous elements x, y ∈W .

Construction 2.2. Each sequence f1, . . . , fc ∈ A yields a Koszul map

dE : E → E defined by the formula

dE(Y ∧β) =
c∑
i=1

(−1)β1+···+βi−1fiY
∧ β−εi

It is a derivation of degree −1 and satisfies d2
E = 0.

Construction 2.3. For every Xi ∈ S1(Q∗) and each Y (β) ∈ Dm(Q) we set

Xi yY
(β) = Y (β−εi) ∈ Dm−1(Q)

Extending this formula by A-bilinearity, we define g y y for all g ∈ S1(Q∗) and
all y ∈ D. It is well known, and easily verified, that the map g y : y 7→ g y y is
a graded derivation D → D of degree −2, and that the derivations associated
with arbitrary g and g′ commute. As a consequence, the formula

Xα yY (β) = (X1 y)α1 · · · (Xc y)αc
(
Y (β)

)
∈ D|β−α|(Q)

extended A-linearly to all u ∈ S, defines on D a structure of graded S-module.
The usual products on S ⊗A E and D ⊗A E and the induced gradings

(S ⊗A E)n =
⊕

`−2k=n

Sk(Q∗)⊗A
∧`(Q∧)

(D ⊗A E)n =
⊕

`+2k=n

D(k)(Q)⊗A
∧`(Q∧)

turn S ⊗A E and D ⊗A E into graded algebras. The second one is a graded
module over the first, for the action (u⊗ z) · (y ⊗ z′) = (u y v)⊗ (z · z′).

Construction 2.4. The element w =
∑c
i=1Xi ⊗ Y ∧i yields a Cartan map

dDE : D ⊗A E → D ⊗A E defined by the formula

dDE(y ⊗ z) = w · (y ⊗ z) =
c∑
i=1

(Xi y y)⊗ (Y ∧i · z)

It is an E-linear derivation of degree −1, and d2
DE = 0 because w2 = 0.

Lemma 2.5. For each integer s define a complex Gs as follows:

· · · → Dk(Q)⊗A
∧s−k(Q∧) w−→ Dk−1(Q)⊗A

∧s−k+1(Q∧)→ · · ·

with D0(Q)⊗A
∧s(Q∧) in degree s. If s > 0, then Gs is split exact.

Proof. Note that for each s ∈ Z there exist isomorphisms of complexes⊕∞
s=1G

s ∼= (D ⊗A E)≥1
∼=
(⊗c

i=1G(i)
)
≥1

, where G(i) is the complex

· · · → AY
(k+1)
i ⊗A A

wi−→ (AY (k)
i )⊗ (AY ∧i ) 0−→ (AY (k)

i )⊗A A→ · · ·

and wi is left multiplication withXi⊗Y ∧i . This map bijective, so each complex
G(i)≥1 is split exact. The assertion follows. ut

3 Universal Homotopies

This section contains the main new mathematical result of the paper.
We introduce a universal construction, that takes as input a projective

resolution C of an A-module M and a finite set f of elements annihilating M ;
the output is a new projective resolution of M over A. If f 6= ∅, then the new
resolution is infinite—even when C is finite—because it encodes additional
data: the null-homotopies for f · 1C for all f ∈ f , all compositions of such
homotopies, and all relations between those compositions. This higher-order
information tracks the transformation of the homological properties of M
when its ring of operators is changed from A to A/(f).

Our construction is motivated by, and is similar to, one due to Shamash
[15] and Eisenbud [11]: assuming that the elements of f form an A-regular
sequence, they produce a projective resolution of M over A/(f). By contrast,
we make no assumption whatsoever on f . With the additional hypothesis, in
the next section we quickly recover the original result from the one below. As
an added benefit, we eliminate the use of spectral sequences from the proof.

Theorem 3.1. Let A be a commutative ring, let f1, . . . , fc be a sequence of
elements of A, let M be an A-module annihilated by fi for i = 1, . . . , c, and let
r : C →M be a resolution of M by projective (respectively, free) A-modules.

There exists a family of homogeneous A-linear maps

{dγ : C → C | deg(dγ) = 2|γ| − 1}γ∈Nc

satisfying the following conditions

do = dC is the differential of C

[do, dγ ] =

−fi · 1C if γ = εi for i = 1, . . . , c
−
∑ +

α+β=γ

dαdβ if |γ| ≥ 2
(1)

where
∑+ denotes a summation restricted to indices in Nc r {o}.

Any family {dγ}γ∈Nc as above defines an A-linear map of degree −1,

dCD : C ⊗A D → C ⊗A D given by

dCD(x⊗ y) =
∑
γ∈Nc

dγ(x)⊗ (Xγ y y) (2)

where D is the divided powers algebra defined in Construction 2.1, and the
action of Xγ on D is defined in Construction 2.3.

With dE and dDE defined in Constructions 2.2 and 2.4 and the tensor
product of maps of graded modules defined as in Remark 3.4, the map

d : C ⊗A D ⊗A E → C ⊗A D ⊗A E given by
d = dCD ⊗ 1E + 1C ⊗ dDE + 1C ⊗ 1D ⊗ dE

(3)

is an A-linear differential of degree −1, and the map

q : C ⊗A D ⊗A E →M given by

q(x⊗ y ⊗ z) =

{
yz · r(x) if deg(y) = deg(z) = 0
0 otherwise

is a resolution of M by projective (respectively, free) A-modules.

For use in the proof, we bring up a few general homological points.
A bounded filtration of a chain complex F is a sequence

0 = F 0 ⊆ F 1 ⊆ · · · ⊆ F s−1 ⊆ F s ⊆ · · ·

of subcomplexes such that for each n there exists an s with F sn = Fn. As
usual, we let grs(F ) denote the complex of A-modules F s/F s−1.

Lemma 3.2. Let q : F → F ′ be a morphism of complexes with bounded fil-
trations, such that q(F s) ⊆ F ′s for all s ∈ Z. If for each s the induced map
grs(q) : grs(F )→ grs(F ′) is a quasi-isomorphism, then so is q.

Proof. Denoting qs the restriction of q to F s, we first show by induction on
s that Hn(qs) is bijective for all n. The assertion is clear for s = 0, since
F 0 = 0 and F ′ 0 = 0. For the inductive step, we assume that qs−1 is a quasi-
isomorphism for some s ≥ 1. We have a commutative diagram of complexes

0 // F s−1

qs−1

��

// F s

qs

��

// grs(F )

grs(q)

��

// 0

0 // F ′ s−1 // F ′ s // grs(F ′) // 0

By hypothesis and inductive assumption, in the induced diagram

Hn+1(grs(F ))

Hn+1(grs(q)) ∼=
��

// Hn(F s−1)

Hn(qs−1) ∼=
��

// Hn(F s)

Hn(qs)

��

// Hn(grs(F ))

Hn(grs(q)) ∼=
��

// Hn−1(F s−1)

Hn−1(qs−1) ∼=
��

Hn+1(grs(F ′)) // Hn(F ′ s−1) // Hn(F ′ s) // Hn(grsF ′) // Hn−1(F ′ s−1)

the four outer vertical maps are bijective. By the Five-Lemma, so is Hn(qs).
Now we fix an integer n ∈ Z, and pick s so large that

F sk = Fk and F ′ sk = F ′k hold for k = n− 1, n, n+ 1 .

The choice of s implies that Hn(F s) = Hn(F ), Hn(F ′ s) = Hn(F ′), and
Hn(qs) = Hn(q). Since we have already proved that Hn(qs) is an isomorphism,
we conclude that Hn(q) : Hn(F )→ Hn(F ′) is an isomorphism. ut

Remark 3.3. If (F, dF ) is a complex of A-modules, then Homgr
A (F, F ) de-

notes the graded module whose n’th component consists of the A-linear maps
g : F → F with g(Fi) ⊆ Fi+n for all i ∈ Z.

If g, h are homogeneous A-linear maps, then their composition gh is ho-
mogeneous of degree deg(g) + deg(h), and so is their graded commutator

[g, h] = gh− (−1)deg g·deg hhg

Commutation is a graded derivation: for each homogeneous map h′ one has

[g, hh′] = [g, h]h′ + (−1)deg g·deg hh[g, h′]

The map h 7→ [dF , h] has square 0, and transforms Homgr
A (F, F ) into a

complex of A-modules; by definition, its cycles are the chain maps F → F ,
and its boundaries are the null-homotopic maps.

Remark 3.4. If p : F → F ′ and q : G → G′ are graded maps of graded
modules, we define the tensor product p⊗q : F ⊗F ′ → G⊗G′ by the formula
(p ⊗ q)(f ⊗ g) = (−1)deg q·deg f (p(f) ⊗ q(g)). With this convention, when
F = F ′ and G = G′, the graded commutator [1F ⊗ q, p⊗ 1G] vanishes.

Lemma 3.5. Let M be an A-module and let r : C →M be a free resolution.
If g : C → C is an A-linear map with deg(g) > 0, and [dC , g] = 0, then
g = [dC , h] for some A-linear map h : C → C with deg(h) = deg(g) + 1.

Proof. The augmentation r : C →M defines a chain map of degree zero

Homgr
A (C, r) : Homgr

A (C,C)→ Homgr
A (C,M)

The map induced in homology is an isomorphism: to see this, apply the
‘comparison theorem for projective resolutions’. Since A-linear maps C →M
of positive degree are trivial, the conclusion follows from Remark 3.3. ut

Proof (of Theorem 3.1). Recall that D is the divided powers algebra of a free
A-module Q with basis Y1, . . . , Yc, that X1, . . . , Xc is the dual basis of the
free A-module Q∗, and S for the symmetric algebra of Q∗, see Construction
2.1 for details. We set f =

∑c
i=1 fiXi ∈ S1(Q∗).

We first construct the maps dγ by induction on |γ|.
If |γ| = 0, then γ = o, so do = dC is predefined. If |γ| = 1, then γ = εi

for some i with 1 ≤ i ≤ c. Since fi annihilates the B-module M , the map
−fi · 1C lifts the zero map on M , hence is null-homotopic. For each i we take
dεi to be a null-homotopy, that is, [do, dεi ] = −fi · 1C . With these choices,
the desired formulas hold for all γ with |γ| ≤ 1.

Assume by induction that maps dγ satisfying the conclusion of the lemma
have been chosen for all γ ∈ Nc with |γ| < n, for some n ≥ 2. Fix γ ∈ Nc

with |γ| = n. Using Remark 3.3 and the induction hypothesis, we obtain[
do,

∑ +

α+β=γ

dαdβ

]
=
∑ +

α+β=γ

(
[do, dα]dβ − dα[do, dβ ]

)
=
∑ +

α+β=γ

(( ∑ +

α′+α′′=α

dα′dα′′

)
dβ − dα

( ∑ +

β′+β′′=β

dβ′dβ′′

))
=

∑ +

α′+α′′+β=γ

dα′dα′′dβ −
∑ +

α+β′+β′′=γ

dαdβ′dβ′′

= 0

The map −
∑+
α+β=γ dαdβ has degree 2|γ|− 2, so by Lemma 3.5 it is equal to

[do, dγ ] for some A-linear map dγ : C → C of degree 2|γ| − 1. Choosing such
a dγ for each γ ∈ Nc with |γ| = n, we complete the step of the induction.

From the definition of d we obtain an expression

d2 =d2
CD ⊗ 1E + 1C ⊗ [dDE , 1D ⊗ dE ] + [dCD ⊗ 1E , 1C ⊗ dDE ]+

1C ⊗ d2
DE + 1C ⊗ 1D ⊗ d2

E + [dCD ⊗ 1E , 1C ⊗ 1D ⊗ dE ]

Constructions 2.4, 2.2, and Remark 3.4 show that the maps in the second row
are equal to 0, so to prove that d2 = 0 it suffices to establish the equalities

d2
CD = −f · 1C⊗D (4)

[dDE , 1D ⊗ dE ] = f · 1D⊗E (5)
[dCD ⊗ 1E , 1C ⊗ dDE ] = 0 (6)

A direct computation with formula (1) proves equality (4) above:

d2
CD(x⊗ y) = dCD

( ∑
β∈Nc

dβ(x)⊗
(
Xβ y y

))

=
∑
β∈Nc

( ∑
α∈Nc

dαdβ(x)⊗
(
Xα y

(
Xβ y y

)))
=

∑
α+β∈Nc

dαdβ(x)⊗
(
Xα+β y y

)
=

c∑
i=1

−fix⊗ (Xi y y)

= −f · (x⊗ y)

By Constructions 2.2 and 2.4, the maps 1D⊗dE and dDE are derivations
of degree −1, so the commutator [dDE , 1D⊗dE ] is a derivation of degree −2.
Every element of D ⊗A E is a product of elements 1 ⊗ Y ∧i of degree 1 and
Y

(k)
j ⊗ 1 of degree 2k, so it suffices to check that the map on either side of

(6) takes the same value on those elements. For degree reasons, both sides
vanish on 1⊗ Y ∧i . We now complete the proof of equality (5) as follows:

[dDE , 1E ⊗ dE ]
(
Y

(k)
j ⊗ 1

)
= dDE

(
(1D ⊗ dE)

(
Y

(k)
j ⊗ 1

))
+ (1E ⊗ dE)

(
dDE

(
Y

(k)
j ⊗ 1

))
= (1E ⊗ dE)

(
Y

(k−1)
j ⊗ Y ∧j

)
= Y

(k−1)
j ⊗ fj

= f ·
(
Y

(k)
j ⊗ 1

)
To derive equation (6) we use Constructions 2.2 and 2.4 once again:(
(dCD ⊗ 1E)(1C ⊗ dDE)

)
(x⊗ y ⊗ z)

=(−1)deg xdCD

( c∑
i=1

x⊗ (Xi y y)⊗ (Y ∧i · z)
)

=(−1)deg x
c∑
i=1

∑
γ∈Nc

dγ(x)⊗
(
Xγ y(Xi y y)

)
⊗ (Y ∧i · z)

=−
∑
γ∈Nc

c∑
i=1

(−1)deg(dγ(x))dγ(x)⊗
(
Xi y(Xγ y y)

)
⊗ (Y ∧i · z)

=−
(
1C ⊗ dDE

)( ∑
γ∈Nc

dγ(x)⊗ (Xγ y y)⊗ z
)

=−
(
(1C ⊗ dDE)(dCD ⊗ 1E)

)
(x⊗ y ⊗ z)

It remains to show q is a quasi-isomorphism. Setting

F s =
⊕
k+`≤s

C ⊗A Dk(Q)⊗A
∧`(Q∧) for s ∈ Z

we obtain a bounded filtration of the complex F = (C ⊗A D ⊗A E, d). On
the other hand, we let F ′ denote the complex with F ′0 = M and F ′n = 0 for
n 6= 0; the filtration defined by F ′ 0 = 0 and F ′ s = F ′ for s ≥ 1 is obviously
bounded, and q(F s) ⊆ F ′ s holds for all s ≥ 0. By Lemma 3.2 it suffices to
show that the induced map grs(q) : grs(F )→ grs(F ′) is bijective for all s.

Inspection of the differential d of F shows that grs(F ) is isomorphic to
the tensor product of complexes C ⊗A Gs, where Gs is the complex defined
in Lemma 2.5. It is established there that Gs is is split exact for s > 0, hence
Hn(C ⊗A Gs) = 0 for all n ∈ Z. As G0 = A and gr0(q) = r, we are done. ut

4 Cohomology Operators

We present a new approach to the procedure of Shamash [15] and Eisenbud
[11] for building projective resolutions over a complete intersection. We then

use this resolution to prove a fundamental result of Gulliksen [12] on the
structure of Ext modules over complete intersections.

A set f = {f1, . . . , fc} ⊆ A is Koszul-regular if the complex (E, dE) of
Construction 2.2, has Hn(E) = 0 for n > 0. A sufficient condition for Koszul-
regularity is that the elements of f , in some order, form a regular sequence.

Theorem 4.1. Let A be a commutative ring, f = {f1, . . . , fc} ⊆ A a subset,
B = A/(f) the residue ring, M a B-module, and r : C → M a resolution of
M by projective (respectively, free) A-modules.

Let {dγ : C → C}γ∈Nc be a family of A-linear maps provided by Theorem
3.1, set D′ = D ⊗A B, and y′ = y ⊗ 1 for y ∈ D. The map

∂ : C ⊗A D′ → C ⊗A D′ given by

∂(x⊗ y′) =
∑
γ∈Nc

dγ(x)⊗ (Xγ y y)′ (7)

is a B-linear differential of degree −1. If f is Koszul-regular, then the map

q′ : C ⊗A D′ →M given by

q′(x⊗ y′) =

{
y · r(x) if deg(y′) = 0
0 otherwise

is a resolution of M by projective (respectively, free) A-modules.

Remark 4.2. Assume that in the theorem f = {f1}. The module D` is then
trivial if ` is odd, and is free with basis consisting of a single element Y (`/2)

1

if ` is even. Thus, the resolution C ⊗A D′ has the form

· · · ∂2n+1−−−−→
∞⊕
j=0

C2j ⊗A BY (n−j)
1

∂2n−−→
∞⊕
j=1

C2j−1 ⊗A BY (n−j)
1

∂2n−1−−−−→ · · ·

The simplest situation occurs when, in addition, C is a free resolution
with Cn = 0 for n ≥ 2. In this case the differential do has a single non-
zero component, d1 : C1 → C0, the homotopy dε1 between −f · 1C and 0C
has a single non-zero component, s0 : C0 → C1, and all the maps dγ with
γ ∈ N1

r {o, ε1} are trivial for degree reasons. It is now easy to see that the
complex above coincides with the one constructed, ad hoc, in Remark 1.6.

Proof (of the theorem). In the notation of Theorem 3.1, we have equalities

C ⊗A D′ = (C ⊗A D ⊗A E)⊗E B and ∂ = d⊗ 1B

It follows that ∂2 = 0. For each s ≥ 0 consider the subcomplexes

F s =
⊕
k+`≤s

Ck ⊗A D` ⊗A E of F = C ⊗A D ⊗A E

F ′ s =
⊕
k+`≤s

Ck ⊗A D′` of F ′ = C ⊗A D′

They provide bounded filtrations of the complexes F and F ′, respectively,
such that the map p′ = 1C ⊗ 1D ⊗ p : F → F ′ satisfies p′(F s) ⊆ F ′ s for
all s ≥ 0. Setting Gs =

⊕
k+`=s(Ck ⊗A D`), we obtain equalities grs(F ) =

(Gs⊗AE, 1Gs ⊗ dE) and grs(F ′) = (Gs⊗AB, 0) of complexes of A-modules.
If f is Koszul regular, then p : E → B is a quasi-isomorphism, hence so

is 1Gs ⊗ p = grs(p′) for each s ≥ 0. Lemma 3.2 then shows that p′ is a quasi-
isomorphism. The quasi-isomorphism q : F → M of Theorem 3.1 factors as
q = q′(1C ⊗ 1D ⊗ p), so we see that q′ is a quasi-isomorphism, as desired. ut

Let M and N be B-modules, and let Ext•B(M,N) denote the graded B-
module having ExtnB(M,N) as component of degree −n. To avoid negative
numbers, it is customary to regrade Ext•B(M,N) by cohomological degree,
under which the elements of ExtnB(M,N) are assigned degree n; we do not
do it here, in order not to confuse Macaulay 2. Of course, these modules can
be computed from any projective resolution of M over B.

The next couple of remarks collect a few innocuous observations. In hind-
sight, they provide some of the basic tools for studying cohomology of modules
over complete intersections: see Remark 4.6 for some related material.

Remark 4.3. The resolution (C ⊗A D′, ∂) provided by Theorem 4.1 is a
graded module over the graded algebra S, with action defined by the formula

u · (x⊗ y′) = x⊗ (u y y)′

and this action commutes with the differential ∂. The induced action provides
a structure of graded S-module on the complex HomB(C ⊗A D′, N).

The action of S commutes with the differential ∂∗ = HomB(∂,N) of this
complex, hence passes to its homology, making it a graded a S-module. Thus,
each element u ∈ S−2k = Sk(Q) determines homomorphisms

ExtnB(M,N) u−−→ Extn+2k
B (M,N) for all n ∈ Z

For this reason, from now on we refer to the graded ring S as the ring of
cohomology operators determined by the Koszul-regular set f .

Remark 4.4. The canonical isomorphisms of complexes of A-modules

HomB(C ⊗A D′, N) = S ⊗A HomA(C,N) = S ⊗A HomA(C,A)⊗A N

commute with the actions of S.

The following fundamental result shows that in many important cases the
action of the cohomology operators is highly nontrivial.

Theorem 4.5. Let A be a commutative ring, let f be a Koszul regular subset
of A, and let S be the graded ring of cohomology operators defined by f .

If M and N are finitely generated modules over B = A/(f), and M has
finite projective dimension over A (in particular, if A is regular), then the
graded S-module Ext•B(M,N) is finitely generated.

Proof. Choose a resolution r : C →M with Cn a finite projective A-module
for each n and Cn = 0 for all n� 0. By Remark (4.4), the graded S-module
HomB(C ⊗A D′, N) is finitely generated. Since S is noetherian, so is the
submodule Ker(∂∗), and hence the homology module, Ext•B(M,N). ut

Remark 4.6. The resolution of Remark 4.2 was constructed by Shamash
[15, Sect. 3], that of Theorem 4.1 by Eisenbud [11, Sect. 7]. The new aspect
of our approach is indicated at the beginning of Section 3.

As introduced in Remark 4.3, the S-module structure on Ext may appear
ad hoc. In fact, it is independent of all choices of resolutions and maps, it can
be computed from any projective resolution of M over B, it is natural in both
module arguments and—in an appropriate sense—in the ring argument, and
it commutes with Yoneda products from either side. These properties were
proved by Gulliksen [12, Sect. 2], Mehta [13, Ch. 2], Eisenbud [11, Sect. 4],
and Avramov [2, Sect. 2]. However, each author used a different construction
of cohomology operators, and comparison of the different approaches has
turned to be an unexpectedly delicate problem. It was finally resolved in [8],
where complete proofs of the main properties of the operators can be found.

Gulliksen [12, Sect. 3] established a stronger form of Theorem 4.5, without
finiteness hypotheses on the ring A: If the A-module Ext•A(M,N) is noethe-
rian, then the S-module Ext•B(M,N) is noetherian; this can be obtained
from the complexes of Remark (4.4) by means of a spectral sequence, cf. [4,
Sect. 6]. The converse of Gulliksen’s theorem was proved in [6, Sect. 4].

For the rest of the paper we place ourselves in a situation where Mac-
aulay 2 operates best—graded modules over positively graded rings. This
grading is inherited by the various Ext modules, and we keep careful track
of it. Our conventions and bookkeeping procedures are discussed in detail in
an Appendix, which the reader is invited to consult as needed.

For ease of reference, we collect some notation.

Notation 4.7. The following is assumed for the rest of the paper.

• K is a field.
• {xh | deg′(xh) > 0}h=1,...,e is a set of indeterminates over K.
• A = K[x1, . . . , xe], graded by deg′(a) = 0 for a ∈ K.
• f1, . . . , fc is a homogeneous A-regular sequence in (x1, . . . , xe)2.
• ri = deg′(fi) for i = 1, . . . , c.
• {Xi | DegXi = (−2,−ri)}i=1,...,c is a set of indeterminates over A.
• S = A[X1, . . . , Xc], bigraded by Deg(a) = (0,deg′(a)).
• B = A/(f), with degree induced by deg′.
• M and N are finitely generated graded B-modules.
• S acts as bigraded ring of cohomology operators on Ext•B(M,N).
• k = B/(x1, . . . , xe)B, with degree induced by deg′.
• R = S ⊗A k ∼= K[X1, . . . , Xc], with bidegree induced by Deg.

Remark 4.8. Under the conditions above, it is reasonable to ask when the
B-free resolution G of Theorem 4.1, obtained from a minimal A-free reso-
lution C of M , will itself be minimal. Shamash [15, Sect. 3] proves that G
is minimal if fi ∈ (x1, . . . , xe)annA(M) for i = 1, . . . , c. An obvious example
with non-minimal G occurs when M has finite projective dimension over B:
if c > 0 then G is infinite. A more interesting failure of minimality follows.

Example 4.9. Let A, f , B, and M be as in Example 1.5.
i38 : M = B^1/m^3;

i39 : F = resolution(M, LengthLimit=>8)

1 10 16 15 15 15 15 15 15
o39 = B <-- B <-- B <-- B <-- B <-- B <-- B <-- B <-- B

0 1 2 3 4 5 6 7 8

o39 : ChainComplex

Thus, the sequence of Betti numbers βBn (M) is (1, 10, 16, 15, 15, 15, . . . ).
i40 : M’ = restrict1 M;

i41 : C = res M’

1 10 15 6
o41 = A <-- A <-- A <-- A <-- 0

0 1 2 3 4

o41 : ChainComplex

By Remark 4.2, the sequence rankB Fn is (1, 10, 16, 16, 16, 16, . . . ).

In a graded context, all cohomological entities discussed so far in the text
acquire an extra grading, discussed in detail in the Appendix. The notions
below are used, but not named, in [5] in a local situation.

Remark 4.10. We define the reduced Ext module for M and N over B by

ext•B(M,N) = Ext•B(M,N)⊗A k

With the induced bigrading and action, it is a bigraded module over the
bigraded ring R, that we call the reduced ring of cohomology operators.

The dimension of the K-vector space extnB(M,N)s is equal to the number
of generators of bidegree (−n, s) in any minimal set of generators of the
graded B-module ExtnB(M,N). We define the graded (respectively, ungraded)
Ext-generator series of M and N to be the formal power series

GM,N
B (t, u) =

∑
n∈N , s∈Z

dimK extnB(M,N)s tnu−s ∈ Z[u, u−1][[t]]

GM,N
B (t) =

∞∑
n=0

dimK extnB(M,N) tn ∈ Z[[t]]

There is a simple relation between these series: GM,N
B (t) = GM,N

B (t, 1).

Corollary 4.11. In the notation above, ext•B(M,N) is a finitely generated
bigraded R-module, and GM,N

B (t, u) represents a rational function of the form

gM,N
B (t, u)

(1− t2ur1) · · · (1− t2urc)
with gM,N

B (t, u) ∈ Z[t, u, u−1]

Proof. The assertion on finite generation results from Theorem 4.5 and the
one on bigradings from Remark A.2. The form of the power series is then
given by the Hilbert-Serre Theorem. ut

5 Computation of Ext Modules

This section contains the main new computational result of the paper.
We discuss, apply, and present an algorithm that computes, for graded

modules M and N over a graded complete intersection ring B, the graded
B-modules ExtnB(M,N) simultaneously in all degrees n, along with all the
cohomology operators defined in Remark 4.3.

More precisely, the input consists of a field K, a polynomial ring A =
K[x1, . . . , xe] with deg′(xh) > 0, a sequence f1, . . . , fc of elements of A, and
finitely generated modules M , N over B = A/(f1, . . . , fc). The program
checks whether the sequence consists of homogeneous elements, whether it
is regular, and whether the modules M , N are graded, sending the appro-
priate error message if any one of these conditions is violated. If the input
data pass those tests, then the program produces a presentation of the bi-
graded module H = Ext•B(M,N), where the elements of ExtnB(M,N) have
homological degree −n, over the polynomial ring A[X1, . . . , Xc], bigraded by
Deg(a) = (0,deg′(a)) and Deg(Xi) = (−2,−deg′(fi)).

The algorithm is based on the proofs of Theorems 4.1 and 4.5, and is
presented in Code 5.4 below. We start with an informal discussion.

Remark 5.1. The routine resolution of Macaulay 2 finds the matrices
do,n : Cn → Cn−1 of the differential dC of a minimal free resolution C of M
over A. Matrices dγ,n : Cn → Cn+2|γ|−1 satisfying equation (1) for γ ∈ Nc
with |γ| > 0 are computed using the routine nullhomotopy of the Macaulay 2
language in a loop that follows the first part of the proof of Theorem 4.1.

The transposed matrix d∗γ,n yields an endomorphism of the free bigraded
A-module C∗ =

⊕e
n=0 HomA(Cn, A) of rank m, where m =

∑e
n=0 rankA Cn.

The m × m matrix d̃∗γ,n describing this endomorphism is formed using the
routines transpose and sum. The m×m matrix

∆ =
e∑

n=0

(−1)n+2|γ|−1
∑
γ∈Nc

|γ|≤(e−n+1)/2

Xγ · d̃∗γ,n

with entries in S = A[X1, . . . , Xc] defines an endomorphism of the free bi-
graded S-module S ⊗A C∗. It induces an endomorphism ∆ of the bigraded



Complete Intersections 151

S-module S ⊗A C∗ ⊗AN . The bigraded S-module H = Ext•B(M,N) is com-
puted as H = Ker(∆)/ Im(∆) using the routine homology.

In the computations we let H denote the bigraded S-module Ext•B(M,N).
As the graded ring S is zero in odd homological degrees, there is a canonical
direct sum decomposition H = Heven ⊕Hodd of bigraded S-modules, where
‘even’ or ‘odd’ refers to the parity of the first degree in each pair Deg(x).

We begin with an example in codimension 1, where it is possible to con-
struct the infinite resolution and the action of S on it by hand.

Example 5.2. Consider the ring A = K[x] where the variable x is assigned
degree 5, and set B = A/(x3). The bigraded ring of cohomology operators
then is S = A[X,x], where Deg(X) = (−2,−15) and Deg x = (0, 5).

For the B-modules M = B/(x2) and N = B/(x), the bigraded S-module
H = Ext•B(M,N) is described by the isomorphism

H ∼= (S/(x))⊕ (S/(x))[1, 10]

A minimal free resolution of M over B is displayed below.

F = . . . −→ B[−30] −x−−→ B[−25] x2

−−→ B[−15] −x−−→ B[−10] x2

−−→ B −→ 0

This resolution is actually isomorphic to the resolution C⊗AD′ described in
Remark 4.2, formed from the free resolution

C = 0 −→ A[−10] x2

−−→ A −→ 0

of M over A and the nullhomotopy dε1 displayed in the diagram

0 // A[−10] x2
//

−x3

��

A //

−x3

��

−x

||yyyyyyyyy
0

0 // A[−10] x2
// A // 0

The isomorphism of F with C ⊗A D′ endows F with a structure of bigraded
module over S, where the action of X on F is the chain map F → F of ho-
mological degree −2 and internal degree −15 that corresponds to the identity
map of B in each component.

The bigraded S-module H = Ext•B(M,N) is the homology of the complex

HomR(F,N) = 0 −→ N
0−−→ N [10] 0−→ N [15] 0−−→ N [25] 0−→ N [30] −→ · · ·

where multiplication by x ∈ S is the zero map, and for each i ≥ 0 multi-
plication by X ∈ S sends N [15i] to N [15i + 15] (respectively, N [10i + 10]
to N [10i + 25]) by the identity map. This description provides the desired
isomorphisms of bigraded S-modules.

Here is how to compute H with Macaulay 2.
Create the rings and modules.



152 L. L. Avramov and D. R. Grayson

i42 : K = ZZ/103;

i43 : A = K[x,Degrees=>{5}];

i44 : B = A/(x^3);

i45 : M = B^1/(x^2);

i46 : N = B^1/(x);

Use the function Ext to compute H = Ext•B(M,N) (the semicolon at the end
of the line will suppress printing until we have assigned the name S to the
ring of cohomology operators constructed by Macaulay 2.)

i47 : H = Ext(M,N);

We may look at the ring.
i48 : ring H

o48 = K [$X , x, Degrees => {{-2, -15}, {0, 5}}]
1

o48 : PolynomialRing

Macaulay 2 has assigned the name $X_1 to the variable X. The dollar sign
indicates an internal name that cannot be entered from the keyboard: if neces-
sary, obtain the variable by entering S_0; notice that indexing in Macaulay 2
starts with 0 rather than 1. Notice also the appearance of braces rather than
parentheses in Macaulay 2’s notation for bidegrees.

i49 : degree \ gens ring H

o49 = {{-2, -15}, {0, 5}}

o49 : List

Assign the ring a name.
i50 : S = ring H;

We can now look at the S-module H.
i51 : H

o51 = cokernel {0, 0} | 0 x |
{-1, -10} | x 0 |

2
o51 : S-module, quotient of S

Each row in the display above is labeled with the bidegree of the correspond-
ing generator of H. This presentation gives the isomorphisms of bigraded
S-modules, already computed by hand earlier.

Let’s try an example with a complete intersection of codimension 2. It is
not so easy to do by hand, but can be checked using the theory in [4].

Example 5.3. Begin by constructing a polynomial ring A = K[x, y].
i52 : A = K[x,y];

Now we produce a complete intersection quotient ring B = A/(x3, y2).

i53 : J = ideal(x^3,y^2);

o53 : Ideal of A

i54 : B = A/J;

We take N to be the B-module B/(x2, xy).
i55 : N = cokernel matrix{{x^2,x*y}}

o55 = cokernel | x2 xy |

1
o55 : B-module, quotient of B

Remark A.1 shows that H = Ext•B(N,N) is a bigraded module over the
bigraded ring S = A[X1, X2] = K[X1, X2, x, y] where

Deg(X1) = (−2,−3) Deg(X2) = (−2,−2)
Deg(x) = (0, 1) Deg(y) = (0, 1)

Using Macaulay 2 (below) we obtain an isomorphism of bigraded S-modules

Heven ∼=
S

(x2, xy, y2, xX1, yX1)
⊕ S

(x, y)
[2, 2]

Hodd ∼=
(

S

(x, y,X1)
⊕ S

(x, y)

)2

[1, 1]

These isomorphisms also yield expressions for the graded B-modules:

Ext2i
B (N,N) ∼= N ·Xi

2 ⊕
i⊕

h=1

k ·Xh
1X

i−h
2 ⊕

i−1⊕
h=0

k[2] ·Xh
1X

i−1−h
2

Ext2i+1
B (N,N) ∼=

(
k[1] ·Xi

2 ⊕
i⊕

h=0

k[1] ·Xi−h
1 Xh

2

)2

Now we follow in detail the computation of the bigraded S-module H.
i56 : time H = Ext(N,N);

-- used 0.2 seconds

i57 : ring H

o57 = K [$X , $X , x, y, Degrees => {{-2, -2}, {-2, -3}, {0, 1}, {0, 1}}]
1 2

o57 : PolynomialRing

i58 : S = ring H;

One might wish to have a better view of the bidegrees of the variables of the
ring S. An easy way to achieve this, with signs reversed, is to display the
transpose of the matrix of variables.

i59 : transpose vars S

o59 = {2, 2} | $X_1 |
{2, 3} | $X_2 |
{0, -1} | x |
{0, -1} | y |

4 1
o59 : Matrix S <--- S

The internal degrees displayed for the cohomology operators may come as
a surprise. To understand what is going on, recall that these degrees are
determined by a choice of minimal generators for J . At this point we do
not know what is the sequence of generators that Macaulay 2 used, so let’s
compute those generators the way the program did.

i60 : trim J

2 3
o60 = ideal (y , x )

o60 : Ideal of A

Notice that Macaulay 2 has reordered the original sequence of generators.
Now we see that our variable X1, which corresponds to x3, is denoted X_2
by Macaulay 2, and that X2, which corresponds to y2 is denoted X_1. This
explains the bidegrees used by the program.

Display H.
i61 : H

o61 = cokernel {-2, -2} | 0 0 0 0 0 0 0 0 0 0 0 y x 0 0 0 · · ·
{-1, -1} | y 0 0 0 0 x 0 0 0 0 0 0 0 $X_1 0 0 · · ·
{-1, -1} | 0 0 0 y 0 0 0 x 0 0 0 0 0 0 $X_1 0 · · ·
{-1, -1} | 0 y 0 0 x 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-1, -1} | 0 0 y 0 0 0 x 0 0 0 0 0 0 0 0 0 · · ·
{0, 0} | 0 0 0 0 0 0 0 0 y2 xy x2 0 0 0 0 $X_1y · · ·

6
o61 : S-module, quotient of S

That’s a bit large, so we want to look at the even and odd parts separately.
We may compute the even and odd parts of H as the span of the genera-

tors of H with the appropriate parity. Since the two desired functions differ
only in the predicate to be applied, we can generate them both by writing a
function that accepts the predicate as its argument and returns a function.

i62 : partSelector = predicate -> H -> (
R := ring H;
H’ := prune image matrix {

select(
apply(numgens H, i -> H_{i}),
f -> predicate first first degrees source f
)

};
H’);

i63 : evenPart = partSelector even; oddPart = partSelector odd;

Now to obtain the even part, Heven, simply type
i65 : evenPart H



Complete Intersections 155

o65 = cokernel {-2, -2} | 0 0 0 y x 0 0 |
{0, 0} | y2 xy x2 0 0 $X_1y $X_1x |

2
o65 : S-module, quotient of S

Do the same thing for the odd part, Hodd.
i66 : oddPart H

o66 = cokernel {-1, -1} | 0 0 y 0 0 0 x 0 0 0 |
{-1, -1} | 0 y 0 0 x 0 0 0 0 0 |
{-1, -1} | 0 0 0 y 0 0 0 x 0 $X_1 |
{-1, -1} | y 0 0 0 0 x 0 0 $X_1 0 |

4
o66 : S-module, quotient of S

These presentations yield the desired isomorphism of bigraded S-modules.

Here is the source code which implements the routine Ext. It is incorpo-
rated into Macaulay 2.

Code 5.4. The function Ext(M,N) computes Ext•B(M,N) for graded mod-
ules M , N over a graded complete intersection ring B. The function code
can be used to obtain a copy of the source code.

i67 : print code(Ext,Module,Module)
-- ../../../m2/ext.m2:82-171
Ext(Module,Module) := Module => (M,N) -> (

cacheModule := youngest(M,N);
cacheKey := (Ext,M,N);
if cacheModule#?cacheKey then return cacheModule#cacheKey;
B := ring M;
if B =!= ring N
then error "expected modules over the same ring";
if not isCommutative B
then error "’Ext’ not implemented yet for noncommutative rings.";
if not isHomogeneous B
then error "’Ext’ received modules over an inhomogeneous ring";
if not isHomogeneous N or not isHomogeneous M
then error "’Ext’ received an inhomogeneous module";
if N == 0 then B^0
else if M == 0 then B^0
else (

p := presentation B;
A := ring p;
I := ideal mingens ideal p;
n := numgens A;
c := numgens I;
if c =!= codim B
then error "total Ext available only for complete intersections";
f := apply(c, i -> I_i);
pM := lift(presentation M,A);
pN := lift(presentation N,A);
M’ := cokernel ( pM | p ** id_(target pM) );
N’ := cokernel ( pN | p ** id_(target pN) );
C := complete resolution M’;
X := local X;
K := coefficientRing A;
-- compute the fudge factor for the adjustment of bidegrees

fudge := if #f > 0 then 1 + max(first \ degree \ f) // 2 else 0;
S := K(monoid [X_1 .. X_c, toSequence A.generatorSymbols,

Degrees => {
apply(0 .. c-1, i -> {-2, - first degree f_i}),
apply(0 .. n-1, j -> { 0, first degree A_j})
},

Adjust => v -> {- fudge * v#0 + v#1, - v#0},
Repair => w -> {- w#1, - fudge * w#1 + w#0}
]);

-- make a monoid whose monomials can be used as indices
Rmon := monoid [X_1 .. X_c,Degrees=>{c:{2}}];
-- make group ring, so ’basis’ can enumerate the monomials
R := K Rmon;
-- make a hash table to store the blocks of the matrix
blks := new MutableHashTable;
blks#(exponents 1_Rmon) = C.dd;
scan(0 .. c-1, i ->

blks#(exponents Rmon_i) = nullhomotopy (- f_i*id_C));
-- a helper function to list the factorizations of a monomial
factorizations := (gamma) -> (

-- Input: gamma is the list of exponents for a monomial
-- Return a list of pairs of lists of exponents showing the
-- possible factorizations of gamma.
if gamma === {} then { ({}, {}) }
else (

i := gamma#-1;
splice apply(factorizations drop(gamma,-1),

(alpha,beta) -> apply (0..i,
j -> (append(alpha,j), append(beta,i-j))))));

scan(4 .. length C + 1,
d -> if even d then (

scan( exponents \ leadMonomial \ first entries basis(d,R),
gamma -> (

s := - sum(factorizations gamma,
(alpha,beta) -> (

if blks#?alpha and blks#?beta
then blks#alpha * blks#beta
else 0));

-- compute and save the nonzero nullhomotopies
if s != 0 then blks#gamma = nullhomotopy s;
))));

-- make a free module whose basis elements have the right degrees
spots := C -> sort select(keys C, i -> class i === ZZ);
Cstar := S^(apply(spots C,

i -> toSequence apply(degrees C_i, d -> {i,first d})));
-- assemble the matrix from its blocks.
-- We omit the sign (-1)^(n+1) which would ordinarily be used,
-- which does not affect the homology.
toS := map(S,A,apply(toList(c .. c+n-1), i -> S_i),

DegreeMap => prepend_0);
Delta := map(Cstar, Cstar,

transpose sum(keys blks, m -> S_m * toS sum blks#m),
Degree => {-1,0});

DeltaBar := Delta ** (toS ** N’);
assert isHomogeneous DeltaBar;
assert(DeltaBar * DeltaBar == 0);
-- now compute the total Ext as a single homology module
cacheModule#cacheKey = prune homology(DeltaBar,DeltaBar)))

Remark 5.5. The bigraded module TorB• (M,N) is the homology of the
complex (C ⊗A D′) ⊗B N , where C ⊗A D′ is the complex from Theorem
4.1. Observations parallel to Remarks 4.3 and A.1 show that TorB• (M,N)
inherits from D′ a structure of bigraded S-module.

It would be desirable also to have algorithms to compute TorB• (M,N)
in the spirit of the algorithm presented above for Ext•B(M,N). If one of
the modules has finite length, then each TorBn (M,N) is a B-module of finite
length, and the computation of TorB• (M,N) can be reduced to a computation
of Ext by means of Matlis duality, which here can be realized as vector space
duality over the field K. However, in homology there is no equivalent for the
finiteness property described in Remark 4.4; it is an open problem to devise
algorithms that would compute TorB• (M,N) in general.

6 Invariants of Modules

In this section we apply our techniques to develop effective methods for com-
putation (for graded modules over a graded complete intersection) of invari-
ants such as cohomology modules, Poincaré series, Bass series, complexity,
critical degree, and support varieties. For each invariant we produce code that
computes it, and illustrate the action of the code on some explicit example.

Whenever appropriate, we describe open problems on which the com-
putational power of Macaulay 2 could be unleashed.

Notation 4.7 is used consistently throughout the section.

6.1 Cohomology Modules

We call the bigraded R-module P = Ext•B(M,k) the contravariant cohomol-
ogy module of M over B, and the bigraded R-module I = Ext•B(k,M) the
covariant cohomology module of M . Codes that display presentations of the
cohomology modules are presented after a detailed discussion of an example.

Example 6.1.1. Let us create the ring B = K[x, y, z]/(x3, y4, z5).
i68 : A = K[x,y,z];

i69 : J = trim ideal(x^3,y^4,z^5)

3 4 5
o69 = ideal (x , y , z )

o69 : Ideal of A

i70 : B = A/J;

We trimmed the ideal, so that we know the generators Macaulay 2 will use.
This time we want a graded B-module M about whose homology we know

nothing a priori. One way to proceed is to create M as the cokernel of some
random matrix of forms; let’s try a 3 by 2 matrix of quadratic forms.

i71 : f = random (B^3, B^{-2,-3})

o71 = | 27x2+49xy-14y2-23xz-6yz-19z2 38x2y-34xy2+4y3+x2z+16xyz-y2z-5xz · · ·
| -5x2+44xy+38y2+40xz+15yz+4z2 -37x2y+51xy2-36y3+26x2z-38xyz-17y · · ·
| 21x2-30xy+32y2-47xz+7yz-50z2 -6x2y-14xy2-26y3-7x2z+41xyz+50y2z · · ·

3 2
o71 : Matrix B <--- B

We can’t read the second column of that matrix, so let’s display it separately.
i72 : f_{1}

o72 = | 38x2y-34xy2+4y3+x2z+16xyz-y2z-5xz2-6yz2+47z3 |
| -37x2y+51xy2-36y3+26x2z-38xyz-17y2z+17xz2-11yz2+8z3 |
| -6x2y-14xy2-26y3-7x2z+41xyz+50y2z+26xz2+46yz2-44z3 |

3 1
o72 : Matrix B <--- B

Now let’s make the module M .
i73 : M = cokernel f;

We are going to produce isomorphisms of bigraded modules

P even ∼= R[4, 10]⊕ (X1, X2)[2, 7]⊕
(

R

(X1, X2, X3)

)3

⊕R4[2, 7]

P odd ∼=
R

(X1, X2, X3)
[1, 2]⊕

(
R

(X1)

)3

[3, 9]⊕ R

(X1, X2)
[1, 3]⊕R6[3, 9]

over the polynomial ring R = K[X1, X2, X3] over K, bigraded by

Deg(X1) = (−2,−3) Deg(X2) = (−2,−4) Deg(X3) = (−2,−5)

Let’s compute Ext•B(M,B/(x, y, z)) by the routine from Section 5.
i74 : time P = Ext(M,B^1/(x,y,z));

-- used 1.64 seconds

i75 : S = ring P;

Examine the variables of S; due to transposing, their bidegrees are displayed
with the opposite signs.

i76 : transpose vars S

o76 = {2, 3} | $X_1 |
{2, 4} | $X_2 |
{2, 5} | $X_3 |
{0, -1} | x |
{0, -1} | y |
{0, -1} | z |

6 1
o76 : Matrix S <--- S

The variables x, y, and z of A annihilate P , and so appear in many places in
a presentation of P . To reduce the size of such a presentation, we pass to a
ring which eliminates those variables.

i77 : R = K[X_1..X_3,Degrees => {{-2,-3},{-2,-4},{-2,-5}},
Adjust => S.Adjust, Repair => S.Repair];

i78 : phi = map(R,S,{X_1,X_2,X_3,0,0,0})

o78 = map(R,S,{X , X , X , 0, 0, 0})
1 2 3

o78 : RingMap R <--- S

i79 : P = prune (phi ** P);

i80 : transpose vars ring P

o80 = {2, 3} | X_1 |
{2, 4} | X_2 |
{2, 5} | X_3 |

3 1
o80 : Matrix R <--- R

As we planned, the original variables x, y, z, which act trivially on the coho-
mology, are no longer present in the ring. Next we compute presentations

i81 : evenPart P

o81 = cokernel {-4, -10} | 0 0 0 0 0 0 0 0 0 0 |
{-4, -10} | 0 0 0 0 0 0 0 0 0 -X_2 |
{-4, -11} | 0 0 0 0 0 0 0 0 0 X_1 |
{0, 0} | 0 0 X_3 0 0 X_2 0 X_1 0 0 |
{0, 0} | 0 X_3 0 0 X_2 0 X_1 0 0 0 |
{0, 0} | X_3 0 0 X_2 0 0 0 0 X_1 0 |
{-2, -7} | 0 0 0 0 0 0 0 0 0 0 |
{-2, -7} | 0 0 0 0 0 0 0 0 0 0 |
{-2, -7} | 0 0 0 0 0 0 0 0 0 0 |
{-2, -7} | 0 0 0 0 0 0 0 0 0 0 |

10
o81 : R-module, quotient of R

i82 : oddPart P

o82 = cokernel {-1, -2} | X_3 0 X_2 0 0 0 0 X_1 |
{-3, -9} | 0 0 0 0 0 0 X_1 0 |
{-3, -9} | 0 0 0 0 0 X_1 0 0 |
{-3, -9} | 0 0 0 0 X_1 0 0 0 |
{-1, -3} | 0 X_2 0 X_1 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |

11
o82 : R-module, quotient of R

These presentations yield the desired isomorphisms of bigraded R-modules.

The procedure above can be automated by installing a method that will
be run when Ext is presented with a module M and the residue field k. It
displays a presentation of Ext•B(M,k) as a bigraded R-module.

Code 6.1.2. The function changeRing H takes an S-module H and tensors
it with R. It does this by constructing R and a ring homomorphism

ϕ : A[X1, . . . , Xc] = S → R = K[X1, . . . , Xc]

i83 : changeRing = H -> (
S := ring H;
K := coefficientRing S;
degs := select(degrees source vars S,

d -> 0 != first d);
R := K[X_1 .. X_#degs, Degrees => degs,

Repair => S.Repair, Adjust => S.Adjust];
phi := map(R,S,join(gens R,(numgens S - numgens R):0));
prune (phi ** H)
);

Code 6.1.3. The function Ext(M,k) computes Ext•B(M,k) when B is a
graded complete intersection, M a graded B-module, and k is the residue
field of B. The result is presented as a module over the ring k[X1, . . . , Xc].

i84 : Ext(Module,Ring) := (M,k) -> (
B := ring M;
if ideal k != ideal vars B
then error "expected the residue field of the module";
changeRing Ext(M,coker vars B)
);

Example 6.1.4. For a test, we run again the computation for P odd.
i85 : use B;

i86 : k = B/(x,y,z);

i87 : use B;

i88 : P = Ext(M,k);

i89 : time oddPart P
-- used 0.09 seconds

o89 = cokernel {-1, -2} | X_3 0 X_2 0 0 0 0 X_1 |
{-3, -9} | 0 0 0 0 0 0 X_1 0 |
{-3, -9} | 0 0 0 0 0 X_1 0 0 |
{-3, -9} | 0 0 0 0 X_1 0 0 0 |
{-1, -3} | 0 X_2 0 X_1 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |
{-3, -9} | 0 0 0 0 0 0 0 0 |

· · ·
o89 : K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]-module · · ·

1 2 3 · · ·

We also introduce code for computing the covariant cohomology modules.

Code 6.1.5. The function Ext(k,M) computes Ext•B(k,M) when B is a
graded complete intersection, M a graded B-module, and k is the residue
field of B. The result is presented as a module over the ring k[X1, . . . , Xc].

i90 : Ext(Ring,Module) := (k,M) -> (
B := ring M;
if ideal k != ideal vars B
then error "expected the residue field of the module";
changeRing Ext(coker vars B,M)
);

Let’s see the last code in action.

Example 6.1.6. For B and M from Example 6.1.1 we compute the odd part
of the covariant cohomology module Ext•B(k,M).

i91 : time I = Ext(k,M);
-- used 14.81 seconds

i92 : evenPart I

o92 = cokernel {0, 6} | 37X_2 37X_1 |
{0, 6} | -18X_2 -18X_1 |
{0, 6} | -13X_2 -13X_1 |
{0, 6} | -37X_2 -37X_1 |
{0, 6} | 22X_2 22X_1 |
{0, 6} | 0 0 |
{0, 6} | X_2 X_1 |

· · ·
o92 : K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]-module · · ·

1 2 3 · · ·
i93 : oddPart I

o93 = cokernel {-1, 5} | -48X_3 13X_3 34X_3 3X_3 0 0 0 · · ·
{-1, 5} | 3X_3 -40X_3 8X_3 8X_3 0 0 0 · · ·
{-1, 5} | -X_3 37X_3 -13X_3 -35X_3 0 0 0 · · ·
{-1, 4} | 4X_2 20X_2 3X_2 -47X_2 4X_1 20X_1 3X_1 · · ·
{-1, 4} | 0 51X_2 0 -30X_2 0 51X_1 0 · · ·
{-1, 4} | 0 12X_2 0 -3X_2 0 12X_1 0 · · ·
{-1, 4} | 42X_2 12X_2 46X_2 25X_2 42X_1 12X_1 46X_ · · ·
{-1, 4} | 45X_2 24X_2 -14X_2 -35X_2 45X_1 24X_1 -14X · · ·
{-1, 4} | 0 0 X_2 0 0 0 X_1 · · ·
{-1, 4} | X_2 0 0 0 X_1 0 0 · · ·
{-1, 4} | 0 -40X_2 0 10X_2 0 -40X_1 0 · · ·
{-1, 4} | 0 X_2 0 0 0 X_1 0 · · ·
{-1, 3} | 0 0 0 X_1 0 0 0 · · ·

· · ·
o93 : K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]-module · · ·

1 2 3 · · ·

6.2 Poincaré Series

The graded Betti number of M over B is the number βBns(M) of direct sum-
mands isomorphic to the free module B[−s] in the n’th module of a minimal
free resolution of M over B. It can be computed from the equality

βBns(M) = dimK ExtnB(M,k)s

The graded Poincaré series of M over B is the generating function

PBM (t, u) =
∑

n∈N , s∈Z

βBns(M) tnu−s ∈ Z[u, u−1][[t]]

It is easily computable with Macaulay 2 from the contravariant cohomology
module, by using the hilbertSeries routine.

Code 6.2.1. The function poincareSeries2 M computes the graded Poin-
caré series of a graded module M over a graded complete intersection B.

First we set up a ring whose elements can serve as Poincaré series.
i94 : T = ZZ[t,u,Inverses=>true,MonomialOrder=>RevLex];

i95 : poincareSeries2 = M -> (
B := ring M;
k := B/ideal vars B;
P := Ext(M,k);
h := hilbertSeries P;
T’:= degreesRing P;
substitute(h, {T’_0=>t^-1,T’_1=>u^-1})
);

The last line in the code above replaces the variables in the Poincaré series
provided by the hilbertSeries function with the variables in our ring T.

The nth Betti number βBn (M) of M over B is the rank of the nth module
in a minimal resolution of M by free B-modules. The Poincaré series PBM (t)
is the generating function of the Betti numbers. There are expressions

βBn (M) =
∞∑
s=0

βBns(M) and PBM (t) = PBM (t, 1)

Accordingly, the code for PBM (t) just replaces in PBM (t, u) the variable u by 1.

Code 6.2.2. The function poincareSeries1 M computes the Poincaré se-
ries of a graded module M over a graded complete intersection B.

i96 : poincareSeries1 = M -> (
substitute(poincareSeries2 M, {u=>1_T})
);

Now let’s use these codes in computations.

Example 6.2.3. To get a module whose Betti sequence initially decreases,
we form an artinian complete intersection B′ and take M ′ to be a cosyzygy in
a minimal injective resolution of the residue field k. Since B′ is self-injective,
this can be achieved by taking a syzygy of k, then transposing its presentation
matrix. Of course, we ask Macaulay 2 to carry out these steps.

i97 : A’ = K[x,y,z];

i98 : B’ = A’/(x^2,y^2,z^3);

i99 : C’ = res(B’^1/(x,y,z), LengthLimit => 6)

1 3 6 10 15 21 28
o99 = B’ <-- B’ <-- B’ <-- B’ <-- B’ <-- B’ <-- B’

0 1 2 3 4 5 6

o99 : ChainComplex

i100 : M’ = coker transpose C’.dd_5

o100 = cokernel {-5} | -y 0 0 0 z 0 0 0 0 0 0 0 0 0 0 |
{-5} | -x -y 0 0 0 z 0 0 0 0 0 0 0 0 0 |
{-5} | 0 x -y 0 0 0 z 0 0 0 0 0 0 0 0 |
{-5} | 0 0 x -y 0 0 0 z 0 0 0 0 0 0 0 |
{-5} | 0 0 0 -x 0 0 0 0 z 0 0 0 0 0 0 |
{-5} | 0 0 0 0 y 0 0 0 0 0 0 0 0 0 0 |
{-5} | 0 0 0 0 -x y 0 0 0 0 0 0 0 0 0 |
{-5} | 0 0 0 0 0 x y 0 0 0 0 0 0 0 0 |
{-5} | 0 0 0 0 0 0 x y 0 0 0 0 0 0 0 |
{-5} | 0 0 0 0 0 0 0 -x y 0 0 0 0 0 0 |
{-5} | 0 0 0 0 0 0 0 0 x 0 0 0 0 0 0 |
{-6} | 0 0 0 0 0 0 0 0 0 -y 0 z 0 0 0 |
{-6} | 0 0 0 0 0 0 0 0 0 x -y 0 z 0 0 |
{-6} | 0 0 0 0 0 0 0 0 0 0 -x 0 0 z 0 |
{-6} | z2 0 0 0 0 0 0 0 0 0 0 y 0 0 0 |
{-6} | 0 -z2 0 0 0 0 0 0 0 0 0 x y 0 0 |
{-6} | 0 0 z2 0 0 0 0 0 0 0 0 0 -x y 0 |
{-6} | 0 0 0 z2 0 0 0 0 0 0 0 0 0 x 0 |
{-7} | 0 0 0 0 0 0 0 0 0 0 0 0 0 0 z |
{-7} | 0 0 0 0 0 0 0 0 0 z2 0 0 0 0 y |
{-7} | 0 0 0 0 0 0 0 0 0 0 z2 0 0 0 x |

21
o100 : B’-module, quotient of B’

Compute the Poincaré series in two variables PB
′

M ′(t, u).
i101 : poincareSeries2 M’

-7 -6 -5 -6 -5 -4 2 -5 2 - · · ·
3u + 7u + 11u + t*u + 5t*u + 9t*u - 6t u - 14t u · · ·

o101 = --------------------------------------------------------------- · · ·
· · ·
· · ·

o101 : Divide

Example 6.2.4. We compute PBM (t) for the module M from Example 6.1.1.
i102 : p = poincareSeries1 M

2 3 4 5 6 7
3 + 2t - 5t + 4t + 12t + t - 4t - t

o102 = -----------------------------------------
2 2 2

(1 - t )(1 - t )(1 - t )

o102 : Divide

We have written some rather näıve code for simplifying rational functions as
above. It locates factors of the form 1 − tn in the denominator, factors out

1−t, and factors out 1+t if n is even. Keeping the factors of the denominator
separate, it then cancels as many of them as it can with the numerator.

i103 : load "simplify.m2"

i104 : simplify p

2 3 4 5 6
3 - t - 4t + 8t + 4t - 3t - t

o104 = ----------------------------------
2 3

(1 + t) (1 - t)

o104 : Divide

In this case, it succeeded in canceling a factor of 1 + t.

Example 6.2.5. We compute some Betti numbers for M . We use the di-
vision operation in the Euclidean domain T ′ = Q[t, t−1] with the reverse
monomial ordering to compute power series expansions.

i105 : T’ = QQ[t,Inverses=>true,MonomialOrder=>RevLex];

i106 : expansion = (n,q) -> (
t := T’_0;
rho := map(T’,T,{t,1});
num := rho value numerator q;
den := rho value denominator q;
n = n + first degree den;
n = max(n, first degree num + 1);
(num + t^n) // den
);

Now let’s expand the Poincaré series up to t20.
i107 : expansion(20,p)

2 3 4 5 6 7 8 9 · · ·
o107 = 3 + 2t + 4t + 10t + 15t + 25t + 32t + 46t + 55t + 73t + · · ·

o107 : T’

Just to make sure, let’s compare the first few coefficients with the more
pedestrian way of doing the computation, one Ext module at a time.

i108 : psi = map(K,B)

o108 = map(K,B,{0, 0, 0})

o108 : RingMap K <--- B

i109 : apply(10, i -> rank (psi ** Ext^i(M,coker vars B)))

o109 = {3, 2, 4, 10, 15, 25, 32, 46, 55, 73}

o109 : List

Now we restore t to its original use.
i110 : use T;

6.3 Complexity

The complexity of M is the least d ∈ N such that the function

n 7→ dimK ExtnB(M,k)

is bounded above by a polynomial of degree d− 1 (with the convention that
the zero polynomial has degree −1). This number, denoted cxB(M), was
introduced in [2] to measures on a polynomial scale the rate of growth of the
Betti numbers of M . It is calibrated so that cxB(M) = 0 if and only if M
has finite projective dimension. Corollary 4.11 yields

PBM (t) =
pBM (t)

(1− t2)c
for some pBM (t) ∈ Z[t]

Decomposing the right hand side into partial fractions, one sees that cxR(M)
equals the order of the pole of PBM (t) at t = 1; in particular, cxR(M,N) ≤ c.
However, since we get PBM (t) from a computation of the R-module P =
Ext•B(M,k), it is natural to obtain cxR(M) as the Krull dimension of P .

Code 6.3.1. The function complexity M yields the complexity of a graded
module M over a graded complete intersection ring B.

i111 : complexity = M -> dim Ext(M,coker vars ring M);

Example 6.3.2. We compute cxB(M) for M from Example 6.1.1.
i112 : complexity M

o112 = 3

6.4 Critical Degree

The critical degree of M is the least integer ` for which the minimal reso-
lution F of M admits a chain map g : F → F of degree m < 0, such that
gm+n : Fm+n → Fn is surjective for all n > `. This number, introduced in [6]
and denoted crdegBM , is meaningful over every graded ring B. It is equal
to the projective dimension whenever the latter is finite.

WhenB is a complete intersection it is proved in [6, Sect. 7] that crdegBM
is finite and yields important information on the Betti sequence:

• if cxBM ≤ 1, then βBn (M) = βBn+1(M) for all n > crdegBM .
• if cxBM ≥ 2, then βBn (M) < βBn+1(M) for all n > crdegBM .

Thus, it is interesting to know crdegBM , or at least to have a good upper
bound. Here is what is known, in terms of h = depthB − depthBM .

• if cxBM = 0, then crdegBM = h.
• if cxBM = 1, then crdegBM ≤ h.
• if cxBM = 2, then crdegBM ≤ h+ 1 + max{2βBh (M)− 1 , 2βBh+1(M)}.

The first part is the Auslander-Buchsbaum Equality, the second part is proved
in [11, Sect. 6], the third is established in [4, Sect. 7].

These upper bounds are realistic: there exist examples in complexity 1
when they are reached, and examples in complexity 2 when they are not
more than twice the actual value of the critical degree. If cxRM ≥ 3, then
it is an open problem whether the critical degree of M can be bounded in
terms that do not depend on the action of the cohomology operators.

However, in every concrete case crdegRM can be computed explicitly by
using Macaulay 2. Indeed, it is proved in [6, Sect. 7] that crdegRM is equal
to the highest degree of a non-zero element in the socle of the R-module
Ext•B(M,k), that is, the submodule consisting of elements annihilated by
(X1, . . . , Xc). The socle is naturally isomorphic to HomB(k,Ext•B(M,k)), so
it can be obtained by standard Macaulay 2 routines.

For instance, for the module M from Example 6.1.1, we get
i113 : k = coker vars ring H;

i114 : prune Hom(k,H)

o114 = 0

o114 : K [$X , $X , x, y, Degrees => {{-2, -2}, {-2, -3}, {0, 1}, {0, · · ·
1 2

The degrees displayed above show that crdegRM = 1.

Of course, one might prefer to see the number crdegBM directly.

Code 6.4.1. The function criticalDegree M computes the critical degree
of a graded module M over a graded complete intersection ring B.

i115 : criticalDegree = M -> (
B := ring M;
k := B / ideal vars B;
P := Ext(M,k);
k = coker vars ring P;
- min ( first \ degrees source gens prune Hom(k,P))
);

Let’s test the new code in a couple of cases.

Example 6.4.2. For the module M of Example 6.1.1 we have
i116 : criticalDegree M

o116 = 1

in accordance with what was already observed above.
For the module M ′ of Example 6.2.3 we obtain
i117 : criticalDegree M’

o117 = 5

6.5 Support Variety

Let K denote an algebraic closure of K. The support variety V∗B(M) is
the algebraic set in Kc defined by the annihilator of Ext•B(M,k) over R =
K[X1, . . . , Xc]. This ‘geometric image’ of the contravariant cohomology mod-
ule was introduced in [2] and used to study the minimal free resolution of
M . The dimension of the support variety is equal to the complexity cxR(M),
that we can already compute. There is no need to associate a variety to the
covariant cohomology module, see 7.4.

Since V∗B(M) is defined by homogeneous equations, it is a cone in Kc. An
important open problem is whether every cone in Kc that can be defined
over K is the variety of some B-module M . By [2, Sect. 6] all linear subspaces
and all hypersurfaces arise in this way, but little more is known in general.

Feeding our computation of Ext•B(M,k) to standard Macaulay 2 routines
we write code for determining a set of equations defining V∗B(M).

Code 6.5.1. The function supportVarietyIdeal M yields a set of polyno-
mial equations with coefficients in K, defining the support variety V∗B(M) in
Kc for a graded module M over a graded complete intersection B.

i118 : supportVarietyIdeal = M -> (
B := ring M;
k := B/ideal vars B;
ann Ext(M,k)
);

As before, we illustrate the code with explicit computations. In view of
the open problem mentioned above, we fix a ring and a type of presentation,
then change randomly the presentation matrix in the hope of finding an
‘interesting’ variety. The result of the experiment is assessed in Remark 6.5.3.

Example 6.5.2. Let F7 denote the prime field with 7 elements, and form
the zero-dimensional complete intersection B′′ = F7[x, y, z]/(x7, y7, z7).

i119 : K’’ = ZZ/7;

i120 : A’’ = K’’[x,y,z];

i121 : J’’ = ideal(x^7,y^7,z^7);

o121 : Ideal of A’’

i122 : B’’ = A’’/J’’;

We apply the code above to search, randomly, for some varieties. Using scan
we print the results from several runs with one command.

i123 : scan((1,1) .. (3,3), (r,d) -> (
V := cokernel random (B’’^r,B’’^{-d});
<< "--------------------------------------------------- · · ·
<< endl
<< "V = " << V << endl
<< "support variety ideal = "
<< timing supportVarietyIdeal V
<< endl))

------------------------------------------------------------------
V = cokernel | -2x+3y+2z |
support variety ideal = ideal (X - 2X , X + X )

2 3 1 3
-- 0.7 seconds

------------------------------------------------------------------
V = cokernel | 3x2-2xy+xz-3yz |
support variety ideal = ideal(X + 3X + 2X )

1 2 3
-- 0.48 seconds

------------------------------------------------------------------
V = cokernel | -2x3+3x2y+y3-x2z-3y2z-xz2-3z3 |
support variety ideal = 0

-- 1.54 seconds
------------------------------------------------------------------
V = cokernel | -3y+3z |

| -2x-2y |
support variety ideal = ideal(X + X - X )

1 2 3
-- 0.86 seconds

------------------------------------------------------------------
V = cokernel | -x2+2y2-xz+yz+3z2 |

| 2xy-3xz-3yz-2z2 |
support variety ideal = 0

-- 1.31 seconds
------------------------------------------------------------------
V = cokernel | -x3-2x2y-xy2-2xyz+3y2z+2xz2-yz2-2z3 |

| 2xy2+3y3-3x2z-2y2z+2xz2+2yz2 |
support variety ideal = 0

-- 2.21 seconds
------------------------------------------------------------------
V = cokernel | 3x-y-z |

| -3x-y+2z |
| x-2y+3z |

support variety ideal = 0
-- 1.1 seconds

------------------------------------------------------------------
V = cokernel | 2x2-2xy+2y2+2xz-3z2 |

| -x2+2xy+y2+3xz+3yz-z2 |
| -2xz+2yz+2z2 |

support variety ideal = 0
-- 1.67 seconds

------------------------------------------------------------------
V = cokernel | 2x3-x2y+2xy2-y3-2xyz+3y2z+xz2+3yz2+z3 |

| -3x3-3x2y+3xy2+2x2z+3xyz-3y2z-xz2 |
| -3x3-2x2y-xy2-2y3-2xyz+y2z+xz2+3yz2-z3 |

support variety ideal = 0
-- 1.92 seconds

Remark 6.5.3. The (admittedly short) search above did not turn up any
non-linear variety. This should be contrasted with the known result that every
cone in F7

3 is the support variety of some B′′-module.
Indeed, B′′ is isomorphic to the group algebra F7[G] of the elementary

abelian group G = C7 × C7 × C7, where C7 is a cyclic group of order 7. It
is shown in [2, Sect. 7] that V∗B′′(V ) is equal to a variety V∗G(V ), defined
in a different way in [9] by Carlson. He proves in [10] that if K is a field of

characteristic p > 0, and G is an elementary abelian p-group of rank c, then
every cone in Kc is the rank variety of a finitely generated module over K[G].

6.6 Bass Series

The graded Bass number µnsB (M) of M over B is the number of direct sum-
mands isomorphic to U [s] in the n’th module of a minimal graded injective
resolution of M over B, where U is the injective envelope of k. It satisfies

µnsB (M) = dimK ExtnB(k,M)s

The graded Bass series of M over B is the generating function

IMB (t, u) =
∑

n∈N , s∈Z

µnsB (M) tnus ∈ Z[u, u−1][[t]]

It is easily computable with Macaulay 2 from the covariant cohomology mod-
ule, by using the hilbertSeries routine.

Code 6.6.1. The function bassSeries2 M computes the graded Bass series
of a graded module M over a graded complete intersection B.

i124 : bassSeries2 = M -> (
B := ring M;
k := B/ideal vars B;
I := Ext(k,M);
h := hilbertSeries I;
T’:= degreesRing I;
substitute(h, {T’_0=>t^-1, T’_1=>u})
);

As with Betti numbers and Poincaré series, there are ungraded versions
of Bass numbers and Bass series; they are given, respectively, by

µnB(M) =
∞∑
s=0

µnsB (M) and IMB (t) = IMB (t, 1)

Code 6.6.2. The function bassSeries1 M computes the Bass series of a
graded module M over a graded complete intersection B.

i125 : bassSeries1 = M -> (
substitute(bassSeries2 M, {u=>1_T})
);

Now let’s use these codes in computations.

Example 6.6.3. For k, the residue field of B, the contravariant and covari-
ant cohomology modules coincide. For comparison, we compute side by side
the Poincaré series and the Bass series of k, when B = K[x, y, z]/(x3, y4, z5)
is the ring defined in Example 6.1.1.

i126 : use B;

i127 : L = B^1/(x,y,z);

i128 : p = poincareSeries2 L

2 2 3 3
1 + 3t*u + 3t u + t u

o128 = ------------------------------
2 3 2 4 2 5

(1 - t u )(1 - t u )(1 - t u )

o128 : Divide

i129 : b = bassSeries2 L

-1 2 -2 3 -3
1 + 3t*u + 3t u + t u

o129 = ---------------------------------
2 -3 2 -4 2 -5

(1 - t u )(1 - t u )(1 - t u )

o129 : Divide

The reader would have noticed that the two series are different, and that one
is obtained from the other by the substitution u 7→ u−1. This underscores
the different meanings of the graded Betti numbers and Bass numbers.

Example 6.6.4. Here we compute the graded and ungraded Bass series of
the B-module M of Example 6.1.1.

i130 : b2 = bassSeries2 M

6 3 4 5 2 2 2 3 3 3 3 2 · · ·
7u + t*u + 9t*u + 3t*u - t u - t u - 4t - 3t u - 3t u + · · ·

o130 = --------------------------------------------------------------- · · ·
2 -3 2 -4 2 -5

(1 - t u )(1 - t u )(1 - t u )

o130 : Divide

i131 : b1 = bassSeries1 M;

i132 : simplify b1

2 3 4
7 + 6t - 8t - 2t + 3t

o132 = ------------------------
2 3

(1 + t) (1 - t)

o132 : Divide

7 Invariants of Pairs of Modules

In this final section we compute invariants of a pair (M,N) of graded modules
over a graded complete intersection B, derived from the reduced Ext module
ext•B(M,N) defined in Remark 4.10. The treatment here is parallel to that
in Section 6. When one of the modules M or N is equal to the residue field
k, the invariants discussed below reduce to those treated in that section.

7.1 Reduced Ext Module

The reduced Ext module ext•B(M,N) = Ext•B(M,N)⊗Ak defined in Remark
4.10 is computed from our basic routine Ext(M,N) by applying the function
changeRing defined in Code 6.1.2.

Code 7.1.1. The function ext(M,N) computes ext•B(M,N) when M and N
are graded modules over a graded complete intersection B.

i133 : ext = (M,N) -> changeRing Ext(M,N);

Example 7.1.2. Using the ring B = K[x, y, z]/(x3, y4, z5) and the module
M created in Example 6.1.1, we make new modules

N = B/(x2 + z2 , y3) and N ′ = B/(x2 + z2 , y3 − 2z3)

i134 : use B;

i135 : N = B^1/(x^2 + z^2,y^3);

i136 : time rH = ext(M,N);
-- used 15.91 seconds

i137 : evenPart rH

o137 = cokernel {-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{0, 2} | 0 0 0 0 0 0 0 X_3 0 0 0 · · ·
{0, 2} | 0 0 0 0 0 X_3 0 0 0 0 0 · · ·
{0, 2} | 0 0 0 0 0 0 X_3 0 0 0 0 · · ·
{0, 2} | 0 0 0 0 X_3 0 0 0 0 0 0 · · ·
{0, 2} | 0 0 0 X_3 0 0 0 0 0 0 0 · · ·
{0, 2} | 0 0 X_3 0 0 0 0 0 0 0 0 · · ·
{0, 2} | 0 X_3 0 0 0 0 0 0 0 0 X_2 · · ·
{0, 2} | X_3 0 0 0 0 0 0 0 0 X_2 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{0, 1} | 0 0 0 0 0 0 0 0 X_2 0 0 · · ·

· · ·
o137 : K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]-modul · · ·

1 2 3 · · ·
i138 : oddPart rH

o138 = cokernel {-3, -6} | 0 0 0 0 0 0 0 0 0 0 · · ·
{-3, -6} | 0 0 0 0 0 0 0 0 0 0 · · ·
{-3, -6} | 0 0 0 0 0 0 0 0 0 0 · · ·
{-3, -6} | 0 0 0 0 0 0 0 0 0 0 · · ·
{-3, -6} | 0 0 0 0 0 0 0 0 0 0 · · ·
{-1, -1} | -39X_3 0 0 0 0 0 0 0 X_2 0 · · ·
{-1, -1} | 31X_3 0 0 0 0 0 0 X_2 0 0 · · ·
{-1, -1} | -34X_3 0 0 0 0 0 X_2 0 0 0 · · ·
{-1, -1} | -35X_3 0 0 0 0 X_2 0 0 0 0 · · ·

{-1, -1} | -29X_3 0 0 0 X_2 0 0 0 0 0 · · ·
{-1, -1} | 12X_3 0 0 X_2 0 0 0 0 0 0 · · ·
{-1, -1} | -8X_3 0 X_2 0 0 0 0 0 0 0 · · ·
{-1, -1} | X_3 X_2 0 0 0 0 0 0 0 0 · · ·
{-3, -7} | 0 0 0 0 0 0 0 0 0 X_1 · · ·

· · ·
o138 : K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]-modul · · ·

1 2 3 · · ·
i139 : N’ = B^1/(x^2 + z^2,y^3 - 2*z^3);

i140 : time rH’ = ext(M,N’);
-- used 20.26 seconds

i141 : evenPart rH’

o141 = cokernel {-4, -8} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-4, -8} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-4, -9} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{0, 2} | 0 0 0 0 0 0 X_3 0 0 0 0 · · ·
{0, 2} | 0 0 0 0 0 X_3 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{0, 2} | 0 0 0 0 X_3 0 0 0 0 0 0 · · ·
{0, 2} | 0 0 X_3 0 0 0 0 0 0 0 X_2 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{0, 2} | 0 0 0 X_3 0 0 0 0 0 0 0 · · ·
{0, 2} | 0 X_3 0 0 0 0 0 0 0 X_2 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{-2, -4} | 0 0 0 0 0 0 0 0 0 0 0 · · ·
{0, 2} | X_3 0 0 0 0 0 0 0 0 0 0 · · ·
{0, 1} | 0 0 0 0 0 0 0 X_2 0 0 0 · · ·
{0, 1} | 0 0 0 0 0 0 0 0 X_2 0 0 · · ·

· · ·
o141 : K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]-modul · · ·

1 2 3 · · ·
i142 : oddPart rH’

o142 = cokernel {-3, -6} | 0 0 0 0 0 0 0 0 -42X_2 21X_ · · ·
{-3, -6} | 0 0 0 0 0 0 0 0 -6X_2 -32X · · ·
{-3, -6} | 0 0 0 0 0 0 0 0 -8X_2 12X_ · · ·
{-3, -6} | 0 0 0 0 0 0 0 0 26X_2 -36X · · ·
{-3, -6} | 0 0 0 0 0 0 0 0 50X_2 18X_ · · ·

{-3, -6} | 0 0 0 0 0 0 0 0 31X_2 7X_2 · · ·
{-3, -7} | 0 0 0 0 0 0 0 0 0 0 · · ·
{-3, -7} | 0 0 0 0 0 0 0 0 0 0 · · ·
{-3, -7} | 0 0 0 0 0 0 0 0 0 0 · · ·
{-3, -7} | 0 0 0 0 0 0 0 0 0 0 · · ·
{-3, -7} | 0 0 0 0 0 0 0 0 0 X_1 · · ·
{-3, -7} | 0 0 0 0 0 0 0 0 X_1 0 · · ·
{-1, -2} | 0 0 0 X_2 0 0 0 X_1 0 0 · · ·
{-1, -2} | 0 0 X_2 0 0 0 X_1 0 0 0 · · ·
{-1, -2} | 0 X_2 0 0 0 X_1 0 0 0 0 · · ·
{-1, -2} | X_2 0 0 0 X_1 0 0 0 0 0 · · ·

· · ·
o142 : K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]-modul · · ·

1 2 3 · · ·

7.2 Ext-generator Series

The Ext-generator series GM,N
B (t, u) defined in Remark 4.10 generalizes both

the Poincaré series of M and the Bass series of N , as seen from the formulas

PBM (t, u) = GM,k
B (t, u) and INB (t, u) = Gk,NB (t, u−1)

Similar equalities hold for the corresponding series in one variable. Codes for
computing Ext-generator series are easy to produce.

Code 7.2.1. The function extgenSeries2(M,N) computes GM,N
B (t, u) when

M and N are graded modules over a graded complete intersection B, and
presents it as a rational function with denominator (1− t2ur1) · · · (1− t2urc).

i143 : extgenSeries2 = (M,N) -> (
H := ext(M,N);
h := hilbertSeries H;
T’:= degreesRing H;
substitute(h, {T’_0=>t^-1,T’_1=>u^-1})
);

Code 7.2.2. The function extgenSeries1(M,N) computes the Ext-genera-
tor series in one variable for a pair (M,N) of graded modules over a graded
complete intersection B.

i144 : extgenSeries1 = (M,N) -> (
substitute(extgenSeries2(M,N), {u=>1_T})
);

Example 7.2.3. For M , N , and N ′ as in Example 7.1.2 we obtain
i145 : time extgenSeries2(M,N)

-- used 0.44 seconds

-2 -1 2 2 2 2 3 2 4 3 4 3 · · ·
8u + u + 8t*u - 8t u - 9t u - 9t u + 7t u - 8t u - 8t u · · ·

o145 = --------------------------------------------------------------- · · ·
· · ·
· · ·

o145 : Divide

i146 : g=time extgenSeries1(M,N)
-- used 0.13 seconds

2 3 4 5 6 7
9 + 8t - 19t - 11t + 17t + 4t - 7t - t

o146 = --------------------------------------------
2 2 2

(1 - t )(1 - t )(1 - t )

o146 : Divide

i147 : simplify g

2 3 4
9 - t - 9t + 6t + t

o147 = ----------------------
2

(1 + t)(1 - t)

o147 : Divide

i148 : time extgenSeries2(M,N’)
-- used 0.15 seconds

-2 -1 2 2 2 2 2 3 2 4 3 5 · · ·
7u + 2u + 4t*u - 7t u - 9t u - 9t u + 16t u - 4t u + 2 · · ·

o148 = --------------------------------------------------------------- · · ·
· · ·
· · ·

o148 : Divide

i149 : g’=time extgenSeries1(M,N’)
-- used 0.18 seconds

2 3 4 5 6
9 + 4t - 9t + 4t + 8t - 2t - 2t

o149 = ------------------------------------
2 2 2

(1 - t )(1 - t )(1 - t )

o149 : Divide

i150 : simplify g’

2 3 5
9 - 5t - 4t + 8t - 2t

o150 = ------------------------
2 3

(1 + t) (1 - t)

o150 : Divide

7.3 Complexity

The complexity of a pair of B-modules (M,N) is the least d ∈ N such that
there exists a polynomial of degree d− 1 bounding above the function

n 7→ dimK extnB(M,N)



Complete Intersections 175

It is denoted cxB(M,N) and measures on a polynomial scale the rate of
growth of the minimal number of generators of ExtnB(M,N); it vanishes if
and only if ExtnB(M,N) = 0 for all n� 0. Corollary 4.11 yields

GM,N
B (t) =

h(t)
(1− t2)c

for some h(t) ∈ Z[t]

so decomposition into partial fractions shows that cxR(M,N) equals the order
of the pole of GM,N

B (t) at t = 1. Alternatively, cxR(M,N) can be obtained
by computing the Krull dimension of a reduced Ext module over R.

Code 7.3.1. The function complexityPair(M,N) yields the complexity of
a pair (M,N) of graded modules over a graded complete intersection ring B.

i151 : complexityPair = (M,N) -> dim ext(M,N);

Example 7.3.2. For M , N , and N ′ as in Example 7.1.2 we have
i152 : time complexityPair(M,N)

-- used 0.39 seconds

o152 = 2

i153 : time complexityPair(M,N’)
-- used 0.12 seconds

o153 = 3

7.4 Support Variety

Let K be an algebraic closure of K. The support variety V∗B(M,N) is the
algebraic set in Kc defined by the annihilator of ext•B(M,N) over R =
K[X1, . . . , Xc]. It is clear from the definition that V∗B(M,k) is equal to
the variety V∗B(M) defined in 6.5. One of the main results of [5, Sect. 5]
shows that V∗B(M,N) = V∗B(M)∩V∗B(N), and, as a consequence, V∗B(M) =
V∗B(M,M) = V∗B(k,M). The dimension of V∗B(M,N) is equal to the com-
plexity cxR(M,N), already computed above.

Feeding our computation of ext•B(M,N) to standard Macaulay 2 routines
we write code for determining a set of equations defining V∗B(M,N).

Code 7.4.1. The function supportVarietyPairIdeal(M,N) yields a set of
polynomial equations with coefficients in K, defining the variety V∗B(M,N)
in Kc for graded modules M , N over a graded complete intersection B.

i154 : supportVarietyPairIdeal = (M,N) -> ann ext(M,N);

Example 7.4.2. For M , N , and N ′ as in Example 7.1.2 we have
i155 : time supportVarietyPairIdeal(M,N)

-- used 0.97 seconds

o155 = ideal X
1

o155 : Ideal of K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]
1 2 3

i156 : time supportVarietyPairIdeal(M,N’)
-- used 1.73 seconds

o156 = 0

o156 : Ideal of K [X , X , X , Degrees => {{-2, -3}, {-2, -4}, {-2, -5}}]
1 2 3

Appendix A. Gradings

Our purpose here is to set up a context in which the theory of Sections 3 and
4 translates into data that Macaulay 2 can operate with.

A first point is to develop a flexible and consistent scheme within which
to handle the two kinds of degrees we deal with—the internal gradings of the
input, and the homological degrees created during computations.

A purely technological difficulty arises when our data are presented to
Macaulay 2: it only accepts multidegrees whose first component is positive,
which is not the case for rings of cohomology operators.

A final point, mostly notational, tends to generate misunderstanding and
errors if left unaddressed. On the printed page, the difference between ho-
mological and cohomological conventions is handled graphically by switching
between sub- and super-indices, and reversing signs; both authors were used
to it, but Macaulay 2 has so far refused to read TEX printouts.

The raison d’être of the following remarks was to debug communications
between the three of us.

Remark A.1. Only one degree, denoted deg, appears in Section 2, and any-
where in the main text before Notation 4.7; when needed, it will be referred
to as homological degree.

Assume that A =
⊕

h∈ZAh is a graded ring. Any element a of Ah is said
to be homogeneous of internal degree h; the notation for this is deg′ a = h. Let
f = {f1, . . . , fc} be a Koszul-regular set consisting of homogeneous elements.
We give the ring B = A/(f) the induced grading, and extend the notation
for internal degree to all graded B-modules M .

Let M be a graded B-module. For any integer e, we let M [e] denote the
graded module with M [e]d = Md+e. We take a projective resolution C of M
by graded A-modules, with differential dC preserving internal degrees. Recall
that we have been writing deg x = n to indicate that x is an element in Cn;
we refer to this situation also by saying that x has homological degree x. We
combine both degrees in a single bidegree, denoted Deg, as follows:

Deg x = (deg x,deg′ x)

For a bigraded module H and pair of integers (e, e′), we let H[e, e′] denote
the bigraded module with H[e, e′]d,d′ = Hd+e,d′+e′ .

Because deg Yi = 2, the elements of the free B-module Q have homological
degree 2. We introduce an internal grading deg′ on Q by setting deg′ Yi = ri,

where ri = deg′ fi for i = 1, . . . , c. With this choice, the homomorphism
f : Q → A acquires internal degree 0 (of course, this was the reason behind
our choice of grading in the first place). The internal grading on Q defines, in
the usual way, internal gradings on all symmetric and exterior powers of Q
and Q∗; in particular, deg′ Y (α) =

∑
αiri and deg′ Y ∧β =

∑
βiri. Thus, the

ring S = A[X1, . . . , Xc] acquires a bigrading defined by Deg a = (0, h) for all
elements a ∈ Ah and DegXi = (−2,−ri) for i = 1, . . . , c.

In this context, we call S the bigraded ring of cohomology operators.
Since the differential dC has internal degree 0, a null-homotopic chain map

C → C which is homogeneous of internal degree r will have a null-homotopy
that is itself homogeneous of internal degree r. In the proof of Theorem
3.1 we construct maps dγ as null-homotopies, so we may arrange for them
to be homogeneous maps with deg′ dγ =

∑
γidi. Our grading assumptions

guarantee that d is homogeneous with Deg d = (−1, 0).
With these data, the B-free resolution C ⊗A D′ provided by Theorem

4.1 becomes one by graded B-modules, and its differential ∂ is homogeneous
with Deg ∂ = (−1, 0). For any graded B-module N , these properties are
transferred to the complex HomB(C ⊗A D′, N) and its differential.

We sum up the contents of Remarks 4.3 and A.1.

Remark A.2. If A is a graded ring, {f1, . . . , fc} is a Koszul-regular set
consisting of homogeneous elements, B is the residue ring A/(f), and M,N
are graded B-modules, then Ext•B(M,N) is a bigraded module over the ring
S = A[X1, . . . , Xc], itself bigraded by setting Deg a = (0,deg′(a)) for all
homogeneous a ∈ A and DegXi = (−2,−deg′(fi)) for i = 1, . . . , c.

Remark A.3. The core algorithms of the program can handle multi-graded
rings and modules, but only if each variable in the ring has positive first
component of its multi-degree. At the moment, a user who needs a multi-
graded ring R which violates this requirement must provide two linear maps:
R.Adjust, that transforms the desired multi-degrees into ones satisfying this
requirement, as well as its inverse map, R.Repair. The routine Ext, discussed
above, incorporates such adjustments for the rings of cohomology operators
it creates. When we wish to create related rings with some of the same multi-
degrees, we may use the same adjustment operator.

Algorithms for the Toric Hilbert Scheme

Michael Stillman, Bernd Sturmfels, and Rekha Thomas

The toric Hilbert scheme parametrizes all algebras isomorphic to a given semi-
group algebra as a multigraded vector space. All components of the scheme
are toric varieties, and among them, there is a fairly well understood co-
herent component. It is unknown whether toric Hilbert schemes are always
connected. In this chapter we illustrate the use of Macaulay 2 for exploring
the structure of toric Hilbert schemes. In the process we will encounter al-
gorithms from commutative algebra, algebraic geometry, polyhedral theory
and geometric combinatorics.

Introduction

Consider the multigrading of the polynomial ring R = C[x1, . . . , xn] specified
by a non-negative integer d × n-matrix A = (a1, . . . , an) such that degree
(xi) = ai ∈ Nd. This defines a decomposition R =

⊕
b∈NARb, where NA is

the subsemigroup of Nd spanned by a1, . . . , an, and Rb is the C-span of all
monomials xu = xu1

1 · · ·xunn with degree Au = a1u1 + · · · + anun = b. The
toric Hilbert scheme HilbA parametrizes all A-homogeneous ideals I ⊂ R
(ideals that are homogeneous under the multigrading of R by NA) with the
property that (R/I)b is a 1-dimensional C-vector space, for all b ∈ NA. We
call such an ideal I an A-graded ideal. Equivalently, I is A-graded if it is
A-homogeneous and R/I is isomorphic as a multigraded vector space to the
semigroup algebra C[NA] = R/IA, where

IA := 〈xu − xv : Au = Av〉 ⊂ R

is the toric ideal of A. An A-graded ideal is generated by binomials and
monomials in R since, by definition, any two monomials xu and xv of the
same degree Au = Av must be C-linearly dependent modulo the ideal.

We recommend [22, §4, §10] as an introductory reference for the topics
in this chapter. The study of toric Hilbert schemes for d = 1 goes back to
Arnold [1] and Korkina et al.[13], and it was further developed by Sturmfels
([21] and [22, §10]). Peeva and Stillman [17] introduced the scheme structure
that gives the toric Hilbert scheme its universal property, and from this they
derive a formula for the tangent space of a point on HilbA. Maclagan recently
showed that the quadratic binomials in [21, §5] define the same scheme as
the determinantal equations in [17]. Both of these systems of global equations
are generally much too big for practical computations. Instead, most of our
algorithms are based on the local equations given by Peeva and Stillman in
[16] and the combinatorial approach of Maclagan and Thomas in [14].



180 M. Stillman, B. Sturmfels, and R. Thomas

We begin with the computation of a toric ideal using Macaulay 2. Our
running example throughout this chapter is the following 2× 5-matrix:

A =
(

1 1 1 1 1
0 1 2 7 8

)
, (1)

which we input to Macaulay 2 as a list of lists of integers.
i1 : A = {{1,1,1,1,1},{0,1,2,7,8}};

The toric ideal of A lives in the multigraded ring R := C[a, b, c, d, e].
i2 : R = QQ[a..e,Degrees=>transpose A];

i3 : describe R

o3 = QQ [a, b, c, d, e, Degrees => {{1, 0}, {1, 1}, {1, 2}, {1, 7}, {1 · · ·
We use Algorithm 12.3 in [22] to compute IA. The first step is to find a

matrix B whose rows generate the lattice kerZ(A) := {x ∈ Zn : Ax = 0}.
i4 : B = transpose syz matrix A

o4 = | 1 -2 1 0 0 |
| 0 5 -6 1 0 |
| 0 6 -7 0 1 |

3 5
o4 : Matrix ZZ <--- ZZ

Although in theory any basis of kerZ(A) will suffice, in practice it is
more efficient to use a reduced basis [20, §6.2], which can be computed using
the basis reduction package LLL.m2 in Macaulay 2. The command LLL when
applied to the output of syz matrix A will return a matrix of the same size
whose columns form a reduced lattice basis for kerZ(A). The output appears
in compressed form as follows:

i5 : load "LLL.m2";

i6 : LLL syz matrix A

o6 = | 0 1 2 |
| 1 -1 0 |
| -1 0 -3 |
| -1 -1 2 |
| 1 1 -1 |

5 3
o6 : Matrix ZZ <--- ZZ

We recompute B using this package to get the following 3× 5 matrix.
i7 : B = transpose LLL syz matrix A

o7 = | 0 1 -1 -1 1 |
| 1 -1 0 -1 1 |
| 2 0 -3 2 -1 |

3 5
o7 : Matrix ZZ <--- ZZ

The advantage of a reduced basis may not be apparent in small examples.
However, as the size of A increases, it becomes increasingly important for the
termination of Algorithm 12.3 in [22]. (To appreciate this, consider the matrix
(7) from Section 4.)

A row b = b+− b− of B is then coded as the binomial xb
+ −xb− ∈ R, and

we let J be the ideal generated by all such binomials.
i8 : toBinomial = (b,R) -> (

top := 1_R; bottom := 1_R;
scan(#b, i -> if b_i > 0 then top = top * R_i^(b_i)

else if b_i < 0 then bottom = bottom * R_i^(-b_i));
top - bottom);

i9 : J = ideal apply(entries B, b -> toBinomial(b,R))

2 2 3
o9 = ideal (- c*d + b*e, - b*d + a*e, a d - c e)

o9 : Ideal of R

The toric ideal equals (J : (x1 · · ·xn)∞), which is computed via n successive
saturations as follows:

i10 : scan(gens ring J, f -> J = saturate(J,f))

Putting the above pieces of code together, we get the following procedure
for computing the toric ideal of a matrix A.

i11 : toricIdeal = (A) -> (
n := #(A_0);
R = QQ[vars(0..n-1),Degrees=>transpose A,MonomialSize=>16];
B := transpose LLL syz matrix A;
J := ideal apply(entries B, b -> toBinomial(b,R));
scan(gens ring J, f -> J = saturate(J,f));
J
);

See [2], [11] and [22, §4, §12] for other algorithms for computing toric
ideals and various ideas for speeding up the computation.

In our example, IA = 〈cd− be, bd− ae, b2 − ac, a2d2 − c3e, c4 − a3e, bc3 −
a3d, ad4 − c2e3, d6 − ce5〉, which we now compute using this procedure.

i12 : I = toricIdeal A;

o12 : Ideal of R

i13 : transpose mingens I

o13 = {-2, -9} | cd-be |
{-2, -8} | bd-ae |
{-2, -2} | b2-ac |
{-4, -14} | a2d2-c3e |
{-4, -8} | c4-a3e |
{-4, -7} | bc3-a3d |
{-5, -28} | ad4-c2e3 |
{-6, -42} | d6-ce5 |

8 1
o13 : Matrix R <--- R

This ideal defines an embedding of P1 as a degree 8 curve into P4. We
will see in Section 3 that its toric Hilbert scheme HilbA has a non-reduced
component.

This chapter is organized into four sections and two appendices as fol-
lows. The main goal in Section 1 is to describe an algorithm for generating
all monomial A-graded ideals for a given A. These monomial ideals are the
vertices of the flip graph of A whose connectivity is equivalent to the connec-
tivity of HilbA. We describe how all neighbors of a given vertex of this graph
can be calculated. In Section 2, we explain the role of polyhedral geometry
in the study of HilbA. Our first algorithm tests for coherence in a monomial
A-graded ideal. We then show how to compute the polyhedral complexes sup-
porting A-graded ideals, which in turn relate the flip graph of A to the Baues
graph of A. For unimodular matrices, these two graphs coincide and hence our
method of computing the flip graph can be used to compute the Baues graph.
Section 3 explores the components of HilbA via local equations around the
torus fixed points of the scheme. We include a combinatorial interpretation
of these local equations from the point of view of integer programming. The
scheme HilbA has a coherent component, which is examined in detail in Sec-
tion 4. We prove that this component is, in general, not normal and that its
normalization is the toric variety of the Gröbner fan of IA. We conclude the
chapter with two appendices, each containing one large piece of Macaulay 2
code that we use in this chapter. Appendix A displays code from the Mac-
aulay 2 file polarCone.m2 that is used to convert a generator representation
of a polyhedron to an inequality representation and vice versa. Appendix B
displays code from the file minPres.m2 used for computing minimal presen-
tations of polynomial quotient rings. The main ingredient of this package is
the subroutine removeRedundantVariables, which is what we use in this
chapter.

1 Generating Monomial Ideals

We start out by computing the Graver basis GrA, which is the set of binomials
in IA that are minimal with respect to the partial order defined by

xu − xv ≤ xu
′
− xv

′
⇐⇒ xu divides xu

′
and xv divides xv

′
.

The set GrA is a universal Gröbner basis of IA and has its origins in the
theory of integer programming [9]. It can be computed using [22, Algorithm
7.2], a Macaulay 2 version of which is given below.

i14 : graver = (I) -> (
R := ring I;
k := coefficientRing R;
n := numgens R;
-- construct new ring S with 2n variables
S := k[Variables=>2*n,MonomialSize=>16];
toS := map(S,R,(vars S)_{0..n-1});

toR := map(R,S,vars R | matrix(R, {toList(n:1)}));
-- embed I in S
m := gens toS I;
-- construct the toric ideal of the Lawrence
-- lifting of A
i := 0;
while i < n do (

wts := join(toList(i:0),{1},toList(n-i-1:0));
wts = join(wts,wts);
m = homogenize(m,S_(n+i),wts);
i=i+1;
);

J := ideal m;
scan(gens ring J, f -> J = saturate(J,f));
-- apply the map toR to the minimal generators of J
f := matrix entries toR mingens J;
p := sortColumns f;
f_p) ;

The above piece of code first constructs a new polynomial ring S in n
more variables than R. Assume S = C[x1, . . . , xn, y1, . . . , yn]. The inclusion
map toS : R→ S embeds the toric ideal I in S and collects its generators in
the matrix m. A binomial xa − xb lies in GrA if and only if xayb − xbya is a
minimal generator of the toric ideal in S of the (d+ n)× 2n matrix

Λ(A) :=
(
A 0
In In

)
,

which is called the Lawrence lifting of A. Since u ∈ kerZ(A) ⇔ (u,−u) ∈
kerZ(Λ(A)), we use the while loop to homogenize the binomials in m with
respect to Λ(A), using the n new variables in S. This converts a binomial
xa − xb ∈ m to the binomial xayb − xbya. The ideal generated by these new
binomials is labeled J . As before, we can now successively saturate J to get
the toric ideal of Λ(A) in S. The image of the minimal generators of this toric
ideal under the map toR : S → R such that xi 7→ xi and yi 7→ 1 is precisely
the Graver basis GrA. These binomials are the entries of the matrix f and is
output by the program.

In our example GrA consists of 42 binomials.
i15 : Graver = graver I

o15 = | -cd+be -bd+ae -b2+ac -cd2+ae2 -a2d2+c3e -c4+a2bd -c4+a3e -bc3+ · · ·
1 42

o15 : Matrix R <--- R

Returning to the general case, an element b of NA is called a Graver
degree if there exists a binomial xu − xv in the Graver basis GrA such that
Au = Av = b. If b is a Graver degree then the set of monomials in Rb is
the corresponding Graver fiber. In our running example there are 37 distinct
Graver fibers. We define the ProductIdeal of A as PI := 〈xaxb : xa − xb ∈
GrA〉. This ideal is contained in every monomial ideal of HilbA and hence no
monomial in PI can be a standard monomial of a monomial A-graded ideal.
Since our purpose in constructing Graver fibers is to use them to generate all

monomial A-graded ideals, we will be content with listing just the monomials
in each Graver fiber that do not lie in PI. Since R is multigraded by A, we
can obtain such a presentation of a Graver fiber by simply asking for the
basis of R in degree b modulo PI.

i16 : graverFibers = (Graver) -> (
ProductIdeal := (I) -> ( trim ideal(

apply(numgens I, a -> (
f := I_a; leadTerm f * (leadTerm f - f)))));

PI := ProductIdeal ideal Graver;
R := ring Graver;
new HashTable from apply(

unique degrees source Graver,
d -> d => compress (basis(d,R) % PI) ));

i17 : fibers = graverFibers Graver

o17 = HashTable{{2, 2} => | ac b2 | }
{2, 8} => | ae bd |
{2, 9} => | be cd |
{3, 16} => | ae2 bde cd2 |
{4, 14} => | a2d2 c3e |
{4, 7} => | a3d bc3 |
{4, 8} => | a3e a2bd c4 |
{5, 10} => | a3ce a2b2e a2bcd ab3d c5 |
{5, 14} => | a3d2 ac3e b2c2e bc3d |
{5, 16} => | a3e2 a2cd2 ab2d2 c4e |
{5, 21} => | a2d3 bc2e2 c3de |
{5, 22} => | a2d2e abd3 c3e2 |
{5, 28} => | ad4 c2e3 |
{5, 7} => | a4d abc3 b3c2 |
{5, 8} => | a4e a3bd ac4 b2c3 |
{6, 12} => | a3c2e a2bc2d ab4e b5d c6 |
{6, 14} => | a4d2 a2c3e abc3d b4ce b3c2d |
{6, 18} => | a3ce2 a2b2e2 a2c2d2 b4d2 c5e |
{6, 21} => | a3d3 abc2e2 ac3de b3ce2 bc3d2 |
{6, 24} => | a3e3 a2cd2e abcd3 b3d3 c4e2 |
{6, 28} => | a2d4 ac2e3 b2ce3 c3d2e |
{6, 30} => | a2d2e2 acd4 b2d4 c3e3 |
{6, 35} => | ad5 bce4 c2de3 |
{6, 36} => | ad4e bd5 c2e4 |
{6, 42} => | ce5 d6 |
{6, 7} => | a5d a2bc3 b5c |
{6, 8} => | a5e a4bd a2c4 b4c2 |
{7, 14} => | a5d2 a3c3e a2bc3d b6e b5cd c7 |
{7, 21} => | a4d3 a2bc2e2 a2c3de abc3d2 b5e2 b3c2d2 |
{7, 28} => | a3d4 a2c2e3 ac3d2e b4e3 bc3d3 |
{7, 35} => | a2d5 abce4 ac2de3 b3e4 c3d3e |
{7, 42} => | ace5 ad6 b2e5 c2d2e3 |
{7, 49} => | be6 cde5 d7 |
{7, 7} => | a6d a3bc3 b7 |
{7, 8} => | a6e a5bd a3c4 b6c |
{8, 56} => | ae7 bde6 cd2e5 d8 |
{8, 8} => | a7e a6bd a4c4 b8 |

o17 : HashTable

For example, the Graver degree (8, 8) corresponds to the Graver fiber{
a7e, a6bd, a4c4, a3b2c3, a2b4c2, ab6c, b8

}
.

Our Macaulay 2 code outputs only the four underlined monomials, in the
format | a7e a6bd a4c4 b8 |. The three non-underlined monomials lie in
the ProductIdeal. Graver degrees are important because of the following
result.

Lemma 1.1 ([22, Lemma 10.5]). The multidegree of any minimal gener-
ator of any ideal I in HilbA is a Graver degree.

The next step in constructing the toric Hilbert scheme is to compute
all its fixed points with respect to the scaling action of the n-dimensional
algebraic torus (C∗)n. (The torus (C∗)n acts on R by scaling variables :
λ 7→ λ · x := (λ1x1, . . . , λnxn).) These fixed points are the monomial ideals
M lying on HilbA. Every term order ≺ on the polynomial ring R gives such
a monomial ideal: M = in≺(IA), the initial ideal of the toric ideal IA with
respect to ≺. Two ideals J and J ′ are said to be torus isomorphic if J = λ ·J ′
for some λ ∈ (C∗)n. Any monomial A-graded ideal that is torus isomorphic
to an initial ideal of IA is said to be coherent. In particular, the initial ideals
of IA are coherent and they can be computed by [22, Algorithm 3.6] applied
to IA. A refinement and fast implementation can be found in the software
package TiGERS by Huber and Thomas [12].

Now we wish to compute all monomial ideals M on HilbA regardless of
whether M is coherent or not. For this we use the procedure generateAmonos
given below. This procedure takes in the Graver basis GrA and records the
numerator of the Hilbert series of IA in trueHS. It then computes the Graver
fibers of A, sorts them and calls the subroutine selectStandard to generate
a candidate for a monomial ideal on HilbA.

i18 : generateAmonos = (Graver) -> (
trueHS := poincare coker Graver;
fibers := graverFibers Graver;
fibers = apply(sort pairs fibers, last);
monos = {};
selectStandard := (fibers, J) -> (
if #fibers == 0 then (

if trueHS == poincare coker gens J
then (monos = append(monos,flatten entries mingens J));

) else (
P := fibers_0;
fibers = drop(fibers,1);
P = compress(P % J);
nP := numgens source P;
-- nP is the number of monomials not in J.
if nP > 0 then (

if nP == 1 then selectStandard(fibers,J)
else (--remove one monomial from P,take the rest.

P = flatten entries P;
scan(#P, i -> (

J1 := J + ideal drop(P,{i,i});
selectStandard(fibers, J1)))));

));
selectStandard(fibers, ideal(0_(ring Graver)));
) ;

The arguments to the subroutine selectStandard are the Graver fibers
given as a list of matrices and a monomial ideal J that should be included in
every A-graded ideal that we generate. The subroutine then loops through
each Graver fiber, and at each step selects a standard monomial from that
fiber and updates the ideal J by adding the other monomials in this fiber to
J . The final J output by the subroutine is the candidate ideal that is sent
back to generateAmonos. It is stored by the program if its Hilbert series
agrees with that of IA. All the monomial A-graded ideals are stored in the
list monos. Below, we ask Macaulay 2 for the cardinality of monos and its
first ten elements.

i19 : generateAmonos Graver;

i20 : #monos

o20 = 281

i21 : scan(0..9, i -> print toString monos#i)
{c*d, b*d, b^2, c^3*e, c^4, b*c^3, c^2*e^3, b*c^2*e^2, b*c*e^4, d^6}
{c*d, b*d, b^2, c^3*e, c^4, b*c^3, c^2*e^3, b*c^2*e^2, c*e^5, b*c*e^4, · · ·
{c*d, b*d, b^2, c^3*e, c^4, b*c^3, c^2*e^3, b*c^2*e^2, c*e^5, b*c*e^4, · · ·
{c*d, b*d, b^2, c^3*e, c^4, b*c^3, c^2*e^3, b*c^2*e^2, c*e^5, b*c*e^4, · · ·
{c*d, b*d, b^2, c^3*e, c^4, b*c^3, c^2*e^3, b*c^2*e^2, d^6, a*d^5}
{c*d, b*d, b^2, c^3*e, c^4, b*c^3, b*c^2*e^2, a*d^4, d^6}
{c*d, b*d, b^2, c^3*e, c^4, b*c^3, a*d^4, a^2*d^3, d^6}
{c*d, b*d, b^2, a^2*d^2, c^4, b*c^3, a*d^4, d^6}
{c*d, b*d, b^2, a^2*d^2, a^3*d, c^4, a*d^4, d^6}
{c*d, b*d, b^2, a^3*e, a^2*d^2, a^3*d, a*d^4, d^6}

The monomial ideals (torus-fixed points) on HilbA form the vertices of
the flip graph of A whose edges correspond to the torus-fixed curves on HilbA.
This graph was introduced in [14] and provides structural information about
HilbA. The edges emanating from a monomial ideal M can be constructed as
follows: For any minimal generator xu of M , let xv be the unique monomial
with xv 6∈ M and Au = Av. Form the wall ideal, which is generated by
xu − xv and all minimal generators of M other than xu, and let M ′ be the
initial monomial ideal of the wall ideal with respect to any term order �
for which xv � xu. It can be shown that M ′ is the unique initial monomial
ideal of the wall ideal that contains xv. If M ′ lies on HilbA then {M,M ′}
is an edge of the flip graph. We now illustrate the Macaulay 2 procedure for
computing all flip neighbors of a monomial A-graded ideal.

i22 : findPositiveVector = (m,s) -> (
expvector := first exponents s - first exponents m;
n := #expvector;
i := first positions(0..n-1, j -> expvector_j > 0);
splice {i:0, 1, (n-i-1):0}
);

i23 : flips = (M) -> (
R := ring M;
-- store generators of M in monoms
monoms := first entries generators M;
result := {};
-- test each generator of M to see if it leads to a neighbor



Toric Hilbert Schemes 187

scan(#monoms, i -> (
m := monoms_i;
rest := drop(monoms,{i,i});
b := basis(degree m, R);
s := (compress (b % M))_(0,0);
J := ideal(m-s) + ideal rest;
if poincare coker gens J == poincare coker gens M then (

w := findPositiveVector(m,s);
R1 := (coefficientRing R)[generators R, Weights=>w];
J = substitute(J,R1);
J = trim ideal leadTerm J;
result = append(result,J);
)));

result
);

The code above inputs a monomial A-graded ideal M whose minimal
generators are stored in the list monoms. The flip neighbors of M will be
stored in result. For each monomial xu in monoms we need to test whether
it yields a flip neighbor of M or not. At the i-th step of this loop, we let
m be the i-th monomial in monoms. The list rest contains all monomials in
monoms except m. We compute the standard monomial s of M of the same
degree as m. The wall ideal of m − s is the binomial ideal J generated by
m− s and the monomials in rest. We then check whether J is A-graded by
comparing its Hilbert series with that of M . (Alternately, one could check
whether M is the initial ideal of the wall ideal with respect to m � s.) If this
is the case, we use the subroutine findPositiveVector to find a unit vector
w = (0, . . . , 1, . . . , 0) such that w · s > w ·m. The flip neighbor is then the
initial ideal of J with respect to w and it is stored in result. The program
outputs the minimal generators of each flip neighbor. Here is an example.

i24 : R = QQ[a..e,Degrees=>transpose A];

i25 : M = ideal(a*e,c*d,a*c,a^2*d^2,a^2*b*d,a^3*d,c^2*e^3,
c^3*e^2,c^4*e,c^5,c*e^5,a*d^5,b*e^6);

o25 : Ideal of R

i26 : F = flips M

2 2 3 4 2 3 3 2 5 5 · · ·
o26 = {ideal (a*e, c*d, a*c, a d , a d, c , c e , c e , a*d , c*e , b* · · ·
o26 : List

i27 : #F

o27 = 4

i28 : scan(#F, i -> print toString entries mingens F_i)
{{a*e, c*d, a*c, a^2*d^2, a^3*d, c^4, c^2*e^3, c^3*e^2, a*d^5, c*e^5, · · ·
{{c*d, a*e, a*c, a^2*d^2, a^2*b*d, a^3*d, c^3*e^2, c^4*e, c^5, a*d^4, · · ·
{{a*e, c*d, a*c, a^2*d^2, a^3*d, a^2*b*d, c^2*e^3, c^3*e^2, c^4*e, c^5 · · ·
{{a*e, a*c, c*d, a^2*b*d, a^3*d, a^2*d^2, c^2*e^3, c^3*e^2, c^4*e, c^5 · · ·
It is an open problem whether the toric Hilbert scheme HilbA is con-

nected. Recent work in geometric combinatorics [19] suggests that this is
probably false for some A. This result and its implications for HilbA will

be discussed further in Section 2. The following theorem of Maclagan and
Thomas [14] reduces the connectivity of HilbA to a combinatorial problem.

Theorem 1.2. The toric Hilbert scheme HilbA is connected if and only if
the flip graph of A is connected.

We now have two algorithms for listing monomial ideals on HilbA. First,
there is the backtracking algorithm whose Macaulay 2 implementation was
described above. Second, there is the flip search algorithm, which starts with
any coherent monomial ideal M and then constructs the connected compo-
nent of M in the flip graph of A by carrying out local flips as above. This
procedure is also implemented in TiGERS [12]. Clearly, the two algorithms will
produce the same answer if and only if HilbA is connected. In other words,
finding an example where HilbA is disconnected is equivalent to finding a
matrix A for which the flip search algorithm produces fewer monomial ideals
than the backtracking algorithm.

2 Polyhedral Geometry

Algorithms from polyhedral geometry are essential in the study of the toric
Hilbert scheme. Consider the problem of deciding whether or not a given
monomial ideal M in HilbA is coherent. This problem gives rise to a sys-
tem of linear inequalities as follows: Let xu1 , . . . , xur be the minimal gener-
ators of M , and let xvi be the unique standard monomial with Aui = Avi.
Then M is coherent if and only if there exists a vector w ∈ Rn such that
w · (ui−vi) > 0 for i = 1, . . . , r. Thus the test for coherence amounts to solv-
ing a feasibility problem of linear programming, and there are many highly
efficient algorithms (based on the simplex algorithms or interior point meth-
ods) available for this task. For our experimental purposes, it is convenient
to use the code polarCone.m2, given in Appendix A, which is based on the
(inefficient but easy-to-implement) Fourier-Motzkin elimination method (see
[25] for a description). This code converts the generator representation of a
polyhedron to its inequality representation and vice versa. A simple example
is given in Appendix A. In particular, given a Gröbner basis G of IA, the
function polarCone will compute all the extreme rays of the Gröbner cone
{w ∈ Rn : w · (ui − vi) ≥ 0 for each xui − xvi ∈ G}.

We now show how to use Macaulay 2 to decide whether a monomial A-
graded ideal M is coherent. The first step in this calculation is to compute all
the standard monomials of M of the same degree as the minimal generators
of M . We do this using the procedure stdMonomials.

i29 : stdMonomials = (M) -> (
R := ring M;
RM := R/M;
apply(numgens M, i -> (

s := basis(degree(M_i),RM); lift(s_(0,0), R)))
);

As an example, consider the following monomial A-graded ideal.
i30 : R = QQ[a..e,Degrees => transpose A ];

i31 : M = ideal(a^3*d, a^2*b*d, a^2*d^2, a*b^3*d, a*b^2*d^2, a*b*d^3,
a*c, a*d^4, a*e, b^5*d, b^4*d^2, b^3*d^3, b^2*d^4,
b*d^5, b*e, c*e^5);

o31 : Ideal of R

i32 : toString stdMonomials M

o32 = {b*c^3, c^4, c^3*e, c^5, c^4*e, c^3*e^2, b^2, c^2*e^3, b*d, c^6, · · ·
From the pairs xu, xv of minimal generators xu and the corresponding

standard monomials xv, the function inequalities creates a matrix whose
columns are the vectors u− v.

i33 : inequalities = (M) -> (
stds := stdMonomials(M);
transpose matrix apply(numgens M, i -> (

flatten exponents(M_i) -
flatten exponents(stds_i))));

i34 : inequalities M

o34 = | 3 2 2 1 1 1 1 1 1 0 0 0 0 0 0 0 |
| -1 1 0 3 2 1 -2 0 -1 5 4 3 2 1 1 0 |
| -3 -4 -3 -5 -4 -3 1 -2 0 -6 -5 -4 -3 -2 -1 1 |
| 1 1 2 1 2 3 0 4 -1 1 2 3 4 5 -1 -6 |
| 0 0 -1 0 -1 -2 0 -3 1 0 -1 -2 -3 -4 1 5 |

5 16
o34 : Matrix ZZ <--- ZZ

It is convenient to simplify the output of the next procedure using the
following program to divide an integer vector by the g.c.d. of its components.
We also load polarCone.m2, which is needed in decideCoherence below.

i35 : primitive := (L) -> (
n := #L-1; g := L#n;
while n > 0 do (n = n-1; g = gcd(g, L#n););
if g === 1 then L else apply(L, i -> i // g));

i36 : load "polarCone.m2"

i37 : decideCoherence = (M) -> (
ineqs := inequalities M;
c := first polarCone ineqs;
m := - sum(numgens source c, i -> c_{i});
prods := (transpose m) * ineqs;
if numgens source prods != numgens source compress prods
then false else primitive (first entries transpose m));

Let K be the cone {x ∈ Rn : g · x ≤ 0, for all columns g of ineqs }. The
command polarCone ineqs computes a pair of matrices P and Q such that
K is the sum of the cone generated by the columns of P and the subspace
generated by the columns of Q. Let m be the negative of the sum of the
columns of P . Then m lies in the cone −K. The entries in the matrix prods
are the dot products g ·m for each column g of ineqs. Since M is a monomial
A-graded ideal, it is coherent if and only if K is full dimensional, which is

the case if and only if no dot product g ·m is zero. This is the conditional
in the if .. then statement of decideCoherence. If M is coherent, the
program outputs the primitive representative of m and otherwise returns the
boolean false. Notice that if M is coherent, the cone −K is the Gröbner
cone corresponding to M and the vector m is a weight vector w such that
inw(IA) = M . We now test whether the ideal M from line i29 is coherent.

i38 : decideCoherence M

o38 = {0, 0, 1, 15, 18}

o38 : List

Hence, M is coherent: it is the initial ideal with respect to the weight
vector w = (0, 0, 1, 15, 18) of the toric ideal in our running example (1). Here
is one of the 55 noncoherent monomial A-graded ideals of this matrix.

i39 : N = ideal(a*e,c*d,a*c,c^3*e,a^3*d,c^4,a*d^4,a^2*d^3,c*e^5,
c^2*e^4,d^7);

o39 : Ideal of R

i40 : decideCoherence N

o40 = false

In the rest of this section, we study the connection between A-graded
ideals and polyhedral complexes defined on A. This study relates the flip
graph of the toric Hilbert scheme to the Baues graph of the configuration A.
(See [18] for a survey of the Baues problem and its relatives). Let pos(A) :=
{Au : u ∈ Rn, u ≥ 0} be the cone generated by the columns of A in Rd.
A polyhedral subdivision ∆ of A is a collection of full dimensional subcones
pos(Aσ) of pos(A) such that the union of these subcones is pos(A) and the
intersection of any two subcones is a face of each. Here Aσ := {aj : j ∈
σ ⊆ {1, . . . , n}}. It is customary to identify ∆ with the set of sets {σ :
pos(Aσ) ∈ ∆}. If every cone in the subdivision ∆ is simplicial (the number of
extreme rays of the cone equals the dimension of the cone), we say that ∆ is
a triangulation of A. The simplicial complex corresponding to a triangulation
∆ is uniquely obtained by including in ∆ all the subsets of every σ ∈ ∆. We
refer the reader to [22, §8] for more details.

For each σ ∈ ∆, let Iσ be the prime ideal that is the sum of the toric ideal
IAσ and the monomial ideal 〈xj : j 6∈ σ〉. Recall that two ideals J and J ′ are
said to be torus isomorphic if J = λ · J ′ for some λ ∈ (C∗)n. The following
theorem shows that polyhedral subdivisions of A are related to A-graded
ideals via their radicals.

Theorem 2.1 (Theorem 10.10 [22, §10]). If I is an A-graded ideal, then
there exists a polyhedral subdivision ∆(I) of A such that

√
I = ∩σ∈∆(I)Jσ

where each component Jσ is a prime ideal that is torus isomorphic to Iσ.

We say that ∆(I) supports the A-graded ideal I. When M is a monomial
A-graded ideal, ∆(M) is a triangulation of A. In particular, if M is coherent



Toric Hilbert Schemes 191

(i.e, M = inw(IA) for some weight vector w), then ∆(M) is the regular or
coherent triangulation of A induced by w [22, §8]. The coherent triangulations
of A are in bijection with the vertices of the secondary polytope of A [3], [8].

It is convenient to represent a triangulation ∆ of A by its Stanley-Reisner
ideal I∆ := 〈xi1xi2 · · ·xik : {i1, i2, . . . , ik} is a non-face of ∆〉. If M is a mono-
mial A-graded ideal, Theorem 2.1 implies that I∆(M) is the radical of M .
Hence we will represent triangulations of A by their Stanley-Reisner ideals.
As seen below, the matrix in our running example has eight distinct trian-
gulations corresponding to the eight distinct radicals of the 281 monomial
A-graded ideals computed earlier. All eight are coherent.

{{1, 2}, {2, 3}, {3, 4}, {4, 5}} ↔ 〈ac, ad, ae, bd, be, ce〉
{{1, 3}, {3, 4}, {4, 5}} ↔ 〈b, ad, ae, ce〉
{{1, 2}, {2, 4}, {4, 5}} ↔ 〈c, ad, ae, be〉
{{1, 2}, {2, 3}, {3, 5}} ↔ 〈d, ac, ae, be〉
{{1, 3}, {3, 5}} ↔ 〈b, d, ae〉
{{1, 4}, {4, 5}} ↔ 〈b, c, ae〉
{{1, 2}, {2, 5}} ↔ 〈c, d, ae〉
{{1, 5}} ↔ 〈b, c, d〉

The Baues graph of A is a graph on all the triangulations of A in which
two triangulations are adjacent if they differ by a single bistellar flip [18].
The Baues problem from discrete geometry asked whether the Baues graph
of a point configuration can be disconnected for some A. Every edge of the
secondary polytope of A corresponds to a bistellar flip, and hence the sub-
graph of the Baues graph that is induced by the coherent triangulations of A
is indeed connected: it is precisely the edge graph of the secondary polytope
of A. The Baues problem was recently settled by Santos [19] who gave an
example of a six dimensional point configuration with 324 points for which
there is an isolated (necessarily non-regular) triangulation.

Santos’ configuration would also have a disconnected flip graph and hence
a disconnected toric Hilbert scheme if it were true that every triangulation
of A supports a monomial A-graded ideal. However, Peeva has shown that
this need not be the case (Theorem 10.13 in [22, §10]). Hence, the map from
the set of all monomial A-graded ideals to the set of all triangulations of A
that sends M 7→ ∆(M) is not always surjective, and it is unknown whether
Santos’ 6× 324 configuration has a disconnected toric Hilbert scheme.

Thus, even though one cannot in general conclude that the existence of a
disconnected Baues graph implies the existence of a disconnected flip graph,
there is an important special situation in which such a conclusion is possible.
We call an integer matrix A of full row rank unimodular if the absolute value
of each of its non-zero maximal minors is the same constant. A matrix A is
unimodular if and only if every monomial A-graded ideal is square-free. For
a unimodular matrix A, the Baues graph of A coincides with the flip graph
of A. As you might expect, Santos’ configuration is not unimodular.

Theorem 2.2 (Lemma 10.14 [22, §10]). If A is unimodular, then each
triangulation of A supports a unique (square-free) monomial A-graded ideal.
In this case, a monomial A-graded ideal is coherent if and only if the trian-
gulation supporting it is coherent.

Using Theorem 2.2 we can compute all the triangulations of a unimodular
matrix since they are precisely the polyhedral complexes supporting mono-
mial A-graded ideals. Then we could enumerate the connected component of
a coherent monomial A-graded ideal in the flip graph of A to decide whether
the Baues/flip graph is disconnected.

Let ∆r be the standard r-simplex that is the convex hull of the r+ 1 unit
vectors in Rr+1, and let A(r, s) be the (r + s + 2) × (r + 1)(s + 1) matrix
whose columns are the products of the vertices of ∆r and ∆s. All matrices
of type A(r, s) are unimodular. From the product of two triangles we get

A(2, 2) :=


1 1 1 0 0 0 0 0 0
0 0 0 1 1 1 0 0 0
0 0 0 0 0 0 1 1 1
1 0 0 1 0 0 1 0 0
0 1 0 0 1 0 0 1 0
0 0 1 0 0 1 0 0 1

 .

We can now use our algebraic algorithms to compute all the triangulations
of A(2, 2). Since Macaulay 2 requires the first entry of the degree of every
variable in a ring to be positive, we use the following matrix with the same
row space as A(2, 2) for our computation:

i41 : A22 =
{{1,1,1,1,1,1,1,1,1},{0,0,0,1,1,1,0,0,0},{0,0,0,0,0,0,1,1,1},
{1,0,0,1,0,0,1,0,0},{0,1,0,0,1,0,0,1,0},{0,0,1,0,0,1,0,0,1}};

i42 : I22 = toricIdeal A22

o42 = ideal (f*h - e*i, c*h - b*i, f*g - d*i, e*g - d*h, c*g - a*i, b* · · ·
o42 : Ideal of R

The ideal I22 is generated by the 2 by 2 minors of a 3 by 3 matrix of
indeterminates. This is the ideal of P2 × P2 embedded in P8 via the Segre
embedding.

i43 : Graver22 = graver I22;

1 15
o43 : Matrix R <--- R

i44 : generateAmonos(Graver22);

i45 : #monos

o45 = 108

i46 : scan(0..9,i->print toString monos#i)
{f*h, c*h, f*g, e*g, c*g, b*g, c*e, c*d, b*d}
{f*h, d*h, c*h, f*g, c*g, b*g, c*e, c*d, b*d}
{d*i, f*h, d*h, c*h, c*g, b*g, c*e, c*d, b*d}
{e*i, c*h, f*g, e*g, c*g, b*g, c*e, c*d, b*d}
{e*i, d*i, c*h, e*g, c*g, b*g, c*e, c*d, b*d}
{e*i, d*i, d*h, c*h, c*g, b*g, c*e, c*d, b*d}
{f*h, c*h, f*g, e*g, c*g, b*g, c*e, a*e, c*d}
{e*i, c*h, f*g, e*g, c*g, b*g, c*e, a*e, c*d, b*d*i}
{e*i, c*h, f*g, e*g, c*g, b*g, c*e, a*e, c*d, a*f*h}
{e*i, d*i, c*h, e*g, c*g, b*g, c*e, a*e, c*d}

Thus there are 108 monomial A(2, 2)-graded ideals and decideCoherence
will check that all of them are coherent. Since A(2, 2) is unimodular, each
monomial A(2, 2)-graded ideal is square-free and is hence radical. These 108
ideals represent the 108 triangulations of A(2, 2) and we have listed ten of
them above. The flip graph (equivalently, Baues graph) of A(2, 2) is con-
nected. However, it is unknown whether the Baues graph of A(r, s) is con-
nected for all values of (r, s).

3 Local Equations

Consider the reduced Gröbner basis of a toric ideal IA for a term order w:{
xu1 − xv1 , xu2 − xv2 , . . . , xur − xvr

}
. (2)

The initial ideal M = inw(IA) = 〈xu1 , xu2 , . . . , xur 〉 is a coherent monomial
A-graded ideal. In particular, it is a (C∗)n-fixed point on the toric Hilbert
scheme HilbA. We shall explain a method, due to Peeva and Stillman [16],
for computing local equations of HilbA around such a fixed point. A variant
of this method also works for computing the local equations around a non-
coherent monomial ideal M , but that variant involves local algebra, specifi-
cally Mora’s tangent cone algorithm, which is not yet fully implemented in
Macaulay 2. See [16] for details.

We saw how to compute the flip graph of A in Section 1. The vertices
of this graph are the (C∗)n-fixed points M and its edges correspond to
the (C∗)n-fixed curves. By computing and decomposing the local equations
around each M , we get a complete description of the scheme HilbA.

The first step is to introduce a new variable zi for each binomial in our
Gröbner basis (2) and to consider the following r binomials:

xu1 − z1 · xv1 , xu2 − z2 · xv2 , . . . , xur − zr · xvr (3)

in the polynomial ring C[x, z] in n+ r indeterminates. The term order w can
be extended to an elimination term order in C[x, z] so that xui is the leading
term of xui − zi · xvi for all i. We compute the minimal first syzygies of the
monomial ideal M , and form the corresponding S-pairs of binomials in (3).
For each S-pair

lcm(xui , xuj )
xui

· (xui − zi · xvi) −
lcm(xui , xuj )

xuj
· (xuj − zj · xvj )

we compute a normal form with respect to (3) using the extended term order
w. The result is a binomial in C[x, z] that factors as

xα · zβ · (zγ − zδ),

where α ∈ Nn and β, γ, δ ∈ Nr. Note that this normal form is not unique
but depends on our choice of a reduction path. Let JM denote the ideal in
C[z1, . . . , zr] generated by all binomials zβ · (zγ − zδ) gotten from normal
forms of all the S-pairs considered above.

Proposition 3.1 ([16]). The ideal JM is independent of the reduction paths
chosen. It defines a subscheme of Cr isomorphic to an affine open neighbor-
hood of the point M on the toric Hilbert scheme HilbA.

We apply this technique to compute a particularly interesting affine chart
of HilbA for our running example. Consider the following set of 13 binomials:{

ae− z1bd, cd− z2be, ac− z3b
2, a2d2 − z4c

3e, a2bd− z5c
4,

a3d− z6bc
3, c2e3 − z7ad

4, c3e2 − z8abd
3, c4e− z9ab

2d2,

c5 − z10ab
3d, ce5 − z11d

6, ad5 − z12bce
4, be6 − z13d

7
}
.

If we set z1 = z2 = · · · = z13 = 1 then we get a generating set for the toric
ideal IA. The 13 monomials obtained by setting z1 = z2 = · · · = z13 = 0
generate the initial monomial ideal M = inw(IA) with respect to the weight
vector w = (9, 3, 5, 0, 0). Thus M is one of the 226 coherent monomial A-
graded ideals of our running example. The above set of 13 binomials in C[x, z]
give the universal family for HilbA around this M .

The local chart of HilbA around the point M is a subscheme of affine space
C

13 with coordinates z1, . . . , z13, whose defining equations are obtained as
follows: Extend the weight vector w by assigning weight zero to all variables
zi, so that the first term in each of the above 13 binomials is the leading
term. For each pair of binomials corresponding to a minimal syzygy of M ,
form their S-pair and then reduce it to a normal form with respect to the 13
binomials above. For instance,

S
(
c5−z10ab

3d, ce5−z11d
6
)

= z11c
4d6−z10ab

3de5 −→ b4d2e4·(z4
2z11−z1z10).

Each such normal form is a monomial in a, b, c, d, e times a binomial in
z1, . . . , z13. The set of all these binomials, in the z-variables, generates the
ideal JM of local equations of HilbA around M . In our example, JM is gener-
ated by 27 nonzero binomials. This computation can be done in Macaulay 2
using the procedure localCoherentEquations.

i47 : localCoherentEquations = (IA) -> (
-- IA is the toric ideal of A living in a ring equipped
-- with weight order w, if we are computing the local
-- equations about the initial ideal of IA w.r.t. w.
R := ring IA;

w := (monoid R).Options.Weights;
M := ideal leadTerm IA;
S := first entries ((gens M) % IA);
-- Make the universal family J in a new ring.
nv := numgens R; n := numgens M;
T = (coefficientRing R)[generators R, z_1 .. z_n,

Weights => flatten splice{w, n:0},
MonomialSize=>16];

M = substitute(generators M,T);
S = apply(S, s -> substitute(s,T));
J = ideal apply(n, i ->

M_(0,i) - T_(nv + i) * S_i);
-- Find the ideal Ihilb of local equations about M:
spairs := (gens J) * (syz M);
g := forceGB gens J;
B = (coefficientRing R)[z_1 .. z_n,MonomialSize=>16];
Fones := map(B,T, matrix(B,{splice {nv:1}}) | vars B);
Ihilb := ideal Fones (spairs % g);
Ihilb
);

Suppose we wish to calculate the local equations about M = inw(IA). The
input to localCoherentEquations is the toric ideal IA living in a polynomial
ring equipped with the weight order specified by w. This is done as follows:

i48 : IA = toricIdeal A;

o48 : Ideal of R

i49 : Y = QQ[a..e, MonomialSize => 16,
Degrees => transpose A, Weights => {9,3,5,0,0}];

i50 : IA = substitute(IA,Y);

o50 : Ideal of Y

The initial ideal M is calculated in the third line of the algorithm, and S
stores the standard monomials ofM of the same degrees as the minimal gener-
ators ofM . We could have calculated S using our old procedure stdMonomials
but this involves computing the monomials in Rb for various values of b, which
can be slow on large examples. As by-products, localCoherentEquations
also gets J, the ideal of the universal family for HilbA about M , the ring T
of this ideal, and the ring B of Ihilb, which is the ideal of the affine patch
of HilbA about M . The matrix spairs contains all the S-pairs between gen-
erators of J corresponding to the minimal first syzygies of M . The command
forceGB is used to declare the generators of J to be a Gröbner basis, and
Fones is the ring map from T to B that sends each of a, b, c, d, e to one and
the z variables to themselves. The columns of the matrix (spairs % g) are
the normal forms of the polynomials in spairs with respect to the forced
Gröbner basis g and the ideal Ihilb of local equations is generated by the
image of these normal forms in the ring B under the map Fones.

i51 : JM = localCoherentEquations(IA)

· · ·
o51 = ideal (z z - z , z z - z , - z z + z , - z z + z , - z z + · · ·

1 2 3 1 2 3 4 7 2 5 8 2 1 5 · · ·

o51 : Ideal of B

Removing duplications among the generators:

JM = 〈z1 − z10z11, z2 − z4z7, z2 − z5z8, z2 − z11z12, z2 − z1z11z13,
z3 − z1z2, z3 − z5z9, z4 − z1z5, z6 − z3z5, z6 − z1z2z5, z7 − z1z10, z8 − z1z7,
z9− z1z8, z12− z1z13, z1z2− z5z9, z1z2− z1z5z8, z1z2− z2

1z4z10, z1z2− z2
1z5z7,

z1z2 − z1z11z12, z1z2 − z2z10z11, z
3
1z4 − z3z11, z1z5z8 − z4z8, z2z10 − z1z12,

z3z4 − z1z6, z3z7 − z2z8, z3z8 − z2z9, z3z10 − z2z7〉.
Notice that there are many generators of JM that have a single vari-

able as one of its terms. Using these generators we can remove variables
from other binomials. This is done in Macaulay 2 using the subroutine
removeRedundantVariables, which is the main ingredient of the package
minPres.m2 for computing the minimal presentations of polynomial quotient
rings. Both removeRedundantVariables and minPres.m2 are explained in
Appendix B. The command removeRedundantVariables applied to an ideal
in a polynomial ring (not quotient ring) creates a ring map from the ring to
itself that sends the redundant variables to polynomials in the non-redundant
variables and the non-redundant variables to themselves. Applying this to our
ideal JM we obtain the following simplifications.

i52 : load "minPres.m2";

i53 : G = removeRedundantVariables JM

3 2 4 3 2 4 3 2 · · ·
o53 = map(B,B,{z z , z z z , z z z , z z z , z , z z z , z · · ·

10 11 5 10 11 5 10 11 5 10 11 5 5 10 11 10 · · ·
o53 : RingMap B <--- B

i54 : ideal gens gb(G JM)

3 2 2
o54 = ideal(z z z - z z z )

5 10 11 10 11 13

o54 : Ideal of B

Thus our affine patch of HilbA has the coordinate ring

C[z1, z2, . . . , z13]/JM ' C[z5, z10, z11, z13]
〈z5z3

10z
2
11 − z10z2

11z13〉
=

C[z5, z10, z11, z13]
〈(z5z2

10 − z13)z10z2
11〉

.

Hence, we see immediately that there are three components through the point
M on HilbA. The restriction of the coherent component to the affine neigh-
borhood of M on HilbA is defined by the ideal quotient (JM : (z1z2 · · · z13)∞)
and hence the first of the above components is an affine patch of the coherent
component. Locally near M it is given by the single equation z5z

2
10− z13 = 0

in A4. It is smooth and, as expected, has dimension three. The second com-
ponent, z10 = 0, is also of dimension three and is smooth at M . The third
component, given by z2

11 = 0 is more interesting. It has dimension three as
well, but is not reduced. Thus we have proved the following result.

Proposition 3.2. The toric Hilbert scheme HilbA of the matrix

A =
(

1 1 1 1 1
0 1 2 7 8

)
is not reduced.

We can use the ring map G from above to simplify J so as to involve only
the four variables z5, z10, z11 and z13.

i55 : CX = QQ[a..e, z_5,z_10,z_11,z_13, Weights =>
{9,3,5,0,0,0,0,0,0}];

i56 : F = map(CX, ring J, matrix{{a,b,c,d,e}} |
substitute(G.matrix,CX))

3 2 4 3 · · ·
o56 = map(CX,T,{a, b, c, d, e, z z , z z z , z z z , z z z , z · · ·

10 11 5 10 11 5 10 11 5 10 11 · · ·
o56 : RingMap CX <--- T

Applying this map to J we get the ideal J1,
i57 : J1 = F J

3 2 2 4 3 · · ·
o57 = ideal (c*d - b*e*z z , a*e - b*d*z z z , a*c - b z z z , a · · ·

10 11 5 10 11 5 10 11 · · ·
o57 : Ideal of CX

and adding the ideal 〈z2
11〉 to J1 we obtain the universal family for the non-

reduced component of HilbA about M .
i58 : substitute(ideal(z_11^2),CX) + J1

2 3 2 2 4 · · ·
o58 = ideal (z , c*d - b*e*z z , a*e - b*d*z z z , a*c - b z z z · · ·

11 10 11 5 10 11 5 10 · · ·
o58 : Ideal of CX

In the rest of this section, we present an interpretation of the ideal JM in
terms of the combinatorial theory of integer programming. See, for instance,
[22, §4] or [24] for the relevant background. Our reduced Gröbner basis (2)
is the minimal test set for the family of integer programs

Minimize w · u subject to A · u = b and u ∈ Nn, (4)

where A ∈ Nd×n and w ∈ Zn are fixed and b ranges over Nd. If u′ ∈ Nn is
any feasible solution to (4), then the corresponding optimal solution u ∈ Nn
is computed as follows: the monomial xu is the unique normal form of xu

′

modulo the Gröbner basis (2).
Suppose we had reduced xu

′
modulo the binomials (3) instead of (2). Then

the output has a z-factor that depends on our choice of reduction path. To
be precise, suppose the reduction path has length m and at the j-th step we

had used the reduction xuµj → zµj · x
vµj . Then we would obtain the normal

form
zµ1zµ2zµ3 · · · zµm · xu.

Reduction paths can have different lengths. If we take another path that has
length m′ and uses xuνj → zνj ·x

vνj at the j-th step, then the output would
be

zν1zν2zν3 · · · zνm′ · x
u.

Theorem 3.3. The ideal JM of local equations on HilbA is generated by the
binomials

zµ1zµ2zµ3 · · · zµm − zν1zν2zν3 · · · zνm′
each encoding a pair of distinct reduction sequences from a feasible solution
of an integer program of the type (4) to the corresponding optimal solution
using the minimal test set in (2).

Proof. The given ideal is contained in JM because its generators are differ-
ences of monomials arising from the possible reduction paths of lcm(xui , xuj ),
for 1 ≤ i, j ≤ r. Conversely, any reduction sequence can be transformed into
an equivalent reduction sequence using S-pair reductions. This follows from
standard arguments in the proof of Buchberger’s criterion [5, §2.6, Theorem
6], and it implies that the binomials zµ1 · · · zµm − zν1 · · · zνm′ are C[z]-linear
combinations of the generators of JM . ut

A given feasible solution of an integer program (4) usually has many
different reduction paths to the optimal solution using the reduced Gröbner
basis (2). For our matrix 1 and cost vector w = (9, 3, 5, 0, 0), the monomial
a2bde6 encodes the feasible solution (2, 1, 0, 1, 6) of the integer program

Minimize w · u subject to A · u =
(

10
56

)
and u ∈ N5.

There are 19 different paths from this feasible solution to the optimal solution
(0, 3, 0, 3, 4) encoded by the monomial b3d3e4. The generating function for
these paths is:

z2
1 + 3z1z

2
2z5z7 + 2z1z2z5z

2
7z12 + 2z1z2z5z8

+ 2z1z2z12z13 + z1z5z9 + z3
2z4z5z

2
7 + z3

2z4z13 + z3
2z5z11

+ 2z2z3z5z7 + z3z5z
2
7z12 + z3z5z8 + z3z12z13.

The difference of any two monomials in this generating function is a valid
local equation for the toric Hilbert scheme of (1). For instance, the binomial
z3z5z

2
7z12−z3z12z13 lies in JM , and, conversely, JM is generated by binomials

obtained in this manner.
The scheme structure of JM encodes obstructions to making certain re-

ductions when solving our family of integer programs. For instance, the vari-
able z3 is a zero-divisor modulo JM . If we factor it out from the binomial

z3z5z
2
7z12−z3z12z13 ∈ JM , we get z5z

2
7z12−z12z13 , which does not lie in JM .

Thus there is no monomial ai1bi2ci3di4ei5 for which both the paths z5z
2
7z12

and z12z13 are used to reach the optimum. It would be a worthwhile combi-
natorial project to study the path generating functions and their relation to
the ideal JM in more detail.

It is instructive to note that the binomials zµ1zµ2 · · · zµm − zν1zν2 · · · zνm′
in Theorem 3.3 do not form a vector space basis for the ideal JM . We demon-
strate this for the lexicographic Gröbner basis (with a � b � c � d � e) of
the toric ideal defining the rational normal curve of degree 4. In this case, we

can take A =
(

1 1 1 1 1
0 1 2 3 4

)
and the universal family in question is :

{
ac− z1b

2, ad− z2bc, ae− z3c
2, bd− z4c

2, be− z5cd, ce− z6d
2
}
.

The corresponding ideal of local equations is JM = 〈z3− z2z5, z2− z1z4, z5−
z4z6〉, from which we see that M is a smooth point of HilbA. The binomial
z1z5−z1z4z6 lies in JM but there is no monomial that has the reduction path
z1z5 or z5z1 to optimality. Indeed, any monomial that admits the reductions
z1z5 or z5z1 must be divisible by either ace or abe. The path generating
functions for these two monomials are

abe → (z3 + z1z4z5 + z2z5) · bc2

ace → (z3 + z1z4z5 + z2z4z6) · c3.

Thus every reduction to optimality using z1 and z5 must also use z4, and we
conclude that z1z5 − z1z4z6 is not in the C-span of the binomials listed in
Theorem 3.3.

4 The Coherent Component of the Toric Hilbert
Scheme

In this section we study the component of the toric Hilbert scheme HilbA that
contains the point corresponding to the toric ideal IA. An A-graded ideal is
coherent if and only if it is isomorphic to an initial ideal of IA under the action
of the torus (C∗)n. All coherent A-graded ideals lie on the same component
of HilbA as IA. We will show that this component need not be normal, and
we will describe how its local and global equations can be computed using
Macaulay 2. Every term order for the toric ideal IA can be realized by a weight
vector that is an element in the lattice N = HomZ(kerZ(A),Z) ' Z

n−d.
Two weight vectors w and w′ in N are considered equivalent if they define
the same initial ideal inw(IA) = inw′(IA). These equivalence classes are the
relatively open cones of a projective fan ΣA called the Gröbner fan of IA [15],
[23]. This fan lies in Rn−d, the real vector space spanned by the lattice N .

Theorem 4.1. The toric ideal IA lies on a unique irreducible component of
the toric Hilbert scheme HilbA, called the coherent component. The normal-
ization of the coherent component is the projective toric variety defined by
the Gröbner fan of IA.

Proof. The divisor at infinity on the toric Hilbert scheme HilbA consists of all
points at which at least one of the local coordinates (around some monomial
A-graded ideal) is zero. This is a proper closed codimension one subscheme
of HilbA, parametrizing all those A-graded ideals that contain at least one
monomial. The complement of the divisor at infinity in HilbA consists of
precisely the orbit of IA under the action of the torus (C∗)n. This is the
content of [22, Lemma 10.12].

The closure of the (C∗)n-orbit of IA is a reduced and irreducible compo-
nent of HilbA. It is reduced because IA is a smooth point on HilbA, as can be
seen from the local equations, and it is irreducible since (C∗)n is a connected
group. It is a component of HilbA because its complement lies in a divisor.
We call this irreducible component the coherent component of HilbA.

Identifying (C∗)n with HomZ(Zn,C∗), we note that the stabilizer of IA
consists of those linear forms w that restrict to zero on the kernel of A.
Therefore the coherent component is the closure in HilbA of the orbit of the
point IA under the action of the torus N ⊗ C∗ = HomZ(kerZ(A),C∗). The
(N⊗C∗)-fixed points on this component are precisely the coherent monomial
A-graded ideals, and the same holds for the toric variety of the Gröbner fan.

Fix a maximal cone σ in the Gröbner fan ΣA, and let M = 〈xu1 , . . . , xur 〉
be the corresponding (monomial) initial ideal of IA. As before we write

{xu1 − z1 · xv1 , xu2 − z2 · xv2 , . . . , xur − zr · xvr}

for the universal family arising from the corresponding reduced Gröbner basis
of IA. Let JM be the ideal in C[z1, z2, . . . , zr] defining this family.

The restriction of the coherent component to the affine neighborhood of
M on HilbA is defined by JM : (z1z2 · · · zr)∞. It then follows from our com-
binatorial description of the ideal JM that this ideal quotient is a binomial
prime ideal. In fact, it is the ideal of algebraic relations among the Lau-
rent monomials xu1−v1 , . . . , xur−vr . We conclude that the restriction of the
coherent component to the affine neighborhood of M on HilbA equals

Spec C
[
xu1−v1 , xu2−v2 , . . . , xur−vr

]
. (5)

The abelian group generated by the vectors u1 − v1, . . . , ur − vr equals
kerZ(A) = HomZ(N,Z). This follows from [21, Lemma 12.2] because the
binomials xui − xvi generate the toric ideal IA. The cone generated by the
vectors u1 − v1, . . . , ur − vr is precisely the polar dual σ∨ to the Gröbner
cone σ. This follows from equation (2.6) in [21]. We conclude that the nor-
malization of the affine variety (5) is the normal affine toric variety

Spec C
[
kerZ(A) ∩ σ∨

]
. (6)

The normalization morphism from (6) to (5) maps the identity point in
the toric variety (6) to the point IA in the affine chart (5) of the toric Hilbert
scheme HilbA. Clearly, this normalization morphism is equivariant with re-
spect to the action by the torus N ⊗C∗. These two properties hold for every
maximal cone σ of the Gröbner fan ΣA. Hence there exists a unique N ⊗C∗-
equivariant morphism φ from the projective toric variety associated with ΣA
onto the coherent component of HilbA, such that φ maps the identity point
to the point IA on HilbA, and φ restricts to the normalization morphism (6)
→ (5) on each affine open chart. We conclude that φ is the desired normal-
ization map from the projective toric variety associated with the Gröbner fan
of IA onto the coherent component of the toric Hilbert scheme HilbA. ut

We now present an example that shows that the coherent component of
HilbA need not be normal. This example is derived from the matrix that ap-
pears in Example 3.15 of [10]. This example is also mentioned in [17] without
details. Let d = 4 and n = 7 and fix the matrix

A =


1 1 1 1 1 1 1
0 6 7 5 8 4 3
3 7 2 0 7 6 1
6 5 2 6 5 0 0

 . (7)

The lattice N = HomZ(kerZ(A),Z) is three-dimensional. The toric ideal
IA is minimally generated by 30 binomials of total degree between 6 and 93.

i59 : A = {{1,1,1,1,1,1,1},{0,6,7,5,8,4,3},{3,7,2,0,7,6,1},
{6,5,2,6,5,0,0}};

i60 : IA = toricIdeal A

2 3 3 2 2 4 4 8 4 4 3 6 7 2 4 4 · · ·
o60 = ideal (a c e - b*d f , a c*d*e f - b g , d e f - b c g , a*b c · · ·
o60 : Ideal of R

We fix the weight vector w = (0, 0, 276, 220, 0, 0, 215) in N and compute
the initial ideal M = inw(IA). This initial ideal has 44 minimal generators.

i61 : Y = QQ[a..g, MonomialSize => 16,
Weights => {0,0,276,220,0,0,215},
Degrees =>transpose A];

i62 : IA = substitute(IA,Y);

o62 : Ideal of Y

i63 : M = ideal leadTerm IA

2 3 8 4 7 2 4 4 7 3 5 4 3 5 2 6 5 4 3 3 1 · · ·
o63 = ideal (a c e, b g , b c g , a*b c f , b c d f , a b c g , a b c · · ·
o63 : Ideal of Y

Proposition 4.2. The three dimensional affine variety (5), for the initial
ideal M with respect to w = (0, 0, 276, 220, 0, 0, 215) of the toric ideal of A in
(7), is not normal.

Proof. The universal family for the toric Hilbert scheme HilbA at M is:

{ a2e15g18 − z1b
3c6d10f16, b13d15f16 − z2a

8ce21g14,

c59d57f110 − z3e
92g134, ac14d11f23 − z4be

19g29,

b7c2g4 − z5d
4e3f6, . . . , bc34d32f62 − z44e

53g76}.

The semigroup algebra in (5) is generated by 44 Laurent monomials gotten
from this family. It turns out that the first four monomials suffice to gen-
erate the semigroup. In other words, for all j ∈ {5, 6, . . . , 44} there exist
i1, i2, i3, i4 ∈ N such that zj − zi11 z

i2
2 z

i3
3 z

i4
4 ∈ JM : (z1 · · · z44)∞. Hence the

semigroup algebra in (5) is:

C

[ a2e15g18

b3c6d10f16
,
b13d15f16

a8ce21g14
,
c59d57f110

e92g134
,
ac14d11f23

be19g29

]
' C[z1, z2, z3, z4]
〈z5

1z2z3 − z2
4〉
.

This algebra is not integrally closed, since a toric hypersurface is normal if
and only if at least one of the two monomials in the defining equation is
square-free. Its integral closure in C[kerZ(A)] is generated by the Laurent
monomial

z4

z2
1

= (z1z2z3)
1
2 =

b5c26d31f55

a3e49g65
. (8)

Hence the affine chart (6) of the toric variety of the Gröbner fan of IA is the
spectrum of the normal domain C[z1, z2, z3, y]/〈z1z2z3 − y2〉, where y maps
to (8). ut

We now examine the local equations of HilbA about M for this example.
i64 : JM = localCoherentEquations(IA)

· · ·
o64 = ideal (z z - z , z z - z , z z - z , z z - z , z z - z , z · · ·

1 2 3 1 2 3 1 5 4 1 3 6 1 3 6 1 · · ·
o64 : Ideal of B

i65 : G = removeRedundantVariables JM;

o65 : RingMap B <--- B

i66 : toString ideal gens gb(G JM)

o66 = ideal(z_32*z_42^2*z_44-z_37^2*z_42,z_32^3*z_35*z_37^2-z_42^2*z_4 · · ·
This ideal has six generators and decomposing it we see that there are

five components through the monomial ideal M on this toric Hilbert scheme.
They are defined by the ideals:

– 〈z32z42z44 − z2
37, z

4
32z35 − z42, z

3
32z35z

2
37 − z2

42z44, z
2
32z35z

4
37 − z3

42z
2
44,

z32z35z
6
37 − z4

42z
3
44, z35z

8
37 − z5

42z
4
44〉

– 〈z44, z37〉
– 〈z37, z

2
42〉

– 〈z42, z35〉

– 〈z42, z
3
32〉.

All five components are three dimensional. The first component is an affine
patch of the coherent component and two of the components are not reduced.
Let K be the first of these ideals.

i67 : K = ideal(z_32*z_42*z_44-z_37^2,z_32^4*z_35-z_42,
z_32^3*z_35*z_37^2-z_42^2*z_44,z_32^2*z_35*z_37^4-z_42^3*z_44^2,
z_32*z_35*z_37^6-z_42^4*z_44^3,z_35*z_37^8-z_42^5*z_44^4);

o67 : Ideal of B

Applying removeRedundantVariables to K we see that the affine patch
of the coherent component is, locally at M , a non-normal hypersurface sin-
gularity (agreeing with (8)). The labels on the variables depend on the order
of elements in the initial ideal M computed by Macaulay 2 in line i61.

i68 : GG = removeRedundantVariables K;

o68 : RingMap B <--- B

i69 : ideal gens gb (GG K)

5 2
o69 = ideal(z z z - z )

32 35 44 37

o69 : Ideal of B

There is a general algorithm due to de Jong [6] for computing the normal-
ization of any affine variety. In the toric case, the problem of normalization
amounts to computing the minimal Hilbert basis of a given convex rational
polyhedral cone [20]. An efficient implementation can be found in the software
package Normaliz by Bruns and Koch [4].

Our computational study of the toric Hilbert scheme in this chapter was
based on local equations rather than global equations (arising from a pro-
jective embedding of HilbA), because the latter system of equations tends
to be too large for most purposes. Nonetheless, they are interesting. In the
remainder of this section, we present a canonical projective embedding of the
coherent component of HilbA.

Let G1, G2, G3, . . . , Gs denote all the Graver fibers of the matrix A. In
Section 1 we showed how to compute them in Macaulay 2. Each set Gi
consists of the monomials in C[x1, . . . , xn] that have a fixed Graver degree.
Consider the set G := G1G2G3 · · ·Gs that consists of all monomials that
are products of monomials, one from each of the distinct Graver fibers. Let
t denote the cardinality of G. We introduce an extra indeterminate z, and
we consider the N-graded semigroup algebra C[zG], which is a subalgebra of
C[x1, . . . , xn, z]. The grading of this algebra is deg(z) = 1 and deg(xi) = 0.
Labeling the elements of G with indeterminates yi, we can write

C[zG] = C[y1, y2, . . . , yt]/PA,

where PA is a homogeneous toric ideal associated with a configuration of t
vectors in Zn+1. We note that the torus (C∗)n acts naturally on C[zG].

Example 4.3. Let n = 4, d = 2 and A =
(

3 2 1 0
0 1 2 3

)
, so that IA is the ideal

of the twisted cubic curve. There are five Graver fibers:
i70 : A = {{1,1,1,1},{0,1,2,3}};

i71 : I = toricIdeal A;

o71 : Ideal of R

i72 : Graver = graver I;

1 5
o72 : Matrix R <--- R

i73 : fibers = graverFibers Graver;

i74 : peek fibers

o74 = HashTable{{2, 2} => | ac b2 | }
{2, 3} => | ad bc |
{2, 4} => | bd c2 |
{3, 3} => | a2d abc b3 |
{3, 6} => | ad2 bcd c3 |

The set G = G1G2G3G4G5 consists of 22 monomials of degree 14.
i75 : G = trim product(values fibers, ideal)

5 5 4 3 5 5 3 4 4 2 2 4 3 4 4 2 6 4 4 · · ·
o75 = ideal (a b*c*d , a b d , a c d , a b c d , a b c*d , a b d , a b · · ·
o75 : Ideal of R

i76 : numgens G

o76 = 22

We introduce a polynomial ring in 22 variables y1, y2, . . . , y22, and we
compute the ideal PA. It is generated by 180 binomial quadrics.

i77 : z = symbol z;

i78 : S = QQ[a,b,c,d,z];

i79 : zG = z ** substitute(gens G, S);

1 22
o79 : Matrix S <--- S

i80 : R = QQ[y_1 .. y_22];

i81 : F = map(S,R,zG)

5 5 4 3 5 5 3 4 4 2 2 4 3 4 4 2 6 · · ·
o81 = map(S,R,{a b*c*d z, a b d z, a c d z, a b c d z, a b c*d z, a b · · ·
o81 : RingMap S <--- R

i82 : PA = trim ker F

2 · · ·
o82 = ideal (y - y y , y y - y y , y y - y y , y y - · · ·

21 20 22 19 21 18 22 18 21 17 22 17 21 · · ·
o82 : Ideal of R

These equations define a toric surface of degree 30 in projective 21-space.
i83 : codim PA

o83 = 19

i84 : degree PA

o84 = 30

The surface is smooth, but there are too many equations and the codi-
mension is too large to use the Jacobian criterion for smoothness [7, §16.6]
directly. Instead we check smoothness for each open set yi 6= 0.

i85 : Aff = apply(1..22, v -> (
K = substitute(PA,y_v => 1);
FF = removeRedundantVariables K;
ideal gens gb (FF K)));

i86 : scan(Aff, i -> print toString i);
ideal()
ideal()
ideal()
ideal(y_1^4*y_5*y_21-1)
ideal(y_1^4*y_6^6*y_21-1)
ideal()
ideal(y_1^2*y_11^2*y_17-1)
ideal(y_1^3*y_9^2*y_21^2-1)
ideal(y_6^3*y_21-y_10,y_1*y_10^3-y_6^2,y_1*y_6*y_10^2*y_21-1)
ideal(y_6*y_15-1,y_2*y_15^2-y_6*y_14,y_6^2*y_14-y_2*y_15)
ideal()
ideal(y_11*y_13-1,y_1^2*y_21^3-y_13^2)
ideal(y_1^2*y_14^3*y_21^3-1)
ideal(y_10^2*y_21-1,y_1*y_15^4-y_10^3)
ideal()
ideal(y_11*y_20-1,y_3*y_20^2-y_11*y_17,y_11^2*y_17-y_3*y_20)
ideal(y_11*y_18*y_21-1,y_1*y_21^3-y_11*y_18^2,y_11^2*y_18^3-y_1*y_21^2)
ideal(y_1*y_19^4*y_21^4-1)
ideal(y_15*y_22-1)
ideal()
ideal(y_20*y_22-1)
ideal()

By examining these local equations, we see that HilbA is smooth, and
also that there are eight fixed points under the action of the 2-dimensional
torus. They correspond to the variables y1, y2, y3, y6, y11, y15, y20 and y22. By
setting any of these eight variables to 1 in the 180 quadrics above, we obtain
an affine variety isomorphic to the affine plane.

Theorem 4.4. The coherent component of the toric Hilbert scheme HilbA
is isomorphic to the projective spectrum Proj C[zG] of the algebra C[zG].

Proof. The first step is to define a morphism from HilbA to the (t − 1)-
dimensional projective space P(G) = Proj C[y1, y2, . . . , yt]. Consider any
point I on HilbA. We intersect the ideal I with the finite-dimensional vec-
tor space CGi, consisting of all homogeneous polynomials in C[x1, . . . , xn]
that lie in the i-th Graver degree. The definition of A-graded ideal implies
that I ∩ CGi is a linear subspace of codimension 1 in CGi. We represent

this subspace by an equation gi(I) =
∑
u∈Gi cux

u , which is unique up to
scaling. Taking the product of these polynomials for i = 1, . . . , t, we get a
unique (up to scaling) polynomial that is supported on G = G1G2 · · ·Gt. The
map I 7→ g1(I)g2(I) · · · gt(I) defines a morphism from HilbA to P(G). This
morphism is equivariant with respect to the (C∗)n-action on both schemes.

Consider the restriction of this equivariant morphism to the coherent com-
ponent of the toric Hilbert scheme. It maps the (C∗)n-orbit of the toric ideal
IA into the subvariety Proj C[zG] of P(G). This inclusion is an isomorphism
onto the dense torus, as the dimension of the Newton polytope of

g(IA) =
t∏
i=1

(
∑
u∈Gi

xu )

equals the dimension of the kernel of A. Equivalently, the stabilizer of g(IA)
in (C∗)n consists only of those one-parameter subgroups w that restrict to
zero on the kernel of A.

To show that our morphism is an isomorphism between the coherent com-
ponent and Proj C[zG], we consider the affine chart around an initial mono-
mial ideal M = inw(IA). The polynomial g(M) is a monomial, namely, it
is the product of all standard monomials whose degree is a Graver degree.
Moreover, g(M) is the leading monomial of g(IA) with respect to the weight
vector w. The Newton polytope of g(IA) is the Minkowski sum of the Newton
polytopes of the polynomials g1(IA), . . . , gt(IA), and it is a state polytope
for IA, by [22, Theorem 7.5].

Let g(M) = xq, and let σ be the cone of the Gröbner fan ΣA that has w in
its interior. Then σ coincides with the normal cone at the vertex q of the state
polytope described above [22, §3]. Consider the restriction of our morphism
to the affine chart around M of the coherent component, as described in (5).
This restriction defines an isomorphism onto the variety

Spec C[xp−q : xp ∈ G ] (9)

On the other hand, the semigroup algebra in (9) is isomorphic to that in (5)
because each pair of vectors {ui, vi} seen in the reduced Gröbner basis lies in
one of the Graver fibers Gj . Hence our morphism restricts to an isomorphism
from the affine chart around M of the coherent component onto (9). Finally,
note that (9) is the principal affine open subset of Proj C[zG] defined by the
coordinate xq. Hence we get an isomorphism between the coherent component
of HilbA and Proj C[zG]. ut

Appendix A. Fourier-Motzkin Elimination

We now give the Macaulay 2 code for converting the generator/inequality
representation of a rational convex polyhedron to the other. It is based on

the Fourier-Motzkin elimination procedure for eliminating a variable from a
system of inequalities [25]. This code was written by Greg Smith.

Given any cone C ⊂ Rd, the polar cone of C is defined to be

C∨ = {x ∈ Rd | x · y ≤ 0, for all y ∈ C}.

For a d×nmatrix Z, define cone(Z) = {Zx | x ∈ Rn≥0} ⊂ Rd, and affine(Z) =
{Zx | x ∈ Rn} ⊂ Rd. For two integer matrices Z and H, both having d rows,
polarCone(Z,H) returns a list of two integer matrices {A,E} such that

cone(Z) + affine(H) = {x ∈ Rd | Atx ≤ 0, Etx = 0}.

Equivalently, (cone(Z) + affine(H))∨ = cone(A) + affine(E).
We now describe each routine in the package polarCone.m2. We have

simplified the code for readability, sometimes at the cost of efficiency. We
start with three simple subroutines: primitive, toZZ, and rotateMatrix.

The routine primitive takes a list of integers L, and divides each element
of this list by their greatest common denominator.

i87 : code primitive

o87 = -- polarCone.m2:16-20
primitive = (L) -> (

n := #L-1; g := L#n;
while n > 0 do (n = n-1; g = gcd(g, L#n);

if g === 1 then n = 0);
if g === 1 then L else apply(L, i -> i // g));

The routine toZZ converts a list of rational numbers to a list of integers,
by multiplying by their common denominator.

i88 : code toZZ

o88 = -- polarCone.m2:28-32
toZZ = (L) -> (

d := apply(L, e -> denominator e);
R := ring d#0; l := 1_R;
scan(d, i -> (l = (l*i // gcd(l,i))));
apply(L, e -> (numerator(l*e))));

The routine rotateMatrix is a kind of transpose. Its input is a matrix,
and its output is a matrix of the same shape as the transpose. It places the
matrix in the form so that in the routine polarCone, computing a Gröbner
basis will do the Gaussian elimination that is needed.

i89 : code rotateMatrix

o89 = -- polarCone.m2:41-43
rotateMatrix = (M) -> (

r := rank source M; c := rank target M;
matrix table(r, c, (i,j) -> M_(c-j-1, r-i-1)));

The procedure of Fourier-Motzkin elimination as presented by Ziegler in
[25] is used, together with some heuristics that he presents as exercises. The
following, which is a kind of S-pair criterion for inequalities, comes from
Exercise 2.15(i) in [25].

The routine isRedundant determines if a row vector (inequality) is redun-
dant. Its input argument V is the same input that is used in fourierMotzkin:
it is a list of sets of integers. Each entry contains indices of the original rays
that do not vanish at the corresponding row vector. vert is a set of integers;
the original rays for the row vector in question. A boolean value is returned.

i90 : code isRedundant

o90 = -- polarCone.m2:57-65
isRedundant = (V, vert) -> (

-- the row vector is redundant iff ’vert’ contains an
-- entry in ’V’.
x := 0; k := 0;
numRow := #V; -- equals the number of inequalities
while x < 1 and k < numRow do (

if isSubset(V#k, vert) then x = x+1;
k = k+1;);

x === 1);

The main work horse of polarCone.m2 is the subroutine fourierMotzkin,
which eliminates the first variable in the inequalities A using the double de-
scription version of Fourier-Motzkin elimination. The set A is a list of lists
of integers, each entry corresponding to a row vector in the system of in-
equalities. The argument V is a list of sets of integers. Each entry contains
the indices of the original rays that do not vanish at the corresponding row
vector in A. Note that this set is the complement of the set Vi appearing in
exercise 2.15 in [25]. The argument spot is the integer index of the variable
being eliminated.

The routine returns a list {projA,projV} where projA is a list of lists of
integers. Each entry corresponds to a row vector in the projected system of
inequalities. The list projV is a list of sets of integers. Each entry contains
indices of the original rays that do not vanish at the corresponding row vector
in projA.

i91 : code fourierMotzkin

o91 = -- polarCone.m2:89-118
fourierMotzkin = (A, V, spot) -> (

-- initializing local variables
numRow := #A; -- equal to the length of V
numCol := #(A#0); pos := {};
neg := {}; projA := {};
projV := {}; k := 0;
-- divide the inequalities into three groups.
while k < numRow do (

if A#k#0 < 0 then neg = append(neg, k)
else if A#k#0 > 0 then pos = append(pos, k)
else (projA = append(projA, A#k);

projV = append(projV, V#k););
k = k+1;);

-- generate new irredundant inequalities.
scan(pos, i -> scan(neg, j -> (vert := V#i + V#j;

if not isRedundant(projV, vert)
then (iRow := A#i; jRow := A#j;

iCoeff := - jRow#0;
jCoeff := iRow#0;
a := iCoeff*iRow + jCoeff*jRow;
projA = append(projA, a);
projV = append(projV, vert););)));

-- don’t forget the implicit inequalities ’-t <= 0’.
scan(pos, i -> (vert := V#i + set{spot};

if not isRedundant(projV, vert) then (
projA = append(projA, A#i);
projV = append(projV, vert););));

-- remove the first column
projA = apply(projA, e -> e_{1..(numCol-1)});
{projA, projV});

As mentioned above, polarCone takes two matrices Z, H, both having d
rows, and outputs a pair of matrices A, E such that (cone(Z)+affine(H))∨ =
cone(A) + affine(E).

i92 : code(polarCone,Matrix,Matrix)

o92 = -- polarCone.m2:137-192
polarCone(Matrix, Matrix) := (Z, H) -> (

R := ring source Z;
if R =!= ring source H then error ("polarCone: " |

"expected matrices over the same ring");
if rank target Z =!= rank target H then error (

"polarCone: expected matrices to have the " |
"same number of rows");

if (R =!= ZZ) then error ("polarCone: expected " |
"matrices over ’ZZ’");

-- expressing ’cone(Y)+affine(B)’ as ’{x : Ax <= 0}’
Y := substitute(Z, QQ); B := substitute(H, QQ);
if rank source B > 0 then Y = Y | B | -B;
n := rank source Y; d := rank target Y;
A := Y | -id_(QQ^d);
-- computing the row echelon form of ’A’
A = gens gb rotateMatrix A;
L := rotateMatrix leadTerm A;
A = rotateMatrix A;
-- find pivots
numRow = rank target A; -- numRow <= d
i := 0; pivotCol := {};
while i < numRow do (j := 0;

while j < n+d and L_(i,j) =!= 1_QQ do j = j+1;
pivotCol = append(pivotCol, j);
i = i+1;);

-- computing the row-reduced echelon form of ’A’
A = ((submatrix(A, pivotCol))^(-1)) * A;
-- converting ’A’ into a list of integer row vectors
A = entries A;
A = apply(A, e -> primitive toZZ e);
-- creating the vertex list ’V’ for double description
-- and listing the variables ’T’ which remain to be
-- eliminated
V := {}; T := toList(0..(n-1));
scan(pivotCol, e -> (if e < n then (T = delete(e, T);

V = append(V, set{e});)));
-- separating inequalities ’A’ and equalities ’E’
eqnRow := {}; ineqnRow := {};
scan(numRow, i -> (if pivotCol#i >= n then

eqnRow = append(eqnRow, i)
else ineqnRow = append(ineqnRow, i);));

E := apply(eqnRow, i -> A#i);
E = apply(E, e -> e_{n..(n+d-1)});
A = apply(ineqnRow, i -> A#i);
A = apply(A, e -> e_(T | toList(n..(n+d-1))));
-- successive projections eliminate the variables ’T’.
if A =!= {} then scan(T, t -> (

D := fourierMotzkin(A, V, t);
A = D#0; V = D#1;));

-- output formating
A = apply(A, e -> primitive e);
if A === {} then A = map(ZZ^d, ZZ^0, 0)
else A = transpose matrix A;
if E === {} then E = map(ZZ^d, ZZ^0, 0)
else E = transpose matrix E;
(A, E));

If the input matrix H has no columns, it can be omitted. A sequence of
two matrices is returned, as above.

i93 : code(polarCone,Matrix)

o93 = -- polarCone.m2:199-200
polarCone(Matrix) := (Z) -> (

polarCone(Z, map(ZZ^(rank target Z), ZZ^0, 0)));

As a simple example, consider the permutahedron in R3 whose vertices
are the following six points.

i94 : H = transpose matrix{
{1,2,3},
{1,3,2},
{2,1,3},
{2,3,1},
{3,1,2},
{3,2,1}};

3 6
o94 : Matrix ZZ <--- ZZ

The inequality representation of the permutahedron is obtained by calling
polarCone on H: the facet normals of the polytope are the columns of the
matrix in the first argument of the output. The second argument is trivial
since our input is a polytope and hence there are is no non-trivial affine space
contained in it. If we call polarCone on the output, we will get back H as
expected.

i95 : P = polarCone H

o95 = (| 1 1 1 -1 -1 -5 |, 0)
| -1 1 -5 1 -1 1 |
| -1 -5 1 -1 1 1 |

o95 : Sequence

i96 : Q = polarCone P_0

o96 = (| 1 1 2 2 3 3 |, 0)
| 2 3 1 3 1 2 |
| 3 2 3 1 2 1 |

o96 : Sequence

Appendix B. Minimal Presentation of Rings

Throughout this chapter, we have used on several occasions the simple, yet
useful subroutine removeRedundantVariables. In this appendix, we present
Macaulay 2 code for this routine, which is the main ingredient for finding
minimal presentations of quotients of polynomial rings. Our code for this
routine is a somewhat simplified, but less efficient version of a routine in the
Macaulay 2 package, minPres.m2, written by Amelia Taylor.

The routine removeRedundantVariables takes as input an ideal I in
a polynomial ring A. It returns a ring map F from A to itself that sends
redundant variables to polynomials in the non-redundant variables and sends
non-redundant variables to themselves. For example:

i97 : A = QQ[a..e];

i98 : I = ideal(a-b^2-1, b-c^2, c-d^2, a^2-e^2)

2 2 2 2 2
o98 = ideal (- b + a - 1, - c + b, - d + c, a - e )

o98 : Ideal of A

i99 : F = removeRedundantVariables I

8 4 2
o99 = map(A,A,{d + 1, d , d , d, e})

o99 : RingMap A <--- A

The non-redundant variables are d and e. The image of I under F gives the
elements in this smaller set of variables. We take the ideal of a Gröbner basis
of the image:

i100 : I1 = ideal gens gb(F I)

16 8 2
o100 = ideal(d + 2d - e + 1)

o100 : Ideal of A

The original ideal can be written in a cleaner way as
i101 : ideal compress (F.matrix - vars A) + I1

8 4 2 16 8 2
o101 = ideal (d - a + 1, d - b, d - c, d + 2d - e + 1)

o101 : Ideal of A

Let us now describe the Macaulay 2 code. The subroutine findRedundant
takes a polynomial f , and finds a variable xi in the ring of f such that
f = cxi+g for a non-zero constant c and a polynomial g that does not involve

the variable xi. If there is no such variable, null is returned. Otherwise, if
xi is the first such variable , the list {i, c−1g} is returned.

i102 : code findRedundant

o102 = -- minPres.m2:1-12
findRedundant=(f)->(

A := ring(f);
p := first entries contract(vars A,f);
i := position(p, g -> g != 0 and first degree g === 0);
if i === null then

null
else (

v := A_i;
c := f_v;
{i,(-1)*(c^(-1)*(f-c*v))}
)

)

The main function removeRedundantVariables requires an ideal in a
polynomial ring (not a quotient ring) as input. The internal routine findnext
finds the first entry of the (one row) matrix M that contains a redundancy.
This redundancy is used to modify the list xmap, which contains the images
of the redundant variables. The matrix M, and the list xmap are both updated,
and then we continue to look for more redundancies.

i103 : code removeRedundantVariables

o103 = -- minPres.m2:14-39
removeRedundantVariables = (I) -> (

A := ring I;
xmap := new MutableList from gens A;
M := gens I;
findnext := () -> (

p := null;
next := 0;
done := false;
ngens := numgens source M;
while next < ngens and not done do (

p = findRedundant(M_(0,next));
if p =!= null then

done = true
else next=next+1;

);
p);

p := findnext();
while p =!= null do (

xmap#(p#0) = p#1;
F1 := map(A,A,toList xmap);
F2 := map(A,A, F1 (F1.matrix));
xmap = new MutableList from first entries F2.matrix;
M = compress(F2 M);
p = findnext();
);

map(A,A,toList xmap));

10. S. Hoşten and D. Maclagan: The vertex ideal of a lattice. Preprint 2000.

Sheaf Algorithms Using the Exterior Algebra

Wolfram Decker and David Eisenbud

In this chapter we explain constructive methods for computing the cohomol-
ogy of a sheaf on a projective variety. We also give a construction for the
Beilinson monad, a tool for studying the sheaf from partial knowledge of
its cohomology. Finally, we give some examples illustrating the use of the
Beilinson monad.

1 Introduction

In this chapter V denotes a vector space of finite dimension n+ 1 over a field
K with dual space W = V ∗, and S = SymK(W ) is the symmetric algebra of
W , isomorphic to the polynomial ring on a basis for W . We write E for the
exterior algebra on V . We grade S and E by taking elements of W to have
degree 1, and elements of V to have degree −1. We denote the projective
space of 1-quotients of W (or of lines in V ) by Pn = P(W ).

Serre’s sheafification functor M 7→ M̃ allows one to consider a coherent
sheaf on P(W ) as an equivalence class of finitely generated graded S-modules,
where we identify two such modules M and M ′ if, for some r, the truncated
modules M≥r and M ′≥r are isomorphic. A free resolution of M , sheafified,
becomes a resolution of M̃ by sheaves that are direct sums of line bundles on
P(W ) – that is, a description of M̃ in terms of homogeneous matrices over S.
Being able to compute syzygies over S one can compute the cohomology of
M̃ starting from the minimal free resolution of M (see [16], [40] and Remark
3.2 below).

The Bernstein-Gel’fand-Gel’fand correspondence (BGG) is an isomor-
phism between the derived category of bounded complexes of finitely ge-
nerated S-modules and the derived category of bounded complexes of finitely
generated E-modules or of certain “Tate resolutions” of E-modules. In this
chapter we show how to effectively compute the Tate resolution T(F) as-
sociated to a sheaf F , and we use this construction to give relatively cheap
computations of the cohomology of F .

It turns out that by applying a simple functor to the Tate resolution
T(F) one gets a finite complex of sheaves whose homology is the sheaf F
itself. This complex is called a Beilinson monad for F . The Beilinson monad
provides a powerful method for getting information about a sheaf from partial
knowledge of its cohomology. It is a representation of the sheaf in terms of
direct sums of (suitably twisted) bundles of differentials and homomorphisms
between these bundles, which are given by homogeneous matrices over E.

The following recipe for computing the cohomology of a sheaf is typical
of our methods: Suppose that F = M̃ is the coherent sheaf on P(W ) asso-

ciated to a finitely generated graded S-module M = ⊕Mi. To compute the
cohomology of F we consider a sequence of free E-modules and maps

F(M) : · · · - F i−1 φi−1- F i
φi- F i+1 - · · · .

Here we set F i = Mi ⊗K E and define φi : F i - F i+1 to be the map
taking m⊗ 1 ∈Mi ⊗K E to∑

j

xjm⊗ ej ∈Mi+1 ⊗ V ⊂ F i+1,

where {xj} and {ej} are dual bases of W and V respectively. It turns out
that F(M) is a complex; that is, φiφi−1 = 0 for every i (the reader may easily
check this by direct computation; a proof without indices is given in [18]).
If we regard Mi as a vector space concentrated in degree i, so that F i is a
direct sum of copies of E(−i), then these maps are homogeneous of degree 0.

We shall see that if s is a sufficiently large integer then the truncation of
the Tate resolution

F s
φs- F s+1 - · · ·

is exact and is thus the minimal injective resolution of the finitely generated
graded E-module Ps = kerφs+1. (In fact any value of s greater than the
Castelnuovo-Mumford regularity of M will do.)

Because the number of monomials in E in any given degree is small com-
pared to the number of monomials of that degree in the symmetric algebra,
it is relatively cheap to compute a free resolution of Ps over E, and thus to
compute the graded vector spaces TorEt (Ps,K). Our algorithm exploits the
fact, proved in [18], that the j

th
cohomology HjF of F in the Zariski topology

is isomorphic to the degree −n− 1 part of TorEs−j(Ps,K); that is,

HjF ∼= TorEs−j(Ps,K)−n−1.

In addition, the linear parts of the matrices in the complex T(F) determine
the graded S-modules

Hj
∗F := ⊕i∈ZHjF(i) .

In many cases this is the fastest known method for computing cohomology.
Section 2 of this paper is devoted to a sketch of the Eisenbud-Fløystad-

Schreyer approach to the Bernstein-Gel’fand-Gel’fand correspondence, and
the computation of cohomology, together with Macaulay 2 programs that
carry it out, is explained in Section 3.

The remainder of this paper is devoted to an explanation of the Beilinson
monad, how to compute it in Macaulay 2, and what it is good for. This
technique has played an important role in the construction and study of
vector bundles and varieties. In the typical application one constructs or
classifies monads in order to construct or classify sheaves.

The BGG correspondence and Beilinson’s monad were originally formu-
lated in the language of derived categories, and the proofs were rather compli-
cated. The ideas of Eisenbud-Fløystad-Schreyer exposed above allow, for the
first time, an explanation of these matters on a level that can be understood
by an advanced undergraduate.

The Beilinson monad is similar in spirit to the technique of free re-
solutions. That theory essentially describes arbitrary sheaves by comparing
them with direct sums of line bundles. In the Beilinson technique, one uses
a different set of “elementary” sheaves, direct sums of exterior powers of
the tautological sub-bundle. Beilinson’s remarkable observation was that this
comparison has a much more direct connection with cohomology than does
the free resolution method.

Sections 4 and 5 are introductory in nature. In Section 4 we begin with
a preparatory discussion of the necessary vector bundles on projective space
and their cohomology. In Section 5 we define monads, a generalization of
resolutions. We give a completely elementary account which constructs the
Beilinson monad in a very special case, following ideas of Horrocks, and we
use this to sketch part of one of the first striking applications of monads:
the classification of stable rank 2 vector bundles on the projective plane by
Barth, Hulek and Le Potier.

In Section 6 we give the construction of Eisenbud-Fløystad-Schreyer for
the Beilinson monad in general. This is quite suitable for computation, and
we give Macaulay 2 code that does this job.

A natural question for the student at this point is: “Why should I bother
learning Beilinson’s theorem, what is it good for?” In section 7, we describe
two more explicit applications of the theory developed. In the first, the classi-
fication of elliptic conic bundles in P4, computer algebra played a significant
role, demonstrating that several published papers contained serious mistakes
by constructing an example they had excluded! Using the routines developed
earlier in the chapter we give a simpler account of the crucial computation.

In the second application, the construction of abelian surfaces in P4

and the related Horrocks-Mumford bundles, computer algebra allows one to
greatly shorten some of the original arguments made. As the reader will see,
everything follows easily with computation, once a certain 2 × 5 matrix of
exterior monomials, given by Horrocks and Mumford, has been written down.
One might compare the computations here with the original paper of Hor-
rocks and Mumford [25] (for the cohomology) and the papers by Manolache
[32] and Decker [13] (for the syzygies) of the Horrocks-Mumford bundle. A
great deal of effort, using representation theory, was necessary to derive re-
sults that can be computed in seconds using the Macaulay 2 programs here.
Much more theoretical effort, however, is needed to derive classification re-
sults.

Another application of the construction of the Beilinson complex (in a
slightly more general setting) is to compute Chow forms of varieties; see [19].

Perhaps the situation is similar to that in the beginning of the 1980’s
when it became clear that syzygies could be computed by a machine. Though
syzygies had been used theoretically for many years it took quite a while
until the practical computation of syzygies lead to applications, too, mostly
through the greatly increased ability to study examples.

A good open problem of this sort is to extend and make more precise
the very useful criterion given in 4.4: Can the reader find a necessary and
sufficient condition to replace the necessary condition for surjectivity given
there? How about a criterion for exactness?

2 Basics of the Bernstein-Gel’fand-Gel’fand
Correspondence

In this section we describe the basic idea of the BGG correspondence, intro-
duced in [8]. For a more complete treatment along the lines given here, see
the first section of [18].

As a simple example of the construction given in Section 1, consider the
case M = S = SymK(W ). The associated complex, made from the homoge-
neous components Symi(W ) of S, has the form

F(S) : E - W ⊗ E - Sym2(W )⊗ E - · · · ,

where we regard SymiW as concentrated in degree i. It is easy to see that the
kernel of the first map, E - W⊗E, is exactly the socle

∧n+1
V ⊂ E, which

is a 1-dimensional vector space concentrated in degree −n− 1. In fact F(S)
is the minimal injective resolution of this vector space. If we tensor with the
dual vector space

∧n+1
W (which is concentrated in degree n+ 1), we obtain

the minimal injective resolution of the vector space
∧n+1

W ⊗
∧n+1

V , which
may be identified canonically with the residue field K of E. This resolution
is called the Cartan resolution of K. To write it conveniently, we set ωE =∧n+1

W ⊗E. The socle of ωE is K. Since E is injective (as well as projective)
as an E-module, the same goes for ωE , so ωE is the injective envelope of the
residue class field K and we have ωE = HomK(E,K). Thus we can write the
injective resolution of the residue field as

R(S) : ωE - W ⊗ ωE - Sym2(W )⊗ ωE - · · · ,

or again as

HomK(E,K) - HomK(E,W ) - HomK(E,Sym2(W )) - · · · .

Taking our cue from this situation, our primary object of study in the case
of an arbitrary finitely generated graded S-module M = ⊕Mi will be the
complex

R(M) : · · · - Mi ⊗ ωE - Mi+1 ⊗ ωE - · · · ,

which will have a more natural grading than F(M); in any case, it differs
from F(M) only by tensoring over K with the one-dimensional K-vector
space

∧n+1
W , concentrated in degree n + 1, and thus has the same basic

properties. (Writing R(M) in terms of Hom as above suggests that the functor
R might have a left adjoint, and indeed there is a left adjoint that produces
linear free complexes over S from graded E-modules. R and its left adjoint
are used to construct the isomorphisms of derived categories in the BGG
correspondence; see [18] for a treatment in this spirit.)

An important fact for us is that the complex R(M) is eventually exact
(and thus

F i
φi- F i+1 - · · ·

is the minimal injective resolution of kerφi when i� 0). It turns out that the
point at which exactness sets in is a well-known invariant, the Castelnuovo-
Mumford regularity of M , whose definition we briefly recall:

If M = ⊕Mi is a finitely generated graded S-module then for all large
integers r the submodule M≥r ⊂M is generated in degree r and has a linear
free resolution; that is, its first syzygies are generated in degree r + 1, its
second syzygies in degree r+ 2, etc. (see [17, chapter 20]). The Castelnuovo-
Mumford regularity of M is the least integer r for which this occurs.

Theorem 2.1 ([18]). Let M be a finitely generated graded S-module of Cas-
telnuovo-Mumford regularity r. The complex R(M) is exact at HomK(E,Mi)
for all i ≥ s if and only if s > r. ut

More generally, it is shown in [18] that the components of the cohomology
of the complex R(M) can be identified with the Koszul cohomology of M .
An equivalent result was stated in [10].

For instance, it is not hard to show that if M is of finite length, then the
regularity of M is the largest i such that Mi 6= 0. Let us verify Theorem 2.1
directly in a simple example:

Example 2.2. Let S = K[x0, x1, x2], and let M = S/(x2
0, x

2
1, x

2
2). The mod-

ule M≥3 = K · x0x1x2 is a trivial S-module, and its resolution is the Koszul
complex on x0, x1 and x2, which is linear. Thus the Castelnuovo-Mumford
regularity of M is ≤ 3. On the other hand M≥2 is, up to twist, isomorphic
to the dual of S/(x0, x1, x2)2, and it follows that the resolution of M≥2 has
the form

0 - S(−6) - 6S(−4) - 8S(−3) - 3S(−2),

which is not linear, so the Castelnuovo-Mumford regularity of M is exactly
3. Note that the regularity is larger than the degrees of the generators and
relations of M—in general it can be much larger.

Over E the linear free complex corresponding to M has the form

· · · → 0→M0 ⊗ ωE →M1 ⊗ ωE →M2 ⊗ ωE →M3 ⊗ ωE → 0→ · · · ,

where all the terms not shown are 0. Using the isomorphism ωE ∼= E(−3)
this can be written (non-canonically) as

0 - E(−3)


e0

e1

e2


- 3E(−2)


0 e2 e1

e2 0 e0

e1 e0 0


- 3E(−1)

(
e0 e1 e2

)
- E - 0.

One checks easily that this complex is inexact at every non-zero term (despite
its resemblance to a Koszul complex), verifying Theorem 2.1. ut

Another case in which everything can be checked directly occurs when M
is the homogeneous coordinate ring of a point:

Example 2.3. Take M = S/I where I is generated by a codimension 1 space
of linear forms in W , so that I is the homogeneous ideal of a point p ∈ P(W ).
The free resolution of M is the Koszul complex on n linear forms, so M is
0-regular. As Mi is 1-dimensional for every i the terms of the complex R(M)
are all rank 1 free E-modules. One easily checks that R(M) takes the form

R(M) : ωE
a- ωE(−1)

a- ωE(−2)
a- · · · ,

where a ∈ V = W ∗ is a linear functional that vanishes on all the linear
forms in I; that is, a is a generator of the one-dimensional subspace of V
corresponding to the point p. As for any linear form in E, the annihilator of
a is generated by a, and it follows directly that the complex R(M) is acyclic
in this case. ut

We present two Macaulay 2 functions, symExt and bgg, which compute
a differential of the complex R(M) for a finitely generated graded module
M defined over some polynomial ring S = K[x0, . . . , xn] with variables xi of
degree 1. Both functions expect as an additional input the name of an exterior
algebra E with the same number n + 1 of generators, also supposed to be
of degree 1 (and NOT -1). This convention, which makes the cohomology
diagrams more naturally looking when printed in Macaulay 2, necessitates
the adjustment of degrees in the second half of the programs.

The first of the functions, symExt, takes as input a matrix m with linear
entries, which we think of as a presentation matrix for a positively graded S-
module M = ⊕i≥0Mi, and returns a matrix representing the map M0⊗ωE →
M1 ⊗ ωE which is the first differential of the complex R(M).

i1 : symExt = (m,E) ->(
ev := map(E,ring m,vars E);
mt := transpose jacobian m;
jn := gens kernel mt;
q := vars(ring m)**id_(target m);
ans:= transpose ev(q*jn);
--now correct the degrees:
map(E^{(rank target ans):1}, E^{(rank source ans):0},

ans));

If M is a module whose presentation is not linear in the sense above, we can
still apply symExt to a high truncation of M :

i2 : S=ZZ/32003[x_0..x_2];

i3 : E=ZZ/32003[e_0..e_2,SkewCommutative=>true];

i4 : M=coker matrix{{x_0^2, x_1^2}};

i5 : m=presentation truncate(regularity M,M);

4 8
o5 : Matrix S <--- S

i6 : symExt(m,E)

o6 = {-1} | e_2 e_1 e_0 0 |
{-1} | 0 e_2 0 e_0 |
{-1} | 0 0 e_2 e_1 |
{-1} | 0 0 0 e_2 |

4 4
o6 : Matrix E <--- E

The function symExt is a quick-and-dirty tool which requires little compu-
tation. If it is called on two successive truncations of a module the maps it
produces may NOT compose to zero because the choice of bases is not con-
sistent. The second function, bgg, makes the computation in such a way that
the bases are consistent, but does more computation to achieve this end. It
takes as input an integer i and a finitely generated graded S-module M , and
returns the ith map in R(M), which is an “adjoint” of the multiplication map
between Mi and Mi+1.

i7 : bgg = (i,M,E) ->(
S :=ring(M);
numvarsE := rank source vars E;
ev:=map(E,S,vars E);
f0:=basis(i,M);
f1:=basis(i+1,M);
g :=((vars S)**f0)//f1;
b:=(ev g)*((transpose vars E)**(ev source f0));
--correct the degrees (which are otherwise
--wrong in the transpose)
map(E^{(rank target b):i+1},E^{(rank source b):i}, b));

For instance, in Example 2.2:

i8 : M=cokernel matrix{{x_0^2, x_1^2, x_2^2}};

i9 : bgg(1,M,E)

o9 = {-2} | e_1 e_0 0 |
{-2} | e_2 0 e_0 |
{-2} | 0 e_2 e_1 |

3 3
o9 : Matrix E <--- E

3 The Cohomology and the Tate Resolution of a Sheaf

Given a finitely generated graded S-module M we construct a (doubly infi-
nite) E-free complex T(M) with vanishing homology, called the Tate reso-
lution of M , as follows: Let r be the Castelnuovo-Mumford regularity of M .
The truncation T>r(M), the part of T(M) with cohomological degree > r,
is R(M>r). We complete this to an exact complex by adjoining a minimal
projective resolution of the kernel of HomK(E,Mr+1)→ HomK(E,Mr+2).

If, for example, M has finite length as in Example 2.2, the Tate resolution
of M is the complex

· · · → 0→ 0→ 0→ · · · .
At the opposite extreme, take M = S, the free module of rank 1. Since S
has regularity 0, it follows that R(S) is an injective resolution of the residue
field K of E. Applying the exact functor HomK(—,K), and using the fact
that it carries ωE = HomK(E,K) back to E, we see that the Tate resolution
T(S) is the first row of the diagram

· · · // W ∗ ⊗ E // E

��@@@@@@@@
// ωE // W ⊗ ωE // · · ·

K

>>||||||||

Another simple example occurs in the case where M is the homogeneous
coordinate ring of a point p ∈ P(W ). The complex R(M) constructed in
Example 2.3 is periodic, so it may be simply continued to the left, giving

T(M) : · · · a- ωE(i)
a- ωE(i− 1)

a- · · · ,

where again a ∈ V = W ∗ is a non-zero linear functional vanishing on the
linear forms in the ideal of p.

For arbitrary M , by the results of the previous section, R(M>r) has no
homology in cohomological degree > r + 1, so T(M) could be constructed
by a similar recipe from any truncation R(M>s) with s ≥ r. Thus the Tate
resolution depends only on the sheaf M̃ on P(W ) corresponding to M . We
sometimes write T(M) as T(M̃) to emphasize this point.

Using the Macaulay 2 function symExt of the last section, one can com-
pute any finite piece of the Tate resolution.

i10 : tateResolution = (m,E,loDeg,hiDeg)->(
M := coker m;
reg := regularity M;
bnd := max(reg+1,hiDeg-1);
mt := presentation truncate(bnd,M);
o := symExt(mt,E);
--adjust degrees, since symExt forgets them
ofixed := map(E^{(rank target o):bnd+1},

E^{(rank source o):bnd},
o);

res(coker ofixed, LengthLimit=>max(1,bnd-loDeg+1)));

tateResolution takes as input a presentation matrix m of a finitely gener-
ated graded module M defined over some polynomial ring S = K[x0, . . . , xn]
with variables xi of degree 1, the name of an exterior algebra E with the same
number n + 1 of generators, also supposed to be of degree 1, and two inte-
gers, say l and h. If r is the regularity of M , then tateResolution(m,E,l,h)
computes the piece

T l(M)→ · · · → T max(r+2,h)(M)

of T(M). For instance, for the homogeneous coordinate ring of a point in the
projective plane:

i11 : m = matrix{{x_0,x_1}};

1 2
o11 : Matrix S <--- S

i12 : regularity coker m

o12 = 0

i13 : T = tateResolution(m,E,-2,4)

1 1 1 1 1 1 1
o13 = E <-- E <-- E <-- E <-- E <-- E <-- E

0 1 2 3 4 5 6

o13 : ChainComplex

i14 : betti T

o14 = total: 1 1 1 1 1 1 1
-4: 1 1 1 1 1 1 1

i15 : T.dd_1

o15 = {-4} | e_2 |

1 1
o15 : Matrix E <--- E

For arbitrary M we have Mi = H0M̃(i) for large i, so the correspon-
ding term of the complex T(M̃) with cohomological degree i is Mi ⊗ ωE =
H0(M̃(i)) ⊗ ωE . The following result generalizes this to a description of all
the terms of the Tate resolution, and gives the formula for the cohomology
described in the introduction.

Theorem 3.1 ([18]). Let M be a finitely generated graded S-module. The
term of the complex T(M) = T(M̃) with cohomological degree i is

⊕jHjM̃(i− j)⊗ ωE ,

where HjM̃(i− j) is regarded as a vector space concentrated in degree i− j,
so that the summand HjM̃(i−j)⊗ωE is isomorphic to a direct sum of copies

of ωE(j − i). Moreover the subquotient complex

· · · → HjM̃(i− j)⊗ ωE → HjM̃(i+ 1− j)⊗ ωE → · · ·

is R(Hj
∗(M̃(−j)))(j) (up to twists and shifts it is R(Hj

∗M̃). ) ut

Thus each cohomology group of each twist of the sheaf M̃ occurs (exactly
once) in a term of T(M). When we compute a part of T(M), we are comput-
ing the sheaf cohomology of various twists of the associated sheaf together
with maps which describe the S-module structure of Hj

∗M̃ in the sense that
the linear maps in this complex are adjoints of the multiplication maps that
determine the module structure (the multiplication maps themselves could
be computed by a function similar to bgg). The higher degree maps in the
complex T(M) determine certain higher cohomology operations, which we
understand only in very special cases (see [19]).

If M = coker m, then betti tateResolution(m,E,l,h) prints the di-
mensions hjM̃(i− j) = dim HjM̃(i− j) for max(r + 2, h) ≥ i ≥ l, where r is
the regularity of M . Truncating the Tate resolution if necessary allows one
to restrict the size of the output.

i16 : sheafCohomology = (m,E,loDeg,hiDeg)->(
T := tateResolution(m,E,loDeg,hiDeg);
k := length T;
d := k-hiDeg+loDeg;
if d > 0 then

chainComplex apply(d+1 .. k, i->T.dd_(i))
else T);

The expression betti sheafCohomology(m,E,l,h) prints a cohomology ta-
ble for M̃ of the form

h0M̃(h) . . . h0M̃(l)
h1M̃(h− 1) . . . h1M̃(l − 1)

...
...

hnM̃(h− n) . . . hnM̃(l − n) .

As a simple example we consider the cotangent bundle on projective 3-space
(see the next section for the Koszul resolution of this bundle):

i17 : S=ZZ/32003[x_0..x_3];

i18 : E=ZZ/32003[e_0..e_3,SkewCommutative=>true];

The cotangent bundle is the cokernel of the third differential of the Koszul
complex on the variables of S.

i19 : m=koszul(3,vars S);

6 4
o19 : Matrix S <--- S

i20 : regularity coker m

o20 = 2

i21 : betti tateResolution(m,E,-6,2)

o21 = total: 45 20 6 1 4 15 36 70 120 189 280
-4: 45 20 6 . . . . . . . .
-3: . . . 1 . . . . . . .
-2: . . . . . . . . . . .
-1: . . . . 4 15 36 70 120 189 280

i22 : betti sheafCohomology(m,E,-6,2)

o22 = total: 6 1 4 15 36 70 120 189 280
-2: 6 . . . . . . . .
-1: . 1 . . . . . . .
0: . . . . . . . . .
1: . . 4 15 36 70 120 189 280

Of course these two results differ only in the precise point of truncation.

Remark 3.2. There is also a built-in sheaf cohomology function HH in Mac-
aulay 2 which is based on the algorithms in [16]. These algorithms are often
much slower than sheafCohomology. To access it, first execute

M=sheaf coker m;

and pick integers j and d. Then

HH^j(M(>=d))

returns the truncated jth cohomology module Hj
i≥dM̃ . In the above exam-

ple of the cotangent bundle F on projective 3-space we obtain the Koszul
presentation of H1F ∼= K considered as an S-module sitting in degree 0:

i23 : M=sheaf coker m;

i24 : HH^1(M(>=0))

o24 = cokernel | x_3 x_2 x_1 x_0 |

1
o24 : S-module, quotient of S

ut

The Tate resolutions of sheaves are, as the reader may easily check, pre-
cisely the doubly infinite, graded, exact complexes of finitely-generated free
E-modules which are “eventually linear” on the right, in an obvious sense.
What about other doubly exact graded free complexes? For example what if
we take the dual of the Tate resolution of a sheaf? In general it will not be
eventually linear. What is it?

To explain this we must generalize the construction of R(M): If

M• : · · · - M i+1 - M i - M i−1 - · · ·

is a complex of S-modules, then applying the functor R gives a complex of
free complexes over E. By changing some signs we get a double complex.
In general the associated total complex is not minimal; but at least if M•

is a bounded complex then, just as one produces the unique minimal free
resolution of a module from any free resolution, we can construct a unique
minimal complex from it. We call this minimal complex R(M•). (See [18] for
more information. This construction is a necessary part of interpreting the
BGG correspondence as an equivalence of derived categories.)

Again if M• is a bounded complex of finitely generated modules, then as
before one shows that R(M•) is exact from a certain point on, and so we can
form the Tate resolution T(M•) by adjoining a free resolution of a kernel.
Once again, the Tate resolution depends only on the bounded complex of
coherent sheaves F• associated to M•, and we write T(F•) = T(M•).

A variant of the theorem of Bernstein, Gel’fand and Gel’fand shows that
every minimal graded doubly infinite exact sequence of finitely generated free
E-modules is of the form T(F•) for some complex of coherent sheaves F•,
unique up to quasi-isomorphism. The terms of the Tate resolution can be
expressed using hypercohomology by a formula like that of Theorem 3.1.

One way that interesting complexes of sheaves arise is through duality.
For simplicity, write O for the structure sheaf OP(W ). If F = M̃ is a sheaf on
P(W ) then the derived functor RHom(F ,O) may be computed by applying
the functor Hom(—,O) to a sheafified free resolution of M ; it’s value is thus
a complex of sheaves rather than an individual sheaf.

We can now identify the dual of the Tate resolution:

Theorem 3.3. HomK(T(F),K) ∼= T(RHom(F ,O))[1]. ut

Here the [1] denotes a shift by one in cohomological degree. For example,
take F = O. We have RHom(O,O) = O. The Tate resolution is given by

T(O) : · · · // E // ωE // · · ·
−1 0

where the number under each term is its cohomological degree. Taking into
account ωE = HomK(E,K), the dual of the Tate resolution is thus

HomK(T(O),K) : · · · ωEoo Eoo · · ·oo

1 0

which is the same as T(O)[1]. A completely analogous computation gives the
proof of Theorem 3.3 if F = O(a) for some a, and the general case follows
by taking free resolutions.

4 Cohomology and Vector Bundles

In this section we first recall how vector bundles, direct sums of line bundles,
and bundles of differentials can be characterized among all coherent sheaves

on P(W ) in terms of cohomology (as usual we do not distinguish between
vector bundles and locally free sheaves). Then we describe the homomor-
phisms between the suitably twisted bundles of differentials in terms of the
exterior algebra E. This description plays an important role in the context
of Beilinson monads.

Vector bundles on P(W ) are characterized by a criterion of Serre [39]
which can be formulated as follows: A coherent sheaf F on P(W ) is locally
free if and only if its module of sections H0

∗(F) is finitely generated and its
intermediate cohomology modules Hj

∗F , 1 ≤ j ≤ n− 1, are of finite length.
From a cohomological point of view, the simplest vector bundles are the

direct sums of line bundles. Every vector bundle on the projective line splits
into a direct sum of line bundles by Grothendieck’s splitting theorem (see
[37]). Induction yields Horrocks’ splitting theorem (see [5]): A vector bundle
on P(W ) splits into a direct sum of line bundles if and only if its intermediate
cohomology vanishes (originally, this theorem was proved as a corollary to a
more general result, see [23] and [42]).

Just a little bit more complicated are the bundles of differentials. To fix
our notation in this context we writeO = OP(W ), W⊗O for the trivial bundle
on P(W ) with fiber W , U = ΩP(W )(1) for the cotangent bundle twisted by
1, and

U i =
∧i
U =

∧i(ΩP(W )(1)) = ΩiP(W )(i)

for the ith bundle of differentials twisted by i; in particular U0 = O, Un ∼=
O(−1), and U i = 0 if i < 0 or i > n.

Remark 4.1. For each 0 ≤ i ≤ n the pairing

U i ⊗ Un−i ∧−→ Un ∼= O(−1)

induces an isomorphism

Un−i ∼= (U i)∗(−1) . ut

The fiber of U at the point of P(W ) corresponding to the line 〈a〉 ⊂ V is
the subspace (V/〈a〉)∗ ⊂W . Thus U fits into the short exact sequence

0→ U →W ⊗O → O(1)→ 0 .

In fact, U is the tautological subbundle of W ⊗O. Taking exterior powers, we
get the short exact sequences

0→ U i+1 →
∧i+1

W ⊗O → U i ⊗O(1)→ 0 .

Twisting the i
th

sequence by −i − 1, and gluing them together we get the
exact sequence

0 //∧n+1
W ⊗O(−n− 1) // · · · //∧0

W ⊗O //0 .

This sequence is the sheafification of the Koszul complex, which is the free
resolution of the “trivial” graded S-module K = S/(W ).

Remark 4.2. By taking cohomology in the short exact sequences above we
find that

Hj
∗U

i =

{
K(i) j = i,

0 j 6= i,
1 ≤ i, j ≤ n− 1 ,

where K(i) = (S/(W ))(i). Conversely, every vector bundle F on P(W ) with
this intermediate cohomology is stably equivalent to U i; that is, there exists a
direct sum L of line bundles such that F ∼= U i⊕L. This follows by comparing
the sheafified Koszul complex with the minimal free resolution of the dual
bundle F∗. ut

In what follows we describe the homomorphisms between the various
U i, 0 ≤ i ≤ n. Note that since U = U1 ⊂ W ⊗ O each element of V =
HomK(W,K) induces a homomorphism U1 → U0 which is the composite

U1 ⊂W ⊗O → K ⊗O = O = U0.

Similarly, using the diagonal map of the exterior algebra U i =
∧i

U → U ⊗
U i−1, each element of V induces a homomorphism U i → U i−1 which is the
composite

U i → U ⊗ U i−1 →W ⊗ U i−1 → K ⊗ U i−1 = U i−1.

It is not hard to show that these maps induced by elements of V anticommute
with each other (see for example [17, A2.4.1]). Thus we get maps

∧j
V →

Hom(U i, U i−j) which together give a graded ring homomorphism
∧
V →

Hom(⊕iU i,⊕iU i). In fact this construction gives all the homomorphisms
between the U i:

Lemma 4.3. The maps∧j
V → Hom(U i, U i−j), 0 ≤ i, i− j ≤ n ,

described above are isomorphisms. Under these isomorphisms an element e ∈∧j
V acts by contraction on the fibers of the U i:∧i(V/〈a〉)∗

��

� � // ∧iW
e

��∧i−j(V/〈a〉)∗ � � // ∧i−jW .

Proof. Every homomorphism U i → U i−j lifts uniquely to a homomorphism
between shifted Koszul complexes:

0 //∧n+1
W ⊗O(i− n− 1)

��

// · · · //∧jW ⊗O(i− j)

��

// · · ·

· · · //∧n+1−j
W ⊗O(i− n− 1) // · · · //O(i− j) //0

Indeed, the corresponding obstructions vanish by Remarks 4.1 and 4.2. All
results follow since the vertical arrows are necessarily given by contraction
with an element in

Hom(
∧j
W ⊗O(i− j),O(i− j)) ∼=

∧j
V . ut

In practical terms, these results say that a map U i e−→ U i−j is represented
as ∧i+1

W ⊗O(−1)

e

��

// // U i

��∧i−j+1
W ⊗O(−1) // // U i−j

if 0 < i− j ≤ i ≤ n, and as the composite

∧i+1
W ⊗O(−1) // // U i

��

� � // ∧iW ⊗O
e

��
U0 = O

if 0 = i− j < i ≤ n.
A map from a sum of copies of various U i to another such sum is given

by a homogeneous matrix over the exterior algebra E. In general it is an
interesting problem to relate properties of the matrix to properties of the
map. Here is one relation which is easy. We will apply it later on in this
chapter.

Proposition 4.4. If

r U i
B−→ sU i−1

is a homomorphism, that is, if B is an s× r-matrix with entries in V , then
the following condition is necessary for B to be surjective: If (b1, . . . , br) is a
non-trivial linear combination of the rows of B, then

dim span(b1, . . . , br) ≥ i+ 1.

Proof. B is surjective if and only if its dual map is injective on fibers:

s
∧i−1(V/〈a〉) ∧B

t

−→ r
∧i(V/〈a〉)

is injective for any line 〈a〉 ⊂ V . Consider a non-trivial linear combination
(b1, . . . , br)t of the columns of Bt, and write d = dim span(b1, . . . , br). If d = i,
then Bt is not injective at any point of P(W ) corresponding to a vector in
span(b1, . . . , br). If d < i, then Bt is not injective at any point of P(W ). ut

5 Cohomology and Monads

The technique of monads provides powerful tools for problems such as the
construction and classification of coherent sheaves with prescribed invariants.
This section is an introduction to monads. We demonstrate their usefulness,
which is not obvious at first glance, by reviewing the classification of stable
rank 2 vector bundles on the projective plane (see [4], [31], and [26]). Recall
that stable bundles admit moduli (see [22], [33], and [34]).

The basic idea behind monads is to represent arbitrary coherent sheaves
in terms of simpler sheaves such as line bundles or bundles of differentials,
and in terms of homomorphisms between these simpler sheaves. If M is a
finitely generated graded S-module, with associated sheaf F = M̃ , then the
sheafification of the minimal free resolution of M is a monad for F which
involves direct sums of line bundles and thus homogeneous matrices over S.
The Beilinson monad for F , which will be considered in the next section,
involves direct sums of twisted bundles of differentials U i, and thus homoge-
neous matrices over E.

Definition 5.1. A monad on P(W ) is a bounded complex

· · · −→ K−1 −→ K0 −→ K1 −→ · · ·

of coherent sheaves on P(W ) which is exact except at K0. The homology F
at K0 is called the homology of the monad, and the monad is said to be a
monad for F . We say that the type of a monad is determined if the sheaves
Ki are determined. ut

There are different ways of representing a given sheaf as the homology of a
monad, and the type of the monad depends on the way chosen.

When constructing or classifying sheaves in a given class via monads, one
typically proceeds along the following lines.

Step 1. Compute cohomological information which determines the type of
the corresponding monads.
Step 2. Construct or classify the differentials of the monads.

There are no general recipes for either step and some cases require sophisti-
cated ideas and quite a bit of intuition (see Example 7.2 below). If one wants
to classify, say, vector bundles, then a third step is needed:

Step 3. Determine which monads lead to isomorphic vector bundles.

One of the first successful applications of this approach was the classification
of (Gieseker-)stable rank 2 vector bundles with even first Chern class c1 ∈ Z
on the complex projective plane by Barth [4], who detected geometric proper-
ties of the corresponding moduli spaces without giving an explicit description
of the differentials in the second step. The same ideas apply in the case c1
odd which we are going to survey in what follows (see [31], [26], and [37] for
full details and proofs).

In general, rank 2 vector bundles enjoy properties which are not shared
by all vector bundles.

Remark 5.2. Every rank 2 vector bundle F on P(W ) is self-dual, that is,
it admits a symplectic structure. Indeed, the map

F ⊗ F ∧−→
∧2F ∼= OP(W )(c1)

induces an isomorphism ϕ : F
∼=→ F∗(c1) with ϕ = −ϕ∗(c1) (here c1 is the

first Chern class of F). In particular there are isomorphisms

(HjF(i))∗ ∼= Hn−jF(−i− n− 1− c1)

by Serre duality. ut

We will not give a general definition of stability here. For rank 2 vector
bundles stability can be characterized as follows (see [37]).

Remark 5.3. If F is a rank 2 vector bundle on P(W ), then the following
hold:
(1) F is stable if and only if Hom(F ,F) ∼= K. In this case the symplectic
structure on F is uniquely determined up to scalars.
(2) By tensoring with a line bundle we can normalize F so that its first
Chern class is 0 or −1. In this case F is stable if and only if it has no global
sections. ut

Example 5.4. By the results of the previous section the twisted cotangent
bundle U on the projective plane is a stable rank 2 vector bundle with Chern
classes c1 = −1 and c2 = 1. ut

Remark 5.5. The generalized theorem of Riemann-Roch yields a polyno-
mial in Q[c1, . . . , cr] which gives the Euler characteristic χF =

∑
j(−1)jhjF

for every rank r vector bundle F on P(W ) with Chern classes c1, . . . , cr.
This polynomial can be determined by interpreting the generalized theorem
of Riemann-Roch or by computing the Euler characteristic for enough spe-
cial bundles of rank r (like direct sums of line bundles). For a rank 2 vector
bundle on the projective plane, for example, one obtains

χ(F) = (c21 − 2c2 + 3c1 + 4)/2 . ut

We now focus on stable rank 2 vector bundles on the complex projective
plane P2(C) = P(W ) with first Chern class c1 = −1. Let F be such a bundle.

Remark 5.6. Since F is stable and normalized its second Chern class c2
must be ≥ 1. Indeed,

H2F(i− 2) = H0F(−i) = 0 for i ≥ 0

by Remarks 5.2 and 5.3, and χ(F(i)) = (i + 1)2 − c2 by Riemann-Roch.
In particular the dimensions hjF(i) in the range −2 ≤ i ≤ 0 are as in the
following cohomology table (a zero is represented by an empty box):

//
i

OO
j

c2 − 1 c2 c2 − 1

−2 −1 0

2

1

0

ut

We abbreviate O = OP2(C) and go through the three steps above.

Step 1. In this step we show that F is the homology of a monad of type

0→ H1F(−2)⊗ U2 → H1F(−1)⊗ U → H1F ⊗O → 0 ,

where the middle term occurs in cohomological degree 0. This actually fol-
lows from the general construction of Beilinson monads presented in the next
chapter and the fact that H2F(i− 2) = H0F(−i) = 0 for 2 ≥ i ≥ 0 (see Re-
mark 5.6). Here we derive the existence of the monad directly with Horrocks’
technique of killing cohomology [24], which requires further cohomological
information. Such information is typically obtained by restricting the given
bundles to linear subspaces. In our case we consider the Koszul complex on
the equations of a point p ∈ P2(C):

0 // O(−2)

(
−x′
x

)
// 2O(−1)

(x x′) // O // Op // 0 .

By tensoring with F(i+1) and taking cohomology we find that H1F generates
H1
≥0 F . Indeed, the composite map(

x x′
)

: 2H1F(i) −→ H1(Jp ⊗F(i+ 1)) −→ H1F(i+ 1)

is surjective if i ≥ −1. In particular, if c2 = 1, then H1F(i) = 0 for i 6=
−1 (apply Serre duality for the twists ≤ −2), so F ∼= U is the twisted
cotangent bundle by Remark 4.2 since both bundles have the same rank and
intermediate cohomology.

If c2 ≥ 2 then H1F 6= 0, and the identity in

Hom(H1F ,H1F) ∼= Ext1(H1F ⊗O,F)

defines an extension

0→ F → G → H1F ⊗O → 0 ,

where H1
≥0 G = 0, and where G is a vector bundle (apply Serre’s criterion

in Section 4). Similarly, by taking Serre duality into account, we obtain an
extension

0→ H1F(−2)⊗ U2 → H→ F → 0 ,

where H is a vector bundle with H1
≤−2H = 0. The two extensions fit into a

commutative diagram with exact rows and and columns

0

��

0

��
0 // H1F(−2)⊗ U2

||

// H

��

// F //

��

0

0 // H1F(−2)⊗ U2 α // B

β

��

// G

��

// 0

H1F ⊗O

��

= H1F ⊗O

��
0 0

since, for example, the extension in the top row lifts uniquely to an extension
as in the middle row (the obstructions in the corresponding Ext-sequence
vanish). Then B ∼= H1F(−1) ⊗ U since by construction these bundles have
the same rank and intermediate cohomology. What we have is the display of
(the short exact sequences associated to) a monad

0 −→ H1F(−2)⊗ U2 α−→ H1F(−1)⊗ U β−→ H1F ⊗O −→ 0

for F .

Step 2. Our task in this step is to describe what maps α and β could be the
differentials of a monad as above. In fact we give a description in terms of
linear algebra for which it is enough to deal with one of the differentials, say
α, since the self-duality of F and the vanishing of certain obstructions allows
one to represent F as the homology of a “self-dual” monad. Let us abbreviate
A = H1F(−2), B = H1F(−1) and A∗ ∼= H1F . By chasing the displays of a
monad as above and its dual we see that the symplectic structure on F lifts
to a unique isomorphism of monads

0 // A⊗ U2

Φ

��

α // B ⊗ U

Ψ

��

β // A∗ ⊗O //

−Φ∗(−1)

��

0

0 // A⊗O(−1)
β∗(−1)// B∗ ⊗ U∗(−1)

α∗(−1)// A∗ ⊗ (U2)∗(−1) // 0

with Ψ = −Ψ∗(−1). Indeed, the corresponding obstructions vanish (see [5]
and [37, II, 4.1] for a discussion of this argument in a general context). Ψ is
the tensor product of an isomorphism q : B → B∗ and a symplectic form
ι ∈ Hom(U,U∗(−1)) ∼= C on U . Note that q is symmetric since −(q ⊗ ι) =
(q ⊗ ι)∗(−1) = q∗ ⊗ ι∗(−1) = −q∗ ⊗ ι. We may and will now assume that F
is the homology of a self-dual monad, where self-dual means that β = αd :=
α∗(−1) ◦ (q ⊗ ι). The monad conditions

(α1) αd ◦ α = 0, and
(α2) α is a vector bundle monomorphism (αd is an epimorphism)

can be rewritten in terms of linear algebra as follows. The identifications in
Lemma 4.3 allow one to view

α ∈ Hom(A⊗ U2, B ⊗ U) ∼= V ⊗Hom(A,B)

as a homomorphism α : W → Hom(A,B) operating by ξ ⊗ (x ∧ x′) →
α(x)(ξ)⊗ x′ −α(x′)(ξ)⊗ x on the fibers of A⊗U2. Similarly we consider αd

as the homomorphism αd : W → Hom(B,A∗), x 7→ α∗(x) ◦ q, operating by
η ⊗ x→ αd(x)(η) on the fibers of B ⊗ U . Then

(α′1) αd(x) ◦ α(x′) = αd(x′) ◦ α(x) for all x, x′ ∈W , and
(α′2) for every ξ ∈ A \ {0} the map W → B, x→ α(x)(ξ) has rank ≥ 2.

Example 5.7. If c2 = 2, then the monads can be written (non-canonically)
as

0 // U2
(ab) // 2U

(a b) // O // 0 ,

where a, b are two vectors in V . In this case (α1) gives no extra condition and
(α2) means that a and b are linearly independent. If a and b are explicitly
given, then we can compute the homology of the monad with the help of
Macaulay 2:

i25 : S = ZZ/32003[x_0..x_2];

U is obtained from the Koszul complex resolving S/(x0, x1, x2) by tensoring
the cokernel of the differential

∧3
W ⊗ S(−3) →

∧2
W ⊗ S(−2) with S(1)

(and sheafifying).

i26 : U = coker koszul(3,vars S) ** S^{1};

For representing α and αd we also need the differential
∧2

W ⊗ S(−2) →
W ⊗ S(−1) of the Koszul complex.

i27 : k2 = koszul(2,vars S)

o27 = {1} | -x_1 -x_2 0 |
{1} | x_0 0 -x_2 |
{1} | 0 x_0 x_1 |

3 3
o27 : Matrix S <--- S

The expression koszul(2,vars S) computes a matrix representing the dif-
ferential with respect to the monomial bases x0∧x1, x0∧x2, x1∧x2 of

∧2
W

and x0, x1, x2 of W . We pick (a, b) = (e1, e2) and represent the correspond-
ing maps α and αd with respect to the monomial bases (see the discussion
following Lemma 4.3).

i28 : alpha = map(U ++ U, S^{-1}, transpose{{0,-1,0,1,0,0}});

o28 : Matrix

i29 : alphad = map(S^1, U ++ U, matrix{{0,1,0,0,0,1}} * (k2 ++ k2));

o29 : Matrix

Prune computes a minimal presentation.

i30 : F = prune homology(alphad, alpha);

i31 : betti F

o31 = relations : total: 3 1
1: 2 .
2: 1 1

In the next section we will present a more elegant way of computing the
homology of Beilinson monads. ut

We go back to the general case and reverse our construction. Let A and B
be C-vector spaces of the appropriate dimensions, let q be a non-degenerate
quadratic form on B, and let

M̃ = {α ∈ Hom(W,Hom(A,B)) | α satisfies (α′1) and (α′2)} .

Then every α ∈ M̃ defines a self-dual monad as above whose homology is
a stable rank 2 vector bundle on P2(C) with Chern classes c1 = −1 and
c2. In this way we obtain a description of the differentials of the monads
which is not as explicit as we might have hoped (with the exception of the
case c2 = 2). It is, however, enough for detecting geometric properties of the
corresponding moduli spaces.

Step 3. Constructing the moduli spaces means to parametrize the isomor-
phism classes of our bundles in a convenient way. We very roughly out-
line how to do that. Let O(B) be the orthogonal group of (B, q), and let
G := GL(A) × O(B). Then G acts on M̃ by ((Φ, Ψ), α) 7→ ΨαΦ−1, where
ΨαΦ−1(x) := Ψα(x)Φ−1. We may consider an element (Φ, Ψ) ∈ G as an iso-
morphism between the monad defined by α and the monad defined by ΨαΦ−1.
By going back and forth between isomorphisms of bundles and isomorphisms
of monads one shows that the stabilizer of G in each point is {±1}, and that
our construction induces a bijection between the set of isomorphism classes
of stable rank 2 vector bundles on P2(C) with Chern classes c1 = −1 and c2
andM := M̃/G0, where G0 := G/{±1}. With the help of a universal monad

over P2(C)×M̃ one proves that the analytic structure on M̃ descends to an
analytic structure onM so thatM is smooth of dimension h1F∗⊗F = 4c2−4
in each point (the obstructions for smoothness in the point corresponding to
F lie in H2F∗ ⊗ F which is zero). Moreover the homology of the universal
monad tensored by a suitable line bundle descends to a universal family over
M (here one needs c1 = −1). In other words, M is what one calls a fine
moduli space for our bundles. Further efforts show thatM is irreducible and
rational.

Remark 5.8. Horrocks’ technique of killing cohomology always yields 3-
term monads. In general, the bundle in the middle can be pretty complicated.

ut

6 The Beilinson Monad

We can use the Tate resolution associated to a sheaf to give a construction
of a complex first described by Beilinson [6], which gives a powerful method
for deriving information about a sheaf from information about a few of its
cohomology groups. The general idea is the following:

Suppose that A is an additive category and consider a graded object
⊕n+1
i=0 U

i in A. Given a graded ring homomorphism E → EndA(⊕n+1
i=0 U

i) we
can make an additive functor from the category of free E-modules to A: On
objects we take

ωE(i) 7→

{
U i for 0 ≤ i ≤ n+ 1 and;
0 otherwise.

To define the functor on maps, we use

HomE(ωE(i), ωE(j)) = HomE(E(i), E(j))

= Ej−i −→ End(⊕U i)j−i −→ Hom(U i, U j) .

(Note that we could have taken any twist of E in place of ωE ∼= E(−n− 1);
the choice of ωE is made to simplify the statement of Theorem 6.1, below.)

We shall be interested in the special case where A is the category of
coherent sheaves on P(W ) and where U i = ΩiP(W )(i) as in Section 4. Further

examples may be obtained by taking U i to be the i
th

exterior power of the
tautological subbundle Uk on the Grassmannian of k-planes in W for any k;
the case we have taken here is the case k = n. See [19] for more information
on the general case and applications to the computation of resultants and
more general Chow forms.

Applying the functor just defined to the Tate resolution T(F) of a cohe-
rent sheaf F on P(W ), and using Theorem 3.1, we get a complex

Ω(F) : · · · - ⊕j HjF(i− j)⊗ U j−i - . . . ,

where the term we have written down occurs in cohomological degree i. The
resolution T(F) is well-defined up to homotopy, so the same is true of Ω(F).
Since Uk = 0 unless 0 ≤ k ≤ n the only cohomology groups of F that are
actually involved in Ω(F) are HjF(k) with −n ≤ k ≤ 0; Ω(F) is of type

0 //H0F(−n)⊗ Un

||

// · · · // ⊕nj=0 HjF(−j)⊗ U j

||

// · · · //HnF ⊗ U0

||

//0

0 //Ω−n(F) // · · · //Ω0(F) // · · · //Ωn(F) //0 .

For applications it is important to note that instead of working with Ω(F)
one can also work with Ω(F(i)) for some twist i. This gives one some freedom
in choosing the cohomology groups of F to be involved.

To see a simple example, consider again the structure sheaf Op of the
subvariety consisting of a point p ∈ P(W ). Write I for the homogeneous ideal
of p, and let a ∈ V = W ∗ be a non-zero functional vanishing on the linear
forms in I as before. The Tate resolution of the homogeneous coordinate ring
S/I has already been computed, and we have seen that it depends only on
the sheaf S̃/I = Op. From the computation of T(S/I) = T(Op) made in
Section 3 we see that Ω(Op) takes the form

Ω(Op) : 0→ Un
a- Un−1 a- · · · a- U1 a- U0 - 0 ,

with U i in cohomological degree −i.
We have already noted that the map a : U = U1 - U0 = OP(W ) is

the composite of the tautological embedding U ⊂W ⊗OP(W ) with the map
a ⊗ 1 : W ⊗ OP(W ) → OP(W ). Thus the image of a : U1 → OP(W ) is the
ideal sheaf of p, and we see that the homology of the complex Ω(Op) at U0

is Op. One can check further that Ω(Op) is the Koszul complex associated
with the map a : U1 → OP(W ), and it follows that the homology of Ω(Op)
at U i is 0 for i > 0. The following result shows that this is typical.

Theorem 6.1 ([18]). If F is a coherent sheaf on P(W ), then the only non-
vanishing homology of the complex Ω(F) is

H0(Ω(F)) = F . ut

The existence of a complex satisfying the theorem and having the same
terms as Ω(F) was first asserted by Beilinson in [6], and thus we will call
Ω(F) a Beilinson monad for F . Existence proofs via a somewhat less effective
construction than the one given here may be found in [28] and [2].

The explicitness of the construction via Tate resolutions allows one to
detect properties of the differentials of Beilinson monads. Let us write

d
(r)
ij ∈ Hom(HjF(i− j)⊗ U j−i,Hj−r+1F(i− j + r)⊗ U j−i−r)

∼=
∧r
V ⊗Hom(HjF(i− j),Hj−r+1F(i− j + r))

∼= Hom(
∧r
W ⊗HjF(i− j),Hj−r+1F(i− j + r))

for the degree r maps actually occurring in Ω(F).

Remark 6.2. The constant maps d(0)
ij in Ω(F) are zero since T(F) is mini-

mal. ut

Proposition 6.3 ([18]). The linear maps d(1)
ij in Ω(F) correspond to the

multiplication maps

W ⊗HjF(i− j)→ HjF(i− j + 1) . ut

This follows from the identification of the linear strands in T(F) (see the
discussion following Theorem 3.1). The higher degree maps in T(F) and
Ω(F), however, are not yet well-understood.

Since (T(F))[1] = T(F(1)) we can compare the differentials in Ω(F) with
those in Ω(F(1)):

Proposition 6.4 ([18]). If the maps d(r)
ij in Ω(F) and d

(r)
i−1,j in Ω(F(1))

both actually occur, then they correspond to the same element in∧r
V ⊗Hom(HjF(i− j),Hj−r+1F(i− j + r)) . ut

In what follows we present some Macaulay 2 code for computing Beilinson
monads. Our functions sortedBasis, beilinson1, U, and beilinson reflect
what we did in Example 5.7 .

The expression sortedBasis(i,E) sorts the monomials of degree i in E to
match the order of the columns of koszul(i,vars S), where our conventions
with respect to S and E are as in Section 2, and where we suppose that the
monomial order on E is reverse lexicographic, the Macaulay 2 default order.

i32 : sortedBasis = (i,E) -> (
m := basis(i,E);
p := sortColumns(m,MonomialOrder=>Descending);
m_p);

For example:

i33 : S=ZZ/32003[x_0..x_3];

i34 : E=ZZ/32003[e_0..e_3,SkewCommutative=>true];

i35 : koszul(2,vars S)

o35 = {1} | -x_1 -x_2 0 -x_3 0 0 |
{1} | x_0 0 -x_2 0 -x_3 0 |
{1} | 0 x_0 x_1 0 0 -x_3 |
{1} | 0 0 0 x_0 x_1 x_2 |

4 6
o35 : Matrix S <--- S

i36 : sortedBasis(2,E)

o36 = | e_0e_1 e_0e_2 e_1e_2 e_0e_3 e_1e_3 e_2e_3 |

1 6
o36 : Matrix E <--- E

If e ∈ E is homogeneous of degree j, then beilinson1(e,j,i,S) computes
the map U i

e−→ U i−j on Pn = ProjS. If 0 < i − j ≤ i ≤ n, then the result
is a matrix representing the map

∧i+1
W ⊗ S(−1) e⊗1−→

∧i−j+1
W ⊗ S(−1)

defined by contraction with e. If 0 = i− j < i ≤ n, then the result is a matrix
representing the composite of the map

∧i
W ⊗ S

e⊗1−→ S with the Koszul
differential

∧i+1
W ⊗S(−1)→

∧i
W ⊗S. Note that the degrees of the result

are not set correctly since the functions U and beilinson below are supposed
to do that.

i37 : beilinson1=(e,dege,i,S)->(
E := ring e;
mi := if i < 0 or i >= numgens E then map(E^1, E^0, 0)

else if i === 0 then id_(E^1)
else sortedBasis(i+1,E);

r := i - dege;
mr := if r < 0 or r >= numgens E then map(E^1, E^0, 0)

else sortedBasis(r+1,E);
s = numgens source mr;
if i === 0 and r === 0 then

substitute(map(E^1,E^1,{{e}}),S)
else if i>0 and r === i then substitute(e*id_(E^s),S)
else if i > 0 and r === 0 then

(vars S) * substitute(contract(diff(e,mi),transpose mr),S)
else substitute(contract(diff(e,mi), transpose mr),S));

For example:

i38 : beilinson1(e_1,1,3,S)

o38 = {-3} | 0 |
{-3} | 0 |
{-3} | 1 |
{-3} | 0 |

4 1
o38 : Matrix S <--- S

i39 : beilinson1(e_1,1,2,S)

o39 = {-2} | 0 0 0 0 |
{-2} | -1 0 0 0 |
{-2} | 0 0 0 0 |
{-2} | 0 -1 0 0 |
{-2} | 0 0 0 0 |
{-2} | 0 0 0 1 |

6 4
o39 : Matrix S <--- S

i40 : beilinson1(e_1,1,1,S)

o40 = | x_0 0 -x_2 0 -x_3 0 |

1 6
o40 : Matrix S <--- S

The function U computes the bundles U i on ProjS:

i41 : U = (i,S) -> (
if i < 0 or i >= numgens S then S^0
else if i === 0 then S^1
else cokernel koszul(i+2,vars S) ** S^{i});

Finally, if o : ⊕E(−ai) → ⊕E(−bj) is a homogeneous matrix over E, then
beilinson(o,S) computes the corresponding map o : ⊕Uai → ⊕U bj on
ProjS by calling beilinson1 and U.

i42 : beilinson = (o,S) -> (
coldegs := degrees source o;
rowdegs := degrees target o;
mats = table(numgens target o, numgens source o,

(r,c) -> (
rdeg = first rowdegs#r;
cdeg = first coldegs#c;
overS = beilinson1(o_(r,c),cdeg-rdeg,cdeg,S);
-- overS = substitute(overE,S);
map(U(rdeg,S),U(cdeg,S),overS)));

if #mats === 0 then matrix(S,{{}})
else matrix(mats));

With these functions the code in Example 5.7 can be rewritten as follows:

i43 : S=ZZ/32003[x_0..x_2];

i44 : E = ZZ/32003[e_0..e_2,SkewCommutative=>true];

i45 : alphad = map(E^1,E^{-1,-1},{{e_1,e_2}})

o45 = | e_1 e_2 |

1 2
o45 : Matrix E <--- E

i46 : alpha = map(E^{-1,-1},E^{-2},{{e_1},{e_2}})

o46 = {1} | e_1 |
{1} | e_2 |

2 1
o46 : Matrix E <--- E

i47 : alphad=beilinson(alphad,S);

o47 : Matrix

i48 : alpha=beilinson(alpha,S);

o48 : Matrix

i49 : F = prune homology(alphad,alpha);

i50 : betti F

o50 = relations : total: 3 1
1: 2 .
2: 1 1

7 Examples

In this section we give two examples of explicit constructions of Beilinson
monads over P4(C) = P(W ) and of classification results based on these
monads. As in Section 5 we proceed in three steps. Let us write O = OP4(C).

Example 7.1. Our first example is taken from the classification of conic
bundles in P4(C), that is, of smooth surfaces X ⊂ P4(C) which are ruled in
conics in the sense that there exists a surjective morphism π : X → C onto a
smooth curve C such that the general fiber of π is a smooth conic in the given
embedding of X. There are precisely three families of such surfaces (see [20]
and [9]). Two families, the Del Pezzo surfaces of degree 4 and the Castelnuovo
surfaces, are classical. The third family, consisting of elliptic conic bundles
(conic bundles over an elliptic curve) of degree 8, had been falsely ruled out
in two classification papers in the 1980’s (see [36] and [27]). Only recently
Abo, Decker, and Sasakura [1] constructed and classified such surfaces by
considering the Beilinson monads for the suitably twisted ideal sheaves of
the surfaces. Let us explain how this works.
Step 1. In this step we suppose that an elliptic conic bundle X as above
exists, and we determine the type of the Beilinson monad for the suitably
twisted ideal sheaf JX . We know from the classification of smooth surfaces
in P4(C) which are contained in a cubic hypersurface (see [38] and [3]) that
H0JX(i) = 0 for i ≤ 3. It follows from general results such as the theorem
of Riemann-Roch that the dimensions hjJX(i) in range −2 ≤ i ≤ 3 are as
follows (here, again, a zero is represented by an empty box):

//
i

OO j

8 4

1 1 a b

a+ 1 b+ 1

−2 −1 0 1 2 3

4

3

2

1

0

with a := h2JX(2) and b := h2JX(3) still to be determined. The Beilinson
monad for JX(2) is thus of type

0→ 8O(−1)→ 4U3 ⊕ U2 → U ⊕ (a+ 1)O → aO → 0 ,

where (a+ 1)O → aO is the zero map (see Remark 6.2), and where conse-
quently U is mapped surjectively onto aO . By Proposition 4.4 this is only
possible if a = 0. The same idea applied to JX(3) shows that then also b = 0.

The cohomological information obtained so far determines the type of
the Beilinson monad for JX(2) and for JX(3). We decide to concentrate on
the monad for JX(3) since its differentials are smaller in size than those of
the monad for JX(2). In order to ease our calculations further we kill the
4-dimensional space H3JX(−1). Let us write ωX for the dualizing sheaf of
X. Serre duality on P4(C) respectively on X yields canonical isomorphisms

Z := Ext1(JX(−1),O(−5))
∼= (H3JX(−1))∗ ∼= (H2OX(−1))∗ ∼= H0(ωX(1)) .

The identity in

Hom(Z,Z) ∼= Ext1(JX(−1), Z∗ ⊗O(−5))

defines an extension which, twisted by 4, can be written as

0→ 4O(−1)→ G → JX(3)→ 0 .

Let us show that G is a vector bundle. We know from the classification of
scrolls in P4(C) (see [30] and [3]) that X is not a scroll. Hence adjunction
theory implies that ωX(1) is generated by the adjoint linear system H0(ωX(1))
(see [7, Corollary 9.2.2]). It follows by Serre’s criterion ([39], see also [35,
Theorem 2.2]) that G is locally free. By construction G has a cohomology
table as follows:

//
i

OO j

1 1

1 1

−4 −3 −2 −1 0

4

3

2

1

0

So the Beilinson monad of G is of type

0→ U3 α→ U2 ⊕ U β→ O → 0 .

Step 2. Now we proceed the other way around. We show that a rank 5 bundle
G as in the first step exists, and that the dependency locus of four general
sections of G(1) is a surface of the desired type. Differentials which define a
monad as above with a locally free homology can be easily found. By Lemma
4.3 α corresponds to a pair of vectors α = (α1, α2)t ∈ V ⊕

∧2
V . By dualizing

(see Remark 4.1) we find that it is a vector bundle monomorphism if and only

if U2 ⊕ U3 αt−→ U1 is an epimorphism. Equivalently, α1 is non-zero and α2

considered as a vector in
∧2(V/〈α1〉) is indecomposable (argue as in the proof

of Proposition 4.4). Taking the other monad conditions into account we see
that we may pick

α =
(

e4

e0 ∧ e2 + e1 ∧ e3

)
and

β =
(
e0 ∧ e2 + e1 ∧ e3 , −e4

)
,

where e0, . . . , e4 is a basis of V , and that up to isomorphisms of monads
and up to the choice of the basis this is the only possibility. We fix G as the
homology of this monad and compute the syzygies of G with Macaulay 2.

i51 : S = ZZ/32003[x_0..x_4];

i52 : E = ZZ/32003[e_0..e_4,SkewCommutative=>true];

i53 : beta=map(E^1,E^{-2,-1},{{e_0*e_2+e_1*e_3,-e_4}})

o53 = | e_0e_2+e_1e_3 -e_4 |

1 2
o53 : Matrix E <--- E

i54 : alpha=map(E^{-2,-1},E^{-3},{{e_4},{e_0*e_2+e_1*e_3}})

o54 = {2} | e_4 |
{1} | e_0e_2+e_1e_3 |

2 1
o54 : Matrix E <--- E

i55 : beta=beilinson(beta,S);

o55 : Matrix

i56 : alpha=beilinson(alpha,S);

o56 : Matrix

i57 : G = prune homology(beta,alpha);

i58 : betti res G

o58 = total: 10 9 5 1
1: 10 4 1 .
2: . 5 4 1

We see in particular that G(1) is globally generated. Hence the dependency
locus of four general sections of G(1) is indeed a smooth surface in P4(C) by
Kleiman’s Bertini-type result [29]. The smoothness can also be checked with
Macaulay 2 in an example via the built-in Jacobian criterion (see [15] for a
speedier method).

i59 : foursect = random(S^4, S^10) * presentation G;

4 9
o59 : Matrix S <--- S

The function trim computes a minimal presentation.

i60 : IX = trim minors(4,foursect);

o60 : Ideal of S

i61 : codim IX

o61 = 2

i62 : degree IX

o62 = 8

i63 : codim singularLocus IX

o63 = 5

By construction X has the correct invariants and is in fact an elliptic conic
bundle as claimed: Since the adjoint linear system H0(ωX(1)) is base point
free and 4-dimensional by what has been said in the first step, the correspond-
ing adjunction map X → P3 is a morphism which exhibits, as is easy to see,
X as a conic bundle over a smooth elliptic curve in P3 (see [1, Proposition
2.1]).
Step 3. Our discussion in the previous steps gives also a classification result.
Up to projectivities the elliptic conic bundles of degree 8 in P4(C) are pre-
cisely the smooth surfaces arising as the dependency locus of four sections of
the bundle G(1) fixed in Step 2. ut

Example 7.2. This example is concerned with the construction and classi-
fication of abelian surfaces in P4(C), and with the closely related Horrocks-
Mumford bundles [25].
Step 1. Horrocks and Mumford found evidence for the existence of a family
of abelian surfaces in P4(C). Suppose that such a surface X exists. Then
the dualizing sheaf of X is trivial, ωX ∼= OX , and X has degree 10 (see
[21, Example 3.2.15]). The same arguments as in Example 7.1 show that X
arises as the zero scheme of a section of a rank 2 vector bundle: There is an
extension

0→ O → F(3)→ JX(5)→ 0 ,

where F(3) is a rank 2 vector bundle with Chern classes c1 = 5 and c2 =
degX = 10, and where F has a cohomology table as displayed in Figure 1.
In particular F , which has Chern classes c1 = −1 and c2 = 4, is stable by
Remark 5.3. A discussion as in Section 5 shows that the Beilinson monad for
F is of type

0 // A⊗O(−1) α // B ⊗ U2 αd // A∗ ⊗O // 0 ,

//
i

OO j

5

2

5

−4 −3 −2 −1 0

4

3

2

1

0

Fig. 1.

with C-vector spaces A and B of dimension 5 and 2 respectively, and with
αd = α∗(−1) ◦ (q ⊗ ι), where q is a symplectic form on B, and where ι :
U2

∼=−→ (U2)∗(−1) is induced by the pairing U2 ⊗ U2 ∧−→ U4 ∼= O(−1). By
choosing appropriate bases of A and B we may suppose that α is a 2 × 5

matrix with entries in
∧2

V and that αd = αt ·
(

0 1
−1 0

)
.

Step 2. As in Example 7.1 we now proceed the other way around. But this
time it is not obvious how to define α. Horrocks and Mumford remark that
up to projectivities one may suppose that the abelian surfaces in P4(C) are
invariant under the action of the Heisenberg group H5 in its Schrödinger rep-
resentation, and they use the representation theory of H5 and its normalizer
N5 in SL(5,C) to find

α =
(
e2 ∧ e3 e3 ∧ e4 e4 ∧ e0 e0 ∧ e1 e1 ∧ e2

e1 ∧ e4 e2 ∧ e0 e3 ∧ e1 e4 ∧ e2 e0 ∧ e3

)
,

where e0, . . . , e4 is a basis of V . A straightforward computation shows that
with this α the desired monad conditions are indeed satisfied. The resulting
Horrocks-Mumford bundle FHM on P4(C) is essentially the only rank 2 vector
bundle known on Pn(C), n ≥ 4, which does not split as direct sum of two
line bundles. Let us compute the syzygies of FHM with Macaulay 2.

i64 : alphad = matrix{{e_4*e_1, e_2*e_3},{e_0*e_2, e_3*e_4},
{e_1*e_3, e_4*e_0},{e_2*e_4, e_0*e_1},
{e_3*e_0, e_1*e_2}};

5 2
o64 : Matrix E <--- E

i65 : alphad=map(E^5,E^{-2,-2},alphad)

o65 = | -e_1e_4 e_2e_3 |
| e_0e_2 e_3e_4 |
| e_1e_3 -e_0e_4 |
| e_2e_4 e_0e_1 |
| -e_0e_3 e_1e_2 |



246 W. Decker and D. Eisenbud

5 2
o65 : Matrix E <--- E

i66 : alpha=syz alphad

o66 = {2} | e_2e_3 e_0e_4 e_1e_2 -e_3e_4 e_0e_1 |
{2} | e_1e_4 e_1e_3 e_0e_3 e_0e_2 -e_2e_4 |

2 5
o66 : Matrix E <--- E

i67 : alphad=beilinson(alphad,S);

o67 : Matrix

i68 : alpha=beilinson(alpha,S);

o68 : Matrix

i69 : FHM = prune homology(alphad,alpha);

i70 : betti res FHM

o70 = total: 19 35 20 2
3: 4 . . .
4: 15 35 20 .
5: . . . 2

i71 : regularity FHM

o71 = 5

i72 : betti sheafCohomology(presentation FHM,E,-6,6)

o72 = total: 210 100 37 14 10 5 2 5 10 14 37 100 210
-6: 210 100 35 4 . . . . . . . . .
-5: . . 2 10 10 5 . . . . . . .
-4: . . . . . . 2 . . . . . .
-3: . . . . . . . 5 10 10 2 . .
-2: . . . . . . . . . 4 35 100 210

Since H0FHM(i) = 0 for i < 3 every non-zero section of FHM(3) vanishes
along a surface (with the desired invariants). Horrocks and Mumford need
an extra argument to show that the general such surface is smooth (and thus
abelian) since Kleiman’s Bertini-type result does not apply (FHM(3) is not
globally generated). Our explicit construction allows one again to check the
smoothness with Macaulay 2 in an example.

i73 : sect = map(S^1,S^15,0) | random(S^1, S^4);

1 19
o73 : Matrix S <--- S

We compute the equations of X via a mapping cone.

i74 : mapcone = sect || transpose presentation FHM;

36 19
o74 : Matrix S <--- S

i75 : fmapcone = res coker mapcone;

i76 : IX = trim ideal fmapcone.dd_2;

o76 : Ideal of S

i77 : codim IX

o77 = 2

i78 : degree IX

o78 = 10

i79 : codim singularLocus IX

o79 = 5

Step 3. Horrocks and Mumford showed that up to projectivities every abelian
surface in P4(C) arises as the zero scheme of a section of FHM(3). In fact, one
can show much more. By a careful analysis of possible Beilinson monads and
their restrictions to various linear subspaces Decker [12] proved that every
stable rank 2 vector bundle F on P4(C) with Chern classes c1 = −1 and
c2 = 4 is the homology of a monad of the type as in Step 1. From geometric
properties of the “variety of unstable planes” of F Decker and Schreyer [14]
deduced that up to isomorphisms and projectivities the differentials of the
monad coincide with those of FHM. Together with results from [11] this im-
plies that the moduli space of our bundles is isomorphic to the homogeneous
space SL(5,C)/N5. ut

6. A. A. Bĕılinson: Coherent sheaves on Pn and problems in linear algebra. Func-
tional Anal. Appl., 12:214–216, 1978.

7. M.C. Beltrametti and A. Sommese: The adjunction theory of complex projective
varieties. de Gruyter, Berlin, 1995.

Needles in a Haystack:
Special Varieties via Small Fields

Frank-Olaf Schreyer and Fabio Tonoli

In this article we illustrate how picking points over a finite field at random can
help to investigate algebraic geometry questions. In the first part we develop
a program that produces random curves of genus g ≤ 14. In the second part
we use the program to test Green’s Conjecture on syzygies of canonical curves
and compare it with the corresponding statement for Coble self-dual sets of
points. In the third section we apply our techniques to produce Calabi-Yau
3-folds of degree 17 in P6.

Introduction

The advances in speed of modern computers and computer algebra systems
gave life to the idea of solving equations by guessing a solution. Suppose
M ⊂ G is a subvariety of a rational variety of codimension c. Then we expect
that the probability for a point p ∈ G(Fq) to lie in M(Fq) is about 1/qc. Here
Fq denotes the field with q elements.

We will discuss this idea in the following setting: M will be a parameter
space for objects in algebraic geometry, e.g., a Hilbert scheme, a moduli space,
or a space dominating such spaces.

The most basic question we might have in this case is whether M is non-
empty and whether an open part of M corresponds to smooth objects.

Typically in these cases we will not have explicit equations for M ⊂ G
but only an implicit algebraic description of M, and our approach will be
successful if the time required to check p /∈ M(Fq) is sufficiently small com-
pared to qc. The first author applied this method first in [32] to construct
some rational surfaces in P4; see [15,11] for motivation.

In this first section we describe a program that picks curve of genus g ≤ 14
at random. The moduli spaces Mg are known to be unirational for g ≤ 13;
see [33,8].

Our approach based on this result can viewed as a computer aided proof
of the unirationality. Many people might object that this not a proof because
we cannot control every single step in the computation. We however think
that such a proof is much more reliable than a proof based on man-made
computations. A mistake in a computer aided approach most often leads to an
output far away from our expectation, hence it is easy to spot. A substantial
improvement of present computers and computer algebra systems would give
us an explicit unirational parametrization of Mg for g ≤ 13.

In the second part we apply our “random curves” to probe the conse-
quences of Green’s conjecture on syzygies of canonical curves, and compare

these results with the corresponding statements for “Coble self-dual” sets of
2g − 2 points in Pg−2.

In the last section we exploit our method to prove the existence of three
components of the Hilbert scheme of Calabi-Yau 3-folds of degree 17 in P6

over the complex numbers. This is one of the main results of the second
author’s thesis [34, Chapter 4]. Calabi-Yau threefolds of lower degree in P6

are easy to construct, using the Pfaffian construction and a study of their
Hartshorne-Rao modules. For degree 17 the Hartshorne-Rao module has to
satisfy a subtle condition. Explicit examples of such Calabi-Yau 3-folds are
first constructed over a finite field by our probabilistic method. Then a del-
icate semi-continuity argument gives us the existence of such Calabi-Yau
3-folds over some number field.

Acknowledgments. We thank Hans-Christian v. Bothmer and Dan Grayson for
valuable discussions and remarks.

Notation. For a finitely generated graded module M over the polynomial
ring S = k[x0, . . . , xr] we summarize the numerical information of a finite
free resolution

0←M ← F0 ← F1 ← . . .← Fn ← 0

with Fi = ⊕jS(−j)βij in a table of Betti numbers, whose ijth entry is

βi,i+j = dim TorSi (M,k)i+j .

As in the Macaulay 2 command betti we suppress zeroes. For example the
syzygies of the rational normal curve in P3 have the following Betti table.

1 - -
- 3 2

Note that the degrees of the entries of the matrices in the free resolution can
be read off from the relative position of two numbers in consecutive columns.
A pair of numbers in a line corresponds to linear entries. Quadratic entries
correspond to two numbers of a square. Thus

1 - - -
- 5 5 -
- - - 1

corresponds to a 4 term complex with a quadratic, a linear and another
quadratic map. The Grassmannian G(2, 5) in its Plücker embedding has such
a free resolution.

1 How to Make Random Curves up to Genus 14

The moduli space of curves Mg is known to be of general type for g ≥ 24 and
has non-negative Kodaira dimension for g = 23 by work of Harris, Mumford
and Eisenbud [21,13]. For genus g ≤ 13 unirationality is known [8,33]. In this
section we present a Macaulay 2 program that over a finite field Fq picks a
point in Mg(Fq) for g ≤ 14 at random.

By Brill-Noether theory [2] every curve of genus g has a linear system
grd of dimension r and degree d, provided that the Brill-Noether number ρ
satisfies

ρ := ρ(g, d, r) := g − (r + 1)(g − d+ r) ≥ 0.

We utilize this to find appropriate (birational) models for general curves of
genus g.

1.1 Plane Models, g ≤ 10

This case was known to Severi; see [1]. Choose d = g + 2 − bg/3c. Then
ρ(g, d, 2) ≥ 0 i.e., a general curve of genus g has a plane model C ′ of degree
d. We expect that C ′ has

δ =
(
d− 1

2

)
− g

double points. If the double points are in general position, then

s = h0(P2,O(d))− 3δ − 1

is the expected dimension of the linear system of curves of degree d with δ
assigned double points. We have the following table:

g 1 2 3 4 5 6 7 8 9 10 11 12
ρ 1 2 0 1 2 0 1 2 0 1 2 0
d 3 4 4 5 6 6 7 8 8 9 10 10
δ 0 1 0 2 5 4 8 13 12 18 25 24
s 9 11 14 14 12 15 11 5 8 0 -10 -7

Thus for g ≤ 10 we assume that these double points lie in general position.
For g > 10 the double points cannot lie in general position because s < 0.
Since it is difficult to describe the special locus Hδ(g) ⊂ Hilbδ(P2) of double
points of nodal genus g curves, the plane model approach collapses for g > 10.

Random Points. In our program, which picks plane models at random
from an Zariski open subspace of Mg, we start by picking the nodes. How-
ever, over a small field Fq it is not a good idea to pick points individually,
because there might be simply too few: |P2(Fq)| = 1+q+q2. What we should

do is to pick a collection Γ of δ points in P2(F̄q) that is defined over Fq. Gen-
eral points in P2 satisfy the minimal resolution condition, that is, they have
expected Betti numbers. This follows from the Hilbert-Burch theorem [12,
Theorem 20.15]. If the ideal of such Γ has generators in minimal degree k,
then

(
k+1

2

)
≤ δ <

(
k+2

2

)
, which gives δ =

(
k+1

2

)
+ ε with 0 ≤ ε ≤ k. Thus

k = d(−3 +
√

9 + 8δ)/2e. The Betti table is one of the following two tables:

2ε ≤ k :

0 1 - -
1 - - -
...

...
...

...
k − 2 - - -
k − 1 - k + 1− ε k − 2ε
k - - ε

2ε ≥ k :

0 1 - -
1 - - -
...

...
...

...
k − 2 - - -
k − 1 - k + 1− ε -
k - 2ε− k ε

So we can specify a collection Γ of δ points by picking the Hilbert-Burch
matrix of their resolution; see [12, Thm 20.15]. This is a matrix with linear
and quadratic entries only, whose minors of size ε (k − ε if 2ε ≤ k) generate
the homogeneous ideal of Γ .

i1 : randomPlanePoints = (delta,R) -> (
k:=ceiling((-3+sqrt(9.0+8*delta))/2);
eps:=delta-binomial(k+1,2);
if k-2*eps>=0
then minors(k-eps,

random(R^(k+1-eps),R^{k-2*eps:-1,eps:-2}))
else minors(eps,

random(R^{k+1-eps:0,2*eps-k:-1},R^{eps:-2})));

In unlucky cases these points might be infinitesimally near.
i2 : distinctPoints = (J) -> (

singJ:=minors(2,jacobian J)+J;
codim singJ == 3);

The procedure that returns the ideal of a random nodal curve is then straight-
forward:

i3 : randomNodalCurve = method();

i4 : randomNodalCurve (ZZ,ZZ,Ring) := (d,g,R) -> (
delta:=binomial(d-1,2)-g;
K:=coefficientRing R;
if (delta==0)
then ( --no double points

ideal random(R^1,R^{-d}))
else ( --delta double points

Ip:=randomPlanePoints(delta,R);
--choose the curve
Ip2:=saturate Ip^2;
ideal (gens Ip2 * random(source gens Ip2, R^{-d}))));

i5 : isNodalCurve = (I) -> (
singI:=ideal jacobian I +I;delta:=degree singI;
d:=degree I;g:=binomial(d-1,2)-delta;
{distinctPoints(singI),delta,g});

We next ask if we indeed get in this way points in a parameter space
that dominates Mg for g ≤ 10. Let Hilb(d,g)(P2) denote the Hilbert scheme
of nodal plane curves of degree d and geometric genus g. Our construction
starts from a random element in Hilbδ(P2) and picks a random curve in the
corresponding fiber of Hilb(d,g)(P2)→ Hilbδ(P2):

Hilb(d,g)(P2) //

��

Mg

Hilbδ(P2)

.

So the question is whether Hilb(d,g)(P2) dominates Hilbδ(P2). A naive dimen-
sion count suggests that this should be true: the dimension of our parameter
space is given by 2δ + s, which is 3(g − 1) + ρ + dim PGL(3), as it should
be. To conclude this there is more to verify: it could be that the nodal mod-
els of general curves have double points in special position, while all curve
constructed above lie over a subvariety of Mg. One way to exclude this is to
prove that the variety G(g, d, 2) over Mg, whose fiber over a curve C̃ ∈ Mg

is G2
d(C̃) = {g2

d’s}, is irreducible or, to put it differently, that the Severi
Conjecture holds:

Theorem 1.1 (Harris [20]). The space of nodal degree d genus g curves in
P

2 is irreducible.

Another much easier proof for the few (d, g) we are interested in is to
establish that our parameter space M of the construction is smooth of ex-
pected dimension at our random point p ∈M, as in [1]. Consider the following
diagram:

H = Hilb(d,g)(P2)/Aut(P2) π−→Mg.

For a given curve C̃ ∈Mg, the inverse image π−1(C̃) consists of the variety
W 2
d (C̃) ⊂ Picd(C̃). Moreover the choice of a divisor L ∈W 2

d (C̃) is equivalent
to the choice of p ∈ M, modulo Aut(P2): indeed p determines a morphism
ν : C̃ −→ C ⊂ P2 and a line bundle L = ν−1OP2(1), where C̃ is the normal-
ization of the (nodal) curve C. Therefore M is smooth of expected dimension
3(g−1)+ρ+dim PGL(3) at p ∈M if and only if W 2

d (C̃) is smooth of expected
dimension ρ in L. This is well known to be equivalent to the injectivity of
the multiplication map µL

H0(L)⊗H0(KC̃ ⊗ L
−1)

µL−→ H0(KC̃),

which can be easily checked in our cases, see [2, p. 189]. In our cases µL can
be rewritten as

H0(OP2(1))⊗H0(IΓ (d− 4))
µL−→ H0(IΓ (d− 3)).

So we need two conditions:

(1) H0(IΓ (d− 5)) = 0;
(2) there are no linear relations among the generators of H0(IΓ (d − 4)) of

degree d− 3.

We proceed case by case. For genus g ≤ 5 this is clear, since H0(IΓ (d−4)) = 0
for g = 2, 3 and dimH0(IΓ (d − 3)) = 1 for g = 4, 5. For g = 6 we have
dimH0(IΓ (d− 3)) = dimH0(IΓ (2)) = 2 and the Betti numbers of Γ

1 - -
- 2 -
- - 1

shows there are no relations with linear coefficients in H0(IΓ (2)). For 7 ≤ g ≤
10 the method is similar: everything is clear once the Betti table of resolution
of the set of nodal points Γ is computed. As a further example we do here
the case g = 10: we see that dimH0(IΓ (d−3)) = dimH0(IΓ (5)) = 3 and the
Betti numbers of Γ are

1 - -
- - -
- - -
- - -
- 3 -
- 1 3

from which it is clear that there are no linear relations between the quintic
generators of IΓ .

1.2 Space Models and Hartshorne-Rao Modules

The Case of Genus g = 11. In this case we have ρ(11, 12, 3) = 3. Hence
every general curve of genus 11 has a space model of degree 12. Moreover
for a general curve the general space model of this degree is linearly normal,
because ρ(11, 13, 4) = −1 takes a smaller value. If moreover such a curve
C ⊂ P3 has maximal rank, i.e., for each m ∈ Z the map

H0(P3,O(m))→ H0(C,OC(m))

has maximal rank, then the Hartshorne-Rao module M , defined as M =
H1
∗ (IC) = ⊕mH1(P3, IC(m)), has Hilbert function with values (0, 0, 4, 6, 3, 0,

0, . . . ). For readers who want to know more about the Hartshorne-Rao mod-
ule, we refer to the pleasant treatment in [24].

Since being of maximal rank is an open condition, we will try a construc-
tion of maximal rank curves. Consider the vector bundle G on P3 associated
to the first syzygy module of IC :

(1) 0← IC ← ⊕iO(−ai)← G ← 0

In this set-up H2
∗ (G) = H1

∗ (IC). Thus G is, up to direct sum of line bundles,
the sheafified second syzygy module of M ; see e.g., [10, Prop. 1.5].

The expected Betti numbers of M are

4 10 3 - -
- - 8 2 -
- - - 6 3

Thus the F-dual M∗ = HomF(M,F) is presented as F[x0, . . . , x3]-module by a
3×8 matrix with linear and quadratic entries, and a general such matrix will
give a general module (if the construction works, i.e., if the desired space of
modules is non-empty), because all conditions we impose are semi-continuous
and open. Thus M depends on

dimG(6, 3h0O(1)) + dimG(2, 3h0O(2)− 6h0O(1))− dimSL(3) = 36

parameters.
Assuming that C has minimal possible syzygies:

1 - - -
- - - -
- - - -
- - - -
- 6 2 -
- - 6 3

we obtain, by dualizing the sequence (1), the following exact sequence

G∗←6O(5)← O ← 0.

If everything is as expected, i.e., the general curve is of maximal rank and its
syzygies have minimal possible Betti numbers, then the entries of the right
hand matrix are homogeneous polynomials that generate IC . We will compute
IC by determining ker(φ : 6O(5) → G∗). Comparing with the syzygies of M
we obtain the following isomorphism

G∗ ∼= ker(2O(6)⊕6O(7)→ 3O(8)) ∼= image(3O(4)⊕8O(5)→ 2O(6)⊕6O(7)).

and G∗ ← 6O(5) factors over G∗ ← 8O(5)⊕3O(4). A general φ ∈ Hom(6O(5),
G∗) gives a point in G(6, 8) and the Hilbert scheme of desired curves would
have dimension 36 + 12 = 48 = 4 · 12 = 30 + 3 + 15 as expected, c.f. [19].

Therefore the computation for obtaining a random space curve of genus
11 is done as follows:

i6 : randomGenus11Curve = (R) -> (
correctCodimAndDegree:=false;
while not correctCodimAndDegree do (

Mt=coker random(R^{3:8},R^{6:7,2:6});
M=coker (transpose (res Mt).dd_4);
Gt:=transpose (res M).dd_3;
I:=ideal syz (Gt*random(source Gt,R^{6:5}));
correctCodimAndDegree=(codim I==2 and degree I==12););

I);

In general for these problems there is rarely an a priori reason why such
a construction for general choices will give a smooth curve. Kleiman’s global
generation condition [23] is much too strong a hypothesis for many interest-
ing examples. But it is easy to check an example over a finite field with a
computer:

i7 : isSmoothSpaceCurve = (I) -> (
--I generates the ideal sheaf of a pure codim 2 scheme in P3
singI:=I+minors(2,jacobian I);
codim singI==4);

Hence by semi-continuity this is true over Q and the desired unirationality
of G(11, 12, 3)/M11 holds for all fields, except possibly for those whose ground
field has characteristic is in some finite set.

A calculation of an example over the integers would bound the number
of exceptional characteristics, which then can be ruled out case by case, or
by considering sufficiently many integer examples.

As in case of nodal curves, to prove unirationality of M11 by computer
aided computations we have to show the injectivity of

H0(L)⊗H0(KC ⊗ L−1)
µL−→ H0(KC),

where L is the restriction of OP3(1) to the curve C ⊂ P3. The following few
lines do the job:

i8 : K=ZZ/101;

i9 : R=K[x_0..x_3];

i10 : C=randomGenus11Curve(R);

o10 : Ideal of R

i11 : isSmoothSpaceCurve(C)

o11 = true

i12 : Omega=prune Ext^2(coker gens C,R^{-4});

i13 : betti Omega

o13 = relations : total: 5 10
-1: 2 .
0: 3 10

We see that there are no linear relations among the two generators of H0
∗ (ΩC)

in degree -1.

Betti Numbers for Genus g = 12, 13, 14, 15. The approach in these
cases is similar to g = 11. We choose here d = g, so ρ(g, g, 3) = g − 12 ≥ 0.
Under the maximal rank assumption the corresponding space curve has a
Hartshorne-Rao module whose Hilbert function takes values (0, 0, g− 9, 2g−
19, 3g − 34, 0, . . .) in case g = 12, 13 and (0, 0, g − 9, 2g − 19, 3g − 34, 4g −
55, 0, . . .) in case g = 14, 15. Expected syzygies of M have Betti tables:

g = 12 :
3 7 - - -
- - 10 5 -
- - - 3 2

g = 13 :
4 9 1 - -
- - 6 - -
- - 6 13 5

g = 14 :

5 11 2 - -
- - 3 - -
- - 13 17 4
- - - - 1

g = 15 :

6 13 3 - -
- - 3 - -
- - 8 3 -
- - - 9 5

Comparing with the expected syzygies of C

g = 12 :

1 - - -
- - - -
- - - -
- - - -
- 7 5 -
- - 3 2

g = 13 :

1 - - -
- - - -
- - - -
- - - -
- 3 - -
- 6 13 5

g = 14 :

1 - - -
- - - -
- - - -
- - - -
- - - -
- 13 17 4
- - - 1

g = 15 :

1 - - -
- - - -
- - - -
- - - -
- - - -
- 8 3 -
- - 9 5

we see that given M the choice of a curve corresponds to a point in G(7, 10)
or G(3, 6) for g = 12, 13 respectively, while for g = 14, 15 everything is
determined by the Hartshorne-Rao module. For g = 12 Kleiman’s result
guarantees smoothness for general choices, in contrast to the more difficult
cases g = 14, 15. So the construction of M is the crucial step.

Construction of Hartshorne-Rao Modules. In case g = 12 the con-
struction of M is straightforward. It is presented by a sufficiently general
matrix of linear forms:

0←M ← 3S(−2)← 7S(−3).

The procedure for obtaining a random genus 12 curve is:
i14 : randomGenus12Curve = (R) -> (

correctCodimAndDegree:=false;
while not correctCodimAndDegree do (

M:=coker random(R^{3:-2},R^{7:-3});
Gt:=transpose (res M).dd_3;
I:=ideal syz (Gt*random(source Gt,R^{7:5}));
correctCodimAndDegree=(codim I==2 and degree I==12););

I);

In case g = 13 we have to make sure that M has a second linear syzygy.
Consider the end of the Koszul complex:

6R(−2) κ← 4R(−3)← R(−1)← 0.

Any product of a general map 4R(−2) α← 6R(−2) with the Koszul matrix
κ yields 4R(−2) ← 4R(−3) with a linear syzygy, and concatenated with a

general map 4R(−2)
β← 5R(−3) gives a presentation matrix of a module M

of desired type:

0←M ← 4R(−2)← 4R(−3)⊕ 5R(−3).

The procedure for obtaining a random genus 13 curve is:
i15 : randomGenus13Curve = (R) -> (

kappa:=koszul(3,vars R);
correctCodimAndDegree:=false;
while not correctCodimAndDegree do (

test:=false;while test==false do (
alpha:=random(R^{4:-2},R^{6:-2});
beta:=random(R^{4:-2},R^{5:-3});
M:=coker(alpha*kappa|beta);
test=(codim M==4 and degree M==16););

Gt:=transpose (res M).dd_3;
--up to change of basis we can reduce phi to this form
phi:=random(R^6,R^3)++id_(R^6);
I:=ideal syz(Gt_{1..12}*phi);
correctCodimAndDegree=(codim I==2 and degree I==13););

I);

The case of genus g = 14 is about a magnitude more difficult. To start
with, we can achieve two second linear syzygies by the same method as
in the case g = 13. A general matrix 5R(−2) α← 12R(−2) composed with
12R(−2) κ⊕κ← 8R(−3) yields the first component of

5R(−2)← (8 + 3)R(−3).

For a general choice of the second component 5R(−2)
β← 3R(−3) the cokernel

will be a module with Hilbert function (0, 0, 5, 9, 8, 0, 0, . . .) and syzygies

5 11 2 - -
- - 2 - -
- - 17 23 8
- - - - -

What we want is to find α and β so that dimM5 = 1 and dim TorR2 (M,F)5 =
3. Taking into account that we ensured dim TorR2 (M,F)4 = 2 this amounts
to asking that the 100× 102 matrix m(α, β) obtained from

[0← 5R(−2)5 ← 11R(−3)5 ← 2R(−4)5 ← 0] ∼= [0← 100F
m(α,β)← 102F← 0]

drops rank by 1. We do not know a systematic approach to produce such
m(α, β)’s. However, we can find such matrices in a probabilistic way. In the
space of the matrices m(α, β), those which drop rank by 1 have expected
codimension 3. Hence over a finite field F = Fq we expect to find the desired
modules M with a probability of 1/q3. The code to detect bad modules is
rather fast.

i16 : testModulesForGenus14Curves = (N,p) ->(
x := local x;
R := ZZ/p[x_0..x_3];
i:=0;j:=0;
kappa:=koszul(3,vars R);
kappakappa:=kappa++kappa;
utime:=timing while (i<N) do (

test:=false;
alpha:=random(R^{5:-2},R^{12:-2});
beta:=random(R^{5:-2},R^{3:-3});
M:=coker (alpha*kappakappa|beta);
fM:=res (M,DegreeLimit =>3);
if (tally degrees fM_2)_{5}==3 then (

--further checks to pick up the right module
test=(tally degrees fM_2)_{4}==2 and
codim M==4 and degree M==23;);

i=i+1;if test==true then (j=j+1;););
timeForNModules:=utime#0; numberOfGoodModules:=j;
{timeForNModules,numberOfGoodModules});

i17 : testModulesForGenus14Curves(1000,5)

o17 = {41.02, 10}

o17 : List

For timing tests we used a Pentium2 400Mhz with 128Mb of memory running
GNU Linux. On such a machine examples can be tested at a rate of 0.04
seconds per example. Hence an approximate estimation of the CPU-time
required to find a good example is q3 · 0.04 seconds. Comparing this with
the time to verify smoothness, which is about 12 seconds for an example
of this degree, we see that up to |Fq| = q ≤ 13 we can expect to obtain
examples within few minutes. Actually the computations for q = 2 and q = 3
take longer than for q = 5 on average, because examples of “good modules”
tend to give singular curves more often. Here is a table of statistics which
summarizes the situation.

q 2 3 5 7 11 13
smooth curves 100 100 100 100 100 100
1-nodal curves 75 53 31 16 10 8
reduced more singular 1012 142 24 11 2 0
non reduced curves 295 7 0 0 0 0
total number of curves 1482 302 155 127 112 108
percentage of smooth curves 6.7% 33% 65% 79% 89% 93%
approx. time (in seconds) 7400 3100 2700 3400 6500 9500

The procedure for obtaining a random genus 14 curve is
i18 : randomGenus14Curve = (R) -> (

kappa:=koszul(3,vars R);
kappakappa:=kappa++kappa;
correctCodimAndDegree:=false;
count:=0;while not correctCodimAndDegree do (

test:=false;
t:=timing while test==false do (

alpha=random(R^{5:-2},R^{12:-2});
beta=random(R^{5:-2},R^{3:-3});
M:=coker (alpha*kappakappa|beta);
fM:=res (M,DegreeLimit =>3);
if (tally degrees fM_2)_{5}==3 then (

--further checks to pick up the right module
test=(tally degrees fM_2)_{4}==2 and
codim M==4 and degree M==23;);

count=count+1;);
Gt:=transpose (res M).dd_3;
I:=ideal syz (Gt_{5..17});
correctCodimAndDegree=(codim I==2 and degree I==14););

<<" -- "<<t#0<<" seconds used for ";
<<count<<" modules"<<endl;
I);

For g = 15 we do not know a method along these lines that would give
examples over small fields.

Counting Parameters. For genus g = 12 clearly the module M depends
on dimG(7, 3 · h0O(1)) − dimSL(3) = 7 · 5 − 8 = 36 parameters, and the
family of curves has dimension 36 + dimG(7, 10) = 48 = 4 · 12 = 33 + 0 + 15,
as expected.

For genus g = 13 and 14 the parameter count is more difficult. Let us make
a careful parameter count for genus g = 14; the case g = 13 is similar and
easier. The choice of α corresponds to a point in G(5, 12). Then β corresponds
to a point G(3, Bα) where Bα = U ⊗R1/<α> where U denotes the universal
subbundle on G(5, 12) and <α> the subspace generated by the 8 columns
of α ◦ (κ ⊕ κ). So dimBα = 20 − 8 = 12 and G(3, Bα) → G(5, 12) is a
Grassmannian bundle with fiber dimension 27 and total dimension 62. In
this space the scheme of good modules has codimension 3, so we get a 59
dimensional family. This is larger than the expected dimension 56 = 4 · 14 =
39 + 2 + 15 of the Hilbert scheme, c.f. [19]. Indeed the construction gives a
curve together with a basis of TorR2 (M,F)4. Subtracting the dimension of the
group of the projective coordinate changes we arrive at the desired dimension
59− 3 = 56.

The unirationality of M12 and M13 can be proved by computer as in case
M11, while in case g = 14 we don’t know the unirationality of the parameter
space of the modules M with dimM5 = 1 and dim TorR2 (M,F)5 = 3.

2 Comparing Green’s Conjecture for Curves
and Points

2.1 Syzygies of Canonical Curves

One of the most outstanding conjectures about free resolutions is Green’s
prediction for the syzygies of canonical curves.

A canonical curve C ⊂ Pg−1, i.e., a linearly normal curve with OC(1) ≡
ωC , the canonical line bundle, is projectively normal by a result of Max
Noether, and hence has a Gorenstein homogeneous coordinate ring and is
3-regular.

Therefore the Betti numbers of the free resolution of a canonical curve
are symmetric, that is, βj,j+1 = βg−2−j,g−j , and essentially only two rows of
Betti numbers occur. The situation is summarized in the following table.

•

1 - - - - - - - - - - -
- β1,2 β2,3 · · · βp,p+1 · · · · · · βp,p+2 - - - -
- - - - βp,p+2 · · · · · · βp,p+1 · · · β2,3 β1,2 -
- - - - - - - - - - - 1

The first p such that βp,p+2 6= 0 is conjecturally precisely the Clifford
index of the curve.

Conjecture 2.1 (Green [16]). Let C be a smooth canonical curve over C.
Then βp,p+2 6= 0 if and only if ∃L ∈ Picd(C) with h0(C,L), h1(C,L) ≥ 2 and
cliff(L) := d − 2(h0(C,L)) − 1) ≤ p. In particular, βj,j+2 = 0 for j ≤ b g−3

2 c
for a general curve of genus g.

The “if” part is proved by Green and Lazarsfeld in [18] and holds for
arbitrary ground fields. For some partial results see [35,31,30,4,5,22,25]. The
conjecture is known to be false for some (algebraically closed) fields of finite
characteristic, e.g., genus g = 7 and characteristic charF = 2; see [29].

2.2 Coble Self-Dual Sets of Points

The free resolution of a hyperplane section of a Cohen-Macaulay ring has the
same Betti numbers. Thus we may ask for a geometric interpretation of the
syzygies of 2g − 2 points in Pg−2 (hyperplane section of a canonical curve),
or syzygies of a graded Artinian Gorenstein algebra with Hilbert function
(1, g− 2, g− 2, 1, 0, . . .) (twice a hyperplane section). Any collection of 2g− 2
points obtained as a hyperplane section of a canonical curve is special in
the sense that it imposes only 2g − 3 conditions on quadrics. An equivalent
condition for points in linearly uniform position is that they are Coble (or
Gale) self-dual; see [14]. Thus if we distribute the 2g − 2 points into two
collections each of g−1 points, with, say, the first consisting of the coordinate
points and the second corresponding to the rows of a (g− 1)× (g− 1) matrix

A = (aij), then A can be chosen to be an orthogonal matrix, i.e., AtA = 1;
see [14].

To see what the analogue of Green’s Conjecture for the general curve
means for orthogonal matrices we recall a result of [28].

Set n = g − 2. We identify the homogeneous coordinate ring of Pn with
the ring S = F[∂0, . . . , ∂n] of differential operators with constant coefficients,
∂i = ∂

∂xi
. The ring S acts on F[x0, . . . , xn] by differentiation. The annihilator

of q = x2
0 + . . .+ x2

n is a homogeneous ideal J ⊂ S such that S/J is a graded
Artinian Gorenstein ring with Hilbert function (1, n+1, 1) and socle induced
by q, see [27], [12, Section 21.2 and related exercise 21.7]. The syzygy numbers
of S/J are

1 - - - - - -
- n

n+2

(
n+3

2

)
· · · p(n+1−p)

n+2

(
n+3
p+1

)
· · · n

n+2

(
n+3
n+1

)
-

- - - - - - 1

A collection H0, . . . ,Hn of hyperplanes in P
n is said to form a polar

simplex to q if and only if the collection Γ = {p0, . . . , pn} ⊂ P̌
n of the

corresponding points in the dual space has its homogeneous ideal IΓ ⊂ S
contained in J .

In particular the set Λ consisting of the coordinate points correspond to
a polar simplex, because ∂i∂j annihilates q for i 6= j.

For any polar collection of points Γ the free resolution SΓ is a subcomplex
of the resolution SS/J . Green’s conjecture for the generic curve of genus
g = n+ 2 would imply:

Conjecture 2.2. For a general Γ and the given Λ the corresponding Tor-
groups

TorSk (S/IΓ ,F)k+1 ∩ TorSk (S/IΛ,F)k+1 ⊂ TorSk (S/J,F)k+1

intersect transversally.

Proof. A zero-dimensional non-degenerate scheme Γ ⊂ Pn of degree n+1 has
syzygies

1 - - - - -
-
(
n+1

2

)
· · · k

(
n+1
k+1

)
· · · n

Since both Tor groups are contained in TorSk (S/J,F)k+1, the claim is
equivalent to saying that for a general polar simplex Γ the expected dimen-
sion of their intersection is dim TorSk (S/Γ ,F)k+1 + dim TorSk (S/Γ ,F)k+1 −
dim TorSk (S/J,F)k+1, which is

2k
(
g − 1
k + 1

)
− k(g − 1− k)

g

(
g + 1
k + 1

)
.

On the other hand, IΓ∪Λ = IΓ ∩ IΛ, hence

TorSk (S/IΓ ,F)k+1 ∩ TorSk (S/IΛ,F)k+1 = TorSk (S/IΓ∪Λ,F)k+1,

and Green’s conjecture would imply

dim TorSk (S/IΓ∪Λ,F)k+1 = βk,k+1(Γ ∪Λ) = k

(
g − 2
k + 1

)
− (g− 1− k)

(
g − 2
k − 2

)
.

Now a calculation shows that the two dimensions above are equal.

The family of all polar simplices V is dominated by the family defined by
the ideal of 2× 2 minors of the matrix(

∂0 . . . ∂i . . . ∂n∑
j b0j∂j . . .

∑
j bij∂j . . .

∑
j bnj∂j

)
depending on a symmetric matrix B = (bij), i.e., bij = bji as parameters.
For B a general diagonal matrix we get Λ together with a specific element in
TorSn(S/IΛ,F)n+1.

2.3 Comparison and Probes

One of the peculiar consequences of Green’s conjecture for odd genus g =
2k + 1 is that, if βk,k+1 = βk−1,k+1 6= 0, then the curve C lies in the closure
of the locus of k + 1-gonal curve. Any k + 1-gonal curve lies on a rational
normal scroll X of codimension k that satisfies βk,k+1(X) = k. Hence

βk,k+1(C) 6= 0⇒ βk,k+1(C) ≥ k

We may ask whether a result like this is true for the union of two polar
simplices Λ ∪ Γ ⊂ P2k−1. Define

D̃ = {Γ ∈ V |Γ ∪ Λ is syzygy special}

where, as above, V denotes the variety of polar simplices and Λ denotes the
coordinate simplex. If D̃ is a proper subvariety, then it is a divisor, because
βk,k+1(Γ ∪ Λ) = βk−1,k+1(Γ ∪ Λ).

Conjecture 2.3. The subscheme D̃ ⊂ V is an irreducible divisor, for g =
n + 2 = 2k + 1 ∈ {7, 9, 11}. The value of βk,k+1 on a general point of D is
3, 2, 1 respectively.

We can prove this for g = 7 by computer algebra. For g = 9 and g = 11 a
proof is computationally out of reach with our methods, but we can get some
evidence from examples over finite fields.

Evidence. Since D̃ is a divisor, we expect that if we pick symmetric matrices
B over Fq at random, we will hit points on every component of D̃ at a
probability of 1/q. For a general point on D̃ the corresponding Coble self-
dual set of points will have the generic number of extra syzygies of that
component. Points with even more syzygies will occur in higher codimension,
hence only with a probability of 1/q2. Some evidence for irreducibility can be
obtained from the Weil formulas: for sufficiently large q we should see points
on D̃ with probability Cq−1+O(q−

3
2 ), where C is the number of components.

The following tables give for small fields Fq the number si of examples
with i extra syzygies in a test of 1000 examples for g = 9 and 100 examples
for g = 11. The number stot =

∑
i>0 si is the total number of examples with

extra syzygies.
Genus g = 9:

q 1000/q stot s1 s2 s3 s4

2 500 925 0 130 0 63
3 333 782 0 273 0 33
4 250 521 0 279 0 99
5 200 350 0 217 0 74
7 143 197 0 144 0 36
8 125 199 0 147 0 43
9 111 218 0 98 0 0

11 91 118 0 102 0 15
13 77 90 0 79 0 10
16 62 72 0 68 0 4
17 59 76 0 69 0 6

Genus g = 11:

q 100/q stot s1 s2 s3 s4

7 14.3 16 14 0 0 0
17 5.9 7 7 0 0 0

In view of these numbers, it is more likely that the set D̃ of syzygy spe-
cial Coble points is irreducible than that it is reducible. For a more precise
statement we refer to [6].

A test of Green’s Conjecture for Curves. In view of 2.3 it seems plau-
sible that for a general curve of odd genus g ≥ 11 with βk,k+1(C) 6= 0 the
value might be βk,k+1 = 1 contradicting Green’s conjecture. It is clear that
the syzygy exceptional locus has codimension 1 in Mg for odd genus, if it is
proper, i.e., if Green’s conjecture holds for the general curve of that genus.
So picking points at random we might be able to find such curve over a finite
field Fq with probability 1/q, roughly.

Writing code that does this is straightforward. One makes a loop that
picks up randomly a curve, computes its canonical image, and resolves its

ideal, counting the possible values βk,k+1 until a certain amount of special
curves is reached. The result for 10 special curves in F7 is as predicted:

total special possible values of βk,k+1

g seconds curves curves ≤ 2 3 4 5 6 ≥ 7
7 148 75 10 0 10 0 0 0 0
9 253 58 10 0 0 9 0 0 1

11 25640 60 10 0 0 0 9 0 1

(The test for genus 9 and 11 used about 70 and 120 megabytes of memory,
respectively.)

So Green’s conjecture passed the test for g = 9, 11. Shortly after the first
author tried this test for the first time, a paper of Hirschowitz and Ramanan
appeared proving this in general:

Theorem 2.4 ([22]). If the general curve of odd genus g = 2k + 1 satisfies
Green’s conjecture then the syzygy special curves lie on the divisor D = {C ∈
Mg|W 1

k+1(C) 6= ∅}

The theorem gives strong evidence for the full Green’s conjecture in view
of our study of Coble self-dual sets of points.

Our findings suggest that the variety of points arising as hyperplane sec-
tions of smooth canonical curves has the strange property that it intersects
the divisor of syzygy special sets of points D̃ only in its singular locus.

The conjecture for general curves is known to us up to g ≤ 17, which is
as far as a computer allows us to do a ribbon example; see [4].

3 Pfaffian Calabi-Yau Threefolds in P6

Calabi-Yau 3-folds caught the attention of physicists because they can serve
as the compact factor of the Kaluza-Klein model of spacetime in superstring
theory. One of the remarkable things that grows out of the work in physics is
the discovery of mirror symmetry, which associates to a family of Calabi-Yau
3-folds (Mλ), another family (Wµ) whose Hodge diamond is the mirror of the
Hodge diamond of the original family.

Although there is an enormous amount of evidence at present, the ex-
istence of a mirror is still a hypothesis for general Calabi-Yau 3-folds. The
thousands of cases where this was established all are close to toric geometry,
where through the work of Batyrev and others [3,9] rigorous mirror construc-
tions were given and parts of their conjectured properties proved.

From a commutative algebra point of view the examples studied so far
are rather trivial, because nearly all are hypersurfaces or complete intersec-
tions on toric varieties, or zero loci of sections in homogeneous bundles on
homogeneous spaces.

Of course only a few families of Calabi-Yau 3-folds should be of this
kind. Perhaps the easiest examples beyond the toric/homogeneous range are
Calabi-Yau 3-folds in P6. Here examples can be obtained by the Pfaffian
construction of Buchsbaum-Eisenbud [7] with vector bundles; see section 3.1
below. Indeed a recent theorem of Walter [36] says that any smooth Calabi-
Yau in P6 can be obtained in this way. In this section we report on our
construction of such examples.

As is quite usual in this kind of problem, there is a range where the
construction is still quite easy, e.g., for surfaces in P4 the work in [10,26]
shows that the construction of nearly all the 50 known families of smooth
non-general type surfaces is straight forward and their Hilbert scheme com-
ponent unirational. Only in very few known examples is the construction
more difficult and the unirationality of the Hilbert scheme component an
open problem.

The second author did the first “non-trivial” case of a construction of
Calabi-Yau 3-folds in P6. Although in the end the families turned out to be
unirational, the approach utilized small finite field constructions as a research
tool.

3.1 The Pfaffian Complex

Let F be a vector bundle of odd rank rkF = 2r+ 1 on a projective manifold
M , and let L be a line bundle. Let ϕ ∈ H0(M,Λ2F ⊗ L) be a section. We
can think of ϕ as a skew-symmetric twisted homomorphism

F∗ ϕ−→ F ⊗L.

The rth divided power of ϕ is a section ϕ(r) = 1
r! (ϕ ∧ · · · ∧ ϕ) ∈ H0(M,

Λ2rF ⊗ Lr). Wedge product with ϕ(r) defines a morphism

F ⊗ L ψ−→ Λ2r+1F ⊗ Lr+1 = det(F)⊗ Lr+1.

The twisted image I = image(ψ)⊗ det(F∗)⊗ L−r−1 ⊂ OM is called the
Pfaffian ideal of ϕ, because working locally with frames, it is given by the
ideal generated by the 2r× 2r principle Pfaffians of the matrix describing ϕ.
Let D denote the determinant line bundle det(F∗).

Theorem 3.1 (Buchsbaum-Eisenbud [7]). With this notation

0→ D2 ⊗ L−2r−1 ψt−→ D ⊗L−r−1 ⊗F∗ ϕ−→ F ⊗D ⊗ L−r ψ−→ OM

is a complex. X = V (I) ⊂M has codimension ≤ 3 at every point, and in case
equality holds (everywhere along X) then this complex is exact and resolves
the structure sheaf OX = OM/I of the locally Gorenstein subscheme X.

We will apply this to construct Calabi-Yau 3-folds in P6. In that case
we want X to be smooth and det(F)−2 ⊗ L−2r−1 ∼= ωP ∼= O(−7), so we
may conclude that ωX ∼= OX . A result of Walter [36] for Pn guarantees the
existence of a Pfaffian presentation in P6 for every subcanonical embedded 3-
fold. Moreover Walter’s choice of F⊗D⊗L−r for Calabi-Yau 3-foldsX ⊂ P6 is
the sheafified first syzygy module H1

∗ (IX) plus possibly a direct sum of line
bundles (indeed H2

∗ (IX) = 0 because of the Kodaira vanishing theorem).
Under the maximal rank assumption for

H0(P6,O(m))→ H0(X,OX(m))

the Hartshorne-Rao module is zero for d = degX ∈ {12, 13, 14} and an
arithmetically Cohen-Macaulay X is readily found. For d ∈ {15, 16, 17, 18}
the Hartshorne-Rao modules M have Hilbert functions with values (0, 0, 1, 0,
. . .), (0, 0, 2, 1, 0, . . .), (0, 0, 3, 5, 0, . . .) and (0, 0, 4, 9, 0, . . .) respectively.

We do not discuss the cases d ≤ 16 further. The construction in those
cases is obvious; see [34].

3.2 Analysis of the Hartshorne-Rao Module for Degree 17

Denote with F1 the sheaf F⊗D⊗L−r. We try to construct F1 as the sheafified
first syzygy module of M . The construction of a module with the desired
Hilbert function is straightforward. The cokernel of 3S(−2) b←− 16S(−3) for
a general matrix of linear forms has this property. However, for a general
b and F1 = ker(16O(−3) b−→ 3O(−2)) the space of skew-symmetric maps
Homskew(F∗1 (−7),F1) is zero: M has syzygies

3 16 28 - - - -
- - - 70 112 84 32 5

Any map ϕ : F∗1 (−7)→ F1 induces a map on the free resolutions:

0 F1
oo 28O(−4)oo 70O(−6)oo 112O(−7)oo

0 F∗1 (−7)oo

ϕ

OO

16O(−4)oo

ϕ0

OO

3O(−5)oo

ϕ1

OO

0oo

Since ϕ1 = 0 for degree reasons, ϕ = 0 as well, and Hom(F∗1 (−7),F1) = 0
for a general module M .

What we need are special modules M that have extra syzygies

3 16 28 k - - -
- - k 70 112 84 32 5

with k at least 3.

In a neighborhood of o ∈ SpecB, where denotes B the base space of a
semi-universal deformation of M , the resolution above would lift to a complex
over B[x0, . . . , x6] and in the lifted complex there is a k × k matrix ∆ with
entries in the maximal ideal o ⊂ B. By the principal ideal theorem we see that
Betti numbers stay constant in a subvariety of codimension at most k2. To
check for second linear syzygies on a randomly chosen M is a computationally
rather easy task. The following procedure tests the computer speed of this
task.

i19 : testModulesForDeg17CY = (N,k,p) -> (
x:=symbol x;R:=(ZZ/p)[x_0..x_6];
numberOfGoodModules:=0;i:=0;
usedTime:=timing while (i<N) do (

b:=random(R^3,R^{16:-1});
--we put SyzygyLimit=>60 because we expect
--k<16 syzygies, so 16+28+k<=60
fb:=res(coker b,

DegreeLimit =>0,SyzygyLimit=>60,LengthLimit =>3);
if rank fb_3>=k and (dim coker b)==0 then (

fb=res(coker b, DegreeLimit =>0,LengthLimit =>4);
if rank fb_4==0
then numberOfGoodModules=numberOfGoodModules+1;);

i=i+1;);
collectGarbage();
timeForNModules:=usedTime#0;
{timeForNModules,numberOfGoodModules});

Running this procedure we see that it takes not more than 0.64 seconds
per example. Hence we can hope to find examples with k = 3 within a
reasonable time for a very small field, say F3.

The first surprise is that examples with k extra syzygies are found much
more often, as can be seen by looking at the second value output by the
function testModulesForDeg17CY().

This is not only a “statistical” remark, in the sense that the result is
confirmed by computing the semi-universal deformations of these modules.
Indeed define Mk = {M | TorS3 (M,F)5 ≥ k} and consider a module M ∈Mk:
“generically” we obtain codim(Mk)M = k instead of k2 (and in fact one can
diagonalize the matrix ∆ over the algebraic closure F̄).

The procedure is straightforward but a bit long. First we pick up an
example with k-extra syzygies.

i20 : randomModuleForDeg17CY = (k,R) -> (
isGoodModule:=false;i:=0;
while not isGoodModule do (

b:=random(R^3,R^{16:-1});
--we put SyzygyLimit=>60 because we expect
--k<16 syzygies, so 16+28+k<=60
fb:=res(coker b,

DegreeLimit =>0,SyzygyLimit=>60,LengthLimit =>3);
if rank fb_3>=k and (dim coker b)==0 then (

fb=res(coker b, DegreeLimit =>0,LengthLimit =>4);
if rank fb_4==0 then isGoodModule=true;);

i=i+1;);
<<" -- Trial n. " << i <<", k="<< rank fb_3 <<endl;



Needles in a Haystack: Special Varieties via Small Fields 271

b);

Notice that the previous function returns a presentation matrix b of M , and
not M .

Next we compute the tangent codimension of Mk in the given example
M = Coker b by computing the codimension of the space of the infinitesimal
deformations of M that still give an element in Mk. Denote with bi the
maps in the linear strand of a minimal free resolution of M , and with b′2 the
quadratic part in the second map of this resolution. Over B = F[ε]/ε2 let
b1 + εf1 be an infinitesimal deformation of b1. Then f1 lifts to a linear map
f2 : 28S(−4)→ 16S(−3) determined by (b1 +εf1)◦(b2 +εf2) = 0, and f2 to a
map f3⊕∆ : kS(−5)→ 28S(−4)⊕kS(−5) determined by (b2+b′2)◦ε(f3⊕∆) =
0. Therefore we can determine ∆ as:

i21 : getDeltaForDeg17CY = (b,f1) -> (
fb:=res(coker b, LengthLimit =>3);
k:=numgens target fb.dd_3-28; --# of linear syzygies
b1:=fb.dd_1;b2:=fb.dd_2_{0..27};b2’:=fb.dd_2_{28..28+k-1};
b3:=fb.dd_3_{0..k-1}^{0..27};
--the equation for f2 is b1*f2+f1*b2=0,
--so f2 is a lift of (-f1*b2) through b1
f2:=-(f1*b2)//b1;
--the equation for A=(f3||Delta) is -f2*b3 = (b2|b2’) * A
A:=(-f2*b3)//(b2l|b2’);
Delta:=A^{28..28+k-1});

Now we just parametrize all possible maps f1 : 16S(−3)→ 3S(−2), compute
their respective maps ∆, and find the codimension of the condition that ∆
is the zero map:

i22 : codimInfDefModuleForDeg17CY = (b) -> (
--we create a parameter ring for the matrices f1’s
R:=ring b;K:=coefficientRing R;
u:=symbol u;U:=K[u_0..u_(3*16*7-1)];
i:=0;while i<3 do (

<<endl<< " " << i+1 <<":" <<endl;
j:=0;while j<16 do(

<< " " << j+1 <<". "<<endl;
k:=0;while k<7 do (

l=16*7*i+7*j+k; --index parametrizing the f1’s
f1:=matrix(R,apply(3,m->apply(16,n->

if m==i and n==j then x_k else 0)));
Delta:=substitute(getDeltaForDeg17CY(b,f1),U);
if l==0 then (equations=u_l*Delta;) else (

equations=equations+u_l*Delta;);
k=k+1;);

collectGarbage(); --frees up memory in the stack
j=j+1;);

i=i+1;);
codim ideal equations);

The second surprise is that for F1 = syz1(M) we find

dim Homskew(F∗1 (−7),F1) = k = dim TorS3 (M,F)5.

Homskew(F∗1 (−7),F1) is the vector space of skew-symmetric linear matrices
ϕ such that b ◦ ϕ = 0. The following procedure gives a matrix of size

(
16
2

)
×



272 F-O. Schreyer and F. Tonoli

dim Homskew(F∗1 (−7),F1) whose i-th column gives the entries of a 16 × 16
skew-symmetric matrix inducing the i-th basis element of the vector space
Homskew(F∗1 (−7),F1).

i23 : skewSymMorphismsForDeg17CY = (b) -> (
--we create a parameter ring for the morphisms:
K:=coefficientRing ring b;
u:=symbol u;U:=K[u_0..u_(binomial(16,2)-1)];
--now we compute the equations for the u_i’s:
UU:=U**ring b;
equationsInUU:=flatten (substitute(b,UU)*

substitute(genericSkewMatrix(U,u_0,16),UU));
uu:=substitute(vars U,UU);
equations:=substitute(

diff(uu,transpose equationsInUU),ring b);
syz(equations,DegreeLimit =>0));

A morphism parametrized by a column skewSymMorphism is then recovered
by the following code.

i24 : getMorphismForDeg17CY = (SkewSymMorphism) -> (
u:=symbol u;U:=K[u_0..u_(binomial(16,2)-1)];
f=map(ring SkewSymMorphism,U,transpose SkewSymMorphism);
f genericSkewMatrix(U,u_0,16));

Rank 1 Linear Syzygies of M . To understand this phenomenon we con-
sider the multiplication tensor of M :

µ : M2 ⊗ V →M3

where V = H0(P6,O(1)).

Definition 3.2. A decomposable element of M2 ⊗ V in the kernel of µ is
called a rank 1 linear syzygy of M . The (projective) space of rank 1 syzygies
is

Y = (P2 × P6) ∩ P15 ⊂ P20

where P2 = P(M∗2 ),P6 = P(V ∗) and P15 = P(ker(µ)∗) inside the Segre space
P((M2 ⊗ V )∗) ∼= P

20.

Proposition 1.5 of [17] says that, for dimM2 ≤ j, the existence of a jth

linear syzygy implies dimY ≥ j − 1. This is automatically satisfied for j = 3
in our case: dimY ≥ 3 with equality expected.

The projection Y → P
2 has linear fibers, and the general fiber is a P1.

However, special fibers might have higher dimension. In terms of the presen-
tation matrix b a special 2-dimensional fiber (defined over F) corresponds to
a block

b =

0 0 0 ∗ . . .
0 0 0 ∗ . . .
l1 l2 l3 ∗ . . .

 ,

where l1, l2, l3 are linear forms, in the 3× 16 presentation matrix of M . Such
a block gives a

1 3 3 1 - - - -
- - - - - - - -

subcomplex in the free resolution of M and an element s ∈ H0(P6, Λ2F1 ⊗
O(7)) since the syzygy matrix 0 −l3 l2

l3 0 −l1
−l2 l1 0


is skew.

This answers the questions posed by both surprises: we want a module M
with at least k ≥ 3 special fibers and these satisfy h0(P6, Λ2F1 ⊗O(7)) ≥ k,
if the k sections are linearly independent. The condition for k special fibers
is of expected codimension k in the parameter space G(16, 3h0(P6,O(1)) of
the presentation matrices. In a given point M the actual codimension can be
readily computed by a first order deformations and that H0(P6, Λ2F1⊗O(7))
is k-dimensional, and spanned by the k sections corresponding to the k special
fibers can be checked as well.

First we check that M has k distinct points in P(M∗2 ) where the multipli-
cation map drops rank. (Note that this condition is likely to fail over small
fields. However, the check is computationally easy).

i25 : checkBasePtsForDeg17CY = b -> (
--firstly the number of linear syzygies
fb:=res(coker b, DegreeLimit=>0, LengthLimit =>4);
k:=#select(degrees source fb.dd_3,i->i=={3});
--then the check
a=symbol a;A=K[a_0..a_2];
mult:=(id_(A^7)**vars A)*substitute(

syz transpose jacobian b,A);
basePts=ideal mingens minors(5,mult);
codim basePts==2 and degree basePts==k and distinctPoints(

basePts));

Next we check that H0(P6, Λ2F1 ⊗ O(7)) is k-dimensional, by looking at
the numbers of columns of skewSymMorphismsForDeg17CY(b). Finally we
do the computationally hard part of the check, which is to verify that the
k special sections corresponding to the k special fibers of Y → P

2 span
H0(P6, Λ2F1 ⊗O(7)).

i26 : checkMorphismsForDeg17CY = (b,skewSymMorphisms) -> (
--first the number of linear syzygies
fb:=res(coker b, DegreeLimit=>0, LengthLimit =>4);
k:=#select(degrees source fb.dd_3,i->i=={3});
if (numgens source skewSymMorphisms)!=k then (

error "the number of skew-sym morphisms is wrong";);
--we parametrize the morphisms:
R:=ring b;K:=coefficientRing R;
w:=symbol w;W:=K[w_0..w_(k-1)];
WW:=R**W;ww:=substitute(vars W,WW);

genericMorphism:=getMorphismForDeg17CY(
substitute(skewSymMorphisms,WW)*transpose ww);

--we compute the scheme of the 3x3 morphisms:
equations:=mingens pfaffians(4,genericMorphism);
equations=diff(

substitute(symmetricPower(2,vars R),WW),equations);
equations=saturate ideal flatten substitute(equations,W);
CorrectDimensionAndDegree:=(

dim equations==1 and degree equations==k);
isNonDegenerate:=#select(

(flatten degrees source gens equations),i->i==1)==0;
collectGarbage();
isOK:=CorrectDimensionAndDegree and isNonDegenerate;
if isOK then (

--in this case we also look for a skew-morphism f
--which is a linear combination of the special
--morphisms with all coefficients nonzero.
isGoodMorphism:=false;while isGoodMorphism==false do (

evRandomMorphism:=random(K^1,K^k);
itsIdeal:=ideal(

vars W*substitute(syz evRandomMorphism,W));
isGoodMorphism=isGorenstein(

intersect(itsIdeal,equations));
collectGarbage());

f=map(R,WW,vars R|substitute(evRandomMorphism,R));
randomMorphism:=f(genericMorphism);
{isOK,randomMorphism}) else {isOK});

The code above is structured as follows. First we parametrize the skew-
symmetric morphisms with new variables. The ideal of 4 × 4 Pfaffians is
generated by forms of bidegree (2, 2) over P6 × Pk−1. We are interested in
points p ∈ Pk−1 such that the whole fiber P6×{p} is contained in the zero lo-
cus of the Pfaffian ideal. The next two lines produce the ideal of these points
on Pk−1. Since we already know of k distinct points by the previous check,
it suffices to establish that the set consists of collection of k spanning points.
Finally, if this is the case, a further point, i.e., a further skew morphism, is
a linear combination with all coefficients non-zero, if and only if the union
with this point is a Gorenstein set of k + 1 points in Pk−1.

i27 : isGorenstein = (I) -> (
codim I==length res I and rank (res I)_(length res I)==1);

It is clear that all 16 relations should take part in the desired skew ho-
momorphism F∗1 (−7)

ϕ−→ F1. Thus we need k ≥ 6 to have a chance for a
Calabi-Yau. Since 3 · 5 < 16 it easy to guarantee 5 special fibers by suitable
choice of the presentation matrix. So the condition k ≥ 6 is only of codimen-
sion k − 5 on this subspace, and we have a good chance to find a module of
the desired type.

i28 : randomModule2ForDeg17CY = (k,R) -> (
isGoodModule:=false;i:=0;
while not isGoodModule do (

b:=(random(R^1,R^{3:-1})++
random(R^1,R^{3:-1})++
random(R^1,R^{3:-1})|
matrix(R,{{1},{1},{1}})**random(R^1,R^{3:-1})|
random(R^3,R^1)**random(R^1,R^{3:-1})|

random(R^3,R^{1:-1}));
--we put SyzygyLimit=>60 because we expect
--k<16 syzygies, so 16+28+k<=60
fb:=res(coker b,

DegreeLimit =>0,SyzygyLimit=>60,LengthLimit =>3);
if rank fb_3>=k and dim coker b==0 then (

fb=res(coker b, DegreeLimit =>0,LengthLimit =>4);
if rank fb_4==0 then isGoodModule=true;);

i=i+1;);
<<" -- Trial n. " << i <<", k="<< rank fb_3 <<endl;
b);

Some modules M with k = 8, 9, 11 lead to smooth examples of Calabi-Yau
3-folds X of degree 17. To check the smoothness via the Jacobian criterion
is computationally too heavy for a common computer today. For a way to
speed up this computation considerably and to reduce the required amount
of memory to a reasonable value (128MB), we refer to [34].

Since h0(P6, Λ2F1 ⊗ O(7)) = k and codim{M | TorS3 (M,F)5 ≥ k} = k
all three families have the same dimension. In particular no family lies in the
closure of another.

A deformation computation verifies h1(X, T ) = h1(X,Ω2) = 23. Hence a
computation of the Hodge numbers hq(X,Ωp) gives the diamond

1
0 0

0 1 0
1 23 23 1

0 1 0
0 0

1

Example 3.3. The following commands give an example of a Calabi-Yau
3-fold in P6:

i29 : K=ZZ/13;

i30 : R=K[x_0..x_6];

i31 : time b=randomModule2ForDeg17CY(8,R);
-- Trial n. 1757, k=8
-- used 764.06 seconds

3 16
o31 : Matrix R <--- R

i32 : betti res coker b

o32 = total: 3 16 36 78 112 84 32 5
0: 3 16 28 8 . . . .
1: . . 8 70 112 84 32 5

i33 : betti (skewSymMorphisms=skewSymMorphismsForDeg17CY b)

o33 = total: 120 8
-1: 120 8

We check whether the base points in M0 are all distinct.
i34 : checkBasePtsForDeg17CY b

o34 = true

Now we check whether the k sections span the morphisms. If we get true
then this is a good module.

i35 : finalTest=checkMorphismsForDeg17CY(b,skewSymMorphisms);

i36 : finalTest#0

o36 = true

We pick up a random morphism involving all k sections.
i37 : n=finalTest#1;

16 16
o37 : Matrix R <--- R

If all the tests are okay, there should be a high degree syzygy.
i38 : betti (nn=syz n)

o38 = total: 16 4
1: 16 3
2: . .
3: . 1

i39 : n2t=transpose submatrix(nn,{0..15},{3});

1 16
o39 : Matrix R <--- R

i40 : b2:=syz b;

16 36
o40 : Matrix R <--- R

Finally, compute the ideal of the Calabi-Yau 3-fold in P6.
i41 : j:=ideal mingens ideal flatten(n2t*b2);

o41 : Ideal of R

i42 : degree j

o42 = 17

i43 : codim j

o43 = 3

i44 : betti res j

o44 = total: 1 20 75 113 84 32 5
0: 1 . . . . . .
1: . . . . . . .
2: . . . . . . .
3: . 12 5 . . . .
4: . 8 70 113 84 32 5

3.3 Lift to Characteristic Zero

At this point we have constructed Calabi-Yau 3-folds X ⊂ P6 over the finite
field F5 or F7. However, our main interest is the field of complex numbers C.
The existence of a lift to characteristic zero follows by the following argument.

The set Mk = {M | TorS3 (M,F)5 ≥ k} has codimension at most k. A
deformation calculation shows that at our special point M special ∈ M(Fp)
the codimension is achieved and that Mk is smooth at this point. Thus tak-
ing a transversal slice defined over Z through this point we find a number
field K and a prime p in its ring of integers OK with OK/p ∼= Fp such that
M special is the specialization of an OK,p-valued point of Mk. Over the generic
point of SpecOK,p we obtain a K-valued point. From our computations with
checkBasePtsForDeg17CY() and checkMorphismsForDeg17CY(), which ex-
plained why h0(P6, Λ2F special

1 ⊗O(7)) = k, it follows that

H0(P6
Z
× SpecOK,p, Λ2F1 ⊗O(7))

is free of rank k over OK,p. Hence ϕspecial extends to OK,p as well, and by
semi-continuity we obtain a smooth Calabi-Yau 3-fold defined over K ⊂ C.

Theorem 3.4 ([34]). The Hilbert scheme of smooth Calabi-Yau 3-folds of
degree 17 in P6 has at least 3 components. These three components are reduced
and have dimension 23 + 48. The corresponding Calabi-Yau 3-folds differ in
the number of quintic generators of their homogeneous ideals, which are 8, 9
and 11 respectively.

See [34] for more details.

Note that we do not give a bound on the degree [K : Q] of the number
field, and certainly we are far away from a bound of its discriminant.

This leaves the question open whether these parameter spaces of Calabi-
Yau 3-folds are unirational. Actually they are, as the geometric construction
of modules M ∈Mk in [34] shows.

A construction of one or several mirror families of these Calabi-Yau 3-folds
is an open problem.

488, 1983.

2. E. Arbarello, M. Cornalba, P. Griffith, and J. Harris: Geometry of algebraic
curves, vol I. Springer Grundlehren, 267:xvi+386, 1985.

3. V. Batyrev: Dual polyhedra and mirror symmetry for Calabi-Yau hypersurfaces
in algebraic tori. J. Alg. Geom., 3:493–535, 1994.

4. D. Bayer and D. Eisenbud: Ribbons and their canonical embeddings. Trans.
Am. Math. Soc., 347,:719–756, 1995.



278 F-O. Schreyer and F. Tonoli

5. H.-C. v. Bothmer: Geometrische Syzygien kanonischer Kurven. Thesis
Bayreuth, pages 1–123, 2000.

6. H.-C. v. Bothmer and F.-O. Schreyer: A quick and dirty irreduciblity test.
manuscript, to appear.

7. D. Buchsbaum and D. Eisenbud: Algebra structure on finite free resolutions and
some structure theorem for ideals of codimension 3. Am. J. Math., 99:447–485,
1977.

8. M.-C. Chang and Z. Ran: Unirationality of the moduli space of curves genus
11, 13 (and 12). Invent. Math., 76:41–54, 1984.

9. D. Cox and S. Katz: Mirror symmetry and algebraic geometry. AMS, Mathe-
matical Surveys and Monographs, 68:xxii+469, 1999.

10. W. Decker, L. Ein, and F.-O. Schreyer: Construction of surfaces in P4. J. of
Alg. Geom., 2:185–237, 1993.

11. W. Decker and F.-O. Schreyer: Non-general type surfaces in P4 - Remarks on
bounds and constructions. J. Symbolic Comp., 29:545–585, 2000.

12. D. Eisenbud: Commutative Algebra. With a view towards algebraic geometry.
Springer Graduate Texts in Mathematics, 150:xvi+785, 1995.

13. D. Eisenbud and J. Harris: The Kodaira dimension of the moduli space of
curves of genus g ≥ 23. Invent. Math., 87:495–515, 1987.

14. D. Eisenbud and S. Popescu: Gale duality and free resolutions of ideals of
points.. Invent. Math., 136:419–449, 1999.

15. G. Ellingsrud and C. Peskine: Sur les surfaces de P4. Invent. Math., 95:1–11,
1989.

16. M. Green: Koszul cohomology and the geometry of projective varieties. J. Diff.
Geom., 19:125–171, 1984.

17. M. Green: The Eisenbud-Koh-Stillman conjecture on linear syzygies. Invent.
Math, 163:411–418, 1999.

18. M. Green and R. Lazarsfeld: Appendix to Koszul cohomology and the geometry
of projective varieties. J. Diff. Geom., 19:168–171, 1984.

19. J. Harris: Curves in projective space. With collaboration of David Eisenbud.
Séminaire de Mathématiques Supériores, 85:Montreal, 138 pp., 1982.

20. J. Harris: On the Severi problem. Invent. Math., 84:445–461, 1986.
21. J. Harris and D. Mumford: On the Kodaira dimension of the moduli space of

curves. With an appendix by William Fulton. Invent. Math., 67:23–88, 1982.
22. A. Hirschowitz and S. Ramanan: New evidence for Green’s conjecture on syz-

gyies of canonical curves. Ann. Sci. École Norm. Sup. (4), 31:145–152, 1998.

D-modules and Cohomology of Varieties

Uli Walther

In this chapter we introduce the reader to some ideas from the world of
differential operators. We show how to use these concepts in conjunction
with Macaulay 2 to obtain new information about polynomials and their
algebraic varieties.

Gröbner bases over polynomial rings have been used for many years in
computational algebra, and the other chapters in this book bear witness to
this fact. In the mid-eighties some important steps were made in the theory
of Gröbner bases in non-commutative rings, notably in rings of differential
operators. This chapter is about some of the applications of this theory to
problems in commutative algebra and algebraic geometry.

Our interest in rings of differential operators and D-modules stems from
the fact that some very interesting objects in algebraic geometry and com-
mutative algebra have a finite module structure over an appropriate ring of
differential operators. The prime example is the ring of regular functions on
the complement of an affine hypersurface. A more general object is the Čech
complex associated to a set of polynomials, and its cohomology, the local co-
homology modules of the variety defined by the vanishing of the polynomials.
More advanced topics are restriction functors and de Rham cohomology.

With these goals in mind, we shall study applications of Gröbner bases
theory in the simplest ring of differential operators, the Weyl algebra, and de-
velop algorithms that compute various invariants associated to a polynomial
f . These include the Bernstein-Sato polynomial bf (s), the set of differential
operators J(fs) which annihilate the germ of the function fs (where s is a
new variable), and the ring of regular functions on the complement of the
variety of f .

For a family f1, . . . , fr of polynomials we study the associated Čech com-
plex as a complex in the category of modules over the Weyl algebra. The
algorithms are illustrated with examples. We also give an indication what
other invariants associated to polynomials or varieties are known to be com-
putable at this point and list some open problems in the area.

volume.

1 Introduction

1.1 Local Cohomology – Definitions

Let R be a commutative Noetherian ring (always associative, with identity)
and M an R-module. For f ∈ R one defines a Čech complex of R-modules

Č•(f) = (0→ R︸︷︷︸
degree 0

↪→ R[f−1]︸ ︷︷ ︸
degree 1

→ 0) (1.1)

where the injection is the natural map sending g ∈ R to g/1 ∈ R[f−1] and
“degree” refers to cohomological degree. For a family f1, . . . , fr ∈ R one
defines

Č•(f1, . . . , fr) =
r⊗
i=1

Č•(fi), (1.2)

and for an R-module M one sets

Č•(M ; f1, . . . , fr) = M ⊗R Č•(f1, . . . , fr). (1.3)

The i-th (algebraic) local cohomology functor with respect to f1, . . . , fr
is the i-th cohomology functor of Č•(−; f1, . . . , fr). If I = R · (f1, . . . , fr)
then this functor agrees with the i-th right derived functor of the functor
H0
I (−) which sends M to the I-torsion

⋃∞
k=1(0 :M Ik) of M and is denoted

by Hi
I(−). This means in particular, that H•I (−) depends only on the (rad-

ical of the) ideal generated by the fi. Local cohomology was introduced by
A. Grothendieck [13] as an algebraic analog of (classical) relative cohomology.
For instance, if X is a scheme, Y is a closed subscheme and U = X \ Y then
there is a long exact sequence

· · · → Hi(X,F)→ Hi(U,F)→ Hi+1
Y (X,F)→ · · ·

for all quasi-coherent sheaves F on X. (To make sense of this one has to
generalize the definition of local cohomology to be the right derived functor
of H0

Y (−) : F → (U → {f ∈ F(U) : supp(f) ⊆ Y ∩ U}).) An introduction to
algebraic local cohomology theory may be found in [8].

The cohomological dimension of I in R, denoted by cd(R, I), is the small-
est integer c such that Hi

I(M) = 0 for all i > c and all R-modules M . If R
is the coordinate ring of an affine variety X and I ⊆ R is the defining ideal
of the Zariski closed subset Y ⊆ X then the local cohomological dimension
of Y in X is defined as cd(R, I). It is not hard to show that if X is smooth,
then the integer dim(X)− cd(R, I) depends only on Y but neither on X nor
on the embedding Y ↪→ X.

1.2 Motivation

As one sees from the definition of local cohomology, the modules Hi
I(R) carry

information about the sections of the structure sheaf on Zariski open sets,
and hence about the topology of these open sets. This is illustrated by the
following examples. Let I ⊆ R and c = cd(R, I). Then I cannot be generated
by fewer than c elements – in other words, Spec(R)\Var(I) cannot be covered
by fewer than c affine open subsets (i.e., Var(I) cannot be cut out by fewer
than c hypersurfaces). In fact, no ideal J with the same radical as I will be
generated by fewer than c elements, [8].

Let Hi
Sing(−;C) stand for the i-th singular cohomology functor with com-

plex coefficients. The classical Lefschetz Theorem [12] states that if X ⊆ Pn
C

is a variety in projective n-space and Y a hyperplane section of X such that
X \ Y is smooth, then Hi

Sing(X;C) → Hi
Sing(Y ;C) is an isomorphism for

i < dim(X) − 1 and injective for i = dim(X) − 1. The Lefschetz Theorem
has generalizations in terms of local cohomology, called Theorems of Barth
Type. For example, let Y ⊆ Pn

C
be Zariski closed and I ⊆ R = C[x0, . . . , xn]

the defining ideal of Y . Then Hi
Sing(Pn

C
;C)→ Hi

Sing(Y ;C) is an isomorphism
for i < n− cd(R, I) and injective if i = n− cd(R, I) ([16], Theorem III.7.1).

A consequence of the work of Ogus and Hartshorne ([38], 2.2, 2.3 and [16],
Theorem IV.3.1) is the following. If I ⊆ R = C[x0, . . . , xn] is the defining ideal
of a complex smooth variety Y ⊆ Pn

C
then, for i < n− codim(Y ),

dimC socR(H0
m(Hn−i

I (R))) = dimCHi
x(Ỹ ;C)

where Hi
x(Ỹ ;C) stands for the i-th singular cohomology group of the affine

cone Ỹ over Y with support in the vertex x of Ỹ and with coefficients in C
(and socR(M) denotes the socle (0 :M (x0, . . . , xn)) ⊆ M for any R-module
M), [25]. These iterated local cohomology modules have a special structure
(cf. Subsection 4.3).

Local cohomology relates to the connectedness of the underlying spaces
as is shown by the following facts. If Y is a complete intersection of positive
dimension in Pn

C
, then Y cannot be disconnected by the removal of closed

subsets of codimension 2 in Y or higher, [7]. This is a consequence of the
so-called Hartshorne-Lichtenbaum vanishing theorem, see [8].

In a similar spirit one can show that if (A,m) is a complete local domain of
dimension n and f1, . . . , fr are elements of the maximal ideal with r+ 2 ≤ n,
then Var(f1, . . . , fr) \ {m} is connected, [7].

In fact, as we will discuss to some extent in Section 5, over the complex
numbers the complex Č•(R; f1, . . . , fr) for R = C[x1, . . . , xn] determines the
Betti numbers dimC(Hi

Sing(Cn \Var(f1, . . . , fr);C)).

1.3 The Master Plan

The cohomological dimension has been studied by many authors. For an
extensive list of references and some open questions we recommend to consult
the very nice survey article [17].

It turns out that for the determination of cd(R, I) it is in fact enough to
find a test to decide whether or not the local cohomology module Hi

I(R) = 0
for given i, R, I. This is because Hi

I(R) = 0 for all i > c implies cd(R, I) ≤ c
(see [14], Section 1).

Unfortunately, calculations are complicated by the fact that Hi
I(M) is

rarely finitely generated as R-module, even for very nice R and M . In this
chapter we show how in an important class of examples one may still carry
out explicit computations, by enlarging R.

We shall assume that I ⊆ Rn = K[x1, . . . , xn] where K is a computable
field containing the rational numbers. (By a computable field we mean a
subfield K of C such that K is described by a finite set of data and for which
addition, subtraction, multiplication and division as well as the test whether
the result of any of these operations is zero in the field can be executed by the
Turing machine. For example, K could be Q[

√
2] stored as a 2-dimensional

vector space over Q with an appropriate multiplication table.)
The ring of K-linear differential operators D(R,K) of the commutative

K-algebra R is defined inductively: one sets D0(R,K) = R, and for i > 0
defines

Di(R,K) = {P ∈ HomK(R,R) : Pr − rP ∈ Di−1(R,K) for all r ∈ R} .

Here, r ∈ R is interpreted as the endomorphism of R that multiplies by r.
The local cohomology modules Hi

I(Rn) have a natural structure of finitely
generated left D(Rn,K)-modules (see for example [20,25]). The basic reason
for this finiteness is that in this case Rn[f−1] is a cyclic D(Rn,K)-module,
generated by fa for Z 3 a� 0 (compare [5]):

Rn[f−1] = D(Rn,K) • fa. (1.4)

Using this finiteness we employ the theory of Gröbner bases in D(Rn,K)
to develop algorithms that give a presentation of Hi

I(Rn) and Hi
m(Hj

I (Rn))
for all triples i, j ∈ N, I ⊆ Rn in terms of generators and relations over
D(Rn,K) (where m = Rn · (x1, . . . , xn)), see Section 4. This also leads to an
algorithm for the computation of the invariants

λi,j(Rn/I) = dimK socRn(Hi
m(Hn−j

I (Rn)))

introduced in [25].
At the basis for the computation of local cohomology are algorithms that

compute the localization of a D(Rn,K)-module at a hypersurface f ∈ Rn.
That means, if the left module M = D(Rn,K)d/L is given by means of a
finite number of generators for the left module L ⊆ D(Rn,K)d then we want
to compute a finite number of generators for the left module L′ ⊆ D(Rn,K)d

′

which satisfies

D(Rn,K)d
′
/L′ ∼= (D(Rn,K)d/L)⊗Rn Rn[f−1],

which we do in Section 3.
Let L be a left ideal of D(Rn,K). The computation of the localization

of M = D(Rn,K)/L at f ∈ Rn is closely related to the D(Rn,K)[s]-module
Mf generated by

1⊗ 1⊗ fs ∈M ⊗Rn Rn[f−1, s]⊗ fs (1.5)

and the minimal polynomial bf (s) of s on the quotient ofMf by its submod-
uleMf ·f generated over D(Rn,K)[s] by 1⊗f⊗fs, cf. Section 3. Algorithms
for the computation of these objects have been established by T. Oaku in a
sequence of papers [31–33].

Astonishingly, the roots of bf (s) prescribe the exponents a that can be
used in the isomorphism (1.4) between Rn[f−1] and the D(Rn,K)-module
generated by fa. Moreover, any good exponent a can be used to transform
Mf into M ⊗Rn[f−1] by a suitable “plugging in” procedure.

Thus the strategy for the computation of local cohomology will be to
compute Mf and a good a for each f ∈ {f1, . . . , fr}, and then assemble the
Čech complex.

1.4 Outline of the Chapter

The next section is devoted to a short introduction of results on the Weyl
algebra D(Rn,K) and D-modules as they apply to our work. We start with
some remarks on the theory of Gröbner bases in the Weyl algebra.

In Section 3 we investigate Bernstein-Sato polynomials, localizations and
the Čech complex. The purpose of that section is to find a presentation of
M ⊗ Rn[f−1] as a cyclic D(Rn,K)-module if M = D(Rn,K)/L is a given
holonomic D-module (for a definition and some properties of holonomic mod-
ules, see Subsection 2.3 below).

In Section 4 we describe algorithms that for arbitrary i, j, k, I determine
the structure of Hk

I (R),Hi
m(Hj

I (R)) and find λi,j(R/I). The final section
is devoted to comments on implementations, efficiency, discussions of other
topics, and open problems.

2 The Weyl Algebra and Gröbner Bases

D-modules, that is, rings or sheaves of differential operators and modules
over these, have been around for several decades and played prominent roles
in representation theory, some parts of analysis and in algebraic geometry.
The founding fathers of the theory are M. Sato, M. Kashiwara, T. Kawai, J.
Bernstein, and A. Beilinson. The area has also benefited much from the work
of P. Deligne, J.-E. Björk, J.-E. Roos, B. Malgrange and Z. Mebkhout. The
more computational aspects of the theory have been initiated by T. Oaku
and N. Takayama.

The simplest example of a ring of differential operators is given by the
Weyl algebra, the ring of K-linear differential operators on Rn. In charac-
teristic zero, this is a finitely generated K-algebra that resembles the ring of
polynomials in 2n variables but fails to be commutative.

2.1 Notation

Throughout we shall use the following notation: K will denote a computable
field of characteristic zero and Rn = K[x1, . . . , xn] the ring of polynomials
over K in n variables. The K-linear differential operators on Rn are then the
elements of

Dn = K〈x1, ∂1, . . . , xn, ∂n〉,

the n-th Weyl algebra over K, where the symbol xi denotes the operator
“multiply by xi” and ∂i denotes the operator “take partial derivative with
respect to xi”. We therefore have in Dn the relations

xixj = xjxi for all 1 ≤ i, j ≤ n,
∂i∂j = ∂j∂i for all 1 ≤ i, j ≤ n,
xi∂j = ∂jxi for all 1 ≤ i 6= j ≤ n,

and xi∂i + 1 = ∂ixi for all 1 ≤ i ≤ n.

The last relation is nothing but the product (or Leibniz) rule, xf ′+f = (xf)′.
We shall use multi-index notation: xα∂β denotes the monomial

x1
α1 · · ·xnαn · ∂1

β1 · · · ∂nβn

and |α| = α1 + · · ·+ αn.
In order to keep the product ∂ixi ∈ Dn and the application of ∂i ∈ Dn

to xi ∈ Rn apart, we shall write ∂i • (g) to mean the result of the action of
∂i on g ∈ Rn. So for example, ∂ixi = xi∂i + 1 ∈ Dn but ∂i • xi = 1 ∈ Rn.
The action of Dn on Rn takes precedence over the multiplication in Rn
(and is of course compatible with the multiplication in Dn), so for example
∂2 • (x1)x2 = 0 · x2 = 0 ∈ Rn.

The symbol m will stand for the maximal ideal Rn · (x1, . . . , xn) of Rn, ∆
will denote the maximal left ideal Dn · (∂1, . . . , ∂n) of Dn and I will stand for
the ideal Rn · (f1, . . . , fr) in Rn. Every Dn-module becomes an Rn-module
via the embedding Rn ↪→ Dn as D0(Rn,K).

All tensor products in this chapter will be over Rn and all Dn-modules
(resp. ideals) will be left modules (resp. left ideals) unless specified otherwise.

2.2 Gröbner Bases in Dn

This subsection is a severely shortened version of Chapter 1 in [40] (and we
strongly recommend that the reader take a look at this book). The purpose
is to see how Gröbner basis theory applies to the Weyl algebra.

The elements in Dn allow a normally ordered expression. Namely, if P ∈
Dn then we can write it as

P =
∑

(α,β)∈E

cα,βx
α∂β

where E is a finite subset of N2n. Thus, as K-vector spaces there is an iso-
morphism

Ψ : K[x, ξ]→ Dn

(with ξ = ξ1, . . . , ξn) sending xαξβ to xα∂β . We will assume that every P ∈
Dn is normally ordered.

We shall say that (u, v) ∈ R2n is a weight vector for Dn if u+ v ≥ 0, that
is ui + vi ≥ 0 for all 1 ≤ i ≤ n. We set the weight of the monomial xα∂β

under (u, v) to be u · α+ v · β (scalar product). The weight of an operator is
then the maximum of the weights of the nonzero monomials appearing in the
normally ordered expression of P . If (u, v) is a weight vector for Dn, there is
an associated graded ring gr(u,v)(Dn) with

grr(u,v)(Dn) =
{P ∈ Dn : w(P ) ≤ r}
{P ∈ Dn : w(P ) < r}

.

So gr(u,v)(Dn) is the K-algebra on the symbols {xi : 1 ≤ i ≤ n} ∪ {∂i :
ui + vi = 0} ∪ {ξi : ui + vi > 0}. Here all variables commute with each other
except ∂i and xi for which the Leibniz rule holds.

Each P ∈ Dn has an initial form or symbol in(u,v)(P ) in gr(u,v)(Dn)
defined by taking all monomials in the normally ordered expression for P
that have maximal weight, and replacing all ∂i with ui + vi > 0 by the
corresponding ξi.

The inequality ui + vi ≥ 0 is needed to assure that the product of the
initial forms of two operators equals the initial form of their product: one
would not want to have in(∂i · xi) = in(xi · ∂i + 1) = 1.

A weight of particular importance is −u = v = (1, . . . , 1), or more
generally −u = v = (1, . . . , 1, 0, . . . , 0). In these cases gr(u,v)(Dn) ∼= Dn.
On the other hand, if u + v is componentwise positive, then gr(u,v)(Dn)
is commutative (compare the initial forms of ∂ixi and xi∂i) and isomor-
phic to the polynomial ring in 2n variables corresponding to the symbols of
x1, . . . , xn, ∂1, . . . , ∂n.

If L is a left ideal in Dn we write in(u,v)(L) for {in(u,v)(P ) : P ∈ L}.
This is a left ideal in gr(u,v)(Dn). If G ⊂ L is a finite set we call it a (u, v)-
Gröbner basis if the left ideal of gr(u,v)(Dn) generated by the initial forms of
the elements of G agrees with in(u,v)(L).

A multiplicative monomial order on Dn is a total order ≺ on the normally
ordered monomials such that

1. 1 ≺ xi∂i for all i, and
2. xα∂β ≺ xα′∂β′ implies xα+α′′∂β+β′′ ≺ xα′+α′′∂β′+β′′ for all α′′, β′′ ∈ Nn.

A multiplicative monomial order is a term order if 1 is the (unique) smallest
monomial. Multiplicative monomial orders, and more specifically term orders,
clearly abound.

Multiplicative monomial orders (and hence term orders) allow the con-
struction of initial forms just like weight vectors. Now, however, the initial
forms are always monomials, and always elements of K[x, ξ] (due to the to-
tal order requirement on ≺). One defines Gröbner bases for multiplicative
monomial orders analogously to the weight vector case.

For our algorithms we have need to compute weight vector Gröbner bases,
and this can be done as follows. Suppose (u, v) is a weight vector on Dn and
≺ a term order. Define a multiplicative monomial order ≺(u,v) as follows:

xα∂β ≺(u,v) x
α′∂β

′
⇔ [(α− α′) · u+ (β − β′) · v < 0] or[

(α− α′) · u+ (β − β′) · v = 0 and xα∂β ≺ xα
′
∂β
′
]
.

Note that ≺(u,v) is a term order precisely when (u, v) is componentwise non-
negative.

Theorem 2.1 ([40], Theorem 1.1.6.). Let L be a left ideal in Dn, (u, v)
a weight vector for Dn, ≺ a term order and G a Gröbner basis for L with
respect to ≺(u,v). Then

1. G is a Gröbner basis for L with respect to (u, v).
2. in(u,v)(G) is a Gröbner basis for in(u,v)(L) with respect to ≺. ut

We end this subsection with the remarks that Gröbner bases with respect
to multiplicative monomial orders can be computed using the Buchberger
algorithm adapted to the non-commutative situation (thus, Gröbner bases
with respect to weight vectors are computable according to the theorem),
and that the computation of syzygies, kernels, intersections and preimages
in Dn works essentially as in the commutative algebra K[x, ξ]. For precise
statements of the algorithms we refer the reader to [40].

2.3 D-modules

A good introduction to D-modules are the book by J.-E. Björk, [5], the nice
introduction [9] by S. Coutinho, and the lecture notes by J. Bernstein [4].
In this subsection we list some properties of localizations of Rn that are
important for module-finiteness over Dn. Most of this section is taken from
Section 1 in [5].

Let f ∈ Rn. Then the Rn-module Rn[f−1] has a structure as left Dn-
module via the extension of the action •:

xi • (
g

fk
) =

xig

fk
, ∂i • (

g

fk
) =

∂i • (g)f − k∂i • (f)g
fk+1

.

This may be thought of as a special case of localizing a Dn-module: if M is a
Dn-module and f ∈ Rn then M ⊗Rn Rn[f−1] becomes a Dn-module via the
product rule

xi • (m⊗ g

fk
) = m⊗ (

xig

fk
), ∂i • (m⊗ g

fk
) = m⊗ ∂i • (

g

fk
) + ∂im⊗

g

fk
.

Of particular interest are the holonomic modules which are those finitely
generated Dn-modules M for which ExtjDn(M,Dn) vanishes unless j = n.
This innocent looking definition has surprising consequences, some of which
we discuss now.

The holonomic modules form a full Abelian subcategory of the category
of left Dn-modules, closed under the formation of subquotients. Our standard
example of a holonomic module is

Rn = Dn/∆.

This equality may require some thought – it pictures Rn as a Dn-module
generated by 1 ∈ Rn. It is particularly noteworthy that not all elements of
Rn are killed by ∆ – quite impossible if Dn were commutative.

Holonomic modules are always cyclic and of finite length over Dn. These
fundamental properties are consequences of the Bernstein inequality. To un-
derstand this inequality we associate with the Dn-module M = Dn/L the
Hilbert function qL(k) with values in the integers which counts for each k ∈ N
the number of monomials xα∂β with |α|+ |β| ≤ k whose cosets in M are K-
linearly independent. The filtration k 7→ K · {xα∂β mod L : |α|+ |β| ≤ k} is
called the Bernstein filtration. The Bernstein inequality states that qL(k) is
either identically zero (in which case M = 0) or asymptotically a polynomial
in k of degree between n and 2n. This degree is called the dimension of M .
A holonomic module is one of dimension n, the minimal possible value for a
nonzero module.

This characterization of holonomicity can be used quite easily to check
with Macaulay 2 that Rn is holonomic. Namely, let’s say n = 3. Start a
Macaulay 2 session with

i1 : load "D-modules.m2"

i2 : D = QQ[x,y,z,Dx,Dy,Dz, WeylAlgebra => {x=>Dx, y=>Dy, z=>Dz}]

o2 = D

o2 : PolynomialRing

i3 : Delta = ideal(Dx,Dy,Dz)

o3 = ideal (Dx, Dy, Dz)

o3 : Ideal of D

The first of these commands loads the D-module library by A. Leykin, M.
Stillman and H. Tsai, [23]. The second line defines the base ring D3 =
Q〈x, y, z, ∂x, ∂y, ∂z〉, while the third command defines theD3-moduleD3/∆ ∼=
R3.

As one can see, Macaulay 2 thinks of D as a ring of polynomials. This is
using the vector space isomorphism Ψ from Subsection 2.2. Of course, two
elements are multiplied according to the Leibniz rule. To see how Macaulay 2
uses the map Ψ , we enter the following expression.

i4 : (Dx * x)^2

2 2
o4 = x Dx + 3x*Dx + 1

o4 : D

All Weyl algebra ideals and modules are by default left ideals and left modules
in Macaulay 2.

If we don’t explicitly specify a monomial ordering to be used in the Weyl
algebra, then Macaulay 2 uses graded reverse lex (GRevLex), as we can see
by examining the options of the ring.

i5 : options D

o5 = OptionTable{Adjust => identity }
Degrees => {{1}, {1}, {1}, {1}, {1}, {1}}
Inverses => false
MonomialOrder => GRevLex
MonomialSize => 8
NewMonomialOrder =>
Repair => identity
SkewCommutative => false
VariableBaseName =>
VariableOrder =>
Variables => {x, y, z, Dx, Dy, Dz}
Weights => {}
WeylAlgebra => {x => Dx, y => Dy, z => Dz}

o5 : OptionTable

To compute the initial ideal of ∆ with respect to the weight that associates
1 to each ∂ and to each variable, execute

i6 : DeltaBern = inw(Delta,{1,1,1,1,1,1})

o6 = ideal (Dz, Dy, Dx)

o6 : Ideal of QQ [x, y, z, Dx, Dy, Dz]

The command inw can be used with any weight vector for Dn as second
argument. One notes that the output is not an ideal in a Weyl algebra any
more, but in a ring of polynomials, as it should. The dimension of R3, which
is the dimension of the variety associated to DeltaBern, is computed by

i7 : dim DeltaBern

o7 = 3

As this is equal to n = 3, the ideal ∆ is holonomic.

Let Rn[f−1, s]⊗fs be the free Rn[f−1, s]-module generated by the symbol
fs. Using the action • of Dn on Rn[f−1, s] we define an action • of Dn[s] on

Rn[f−1, s]⊗ fs by setting

s •
(
g(x, s)
fk

⊗ fs
)

=
sg(x, s)
fk

⊗ fs,

xi •
(
g(x, s)
fk

⊗ fs
)

=
xig(x, s)
fk

⊗ fs,

∂i •
(
g(x, s)
fk

⊗ fs
)

=
(
∂i •

(
g(x, s)
fk

)
+ s∂i • (f) · g(x, s)

fk+1

)
⊗ fs.

The last rule justifies the choice for the symbol of the generator.
Writing M = Dn/L and denoting by 1 the coset of 1 ∈ Dn in M , this

action extends to an action of Dn[s] on

ML
f = Dn[s] • (1⊗ 1⊗ fs) ⊆M ⊗Rn

(
Rn[f−1, s]⊗ fs

)
(2.1)

by the product rule for all left Dn-modules M . The interesting bit about
ML

f is the following fact. If M = Dn/L is holonomic then there is a nonzero
polynomial b(s) in K[s] and an operator P (s) ∈ Dn[s] such that

P (s) • (1⊗ f ⊗ fs) = 1⊗ b(s)⊗ fs (2.2)

in ML
f . This entertaining equality, often written as

P (s)
(
1⊗ fs+1

)
= b(s)⊗ fs,

says that P (s) is roughly equal to division by f . The unique monic polynomial
that divides all other polynomials b(s) satisfying an identity of this type is
called the Bernstein (or also Bernstein-Sato) polynomial of L and f and
denoted by bLf (s). Any operator P (s) that satisfies (2.2) with b(s) = bf (s) we
shall call a Bernstein operator and refer to the roots of bLf (s) as Bernstein
roots of f on Dn/L. It is clear from (2.2) and the definitions that bLf (s) is the
minimal polynomial of s on the quotient of ML

f by Dn[s] • (1⊗ f ⊗ fs).
The Bernstein roots of the polynomial f are somewhat mysterious, but

related to other algebro-geometric invariants as, for example, the monodromy
of f (see [29]), the Igusa zeta function (see [24]), and the log-canonical thresh-
old (see [21]). For a long time it was also unclear how to compute bf (s) for
given f . In [53] many interesting examples of Bernstein-Sato polynomials are
worked out by hand, while in [1,6,28,41] algorithms were given that compute
bf (s) under certain conditions on f . The general algorithm we are going to
explain was given by T. Oaku. Here is a classical example.

Example 2.2. Let f =
∑n
i=1 xi

2 and M = Rn with L = ∆. One can check
that

n∑
i=1

∂i
2 • (1⊗ 1⊗ fs+1) = 1⊗ 4(s+ 1)(

n

2
+ s)⊗ fs

and hence that 1
4

∑n
i=1 ∂i

2 is a Bernstein operator while the Bernstein roots
of f are −1 and −n/2 and the Bernstein polynomial is (s+ 1)(s+ n

2 ).

Example 2.3. Although in the previous example the Bernstein operator
looked a lot like the polynomial f , this is not often the case and it is usually
hard to guess Bernstein operators. For example, one has(

1
27
∂y

3 +
y

6
∂x

2∂y +
x

8
∂x

3

)
(x2 + y3)s+1 = (s+

5
6

)(s+ 1)(s+
7
6

)(x2 + y3)s.

In the case of non-quasi-homogeneous polynomials, there is usually no resem-
blance between f and any Bernstein operator.

A very important property of holonomic modules is the (somewhat counter-
intuitive) fact that any localization of a holonomic module M = Dn/L at a
single element (and hence at any finite number of elements) of Rn is holo-
nomic ([5], 1.5.9) and in particular cyclic over Dn, generated by 1 ⊗ fa for
sufficiently small a ∈ Z. As a special case we note that localizations of Rn are
holonomic, and hence finitely generated over Dn. Coming back to the Čech
complex we see that the complex Č•(M ; f1, . . . , fr) consists of holonomic
Dn-modules whenever M is holonomic.

As a consequence, local cohomology modules of Rn are Dn-modules and
in fact holonomic. To see this it suffices to know that the maps in the Čech
complex are Dn-linear, which we will explain in Section 4. Since the cat-
egory of holonomic Dn-modules and their Dn-linear maps is closed under
subquotients, holonomicity of Hk

I (Rn) follows.
For similar reasons, Hi

m(Hj
I (Rn)) is holonomic for i, j ∈ N (since Hj

I (Rn)
is holonomic). These modules, investigated in Subsections 4.2 and 4.3, are
rather special Rn-modules and seem to carry some very interesting informa-
tion about Var(I), see [10,52].

The fact that Rn is holonomic and every localization of a holonomic mod-
ule is as well, provides motivation for us to study this class of modules. There
are, however, more occasions where holonomic modules show up. One such
situation arises in the study of linear partial differential equations. More
specifically, the so-called GKZ-systems (which we will meet again in the final
chapter) provide a very interesting class of objects with fascinating combina-
torial and analytic properties [40].

3 Bernstein-Sato Polynomials and Localization

We mentioned in the introduction that for the computation of local cohomol-
ogy the following is an important algorithmic problem to solve.

Problem 3.1. Given f ∈ Rn and a left ideal L ⊆ Dn such that M = Dn/L
is holonomic, compute the structure of the module Dn/L⊗Rn[f−1] in terms
of generators and relations.

This section is about solving Problem 3.1.

3.1 The Line of Attack

Recall for a given Dn-module M = Dn/L the action of Dn[s] on the tensor
product M⊗Rn (Rn[f−1, s]⊗fs) from Subsection 2.3. We begin with defining
an ideal of operators:

Definition 3.2. Let JL(fs) stand for the ideal in Dn[s] that kills 1⊗1⊗fs ∈
(Dn/L)⊗Rn Rn[f−1, s]⊗ fs.

It turns out that it is very useful to know this ideal. If L = ∆ then there
are some obvious candidates for generators of JL(fs). For example, there are
f∂i − ∂i • (f)s for all i. However, unless the affine hypersurface defined by
f = 0 is smooth, these will not generate J∆(fs). For a more general L, there
is a similar set of (somewhat less) obvious candidates, but again finding all
elements of J∆(fs) is far from elementary, even for smooth f .

In order to find JL(fs), we will consider the module (Dn/L)⊗Rn[f−1, s]⊗
fs over the ring Dn+1 = Dn〈t, ∂t〉 by defining an appropriate action of t and
∂t on it. It is then not hard to compute the ideal JLn+1(fs) ⊆ Dn+1 consisting
of all operators that kill 1 ⊗ 1 ⊗ fs, see Lemma 3.5. In Proposition 3.6 we
will then explain how to compute JL(fs) from JLn+1(fs).

This construction gives an answer to the question of determining a pre-
sentation of Dn • (1 ⊗ fa) for “most” a ∈ K, which we make precise as
follows.

Definition 3.3. We say that a property depending on a ∈ Km holds for a in
very general position, if there is a countable set of hypersurfaces in Km such
that the property holds for all a not on any of the exceptional hypersurfaces.

It will turn out that for a ∈ K in very general position JL(fs) “is” the
annihilator for fa: we shall very explicitly identify a countable number of
exceptional values in K such that if a is not equal to one of them, then
JL(fs) evaluates under s 7→ a to the annihilator inside Dn of 1⊗ fa.

For a ∈ Z we have of course Dn • (1 ⊗ fa) ⊆ M ⊗ Rn[f−1] but the
inclusion may be strict (e.g., for L = ∆ and a = 0). Proposition 3.11 shows
how (Dn/L)⊗Rn[f−1] and JL(fs) are related.

3.2 Undetermined Exponents

Consider Dn+1 = Dn〈t, ∂t〉, the Weyl algebra in x1, . . . , xn and the new
variable t. B. Malgrange [29] has defined an action • of Dn+1 on (Dn/L) ⊗
Rn[f−1, s]⊗fs as follows. We require that xi acts as multiplication on the first

factor, and for the other variables we set (with P ∈ Dn/L and g(x, s) ∈ Rn[s])

∂i • (P ⊗ g(x, s)
fk

⊗ fs) =
(
P ⊗

(
∂i • (

g(x, s)
fk

) +
s∂i • (f)g(x, s)

fk+1

)
+ ∂iP ⊗

g(x, s)
fk

)
⊗ fs,

t • (P ⊗ g(x, s)
fk

⊗ fs) = P ⊗ g(x, s+ 1)f
fk

⊗ fs,

∂t • (P ⊗ g(x, s)
fk

⊗ fs) = P ⊗ −sg(x, s− 1)
fk+1

⊗ fs.

One checks that this actually defines a left Dn+1-module structure (i.e., ∂tt
acts like t∂t + 1) and that −∂tt acts as multiplication by s.

Definition 3.4. We denote by JLn+1(fs) the ideal in Dn+1 that annihi-
lates the element 1 ⊗ 1 ⊗ fs in (Dn/L) ⊗ Rn[f−1, s] ⊗ fs with Dn+1 act-
ing as defined above. Then we have an induced morphism of Dn+1-modules
Dn+1/J

L
n+1(fs) → (Dn/L) ⊗ Rn[f−1, s] ⊗ fs sending P + JLn+1(fs) to P •

(1⊗ 1⊗ fs).

We say that an ideal L ⊆ Dn is f-saturated if f · P ∈ L implies P ∈ L
and we say that Dn/L is f-torsion free if L is f -saturated. Rn and all its
localizations are examples of f -torsion free modules for arbitrary f .

The following lemma is a modification of Lemma 4.1 in [29] where the
special case L = Dn · (∂1, . . . , ∂n), Dn/L = Rn is considered (compare also
[47]).

Lemma 3.5. Suppose that L = Dn · (P1, . . . , Pr) is f-saturated. With the
above definitions, JLn+1(fs) is the ideal generated by f − t together with the
images of the Pj under the automorphism φ of Dn+1 induced by xi 7→ xi for
all i, and t 7→ t− f .

Proof. The automorphism sends ∂i to ∂i+∂i•(f)∂t and ∂t to ∂t. So if we write
Pj as a polynomial Pj(∂1, . . . , ∂n) in the ∂i with coefficients in K[x1, . . . , xn],
then

φ(Pj) = Pj(∂1 + ∂1 • (f)∂t, . . . , ∂n + ∂n • (f)∂t).

One checks that (∂i + ∂i • (f)∂t) • (Q ⊗ 1 ⊗ fs) = ∂iQ ⊗ 1 ⊗ fs for all
Q ∈ Dn+1, so that φ(Pj(∂1, . . . , ∂n))•(1⊗1⊗fs) = Pj(∂1, . . . , ∂n)⊗1⊗fs = 0.
By definition, f • (1 ⊗ 1 ⊗ fs) = t • (1 ⊗ 1 ⊗ fs). So t − f ∈ JLn+1(fs) and
φ(Pj) ∈ JLn+1(fs) for j = 1, . . . , r.

Conversely let P • (1⊗1⊗fs) = 0. The proof that P ∈ φ(JLn+1 +Dn+1 · t)
relies on an elimination idea and has some Gröbner basis flavor. We have to
show that P ∈ Dn+1 · (φ(P1), . . . , φ(Pr), t− f). We may assume, that P does
not contain any power of t since we can eliminate t using f − t. Now rewrite
P in terms of ∂t and the ∂i + ∂i • (f)∂t. Say, P =

∑
α,β ∂

α
t x

βQα,β(∂1 + ∂1 •

(f)∂t, . . . , ∂n + ∂n • (f)∂t), where the Qα,β ∈ K[y1, . . . , yn] are polynomial
expressions. Then

P • (1⊗ 1⊗ fs) =
∑
α,β

∂αt • (xβQα,β(∂1, . . . , ∂n)⊗ 1⊗ fs).

Let α be the largest α ∈ N for which there is a nonzero Qα,β occurring in
P =

∑
α,β ∂

α
t x

βQα,β(∂1 + ∂1 • (f)∂t, . . . , ∂n + ∂n • (f)∂t). We show that the
sum of terms that contain ∂αt is in Dn+1 ·φ(L) as follows. In order for P •(1⊗
1⊗fs) to vanish, the sum of terms with the highest s-power, namely sα, must
vanish. Hence

∑
β x

βQα,β(∂1, . . . , ∂n)⊗ (−1/f)α ⊗ fs ∈ L⊗Rn[f−1, s]⊗ fs

as Rn[f−1, s] is Rn[s]-flat. It follows that
∑
β x

βQα,β(∂1, . . . , ∂n) ∈ L (L is
f -saturated!) and hence

∑
β ∂

α
t x

βQα,β(∂1 + ∂1 • (f)∂t, . . . , ∂n + ∂n • (f)∂t) ∈
Dn+1 · φ(L) as announced.

So by the first part, P −
∑
β ∂

α
t x

βQα,β(∂1 +∂1 • (f)∂t, . . . , ∂n+∂n • (f)∂t)
kills 1⊗ 1⊗ fs, but is of smaller degree in ∂t than P was.

The claim follows by induction on α. ut

If we identify Dn[−∂tt] with Dn[s] then JLn+1(fs)∩Dn[−∂tt] is identified
with JL(fs) since, as we observed earlier, −∂tt multiplies by s on ML

f . As
we pointed out in the beginning, the crux of our algorithms is to calculate
JL(fs) = JLn+1(fs) ∩Dn[s]. We shall deal with this computation now.

In Theorem 19 of [33], T. Oaku showed how to construct a generating
set for JL(fs) in the case L = Dn · (∂1, . . . , ∂n). Using his ideas we explain
how one may calculate J ∩Dn[−∂tt] whenever J ⊆ Dn+1 is any given ideal,
and as a corollary develop an algorithm that for f -saturated Dn/L computes
JL(fs) = JLn+1(fs) ∩Dn[−∂tt].

We first review some work of Oaku. On Dn+1 we define the weight vec-
tor w by w(t) = 1, w(∂t) = −1, w(xi) = w(∂i) = 0 and we extend it to
Dn+1[y1, y2] by w(y1) = −w(y2) = 1. If P =

∑
i Pi ∈ Dn+1[y1, y2] and all

Pi are monomials, then we will write (P )h for the operator
∑
i Pi · y

di
1 where

di = maxj(w(Pj))− w(Pi) and call it the y1-homogenization of P .
Note that the Buchberger algorithm preserves homogeneity in the follow-

ing sense: if a set of generators for an ideal is given and these generators
are homogeneous with respect to the weight above, then any new generator
for the ideal constructed with the classical Buchberger algorithm will also be
homogeneous. (This is a consequence of the facts that the yi commute with
all other variables and that ∂tt = t∂t + 1 is homogeneous of weight zero.)
This homogeneity is very important for the following result of Oaku:

Proposition 3.6. Let J = Dn+1 · (Q1, . . . , Qr). Let I be the left ideal in
Dn+1[y1] generated by the y1-homogenizations (Qi)h of the Qi, relative to
the weight w above, and set Ĩ = Dn+1[y1, y2] · (I, 1 − y1y2). Let G be a
Gröbner basis for Ĩ under a monomial order that eliminates y1, y2. For each

P ∈ G∩Dn+1 set P ′ = t−w(P )P if w(P ) < 0 and P ′ = ∂
w(P )
t P if w(P ) ≥ 0.

Set G0 = {P ′ : P ∈ G∩Dn+1}. Then G0 ⊆ Dn[−∂tt] generates J ∩Dn[−∂tt].

Proof. This is in essence Theorem 18 of [33]. (See the remarks in Subsection
2.2 on how to compute such Gröbner bases.) ut

As a corollary to this proposition we obtain an algorithm for the compu-
tation of J∆(fs):

Algorithm 3.7 (Parametric Annihilator).
Input: f ∈ Rn; L ⊆ Dn such that L is f -saturated.
Output: Generators for JL(fs).

1. For each generator Qi of Dn+1 · (L, t) compute the image φ(Qi) under
xi 7→ xi, t 7→ t− f , ∂i 7→ ∂i + ∂i • (f)∂t, ∂t 7→ ∂t.

2. Homogenize all φ(Qi) with respect to the new variable y1 relative to the
weight w introduced before Proposition 3.6.

3. Compute a Gröbner basis for the ideal

Dn+1[y1, y2] · ((φ(Q1))h, . . . , (φ(Qr))h, 1− y1y2)

in Dn+1[y1, y2] using an order that eliminates y1, y2.
4. Select the operators {Pj}b1 in this basis which do not contain y1, y2.
5. For each Pj , 1 ≤ j ≤ b, if w(Pj) > 0 replace Pj by P ′j = ∂

w(Pj)
t Pj .

Otherwise replace Pj by P ′j = t−w(Pj)Pj .
6. Return the new operators {P ′j}b1.

End.

The output will be operators in Dn[−∂tt] which is naturally identified with
Dn[s] (including the action on ML

f ). This algorithm is in effect Proposition
7.1 of [32].

In Macaulay 2, one can compute the parametric annihilator ideal (for
Rn = ∆) by the command AnnFs:

i8 : D = QQ[x,y,z,w,Dx,Dy,Dz,Dw,
WeylAlgebra => {x=>Dx, y=>Dy, z=>Dz, w=>Dw}];

i9 : f = x^2+y^2+z^2+w^2

2 2 2 2
o9 = x + y + z + w

o9 : D

i10 : AnnFs(f)

· · ·
o10 = ideal (w*Dz - z*Dw, w*Dy - y*Dw, z*Dy - y*Dz, w*Dx - x*Dw, z*Dx · · ·

· · ·
o10 : Ideal of QQ [x, y, z, w, Dx, Dy, Dz, Dw, $s, WeylAlgebra => {x = · · ·

If we want to compute JL(fs) for more general L, we have to use the com-
mand AnnIFs:

i11 : L=ideal(x,y,Dz,Dw)

o11 = ideal (x, y, Dz, Dw)

o11 : Ideal of D

i12 : AnnIFs(L,f)

1 1
o12 = ideal (y, x, w*Dz - z*Dw, -*z*Dz + -*w*Dw - $s)

2 2

o12 : Ideal of QQ [x, y, z, w, Dx, Dy, Dz, Dw, $s, WeylAlgebra => {x = · · ·
It should be emphasized that saturatedness of L with respect to f is a must
for AnnIFs.

3.3 The Bernstein-Sato Polynomial

Knowing JL(fs) allows us to get our hands on the Bernstein-Sato polynomial
of f on M :

Corollary 3.8. Suppose L is a holonomic ideal in Dn (i.e., Dn/L is holo-
nomic). The Bernstein polynomial bLf (s) of f on (Dn/L) satisfies

(bLf (s)) =
(
Dn[s] · (JL(fs), f)

)
∩K[s]. (3.1)

Moreover, if L is f-saturated then bLf (s) can be computed with Gröbner basis
computations.

Proof. By definition of bLf (s) we have (bLf (s) − PLf (s) · f) • (1 ⊗ 1 ⊗ fs) = 0
for a suitable PLf (s) ∈ Dn[s]. Hence bLf (s) is in K[s] and in Dn[s](JL(fs), f).
Conversely, if b(s) is in this intersection then b(s) satisfies an equality of the
type of (2.2) and hence is a multiple of bLf (s).

If we use an elimination order for which {xi, ∂i}n1 � s in Dn[s], then if
JL(fs) is known, bLf (s) will be (up to a scalar factor) the unique element in
the reduced Gröbner basis for Dn[s] · (JL(fs), f) that contains no xi nor ∂i.
Since we assume L to be f -saturated, JL(fs) can be computed according to
Proposition 3.6. ut

We therefore arrive at the following algorithm for the Bernstein-Sato poly-
nomial [31].

Algorithm 3.9 (Bernstein-Sato polynomial).
Input: f ∈ Rn; L ⊆ Dn such that Dn/L is holonomic and f -torsion free.
Output: The Bernstein polynomial bLf (s).

1. Determine JL(fs) following Algorithm 3.7.

2. Find a reduced Gröbner basis for the ideal JL(fs) + Dn[s] · f using an
elimination order for x and ∂.

3. Pick the unique element b(s) ∈ K[s] contained in that basis and return
it.

End.

We illustrate the algorithm with two examples. We first recall f which
was defined at the end of the previous subsection.

i13 : f

2 2 2 2
o13 = x + y + z + w

o13 : D

Now we compute the Bernstein-Sato polynomial.
i14 : globalBFunction(f)

2
o14 = $s + 3$s + 2

o14 : QQ [$s]

The routine globalBFunction computes the Bernstein-Sato polynomial of f
on Rn. We also take a look at the Bernstein-Sato polynomial of a cubic:

i15 : g=x^3+y^3+z^3+w^3

3 3 3 3
o15 = x + y + z + w

o15 : D

i16 : factorBFunction globalBFunction(g)

7 8 4 5
o16 = ($s + 1)($s + -)($s + -)($s + 2)($s + -)($s + -)

3 3 3 3

o16 : Product

In Macaulay 2 one can also find bLf (s) for more general L. We will see in the
following remark what the appropriate commands are.

Remark 3.10. It is clear that s+ 1 is always a factor of any Bernstein-Sato
polynomial on Rn, but this is not necessarily the case if L 6= ∆. For example,
bLf (s) = s for n = 1, f = x and L = x∂x + 1 (in which case D1/L ∼= R1[x−1],
generated by 1/x). In particular, it is not true that the roots of bLf (s) are
negative for general holonomic L.

If L is equal to ∆, and if f is nice, then the Bernstein roots are all between
−n and 0 [46]. But for general f very little is known besides a famous theorem
of Kashiwara that states that b∆f (s) factors over Q [19] and all roots are
negative.

For L arbitrary, the situation is more complicated. The Bernstein-Sato
polynomial of any polynomial f on the Dn-module generated by 1⊗ fa with

a ∈ K is related to that of f on Dn/L by a simple shift, and so the Bernstein
roots of f on the Dn-module generated by the function germ fa, a ∈ K, are
still all in K by [19]. Localizing other modules however can easily lead to
nonrational roots. As an example, consider

i17 : D1 = QQ[x,Dx,WeylAlgebra => {x=>Dx}];

i18 : I1 = ideal((x*Dx)^2+1)

2 2
o18 = ideal(x Dx + x*Dx + 1)

o18 : Ideal of D1

This is input defined over the rationals. Even localizing D1/I1 at a very
simple f leads to nonrational roots:

i19 : f1 = x;

i20 : b=globalB(I1, f1)

2
o20 = HashTable{Boperator => - x*Dx + 2Dx*$s + Dx}

2
Bpolynomial => $s + 2$s + 2

o20 : HashTable

The routine globalB is to be used if a Bernstein-Sato polynomial is suspected
to fail to factor over Q. If bLf (s) does factor over Q, one can also use the
routine DlocalizeAll to be discussed below. It would be very interesting to
determine rules that govern the splitting field of bLf (s) in general.

3.4 Specializing Exponents

In this subsection we investigate the result of substituting a ∈ K for s in
JL(fs). Recall that the Bernstein polynomial bLf (s) will exist (i.e., be nonzero)
if Dn/L is holonomic. As outlined in the previous subsection, bLf (s) can be
computed if Dn/L is holonomic and f -torsion free. The following proposition
(Proposition 7.3 in [32], see also Proposition 6.2 in [19]) shows that replacing s
by an exponent in very general position leads to a solution of the localization
problem.

Proposition 3.11. If L is holonomic and a ∈ K is such that no element of
{a−1, a−2, . . .} is a Bernstein root of f on L then we have Dn-isomorphisms

(Dn/L)⊗Rn
(
Rn[f−1]⊗ fa

) ∼= (Dn[s]/JL(fs)
)
|s=a ∼= Dn•(1⊗1⊗fa). (3.2)

ut

One notes in particular that if any a ∈ Z satisfies the conditions of the
proposition, then so does every integer smaller than a. This motivates the
following

Definition 3.12. The stable integral exponent of f on L is the smallest in-
tegral root of bLf (s), and denoted aLf .

In terms of this definition,(
Dn/J

L(fs)
)
|s=aLf

∼= (Dn/L)⊗Rn Rn[f−1],

and the presentation corresponds to the generator 1 ⊗ fa
L
f . If L = ∆ then

Kashiwara’s result tells us that bLf (s) will factor over the rationals, and thus it
is very easy to find the stable integral exponent. If we localize a more general
module, the roots may not even be K-rational anymore as we saw at the end
of the previous subsection.

The following lemma deals with the question of finding the smallest integer
root of a polynomial. We let |s| denote the complex absolute value.

Lemma 3.13. Suppose that in the situation of Corollary 3.8,

bLf (s) = sd + bd−1s
d−1 + · · ·+ b0,

and define B = maxi{|bi|1/(d−i)}. The smallest integer root of bLf (s) is an
integer between −2B and 2B. If in particular L = Dn ·(∂1, . . . , ∂n), it suffices
to check the integers between −bd−1 and −1.

Proof. Suppose |s0| = 2Bρ where B is as defined above and ρ > 1. Assume
also that s0 is a root of bLf (s). We find

(2Bρ)d = |s0|d = | −
d−1∑
i=0

bis0
i| ≤

d−1∑
i=0

Bd−i|s0|i

= Bd
d−1∑
i=0

(2ρ)i ≤ Bd((2ρ)d − 1),

using ρ ≥ 1. By contradiction, s0 is not a root.
The final claim is a consequence of Kashiwara’s work [19] where he proves

that if L = Dn · (∂1, . . . , ∂n) then all roots of bLf (s) are rational and negative,
and hence −bn−1 is a lower bound for each single root. ut

Combining Proposition 3.11 with Algorithms 3.7 and 3.9 we therefore
obtain

Algorithm 3.14 (Localization).
Input: f ∈ Rn; L ⊆ Dn such that Dn/L is holonomic and f -torsion free.
Output: Generators for an ideal J such that (Dn/L)⊗Rn Rn[f−1] ∼= Dn/J .

1. Determine JL(fs) following Algorithm 3.7.
2. Find the Bernstein polynomial bLf (s) using Algorithm 3.9.
3. Find the smallest integer root a of bLf (s).

4. Replace s by a in all generators for JL(fs) and return these generators.

End.

Algorithms 3.9 and 3.14 are Theorems 6.14 and Proposition 7.3 in [32].

Example 3.15. For f = x2+y2+z2+w2, we found a stable integral exponent
of −2 in the previous subsection. To compute the annihilator of f−2 using
Macaulay 2, we use the command Dlocalize which automatically uses the
stable integral exponent. We first change the current ring back to the ring D
which we used in the previous subsection:

i21 : use D

o21 = D

o21 : PolynomialRing

Here is the module to be localized.
i22 : R = (D^1/ideal(Dx,Dy,Dz,Dw))

o22 = cokernel | Dx Dy Dz Dw |

1
o22 : D-module, quotient of D

The localization then is obtained by running
i23 : ann2 = relations Dlocalize(R,f)

o23 = | wDz-zDw wDy-yDw zDy-yDz wDx-xDw zDx-xDz yDx-xDy xDx+yDy+zDz+wD · · ·
1 10

o23 : Matrix D <--- D

The output ann2 is a 1× 10 matrix whose entries generate annD4(f−2).

Remark 3.16. The computation of the annihilator of fa for values of a
such that a − k is a Bernstein root for some k ∈ N+ can be achieved by
an appropriate syzygy computation. For example, we saw above that the
Bernstein-Sato polynomial of f = x2 + y2 + z2 + w2 on R4 is (s+ 1)(s+ 2).
So evaluation of JL(fs) at −1 does not necessarily yield annD4(f−1), as will
be documented in the next remark. On the other hand, evaluation at −2
gives annD4(f−2). It is not hard to see that annD4(f−1) = {P ∈ Dn : Pf ∈
annDn(f−2)} because D4 • f−1 = D4f • f−2 ⊆ D4 • f−2. So we set:

i24 : F = matrix{{f}}

o24 = | x2+y2+z2+w2 |

1 1
o24 : Matrix D <--- D

To find annD4(f−1), we use the command modulo which computes relations:
modulo(M,N) computes for two matrices M,N the set of (vectors of) opera-
tors P such that P ·M ⊆ im(N).

i25 : ann1 = gb modulo(F,ann2)

o25 = {2} | wDz-zDw wDy-yDw zDy-yDz Dx^2+Dy^2+Dz^2+Dw^2 wDx-xDw zDx-xD · · ·
o25 : GroebnerBasis

The generator ∂2
2 +∂2

y+∂2
z+∂2

w is particularly interesting. To see the quotient
of D4 • f−2 by D4 • f−1 we execute

i26 : gb((ideal ann2) + (ideal F))

o26 = | w z y x |

o26 : GroebnerBasis

which shows that D4 • f−2 is an extension of D4/D4(x, y, z, w) by D4 • f−1.
This is not surprising, since (0, 0, 0, 0) is the only singularity of f and hence
the difference between D4•f−2 and D4•f−1 must be supported at the origin.

It is perhaps interesting to note that for a more complicated (but still
irreducible) polynomial f the quotient (Dn • fa)/(Dn • fa+1) can be a non-
simple nonzero Dn-module. For example, let f = x3 + y3 + z3 + w3 and
a = a∆f = −2. A computation similar to the quadric case above shows that
here (Dn • fa)/(Dn • fa+1) is a (x, y, z, w)-torsion module (supported at the
singular locus of f) isomorphic to (D4/D4 · (x, y, z, w))6. The socle elements
of the quotient are the degree 2 polynomials in x, y, z, w.

Example 3.17. Here we show how with Macaulay 2 one can get more in-
formation from the localization procedure.

i27 : D = QQ[x,y,z,Dx,Dy,Dz, WeylAlgebra => {x=>Dx, y=>Dy, z=>Dz}];

i28 : Delta = ideal(Dx,Dy,Dz);

o28 : Ideal of D

We now define a polynomial and compute the localization of R3 at the poly-
nomial.

i29 : f=x^3+y^3+z^3;

i30 : I1=DlocalizeAll(D^1/Delta,f,Strategy=>Oaku)

1 1 1 2 · · ·
o30 = HashTable{annFS => ideal (-*x*Dx + -*y*Dy + -*z*Dz - $s, z Dy - · · ·

3 3 3
2 5 4

Bfunction => ($s + 1) ($s + -)($s + -)($s + 2)
3 3

2 3 2 4 1 2 · · ·
Boperator => --*y*z*Dx Dy*Dz - --*y*z*Dy Dz + ---*z Dx · · ·

81 81 243 · · ·
GeneratorPower => -2
LocMap => | x6+2x3y3+y6+2x3z3+2y3z3+z6 |
LocModule => cokernel | 1/3xDx+1/3yDy+1/3zDz+2 z2Dy-y2 · · ·

o30 : HashTable

i31 : I2=DlocalizeAll(D^1/Delta,f)

o31 = HashTable{GeneratorPower => -2 · · ·
2 2 2 1

IntegrateBfunction => ($s) ($s + 1) ($s + -)($s + -)
3 3

LocMap => | x6+2x3y3+y6+2x3z3+2y3z3+z6 |
LocModule => cokernel | xDx+yDy+zDz+6 z2Dy-y2Dz z2Dx-x · · ·

o31 : HashTable

The last two commands both compute the localization of R3 at f but
follow different localization algorithms. The former uses our Algorithm 3.14
while the latter follows [37].

The output of the command DlocalizeAll is a hashtable, because it con-
tains a variety of data that pertain to the map Rn ↪→ Rn[f−1]. LocMap gives
the element that induces the map on the Dn-module level (by right multi-
plication). LocModule gives the localized module as cokernel of the displayed
matrix. Bfunction is the Bernstein-Sato polynomial and annFS the generic
annihilator JL(fs). Boperator displays a Bernstein operator and the stable
integral exponent is stored in GeneratorPower.

Algorithm 3.14 requires the ideal L to be f -saturated. This property is not
checked by Macaulay 2, so the user needs to make sure it holds. For example,
this is always the case if Dn/L is a localization of Rn. One can check the
saturation property in Macaulay 2, but it is a rather involved computation.
This difficulty can be circumvented by omitting the option Strategy=>Oaku,
in which case the localization algorithm of [37] is used. In terms of complexity,
using the Oaku strategy is much better behaved.

One can address the entries of a hashtable. For example, executing
i32 : I1.LocModule

o32 = cokernel | 1/3xDx+1/3yDy+1/3zDz+2 z2Dy-y2Dz z2Dx-x2Dz y2Dx-x2Dy |

1
o32 : D-module, quotient of D

one can see that R3[f−1] is isomorphic to the cokernel of the LocModule entry
which (for either localization method) is

D3/ D3 · ( x∂x + y∂y + z∂z + 6, z2∂y − y2∂z, x
3∂y + y3∂y + y2z∂z + 6y2,

z2∂x − x2∂z, y
2∂x − x2∂y, x

3∂z + y3∂z + z3∂z + 6z2).

The first line of the hashtable I1 shows that R3[f−1] is generated by f−2

over D3, while I1.LocMap shows that the natural inclusion D3/∆ = R3 ↪→
R3[f−1] = D3/J

∆(fs)|s=a∆f is given by right multiplication by f2, shown as
the third entry of the hashtable I1. It is perhaps useful to point out that the
fourth entry of hashtable I2 is a relative of the Bernstein-Sato polynomial
of f , and is used for the computation of the so-called restriction functor
(compare with [35,48]).

Remark 3.18. Plugging in bad values a for s (such that a−k is a Bernstein
root for some k ∈ N+) can have unexpected results. Consider the case n = 1,

f = x. Then J∆(fs) = D1 · (s + 1 − ∂1x1). Hence b∆f (s) = s + 1 and −1 is
the unique Bernstein root. According to Proposition 3.11,(

D1[s]/J∆(fs)
)
|s=a ∼= R1[x1

−1]⊗ x1
a ∼= D1 • x1

a

for all a ∈ K \N. For a ∈ N+, we also have D1[s]/J∆(fs)|s=a ∼= D1 • xa, but
this is of course not R1[x1

−1] but just R1.
For a = 0 however,

(
D1[s]/J∆(fs)

)
|s=a has x1-torsion! It equals in fact

what is called the Fourier transform of R1[x1
−1] and fits into an exact se-

quence
0→ H1

x1
(R1)→ F(R1[x1

−1])→ R1 → 0.

Remark 3.19. If Dn/L is holonomic but has f -torsion, then (Dn/L) ⊗
Rn[f−1] and ((Dn/L)/H0

(f)(Dn/L)) ⊗ Rn[f−1] are of course isomorphic. So
if we knew how to find M/H0

f (M) for holonomic modules M , our localiza-
tion algorithm could be generalized to all holonomic modules. There are two
different approaches to the problem of f -torsion, presented in [35] and in
[43,44]. The former is based on homological methods and restriction to the
diagonal while the latter aims at direct computation of those P ∈ Dn for
which fkP ∈ L for some k.

There is also another direct method for localizing M = Dn/L at f that
works in the situation where the nonholonomic locus of M is contained in
the variety of f (irrespective of torsion). It was proved by Kashiwara, that
M [f−1] is then holonomic, and in [37] an algorithm based on integration is
given that computes a presentation for it.

4 Local Cohomology Computations

The purpose of this section is to present algorithms that compute for given
i, j, k ∈ N, I ⊆ Rn the structure of the local cohomology modules Hk

I (Rn) and
Hi

m(Hj
I (Rn)), and the invariants λi,j(Rn/I) associated to I. In particular, the

algorithms detect the vanishing of local cohomology modules.

4.1 Local Cohomology

We will first describe an algorithm that takes a finite set of polynomials
{f1, . . . , fr} ⊂ Rn and returns a presentation of Hk

I (Rn) where I = Rn ·
(f1, . . . , fr). In particular, if Hk

I (Rn) is zero, then the algorithm will return
the zero presentation.

Definition 4.1. Let Θrk be the set of k-element subsets of {1, . . . , r} and for
θ ∈ Θrk write Fθ for the product

∏
i∈θ fi.

Consider the Čech complex Č• = Č•(f1, . . . , fr) associated to f1, . . . , fr in
Rn,

0→ Rn →
⊕
θ∈Θr1

Rn[Fθ−1]→
⊕
θ∈Θr2

Rn[Fθ−1]→ · · · → Rn[(f1 · · · fr)−1]→ 0.

(4.1)
Its k-th cohomology group is Hk

I (Rn). The map

Mk :

Čk =
⊕
θ∈Θrk

Rn[Fθ−1]

→
 ⊕
θ′∈Θrk+1

Rn[Fθ′−1] = Čk+1

 (4.2)

is the sum of maps

Rn[(fi1 · · · fik)−1]→ Rn[(fj1 · · · fjk+1)−1] (4.3)

which are zero if {i1, . . . , ik} 6⊆ {j1, . . . , jk+1}, or send 1
1 to 1

1 (up to sign).
With Dn/∆ ∼= Rn, identify Rn[(fi1 · · · fik)−1] with Dn/J

∆((fi1 · · · fik)s)|s=a
and Rn[(fj1 · · · fjk+1)−1] with Dn/J

∆((fj1 · · · fjk+1)s)|s=a′ where a, a′ are suf-
ficiently small integers. By Proposition 3.11 we may assume that a = a′ ≤ 0.
Then the map (4.3) is in the nonzero case multiplication from the right by
(fl)−a where l = {j1, . . . , jk+1}\{i1, . . . , ik}, again up to sign. For example,
consider the inclusion

D2/D2 · (∂xx, ∂y) = R2[x−1] ↪→ R2[(xy)−1] = D2/D2 · (∂xx, ∂yy).

Since 1
x = y

xy , the inclusion on the level of D2-modules maps P + ann(x−1)
to Py + ann((xy)−1).

It follows that the matrix representing the map Čk → Čk+1 in terms
of Dn-modules is very easy to write down once the annihilator ideals and
Bernstein polynomials for all k- and (k+1)-fold products of the fi are known:
the entries are 0 or ±f−al where fl is the new factor. These considerations
give the following

Algorithm 4.2 (Local cohomology).
Input: f1, . . . , fr ∈ Rn; k ∈ N.
Output: Hk

I (Rn) in terms of generators and relations as finitely generated
Dn-module where I = Rn · (f1, . . . , fr).

1. Compute the annihilator ideal J∆((Fθ)s) and the Bernstein polynomial
b∆Fθ (s) for all (k−1)-, k- and (k+1)-fold products Fθ of f1, . . . , fr following
Algorithms 3.7 and 3.9 (so θ runs through Θrk−1 ∪Θrk ∪Θrk+1).

2. Compute the stable integral exponents a∆Fθ , let a be their minimum and
replace s by a in all the annihilator ideals.

3. Compute the two matrices Mk−1,Mk representing the Dn-linear maps
Čk−1 → Čk and Čk → Čk+1 as explained above.

4. Compute a Gröbner basis G for the kernel of the composition⊕
θ∈Θrk

Dn →→
⊕
θ∈Θrk

Dn/J
∆(Fθs)|s=a

Mk−→
⊕

θ′∈Θrk+1

Dn/J
∆(Fθ′s)|s=a.

5. Compute a Gröbner basisG0 for the preimage in
⊕

θ∈Θrk
Dn of the module

im(Mk−1) ⊆
⊕
θ∈Θrk

Dn/J
∆((Fθ)s)|s=a ←←

⊕
θ∈Θrk

Dn

under the indicated projection.
6. Compute the remainders of all elements of G with respect to G0.
7. Return these remainders and G0.

End.

The nonzero elements of G generate the quotient G/G0
∼= Hk

I (Rn) so that in
particular Hk

I (Rn) = 0 if and only if all returned remainders are zero.

Example 4.3. Let I be the ideal in R6 = K[x, y, z, u, v, w] that is generated

by the 2 × 2 minors f, g, h of the matrix
(
x y z
u v w

)
. Then Hi

I(R6) = 0 for

i < 2 and H2
I (R6) 6= 0 because I is a height 2 prime, and Hi

I(R6) = 0 for
i > 3 because I is 3-generated, so the only open case is H3

I (R6). This module
in fact does not vanish, and our algorithm provides a proof of this fact by
direct calculation. The Macaulay 2 commands are as follows.

i33 : D= QQ[x,y,z,u,v,w,Dx,Dy,Dz,Du,Dv,Dw, WeylAlgebra =>
{x=>Dx, y=>Dy, z=>Dz, u=>Du, v=>Dv, w=>Dw}];

i34 : Delta=ideal(Dx,Dy,Dz,Du,Dv,Dw);

o34 : Ideal of D

i35 : R=D^1/Delta;

i36 : f=x*v-u*y;

i37 : g=x*w-u*z;

i38 : h=y*w-v*z;

These commands define the relevant rings and polynomials. The following
three compute the localization of R6 at f :

i39 : Rf=DlocalizeAll(R,f,Strategy => Oaku)

o39 = HashTable{annFS => ideal (Dw, Dz, x*Du + y*Dv, y*Dy - u*Du, x*Dy · · ·
Bfunction => ($s + 1)($s + 2)
Boperator => - Dy*Du + Dx*Dv
GeneratorPower => -2
LocMap => | y2u2-2xyuv+x2v2 |
LocModule => cokernel | Dw Dz xDu+yDv yDy-uDu xDy+uDv · · ·

o39 : HashTable

of R6[f−1] at g:

i40 : Rfg=DlocalizeAll(Rf.LocModule,g, Strategy => Oaku)

· · ·
o40 = HashTable{annFS => ideal (Dz*Dv - Dy*Dw, x*Du + y*Dv + z*Dw, z*D · · ·

Bfunction => ($s + 1)($s)
Boperator => - Dz*Du + Dx*Dw
GeneratorPower => -1
LocMap => | -zu+xw |
LocModule => cokernel | DzDv-DyDw xDu+yDv+zDw zDz-uDu- · · ·

o40 : HashTable

and of R6[(fg)−1] at h:

i41 : Rfgh=DlocalizeAll(Rfg.LocModule,h, Strategy => Oaku)

· · ·
o41 = HashTable{annFS => ideal (x*Du + y*Dv + z*Dw, z*Dz - u*Du - v*Dv · · ·

Bfunction => ($s - 1)($s + 1)
Boperator => - Dz*Dv + Dy*Dw
GeneratorPower => -1
LocMap => | -zv+yw |
LocModule => cokernel | xDu+yDv+zDw zDz-uDu-vDv-2 yDy- · · ·

o41 : HashTable

From the output of these commands one sees that R6[(fgh)−1] is generated
by 1/f2gh. This follows from considering the stable integral exponents of the
three localization procedures, encoded in the hashtable entry stored under
the key GeneratorPower: for example,

i42 : Rf.GeneratorPower

o42 = -2

shows that the generator for R6[f−1] is f−2. Now we compute the annihilator
of H3

I (R6). From the Čech complex it follows that H3
I (R6) is the quotient

of the output of Rfgh.LocModule (isomorphic to R6[(fgh)−1]) by the sub-
modules generated by f2, g and h. (These submodules represent R6[(gh)−1],
R6[(fh)−1] and R6[(fg)−1] respectively.)

i43 : Jfgh=ideal relations Rfgh.LocModule;

o43 : Ideal of D

i44 : JH3=Jfgh+ideal(f^2,g,h);

o44 : Ideal of D

i45 : JH3gb=gb JH3

o45 = | w z uDu+vDv+wDw+4 xDu+yDv+zDw yDy-uDu-wDw-1 xDy+uDv uDx+vDy+wD · · ·

o45 : GroebnerBasis

So JH3 is the ideal of D3 generated by

w, z, u∂u + v∂v + w∂w + 4, x∂u + y∂v + z∂w, y∂y − u∂u − w∂w − 1,
x∂y + u∂v, u∂x + v∂y + w∂z, y∂x + v∂u, x∂x − v∂v − w∂w − 1,

v2, uv, yv, u2, yu+ xv, xu, y2, xy,

x2, xv∂v + 2x, v∂y∂u + w∂z∂u − v∂x∂v − w∂x∂w − 3∂x

which form a Gröbner basis. This proves that H3
I (R) 6= 0, because 1 is not

in the Gröbner basis of JH3. (There are also algebraic and topological proofs
to this account. Due to Hochster, and Bruns and Schwänzl, they are quite
ingenious and work only in rather special situations.)

From our output one can see that H3
I (R6) is (x, y, z, u, v, w)-torsion as

JH3 contains (x, y, z, u, v, w)2. The following sequence of commands defines a
procedure testmTorsion which as the name suggests tests a module Dn/L
for being m-torsion. We first replace the generators of L with a Gröbner basis.
Then we pick the elements of the Gröbner basis not using any ∂i. If now the
left over polynomials define an ideal of dimension 0 in Rn, the ideal was
m-torsion and otherwise not.

i46 : testmTorsion = method();

i47 : testmTorsion Ideal := (L) -> (
LL = ideal generators gb L;
n = numgens (ring (LL)) // 2;
LLL = ideal select(first entries gens LL, f->(

l = apply(listForm f, t->drop(t#0,n));
all(l, t->t==toList(n:0))
));

if dim inw(LLL,toList(apply(1..2*n,t -> 1))) == n
then true
else false);

If we apply testmTorsion to JH3 we obtain
i48 : testmTorsion(JH3)

o48 = true

Further inspection shows that the ideal JH3 is in fact the annihilator of the
fraction f/(wzx2y2u2v2) in R6[(xyzuvw)−1]/R6

∼= D6/D6 · (x, y, z, u, v, w),
and that the fraction generates D6/D6·(x1, . . . , x6). Since D6/D6·(x1, . . . , x6)
is isomorphic to ER6(R6/R6 · (x1, . . . , x6)), the injective hull of R6/R6 ·
(x1, . . . , x6) = K in the category of R6-modules, we conclude that H3

I (R6) ∼=
ER6(K). (In the next subsection we will display a way to use Macaulay 2 to
find the length of an m-torsion module.)

In contrast, let I be defined as generated by the three minors, but this time
over a field of finite characteristic. Then H3

I (R6) is zero because Peskine and
Szpiro proved using the Frobenius functor [39] that R6/I Cohen-Macaulay
implies that Hk

I (R6) is nonzero only if k = codim(I).
Also opposite to the above example, but in any characteristic, is the fol-

lowing calculation. Let I be the ideal in K[x, y, z, w] describing the twisted
cubic: I = R4 · (f, g, h) with f = xz − y2, g = yw − z2, h = xw − yz. The

projective variety V2 defined by I is isomorphic to the projective line. It is of
interest to determine whether V2 and other Veronese embeddings of the pro-
jective line are complete intersections. The set-theoretic complete intersection
property can occasionally be ruled out with local cohomology techniques: if
V is of codimension c in the affine variety X and Hc+k

I(V )(O(X)) 6= 0 for any
positive k then V cannot be a set-theoretic complete intersection. In the case
of the twisted cubic, it turns out hat H3

I (R4) = 0 as can be seen from the
following computation:

i49 : D=QQ[x,y,z,w,Dx,Dy,Dz,Dw,WeylAlgebra => {x=>Dx, y=>Dy, z=>Dz,
w=>Dw}];

i50 : f=y^2-x*z;

i51 : g=z^2-y*w;

i52 : h=x*w-y*z;

i53 : Delta=ideal(Dx,Dy,Dz,Dw);

o53 : Ideal of D

i54 : R=D^1/Delta;

i55 : Rf=DlocalizeAll(R,f,Strategy => Oaku)

1 · · ·
o55 = HashTable{annFS => ideal (Dw, x*Dy + 2y*Dz, y*Dx + -*z*Dy, x*Dx · · ·

2 · · ·
3

Bfunction => ($s + -)($s + 1)
2

1 2
Boperator => -*Dy - Dx*Dz

4
GeneratorPower => -1
LocMap => | y2-xz |
LocModule => cokernel | Dw xDy+2yDz yDx+1/2zDy xDx-zDz · · ·

o55 : HashTable

i56 : Rfg=DlocalizeAll(Rf.LocModule,g, Strategy => Oaku);

i57 : Rfgh=DlocalizeAll(Rfg.LocModule,h, Strategy => Oaku);

i58 : Ifgh=ideal relations Rfgh.LocModule;

o58 : Ideal of D

i59 : IH3=Ifgh+ideal(f,g,h);

o59 : Ideal of D

i60 : IH3gb=gb IH3

o60 = | 1 |

o60 : GroebnerBasis

It follows that we cannot conclude from local cohomological considerations
that V2 is not a set-theoretic complete intersection. This is not an acci-
dent but typical, as the second vanishing theorem of Hartshorne, Speiser,

Huneke and Lyubeznik shows [14,15,18]: if a homogeneous ideal I ⊆ Rn de-
scribes an geometrically connected projective variety of positive dimension
then Hn−1

I (Rn) = Hn
I (Rn) = 0.

4.2 Iterated Local Cohomology

Recall that m = Rn · (x1, . . . , xn). As a second application of Gröbner basis
computations over the Weyl algebra we show now how to compute the m-
torsion modules Hi

m(Hj
I (Rn)). Note that we cannot apply Lemma 3.5 to

Dn/L = Hj
I (Rn) since Hj

I (Rn) may well contain some torsion.
Čj(Rn; f1, . . . , fr) denotes the j-th module in the Čech complex to Rn

and {f1, . . . , fr}. Let Č•,• be the double complex

Či,j = Či(Rn;x1, . . . , xn)⊗Rn Čj(Rn; f1, . . . , fr),

with vertical maps φ•,• induced by the identity on the first factor and the
usual Čech maps on the second, and horizontal maps ξ•,• induced by the
Čech maps on the first factor and the identity on the second. Now Či,j is a
direct sum of modules Rn[g−1] where g = xα1 · · ·xαi ·fβ1 · · · fβj . So the whole
double complex can be rewritten in terms of Dn-modules and Dn-linear maps
using Algorithm 3.14:

Či−1,j+1
ξi−1,j+1

// Či,j+1
ξi,j+1

// Či+1,j+1

Či−1,j
ξi−1,j

//

φi−1,j

OO

Či,j
ξi,j //

φi,j

OO

Či+1,j

φi+1,j

OO

Či−1,j−1
ξi−1,j−1

//

φi−1,j−1

OO

Či,j−1
ξi,j−1

//

φi,j−1

OO

Či+1,j−1

φi+1,j−1

OO

Since Či(Rn;x1, . . . , xn) is Rn-flat, the column cohomology of Č•,• at (i, j)
is Či(Rn;x1, . . . , xn) ⊗Rn H

j
I (Rn) and the induced horizontal maps in the

j-th row are simply the maps in the Čech complex Č•(Hj
I (Rn);x1, . . . , xn).

It follows that the row cohomology of the column cohomology at (i0, j0) is
Hi0

m (Hj0
I (Rn)), the object of our interest.

We have, denoting by Xθ′ in analogy to Fθ the product
∏
i∈θ′ xi, the

following

Algorithm 4.4 (Iterated local cohomology).
Input: f1, . . . , fr ∈ Rn; i0, j0 ∈ N.
Output: Hi0

m (Hj0
I (Rn)) in terms of generators and relations as finitely gen-

erated Dn-module where I = Rn · (f1, . . . , fr).

1. For i = i0 − 1, i0, i0 + 1 and j = j0 − 1, j0, j0 + 1 compute the annihila-
tors J∆((Fθ ·Xθ′)s), Bernstein polynomials b∆Fθ·Xθ′ (s), and stable integral
exponents a∆Fθ·Xθ′ of Fθ ·Xθ′ where θ ∈ Θrj , θ′ ∈ Θni .



D-modules and Cohomology of Varieties 311

2. Let a be the minimum of all a∆Fθ·Xθ′ and replace s by a in all the annihi-
lators computed in the previous step.

3. Compute the matrices to the Dn-linear maps φi,j : Či,j → Či,j+1 and
ξk,l : Čk,l → Čk+1,l, for (i, j) ∈ {(i0, j0), (i0 + 1, j0 − 1), (i0, j0 − 1), (i0 −
1, j0)} and (k, l) ∈ {(i0, j0), (i0 − 1, j0)}.

4. Compute a Gröbner basis G for the module

Dn ·G = ker(φi0,j0) ∩
[
(ξi0,j0)−1(im(φi0+1,j0−1))

]
+ im(φi0,j0−1)

and a Gröbner basis G0 for the module

Dn ·G0 = ξi0−1,j0(ker(φi0−1,j0)) + im(φi0,j0−1).

5. Compute the remainders of all elements of G with respect to G0.
6. Return these remainders together with G0.

End.

Note that (Dn ·G)/(Dn ·G0) is isomorphic to

ker
(

ker(φi0,j0)
im(φi0,j0−1)

ξi0,j0−→ ker(φi0+1,j0)
im(φi0+1,j0−1)

)
ξi0−1,j0

(
ker(φi0−1,j0)

im(φi0−1,j0−1)

) ∼= Hi0
m (Hj0

I (Rn)).

The elements of G will be generators for Hi0
m (Hj0

I (Rn)) and the elements of
G0 generate the extra relations that are not syzygies.

The algorithm can of course be modified to compute any iterated local co-
homology groupHj

J(Hi
I(Rn)) for J ⊇ I by replacing the generators x1, . . . , xn

for m by those for J . Moreover, the iteration depth can also be increased by
considering “tricomplexes” etc. instead of bicomplexes.

Again we would like to point out that with the methods of [35] or [37]
one could actually compute first Hi

I(Rn) and from that Hj
J(Hi

I(Rn)), but
probably that is quite a bit more complex a computation.

4.3 Computation of Lyubeznik Numbers

G. Lyubeznik proved in [25] that if K is a field, R = K[x1, . . . , xn], I ⊆ R,
m = R · (x1, . . . , xn) and A = R/I then λi,j(A) = dimK socRHi

m(Hn−j
I (R))

is invariant under change of presentation of A. In other words, it only de-
pends on A and i, j but not the projection R →→ A. Lyubeznik proved that
Hi

m(Hj
I (Rn)) is in fact an injective m-torsion Rn-module of finite socle di-

mension λi,n−j(A) and so isomorphic to (ERn(K))λi,n−j(A) where ERn(K) is
the injective hull of K over Rn. We are now in a position to compute these
invariants of Rn/I in characteristic zero..

Algorithm 4.5 (Lyubeznik numbers).
Input: f1, . . . , fr ∈ Rn; i, j ∈ N.
Output: λi,n−j(Rn/Rn · (f1, . . . , fr)).

1. Using Algorithm 4.4 find g1, . . . , gl ∈ Dn
d and h1, . . . , he ∈ Dn

d such
that Hi

m(Hj
I (Rn)) is isomorphic to Dn · (g1, . . . , gl) modulo H = Dn ·

(h1, . . . , he).
2. Assume that after a suitable renumeration g1 is not in H. If such a g1

cannot be chosen, quit.
3. Find a monomial m ∈ Rn such that m · g1 6∈ H but ximg1 ∈ H for all xi.
4. Replace H by Dnmg1 +H and reenter at Step 2.
5. Return λi,n−j(Rn/I), the number of times Step 3 was executed.

End.

The reason that this works is as follows. We know that (Dn · g1 + H)/H is
m-torsion (as Hi

m(Hj
I (Rn)) is) and so it is possible (with trial and error, or a

suitable syzygy computation) to find the monomial m in Step 3. The element
mg1 mod H ∈ Dn/H has annihilator equal to m over Rn and therefore
generates a Dn-module isomorphic to Dn/Dn ·m ∼= ERn(K). The injection

(Dn ·mg1 +H)/H ↪→ (Dn · (g1, . . . , gl) +H)/H

splits as map of Rn-modules because ERn(K) is injective and so the cokernel
Dn · (g1, . . . , gl)/Dn · (mg1, h1, . . . , he) is isomorphic to (ERn(K))λi,n−j(A)−1.

Reduction of the gi with respect to a Gröbner basis of the new relation
module and repetition will lead to the determination of λi,n−j(A).

Assume that Dn/L is an m-torsion module. For example, we could have
Dn/L ∼= Hi

m(Hj
I (Rn)). Here is a procedure that finds by trial and error the

monomial socle element m of Step 3 in Algorithm 4.4.
i61 : findSocle = method();

i62 : findSocle(Ideal, RingElement):= (L,P) -> (
createDpairs(ring(L));
v=(ring L).dpairVars#0;
myflag = true;
while myflag do (

w = apply(v,temp -> temp*P % L);
if all(w,temp -> temp == 0) then myflag = false
else (

p = position(w, temp -> temp != 0);
P = v#p * P;)

);
P);

For example, if we want to apply this socle search to the ideal JH3 describing
H3
I (R6) of Example 4.3 we do
i63 : D = ring JH3

o63 = D

o63 : PolynomialRing

(as D was most recently the differential operators on Q[x, y, z, w])
i64 : findSocle(JH3,1_D)

o64 = x*v

o64 : D

One can then repeat the socle search and kill the newly found element as
suggested in the explanation above:

i65 : findLength = method();

i66 : findLength Ideal := (I) -> (
l = 0;
while I != ideal 1_(ring I) do (

l = l + 1;
s = findSocle(I,1_(ring I));
I = I + ideal s;);

l);

Applied to JH3 of the previous subsection this yields
i67 : findLength JH3

o67 = 1

and hence JH3 does indeed describe a module isomorphic to ER6(K).

5 Implementation, Examples, Questions

5.1 Implementations and Optimizing

The Algorithms 3.7, 3.9 and 3.14 have first been implemented by T. Oaku
and N. Takayama using the package Kan [42] which is a postscript language
for computations in the Weyl algebra and in polynomial rings. In Macaulay 2
Algorithms 3.7, 3.9 and 3.14 as well as Algorithm 4.2 have been implemented
by A. Leykin, M. Stillman and H. Tsai. They additionally implemented a
wealth of D-module routines that relate to topics which we cannot all cover
in this chapter. These include homomorphisms between holonomic modules
and extension functors, restriction functors to linear subspaces, integration
(de Rham) functors to quotient spaces and others. For further theoretical
information the reader is referred to [35,34,36,40,45,48,50,51].

Computation of Gröbner bases in many variables is in general a time and
space consuming enterprise. In commutative polynomial rings the worst case
performance for the number of elements in reduced Gröbner bases is doubly
exponential in the number of variables and the degrees of the generators. In
the (relatively) small Example 4.3 above R6 is of dimension 6, so that the
intermediate ring Dn+1[y1, y2] contains 16 variables. In view of these facts
the following idea has proved useful.

The general context in which Lemma 3.5 and Proposition 3.11 were stated
allows successive localization of Rn[(fg)−1] in the following way. First one
computes Rn[f−1] according to Algorithm 3.14 as quotient Dn/J

∆(fs)|s=a,
Z 3 a � 0. Then Rn[(fg)−1] may be computed using Algorithm 3.14 again

since Rn[(fg)−1] ∼= Rn[g−1]⊗Rn Dn/J
∆(fs)|s=a. (Note that all localizations

of Rn are automatically f -torsion free for f ∈ Rn so that Algorithm 3.14 can
be used.) This process may be iterated for products with any finite number
of factors. Of course the exponents for the various factors might be different.
This requires some care when setting up the Čech complex. In particular one
needs to make sure that the maps Čk → Čk+1 can be made explicit using the
fi. (In our Example 4.3, this is precisely how we proceeded when we found
Rfgh.)

Remark 5.1. One might hope that for all holonomic fg-torsion free modules
M = Dn/L we have (with M ⊗Rn[g−1] ∼= Dn/L

′):

aLf = min{s ∈ Z : bLf (s) = 0} ≤ min{s ∈ Z : bL
′

f (s) = 0} = aL
′

f . (5.1)

This hope is unfounded. Let R5 = K[x1, . . . , x5], f = x2
1 + x2

2 + x2
3 + x2

4 + x2
5.

One may check that then b∆f (s) = (s+1)(s+5/2). Hence R5[f−1] = D5•f−1,
let L = ker(D5 → D5 • f−1). Set g = x1. Then b∆g (s) = s + 1, let L′ =
ker(D5 → D5 • g−1).

Then bL
′

f (s) = (s+ 1)(s+ 2)(s+ 5/2) and bLg (s) = (s+ 1)(s+ 3) because
of the following computations.

i68 : erase symbol x; erase symbol Dx;

These two commands essentially clear the history of the variables x and Dx
and make them available for future computations.

i70 : D = QQ[x_1..x_5, Dx_1..Dx_5, WeylAlgebra =>
apply(toList(1..5), i -> x_i => Dx_i)];

i71 : f = x_1^2 + x_2^2 + x_3^2 + x_4^2 +x_5^2;

i72 : g = x_1;

i73 : R = D^1/ideal(Dx_1,Dx_2,Dx_3,Dx_4,Dx_5);

As usual, these commands defined the base ring, two polynomials and the
D5-module R5. Now we compute the respective localizations.

i74 : Rf =DlocalizeAll(R,f,Strategy => Oaku);

i75 : Bf = Rf.Bfunction

5
o75 = ($s + -)($s + 1)

2

o75 : Product

i76 : Rfg = DlocalizeAll(Rf.LocModule,g,Strategy => Oaku);

i77 : Bfg = Rfg.Bfunction

o77 = ($s + 1)($s + 3)

o77 : Product

i78 : Rg = DlocalizeAll(R,g,Strategy => Oaku);

i79 : Bg = Rg.Bfunction

o79 = ($s + 1)

o79 : Product

i80 : Rgf = DlocalizeAll(Rg.LocModule,f,Strategy => Oaku);

i81 : Bgf = Rgf.Bfunction

5
o81 = ($s + 2)($s + 1)($s + -)

2

o81 : Product

The output shows that Rn[(fg)−1] is generated by f−2g−1 or f−1g−3

but not by f−1g−2 and in particular not by f−1g−1. This can be seen from
the various Bernstein-Sato polynomials: as for example the smallest integral
root of Bf is −1 and that of Bfg is −3, R3[f−1] is generated by f−1 and
R3[(fg)−1] by f−1g−3. This example not only disproves the above inequality
(5.1) but also shows the inequality to be wrong if Z is replaced by R (as
−3 < min(−5/2,−1)).

Nonetheless, localizing Rn[(fg)−1] as (Rn[f−1])[g−1] is heuristically ad-
vantageous, apparently for two reasons. For one, it allows the exponents of
the various factors to be distinct which is useful for the subsequent coho-
mology computation: it helps to keep the degrees of the maps small. So in
Example 4.3 we can write R6[(fg)−1] as D6•(f−1g−2) instead of D6•(fg)−2.
Secondly, since the computation of Gröbner bases is potentially doubly ex-
ponential it seems to be advantageous to break a big problem (localization
at a product) into several “easy” problems (successive localization).

An interesting case of this behavior is our Example 4.3. If we compute
Rn[(fgh)−1] as ((Rn[f−1])[g−1])[h−1], the calculation uses approximately
6MB and lasts a few seconds using Macaulay 2. If one tries to localize Rn at
the product of the three generators at once, Macaulay 2 runs out of memory
on all machines the author has tried this computation on.

5.2 Projects for the Future

This is a list of theoretical and implementational questions that the author
finds important and interesting.

Prime Characteristic. In [26], G. Lyubeznik gave an algorithm for deciding
whether or not Hi

I(R) = 0 for any given I ⊆ R = K[x1, . . . , xn] where
K is a computable field of positive characteristic. His algorithm is built on
entirely different methods than the ones used in this chapter and relies on
the Frobenius functor. The implementation of this algorithm would be quite
worthwhile.

Ambient Spaces Different from A
n
K . If A equals K[x1, . . . , xn], I ⊆ A,

X = Spec(A) and Y = Spec(A/I), knowledge of Hi
I(A) for all i ∈ N answers

of course the question about the local cohomological dimension of Y in X.
If W ⊆ X is a smooth variety containing Y then Algorithm 4.2 for the
computation ofHi

I(A) also leads to a determination of the local cohomological
dimension of Y in W . Namely, if J stands for the defining ideal of W in X so
that R = A/J is the affine coordinate ring of W and if we set c = ht(J), then
it can be shown that Hi−c

I (R) = HomA(R,Hi
I(A)) for all i ∈ N. As Hi

I(A)
is I-torsion (and hence J-torsion), HomA(R,Hi

I(A)) is zero if and only if
Hi
I(A) = 0. It follows that the local cohomological dimension of Y in W

equals cd(A, I)−c and in fact {i ∈ N : Hi
I(A) 6= 0} = {i ∈ N : Hi−c

I (R) 6= 0}.
If howeverW = Spec(R) is not smooth, no algorithms for the computation

of either Hi
I(R) or cd(R, I) are known, irrespective of the characteristic of

the base field. It would be very interesting to have even partial ideas for
computations in that case.

De Rham Cohomology. In [35,48] algorithms are given to compute de
Rham (in this case equal to singular) cohomology of complements of com-
plex affine hypersurfaces and more general varieties. In [51] an algorithm is
given to compute the multiplicative (cup product) structure, and in [50] the
computation of the de Rham cohomology of open and closed sets in projective
space is explained. Some of these algorithms have been implemented while
others are still waiting.

For example, de Rham cohomology of complements of hypersurfaces, and
partially the cup product routine, are implemented.

Example 5.2. Let f = x3+y3+z3 in R3. One can compute with Macaulay 2
the de Rham cohomology of the complement of Var(f), and it turns out that
the cohomology in degrees 0 and 1 is 1-dimensional, in degrees 3 and 4 2-
dimensional and zero otherwise – here is the input:

i82 : erase symbol x;

Once x gets used as a subscripted variable, it’s hard to use it as a nonsub-
scripted variable. So let’s just erase it.

i83 : R = QQ[x,y,z];

i84 : f=x^3+y^3+z^3;

i85 : H=deRhamAll(f);

H is a hashtable with the entries Bfunction, LocalizeMap, VResolution,
TransferCycles, PreCycles, OmegaRes and CohomologyGroups. For exam-
ple, we have

i86 : H.CohomologyGroups

1
o86 = HashTable{0 => QQ }

1
1 => QQ

2
2 => QQ

2
3 => QQ

o86 : HashTable

showing that the dimensions are as claimed above. One can also extract
information on the generator of R3[f−1] used to represent the cohomology
classes by

i87 : H.LocalizeMap

o87 = | $x_1^6+2$x_1^3$x_2^3+$x_2^6+2$x_1^3$x_3^3+2$x_2^3$x_3^3+$x_3^6 |

o87 : Matrix

which proves that the generator in question is f−2. The cohomology classes
that Macaulay 2 computes are differential forms:

i88 : H.TransferCycles

o88 = HashTable{0 => | -1/12$x_1^4$x_2^3$D_1-1/3$x_1$x_2^6$D_1-1/12$x_ · · ·
1 => | 2/3$x_1^5+2/3$x_1^2$x_2^3+2/3$x_1^2$x_3^3 |

| -2/3$x_1^3$x_2^2-2/3$x_2^5-2/3$x_2^2$x_3^3 |
| 2/3$x_1^3$x_3^2+2/3$x_2^3$x_3^2+2/3$x_3^5 |

2 => | 48$x_1$x_2$x_3^2 600$x_3^4 |
| 48$x_1$x_2^2$x_3 600$x_2$x_3^3 |
| 48$x_1^2$x_2$x_3 600$x_1$x_3^3 |

3 => | -$x_1$x_2$x_3 -$x_3^3 |

o88 : HashTable

So, for example, the left column of the three rows that correspond to H2
dR(C3\

Var(f),C) represent the form (xyz(48zdxdy + 48ydzdx+ 48xdydz))/f2. If
we apply the above elements of D3 to f−2 and equip the results with appro-
priate differentials one arrives (after dropping unnecessary integral factors) at
the results displayed in the table in Figure 1. In terms of de Rham cohomol-
ogy, the cup product is given by the wedge product of differential forms. So
from the table one reads off the cup product relations o∪ t1 = d1, o∪ t2 = d2,
o ∪ o = 0, and that e operates as the identity.

For more involved examples, and the algorithms in [50], an actual implemen-
tation would be necessary since paper and pen are insufficient tools then.

Remark 5.3. The reader should be warned: if f−a generates Rn[f−1] over
Dn, then it is not necessarily the case that each de Rham cohomology class
of U = C

n \Var(f) can be written as a form with a pole of order at most a.
A counterexample is given by f = (x3 + y3 + xy)xy, where H1

dR(U ;C) has
a class that requires a third order pole, although −2 is the smallest integral
Bernstein root of f on R2.

Hom and Ext. In [36,44,45] algorithms are explained that compute homo-
morphisms between holonomic systems. In particular, rational and polyno-
mial solutions can be found because, for example, a polynomial solution to the

Group Dimension Generators

H0
dR 1 e :=

f2

f2

H1
dR 1 o :=

(x2dx− y2dy + z2dz)f

f2

H2
dR 2 t1 :=

xyz(zdxdy + ydzdx+ xdydz)

f2

t2 :=
(zdxdy + ydzdx+ xdydz)z3

f2

H3
dR 2 d1 :=

xyzdxdydz
f2

d2 :=
z3dxdydz

f2

Fig. 1.

system {P1, . . . , Pr} ∈ Dn corresponds to an element of HomDn(Dn/I,Rn)
where I = Dn · (P1, . . . , Pr).

Example 5.4. Consider the GKZ system in 2 variables associated to the ma-
trix (1, 2) ∈ Z1×2 and the parameter vector (5) ∈ C1. Named after Gelfand-
Kapranov-Zelevinski [11], this is the following system of differential equations:

(x∂x + y∂y) • f = 5f,
(∂2
x − ∂y) • f = 0.

With Macaulay 2 one can solve systems of this sort as follows:
i89 : I = gkz(matrix{{1,2}}, {5})

2
o89 = ideal (D - D , x D + 2x D - 5)

1 2 1 1 2 2

o89 : Ideal of QQ [x , x , D , D , WeylAlgebra => {x => D , x => D }]
1 2 1 2 1 1 2 2

This is a simple command to set up the GKZ-ideal associated to a matrix
and a parameter vector. The polynomial solutions are obtained by

i90 : PolySols I

5 3 2
o90 = {x + 20x x + 60x x }

1 1 2 1 2

o90 : List

This means that there is exactly one polynomial solution to the given
GKZ-system, and it is

x5 + 20x3y + 60xy2.

The algorithm for HomDn(M,N) is implemented and can be used to check
whether two given D-modules are isomorphic. Moreover, there are algorithms
(not implemented yet) to compute the ring structure of EndD(M) for a given
D-module M of finite holonomic rank which can be used to split a given
holonomic module into its direct summands. Perhaps an adaptation of these
methods can be used to construct Jordan-Hölder sequences for holonomic
D-modules.

Finiteness and Stratifications. Lyubeznik pointed out in [27] the follow-
ing curious fact.

Theorem 5.5. Let P (n, d;K) denote the set of polynomials of degree at most
d in at most n variables over the field K of characteristic zero. Let B(n, d;K)
denote the set of Bernstein-Sato polynomials

B(n, d;K) = {bf (s) : f ∈ P (n, d;K)}.

Then B(n, d;K) is finite. ut

So P (n, d;K) has a finite decomposition into strata with constant Bernstein-
Sato polynomial. A. Leykin proved in [22] that this decomposition is inde-
pendent of K and computable in the sense that membership in each stratum
can be tested by the vanishing of a finite set of algorithmically computable
polynomials over Q in the coefficients of the given polynomial in P (n, d;K).
In particular, the stratification is algebraic and for each K induced by base
change from Q to K. It makes thus sense to define B(n, d) which is the finite
set of Bernstein polynomials that can occur for f ∈ P (n, d;K) (where K is
in fact irrelevant).

Example 5.6. Consider P (2, 2;K), the set of quadratic binary forms over
K. With Macaulay 2, Leykin showed that there are precisely 4 different
Bernstein polynomials possible:

• bf (s) = 1 iff f ∈ V1 = V ′1 \ V ′′1 , where V ′1 = V (a1,1, a0,1, a0,2, a1,0, a2,0),
while V ′′1 = V (a0,0)
• bf (s) = s + 1 iff f ∈ V2 = (V ′2 \ V ′′2 ) ∪ (V ′3 \ V ′′3 ), where V ′2 = V (0),
V ′′2 = V (γ1), V ′3 = V (γ2, γ3, γ4), V ′′3 = V (γ3, γ4, γ5, γ6, γ7, γ8),
• bf (s) = (s+ 1)2 iff f ∈ V ′4 \ V ′′4 , where V ′4 = V (γ1), V ′′4 = V (γ2, γ3, γ4),
• bf (s) = (s+1)(s+ 1

2 ) iff f ∈ V ′5 \V ′′5 , where V ′5 = V (γ3, γ4, γ5, γ6, γ7, γ8),
V ′′5 = V (a1,1, a0,1, a0,2, a1,0, a2,0).

Here we have used the abbreviations

• γ1 = a0,2a
2
1,0 − a0,1a1,0a1,1 + a0,0a

2
1,1 + a2

0,1a2,0 − 4a0,0a0,2a2,0,
• γ2 = 2a0,2a1,0 − a0,1a1,1,
• γ3 = a1,0a1,1 − 2a0,1a2,0,
• γ4 = a2

1,1 − 4a0,2a2,0,

• γ5 = 2a0,2a1,0 − a0,1a1,1,
• γ6 = a2

0,1 − 4a0,0a0,2,
• γ7 = a0,1a1,0 − 2a0,0a1,1,
• γ8 = a2

1,0 − 4a0,0a2,0.

Similarly, Leykin shows that there are 9 possible Bernstein polynomials for
f ∈ B(2, 3;K):

B(2, 3) =
{

(s+ 1)2(s+
2
3

)(s+
4
3

), (s+ 1)2(s+
1
2

), (s+ 1), 1,

(s+ 1)(s+
2
3

)(s+
1
3

), (s+ 1)2, (s+ 1)(s+
1
2

),

(s+ 1)(s+
7
6

)(s+
5
6

), (s+ 1)2(s+
3
4

)(s+
5
4

)
}
.

It would be very interesting to study the nature of the stratification in larger
cases, and its restriction to hyperplane arrangements.

A generalization of this stratification result is obtained in [49]. There it is
shown that there is an algorithm to give P (n, d;K) an algebraic stratification
defined over Q such that the algebraic de Rham cohomology groups of the
complement of Var(f) do not vary on the stratum in a rather strong sense.
Again, the study and explicit computation of this stratification should be
very interesting.

Hodge Numbers. If Y is a projective variety in Pn
C

then algorithms out-
lined in [50] show how to compute the dimensions not only of the de Rham
cohomology groups of Pn

C
\ Y but also of Y itself. Suppose now that Y is

in fact a smooth projective variety. An interesting set of invariants are the
Hodge numbers, defined by hp,q = dimHp(Y,Ωq), where Ωq denotes the
sheaf of C-linear differential q-forms with coefficients in OY . At present we
do not know how to compute them. Of course there is a spectral sequence
Hp(Y,Ωq)⇒ Hp+q

dR (Y,C) and we know the abutment (or at least its dimen-
sions), but the technique for computing the abutment does not seem to be
usable to compute the E1 term because on an affine patch Hp(Y,Ωq) is either
zero or an infinite dimensional vector space.

Hodge structures and Bernstein-Sato polynomials are related as is for
example shown in [46].

5.3 Epilogue

In this chapter we have only touched a few highlights of the theory of com-
putations in D-modules, most of them related to homology and topology.
Despite this we hardly touched on the topics of integration and restriction,
which are theD-module versions of a pushforward and pullback, [20,30,35,48].

A very different aspect of D-modules is discussed in [40] where at the
center of investigations is the combinatorics of solutions of hypergeometric
differential equations. The combinatorial structure is used to find series solu-
tions for the differential equations which are polynomial in certain logarithmic
functions and power series with respect to the variables.

Combinatorial elements can also be found in the work of Assi, Castro
and Granger, see [2,3], on Gröbner fans in rings of differential operators. An
important (open) question in this direction is the determination of the set of
ideals in Dn that are initial ideals under some weight.

Algorithmic D-module theory promises to be an active area of research
for many years to come, and to have interesting applications to various other
parts of mathematics.

module.. J. Pure Appl. Algebra, 150(1):27–39, 2000.

4. J. Bernstein: Lecture notes on the theory of D-modules. Unpublished.
5. J.-E. Björk: Rings of Differential Operators. North Holland, New York, 1979.

2000.

50. U. Walther: Cohomology with Rational Coefficients of Complex Varieties.
Contemp. Math., to appear.

51. U. Walther: The Cup Product Structure for Complements of Complex Affine
Varieties. J. Pure Appl. Algebra, to appear.

52. U. Walther: On the Lyubeznik numbers of a local ring. Proc. Amer. Math.
Soc., to appear.

53. T. Yano: On the theory of b-functions. Publ. RIMS, Kyoto Univ., 14:111–202,
1978.





Index

(u, v)-Gröbner basis
287

D-modules 288
Dn-modules 286
Dn+1 293
JL(fs), algorithm for

296
Rn as a Dn-module 289
Θrk 304
in(u,v)(L) 287
bLf (s) 291
y1-homogenization 295
-> 44
. 50
:= 44
== 4
=> 49
# 49
# ? 50
% 6, 19

32749 17

abelian surface 244
absolutely irreducible

30
action 289, 293
action (of a differential

operator) 286, 288,
290, 291

adjunction theory 29
Alexander dual 79
append 43
apply 45
arrays 42
artinian, see also ideal,

zero-dimensional 102
assert 137

backtracking algorithm
188

basis reduction 180
Baues graph 182
Baues problem 191

Beilinson monad 215,
237

– applications of 241

– differentials of 237

Bernstein (Bernstein-
Sato) polynomial
291

Bernstein filtration 289

Bernstein inequality
289

Bernstein operator 291

Bernstein polynomial,
algorithm for 297

Bernstein root 291

Bernstein-Gel’fand-
Gel’fand corre-
spondence 215,
218

betti 22

Betti diagram 22

Betti number 162

Bézout number 101,
122, 127

Bézout’s Theorem 27,
63, 101, 111, 117

BGG 215

bidegree 176

bigraded ring of co-
homology operators
177

bilinear form 122

– signature 108, 124

– symmetric 108

bistellar flip 191

blow-up 64, 65

borel 89

Borel-fixed 89

bounded filtration 142

break 46

Brill-Noether number
253

bundle

– Horrocks-Mumford
244

– normalized 231
– self-dual 231
– Serre duality 231
– Serre’s criterion 227
– stable 231
bundles
– of differentials 227

Calabi-Yau 3-folds 267
canonical bundle 26, 32
canonical curve 263
canonical embedding

31, 32
Cartan map 140
Cartan resolution 218
Castelnuovo-Mumford

regularity 219
Čech complex 282
chain complex 21
chain property 95
Chow form 217
classical adjoint 57
Clifford index 31, 263
Coble self-dual sets of

points 263
code 52, 155
codimension 19
Cohen-Macaulay 63
coherent component

200
coherent sheaf
– representation of 18
cohomological dimension

282
cohomology
– intermediate 227
– sheaf 26, 222, 224,

225
coker 21
cokernel 21
complete intersection

116, 131
complex 21
– differentials in 21

complexity 165, 174
computable field 284
conic bundle 241
– elliptic 241
conormal sequence 24
contravariant coho-

mology module
157

cotangent bundle 20,
24, 26

cotangent sequence 20
covariant cohomology

module 157
cover 81
critical degree 165
cubic scroll 66
cubic space curve 18
curve
– general 31
– hyperelliptic 31
– plane quintic 32
– trigonal 32
cylinder 112

data types 41
dd 21, 134
dd i, i-th differential of a

complex 22
decompose 30
decompose a variety 30
degree of a projective

variety 19
degrees 22
determinantal variety

66, 111
diagram, Betti 22
differential operators

264
differentials of a complex

21
dimension (of a
D-module) 289

dimension of a projective
variety 19

discriminant 59, 113
distraction 92
divided powers algebra

139
division command 25

divisor at infinity 200
do 45
dual 79
dual, Alexander 79
duality 26
duality of sheaves 226
dualizing sheaf 26

eliminant 104–106, 114
elimination theory 57,

66, 104
ellipsoid 124
elliptic curve 58
embedded point 62, 64
endl 47
enumerative geometry

102–104, 110, 112, 114
– real 107, 118
enumerative problem

103, 112, 115, 121
– fully real 117, 119
Ext 32, 131, 155
Ext-generator series

149
exterior algebra 139,

215

f -vector 75
factor 50
Fano scheme 68
field
– algebraically closed

101, 113
– splitting 104
files 46
finite fields
– use of 17
flat limit 62, 68
flip graph 186
flip search algorithm

188
for 45
Fourier-Motzkin elimina-

tion 188, 206
functions 44

generates 135
generators 19
generic initial ideal 90,

92

generic monomial ideal
96

gens 19
genus of a curve 30
get 47
gonality 31
graded Artinian

Gorenstein ring 264
graded Bass number

169
graded Bass series 169
graded Betti number

161
graded complete

intersection 131
graded Poincaré series

162
Grassmannian 69, 112,

113, 115, 127
– local coordinates 117
– not a complete

intersection 116
Graver basis 182
Graver degree 183
Graver fiber 183
Green’s conjecture 32,

263
Gröbner basis 74, 104
– reduced 116
– universal 182
Gröbner cone 188
Gröbner fan 199

H 245
Hartshorne-Rao module

256
hash table 49
hessian 59
HH 26
Hilbert basis 203
Hilbert function 116
Hilbert scheme 91, 251
– incidence graph 91
Hilbert series 75
hilbertSeries 162,

169
holonomic module 289
homogeneous coordinate

ring 18

homogeneous derivation
139

homogeneous ideal 18

homological degree 176

homological methods

– introduction to 17

homology 151

homotopy

– Bézout 111

– Gröbner 111, 117

– optimal 111

homotopy continuation
110

Horrocks-Mumford
bundle 244

hyperboloid 124

hyperelliptic curve 31

ideal 19

– A-graded 97, 179

– Borel-fixed 89

– codimension 81

– coherent 185

– degree 102, 103

– generic initial 90, 92

– graph 81

– initial 74

– monomial 73, 74

– of minors 82

– permutahedron 86

– primary 77

– radical 102, 104, 120

– Stanley-Reisner 75

– toric 87

– torus isomorphism
185

– tree 80

– zero-dimensional
102, 119, 123

ideal 51

if 45

image 21

infinitesimal deforma-
tions 271

initial form 287

initial ideal 74, 87

– square-free 111, 117

input 47

integer programming
86, 197

internal degree 176
intersection multiplicities

26
intersection theory 26,

101
irreducible
– absolutely 30
irreducible decomposi-

tion 30
isBorel 89
isMonomialIdeal 75

jacobian 24
Jacobian matrix 24
join 43

kernel of a module map
20

kernel of a ring map 18,
104

keys of a hash table 49
killing cohomology 232
Koszul map 140
Koszul-regular 146

Lawrence lifting 183
Lefschetz Theorem 283
Leibniz rule 139, 286
length of a module 27
Lie algebra 58
lift to characteristic zero

277
linear free resolution

219
lists 42
– appending 43
– joining 43
– prepending 44
load 47
local cohomological

dimension 282
local cohomology 282
– algorithm for 305,

310
– iterated 310
localization
– algorithm for 300

Lyubeznik numbers
312

map of modules 25
map of rings 18
matrix 18
– unimodular 191
matrix factorization

133
maximal rank 256
method 52
method 52
minimal test set 197
minors 19
minPres.m2 211
mirror families 277
modules
– how to represent 20
moduli space 251
moduli space of curves

253
monad 230
– Beilinson 215, 236,

237
– display of 233
– homology of 230
– self-dual 234
– type of 230
monads
– applications of 230
monomial
– support 75
monomial ideal 73, 74
monomial order on Dn

287
monomialCurveIdeal

18
MonomialIdeal 74
monomialIdeal 74
multiplicity 61, 69

nilpotent orbits 56
Normaliz 203
normalization 203
normally ordered

expression 287
NP-complete 81
nullhomotopy 134
number field 277

one-parameter family
68

oo 48

Pfaffian complex 268
Plücker coordinate 115,

121, 122, 127
Plücker embedding 115
Plücker ideal 115
Poincaré series 162
polarCone.m2 207
polarization 92
polyhedral subdivision

190
polynomial equations

101, 103
– deficient 101
– overdetermined 117
positions 34
prepend 44
presentation
– minimal 25
presentation 21
primary decomposition

30, 77, 81, 84
print 46
printing
– to a file 48
product 44
product rule 289, 291
projective space 17
prune 25

quadratic form 113, 121

random curves 253
random nodal curve

254
random points 253
rank 1 linear syzygy

272
rational normal curve

118
rational normal scroll

34
rational parameterization

56
rational quartic curve

63, 93

reduced Ext module
149

reduced form 19
reduced ring of coho-

mology operators
149

Rees algebra 64
res 21
resolution
– free 21
resolution 150
return 45
Riemann-Roch theorem
– geometric 34
ring
– making one 47
ring of cohomology

operators 147

saturate 29
f -saturated 294
saturation 29, 55, 62,

103, 123
scan 45, 167
scheme
– affine 55, 64
– arithmetic 55
– irreducible 56
– non-reduced 62, 69
– over a number field

60
– reduced 62
Schreyer’s algorithm for

free resolutions 22
Schubert calculus 114,

115, 117, 118, 120
scroll
– rational normal 34
secondary polytope 191
Segre embedding 66
sequences 42
Serre correspondence

18
Serre’s intersection

formula 26
set 50
Shape Lemma 104, 120
Shapiros’s Conjecture

118

sheaf

– representation of 18

sheaf 26

sheaf cohomology 26,
222, 224, 225

simplicial complex 75

– f -vector 75

singular locus 58, 65

singular locus of a
scheme 30

solving polynomial
equations 103, 251

– real solutions 107

– via eigenvectors 106

– via elimination 104

– via numerical homo-
topy 110

source 22

sphere 113, 122, 127

splitting theorem

– of Grothendieck 227

– of Horrocks 227

stable equivalence 228

standard pair 83–86

standardPairs 85

Stanley-Reisner ideal
75, 191

stdio 46

Stickelberger’s Theorem
106

strings 42

Sturm sequence 107

sum 44

support variety 167,
175

symbol 287

symbol 52

symbolic computation
102

symbols 41

symmetric algebra 139

Syzygies of canonical
curves 263

table of Betti numbers
252

tally 50

target 22



Index 329

Tate resolution 215,
222

tautological subbundle
227

term order on Dn 288

time 30

toExternalString 48

top dimensional part of
an ideal 30

topology of a projective
variety 29

Tor 28

toric Hilbert scheme
179

toric ideal 87, 179

f -torsion free 294
toString 48
trace form 108
triangulation 190
– regular 191
trigonal curve 32
trim 33
twisted cubic 18

unirationality 251, 258

value 48
values of a hash table

49
variables
– local 44

vector bundle 226
Veronese surface 34
vertex cover 81
vertex ideal 98
visible lists 42

wall ideal 186
weight 287
weight vector 295
weight vector for Dn

287
weight vectors
– equivalent 199
Weyl algebra 286
while 45, 183


